

Domain Specific Languages of Mathematics: Lecture Notes

Patrik Jansson

Cezar Ionescu

December 16, 2017

Abstract

These notes aim to cover the lectures and exercises of the recently introduced course “Domain-Specific Languages of Mathematics” (at Chalmers and University of Gothenburg). The course was developed in response to difficulties faced by third-year computer science students in learning and applying classical mathematics (mainly real and complex analysis). The main idea is to encourage the students to approach mathematical domains from a functional programming perspective: to identify the main functions and types involved and, when necessary, to introduce new abstractions; to give calculational proofs; to pay attention to the syntax of the mathematical expressions; and, finally, to organize the resulting functions and types in domain-specific languages.

Contents

0	Introduction	4
0.1	About this course	6
0.2	Who should read these lecture notes?	7
0.3	Roadmap	7
0.4	Notation and code convention	7
0.5	Acknowledgments	7
1	A DSL for arithmetic expressions and complex numbers	8
1.1	Intro: Pitfalls with traditional mathematical notation	8
1.2	Types of data	9
1.2.1	type / newtype / data	9
1.3	A syntax for simple arithmetical expressions	10
1.3.1	<i>Env</i> , <i>Var</i> , and variable lookup	11
1.4	A case study: complex numbers	11
1.5	A syntax for (complex) arithmetical expressions	14
1.5.1	Other	17
1.6	More Haskell	18
1.6.1	Function composition	18
1.6.2	Notation and abstract syntax for (infinite) sequences	19
1.7	Some helper functions (can be skipped)	21
1.8	Exercises: complex numbers and DSLs	21

2	Logic and calculational proofs	26
2.1	Propositional Calculus	26
2.2	First Order Logic (predicate logic)	27
2.3	An aside: Pure set theory	29
2.4	Back to quantifiers	29
2.5	Proof by contradiction	30
2.6	Proof by cases	30
2.7	Functions as proofs	31
2.8	Proofs for <i>And</i> and <i>Or</i>	31
2.9	Case study: there is always another prime	32
2.10	Existential quantification as a pair type	33
2.11	Basic concepts of calculus	33
2.12	Questions and answers from the exercise sessions week 2	36
2.13	More general code for first order languages	37
2.14	Exercises: abstract FOL	38
2.14.1	Exercises	38
2.14.2	More exercises	43
3	Types in Mathematics	45
3.1	Types in mathematics	45
3.2	Typing Mathematics: derivative of a function	45
3.3	Typing Mathematics: partial derivative	47
3.4	Type inference and understanding: Lagrangian case study	47
3.5	Type in Mathematics (Part II)	50
3.5.1	Type classes	50
3.5.2	Overloaded integers literals	51
3.6	Type classes in Haskell	52
3.7	Computing derivatives	53
3.8	Shallow embeddings	55
3.9	Exercises	56
3.10	Exercises from old exams	56
4	Compositional Semantics and Algebraic Structures	59
4.1	Compositional semantics	59
4.1.1	A simpler example of a non-compositional function	59
4.1.2	Compositional semantics in general	60
4.1.3	Back to derivatives and evaluation	61

4.2	Algebraic Structures and DSLs	62
4.2.1	Algebras, homomorphisms	62
4.2.2	Homomorphism and compositional semantics	64
4.2.3	Other homomorphisms	65
4.3	Summing up: definitions and representation	66
4.3.1	Some helper functions	67
4.4	Exercises	67
5	Polynomials and Power Series	71
5.1	Polynomials	71
5.2	Polynomial degree as a homomorphism	74
5.3	Power Series	75
5.4	Operations on power series	77
5.5	Formal derivative	78
5.6	Helpers	79
5.7	Exercises	79
6	Higher-order Derivatives and their Applications	80
6.1	Review	80
6.2	Higher-order derivatives	83
6.3	Polynomials	84
6.4	Power series	84
6.5	Simple differential equations	85
6.6	The <i>Floating</i> structure of <i>PowerSeries</i>	85
6.7	Taylor series	86
6.8	Associated code	88
6.8.1	Not included to avoid overlapping instances	88
6.8.2	This is included instead	89
6.9	Exercises	89
7	Matrix algebra and linear transformations	92
7.1	Functions on vectors	94
7.2	Examples of matrix algebra	95
7.2.1	Derivative	95
7.2.2	Simple deterministic systems (transition systems)	97
7.2.3	Non-deterministic systems	98
7.2.4	Stochastic systems	100

7.3	Monadic dynamical systems	101
7.4	The monad of linear algebra	103
7.5	Associated code	103
8	Exponentials and Laplace	105
8.1	The Exponential Function	105
8.1.1	Exponential function: Associated code	107
8.2	The Laplace transform	108
9	End	112

0 Introduction

These lecture notes aim to cover the lectures and exercises of the recently introduced BSc-level course “Domain Specific Languages of Mathematics” (at Chalmers University of Technology and University of Gothenburg). The immediate aim of the course is to improve the mathematical education of computer scientists and the computer science education of mathematicians. We believe the course can be the starting point for far-reaching changes, leading to a restructuring of the mathematical training especially for engineers, but perhaps also for mathematicians themselves.

Computer science, viewed as a mathematical discipline, has certain features that set it apart from mainstream mathematics. It places much more emphasis on syntax, tends to prefer formal proofs to informal ones, and views logic as a tool rather than (just) as an object of study. It has long been advocated, both by mathematicians [Wells, 1995, Kraft, 2004] and computer scientists [Gries and Schneider, 1995, Boute, 2009], that the computer science perspective could be valuable in general mathematical education. Until today, this has been convincingly demonstrated (at least since the classical textbook of Gries and Schneider [1993]) only in the field of discrete mathematics. In fact, this demonstration has been so successful, that we increasingly see the discrete mathematics courses being taken over by computer science departments. This is a quite unsatisfactory state of affairs, for at least two reasons.

First, any benefits of the computer science perspective remain within the computer science department and the synergy with the wider mathematical landscape is lost. The mathematics department also misses the opportunity to see more in computer science than just a provider of tools for numerical computations. Considering the increasing dependence of mathematics on software, this can be a considerable loss.

Second, computer science (and other) students are exposed to two quite different approaches to teaching mathematics. For many of them, the formal, tool-oriented style of the discrete mathematics course is easier to follow than the traditional mathematical style. Since, moreover, discrete mathematics tends to be immediately useful to them, this makes the added difficulty of continuous mathematics even less palatable. As a result, their mathematical competence tends to suffer in areas such as real and complex analysis, or linear algebra.

This is a serious problem, because this lack of competence tends to infect the design of the entire curriculum. For example, a course in “Modeling of sustainable energy systems” for Chalmers’ CSE¹ students has to be tailored around this limitation, meaning that the models, methods, and tools that can be presented need to be drastically simplified, to the point where contact with mainstream research becomes impossible.

¹CSE = Computer Science & Engineering = Datateknik = D

We propose that a focus on *domain-specific languages* (DSLs) can be used to repair this unsatisfactory state of affairs. In computer science, a DSL “is a computer language specialized to a particular application domain” (Wikipedia), and building DSLs is increasingly becoming a standard industry practice. Empirical studies show that DSLs lead to fundamental increases in productivity, above alternative modelling approaches such as UML [Tolvanen, 2011]. Moreover, building DSLs also offers the opportunity for interdisciplinary activity and can assist in reaching a shared understanding of intuitive or vague notions (see, for example, the work done at Chalmers in cooperation with the Potsdam Institute for Climate Impact Research in the context of Global Systems Science, Lincke et al. [2009], Ionescu and Jansson [2013a], Jaeger et al. [2013], Ionescu and Jansson [2013b], Botta et al. [2017b,a]).

Thus, a course on designing and implementing DSLs can be an important addition to an engineering curriculum. Our key idea is to combine this with a rich source of domains and applications: mathematics. Indeed, mathematics offers countless examples of DSLs: the language of group theory, say, or the language of probability theory, embedded in that of measure theory. The idea that the various branches of mathematics are in fact DSLs embedded in the “general purpose language” of set theory was (even if not expressed in these words) the driving idea of the Bourbaki project, which exerted an enormous influence on present day mathematics.

The course on *DSLs of Mathematics (DSLM)* allows us to present classical mathematical topics in a way which builds on the experience of discrete mathematics: giving specifications of the concepts introduced, paying attention to syntax and types, and so on. For the mathematics students, used to a more informal style, the increased formality is justified by the need to implement (fragments of) the language. We provide a wide range of applications of the DSLs introduced, so that the new concepts can be seen “in action” as soon as possible.

TODO: update with actual learning outcomes

The course has two major learning outcomes. First, the students should be able to design and implement a DSL in a new domain. Second, they should be able to handle new mathematical areas using the computer science perspective.

To achieve these objectives, the course consists of a sequence of case studies in which a mathematical area is first presented (for example, a fragment of linear algebra, probability theory, interval analysis, or differential equations), followed by a careful analysis that reveals the domain elements needed to build a language for that domain. The DSL is first used informally, in order to ensure that it is sufficient to account for intended applications (for example, solving equations, or specifying a certain kind of mathematical object). It is in this step that the computer science perspective proves valuable for improving the students’ understanding of the mathematical area. The DSL is then implemented in Haskell. The resulting implementation can be compared with existing ones, such as Matlab in the case of linear algebra, or R in the case of statistical computations. Finally, limitations of the DSL are assessed and the possibility for further improvements discussed.

In the first instances, the course is an elective course for the second year within programmes such as CSE, SE, and Math. The potential students will have all taken first-year mathematics courses, and the only prerequisite which some of them will not satisfy will be familiarity with functional programming. However, as the current data structures course (common to the Math and CSE programmes) shows, math students are usually able to catch up fairly quickly, and in any case we aim to keep to a restricted subset of Haskell (no “advanced” features are required).

TODO: rewrite when the evaluation results are available

To assess the impact in terms of increased quality of education, we plan to measure how well the students do in ulterior courses that require mathematical competence (in the case of engineering students) or software competence (in the case of math students). For example, for CS and CSE students we will measure the percentage of students who, having taken DSLM, pass the third-year courses *Transforms, signals and systems* and *Control Theory (Reglerteknik)*, which are

current major stumbling blocks. For math students, we would like to measure their performance in ulterior scientific computing courses.

Since the course is, at least initially, an elective one, we also have the possibility of comparing the results with those of a control group (students who have not taken the course).

The work that lead up to the current course is as follows:

- 2014: in interaction with our colleagues from the various study programmes, we performed an assessment of the current status of potential students for the course in terms of their training (what prerequisites we can reasonably assume) and future path (what mathematical fields they are likely to encounter in later studies), and we worked out a course plan (which we submitted in February 2015, so that the first instance of the course could start in January 2016). We also make a survey of similar courses being offered at other universities, but did not find any close matches.
- 2015: we developed course materials for use within the first instance, wrote a paper [Ionescu and Jansson, 2016] about the course and presented the pedagogical ideas at several events (TODO: perhaps fill in TFPIE'15, DSLDI'15, IFIP WG 2.1 #73 in Göteborg).
- 2016: we ran the first instance of DSLM (partly paid by the regular course budget, partly by this project) with Cezar Ionescu as main lecturer.
- 2017: we ran the second instance of DSLM (paid fully by the regular course budget), now with Patrik Jansson as main lecturer.
- 2016 and 2017: we used the feedback from students following the standard Chalmers course evaluation in order to improve and further develop the course material.

Future work includes involving faculty from CSE and mathematics in the development of other mathematics courses (prel. Linear Algebra, Analysis) with the aim to incorporate these ideas also there. A major concern will be to work together with our colleagues in the mathematics department in order to distill the essential principles that can be “back-ported” to the other mathematics courses, such as Mathematical Analysis or Linear Algebra. Ideally, the mathematical areas used in DSLM will become increasingly challenging, the more the effective aspects of the computer science perspective are adopted in the first-year mathematics courses.

0.1 About this course

Software engineering involves modelling very different domains (e.g., business processes, typesetting, natural language, etc.) as software systems. The main idea of this course is that this kind of modelling is also important when tackling classical mathematics. In particular, it is useful to introduce abstract datatypes to represent mathematical objects, to specify the mathematical operations performed on these objects, to pay attention to the ambiguities of mathematical notation and understand when they express overloading, overriding, or other forms of generic programming. We shall emphasise the dividing line between syntax (what mathematical expressions look like) and semantics (what they mean). This emphasis leads us to naturally organise the software abstractions we develop in the form of domain-specific languages, and we will see how each mathematical theory gives rise to one or more such languages, and appreciate that many important theorems establish “translations” between them.

Mathematical objects are immutable, and, as such, functional programming languages are a very good fit for describing them. We shall use Haskell as our main vehicle, but only at a basic level, and we shall introduce the elements of the language as they are needed. The mathematical topic treated have been chosen either because we expect all students to be familiar with them (for example, limits of sequences, continuous functions, derivatives) or because they can be useful in many applications (e.g., Laplace transforms, linear algebra).

0.2 Who should read these lecture notes?

TODO: Convert formal prerequisites to “what is expected of the reader”

The student should have successfully completed

- a course in discrete mathematics as for example Introductory Discrete Mathematics.
- 15 hec in mathematics, for example Linear Algebra and Calculus
- 15 hec in computer science, for example (Introduction to Programming or Programming with Matlab) and Object-oriented Software Development
- an additional 22.5 hec of any mathematics or computer science courses.

Informally: One full time year (60 hec) of university level study consisting of a mix of mathematics and computer science.

0.3 Roadmap

TODO: write about the lecture plan and internal dependencies

0.4 Notation and code convention

TODO: fill in more about notation

0.5 Acknowledgments

The support from Chalmers Quality Funding 2015 (Dnr C 2014-1712, based on Swedish Higher Education Authority evaluation results) is gratefully acknowledged. Thanks also to Roger Johansson (as Head of Programme in CSE) and Peter Ljunglöf (as Vice Head of the CSE Department for BSc and MSc education) who provided continued financial support when the national political winds changed.

Thanks to Daniel Heurlin who provided many helpful comments during his work as a student research assistant in 2017.

1 A DSL for arithmetic expressions and complex numbers

This chapter is partly based on the paper [Ionescu and Jansson, 2016] from the International Workshop on Trends in Functional Programming in Education 2015. We will implement certain concepts in the functional programming language Haskell and the code for this lecture is placed in a module called *DSLsofMath.W01* that starts here:

```
module DSLsofMath.W01 where
import qualified DSLsofMath.CSem as CSem
import DSLsofMath.CSem (ComplexSem (CS))
import Numeric.Natural (Natural)
import Data.Ratio (Rational, Ratio, (%))
```

1.1 Intro: Pitfalls with traditional mathematical notation

A function or the value at a point? Mathematical texts often talk about “the function $f(x)$ ” when “the function f ” would be more clear. Otherwise there is a risk of confusion between $f(x)$ as a function and $f(x)$ as the value you get from applying the function f to the value bound to the name x .

Examples: let $f(x) = x + 1$ and let $t = 5 * f(2)$. Then it is clear that the value of t is the constant 15. But if we let $s = 5 * f(x)$ it is not clear if s should be seen as a constant or as a function of x .

Paying attention to types and variable scope often helps to sort out these ambiguities.

Scoping The syntax and scoping rules for the integral sign are rarely explicitly mentioned, but looking at it from a software perspective can help. If we start from a simple example, like $\int_1^2 x^2 dx$, it is relatively clear: the integral sign takes two real numbers as limits and then a certain notation for a function, or expression, to be integrated. Comparing the part after the integral sign to the syntax of a function definition $f(x) = x^2$ reveals a rather odd rule: instead of *starting* with declaring the variable x , the integral syntax *ends* with the variable name, and also uses the letter “d”. (There are historical explanations for this notation, and it is motivated by computation rules in the differential calculus, but we will not go there now.) It seems like the scope of the variable “bound” by d is from the integral sign to the final dx , but does it also extend to the limits? The answer is no, as we can see from a slightly extended example:

$$\begin{aligned} f(x) &= x^2 \\ g(x) &= \int_x^{2x} f(x) dx &= \int_x^{2x} f(y) dy \end{aligned}$$

The variable x bound on the left is independent of the variable x “bound under the integral sign”. Mathematics text books usually avoid the risk of confusion by (silently) renaming variables when needed, but we believe this renaming is a sufficiently important operation to be more explicitly mentioned.

Variable names as type hints In mathematical texts there are often conventions about the names used for variables of certain types. Typical examples include i, j, k for natural numbers or integers, x, y for real numbers and z, w for complex numbers.

The absence of explicit types in mathematical texts can sometimes lead to confusing formulations. For example, a standard text on differential equations by Edwards, Penney and Calvis Edwards et al. [2008] contains at page 266 the following remark:

The differentiation operator D can be viewed as a transformation which, when applied to the function $f(t)$, yields the new function $D\{f(t)\} = f'(t)$. The Laplace transformation \mathcal{L} involves the operation of integration and yields the new function $\mathcal{L}\{f(t)\} = F(s)$ of a new independent variable s .

This is meant to introduce a distinction between “operators”, such as differentiation, which take functions to functions of the same type, and “transforms”, such as the Laplace transform, which take functions to functions of a new type. To the logician or the computer scientist, the way of phrasing this difference in the quoted text sounds strange: surely the *name* of the independent variable does not matter: the Laplace transformation could very well return a function of the “old” variable t . We can understand that the name of the variable is used to carry semantic meaning about its type (this is also common in functional programming, for example with the conventional use of a plural “s” suffix, as in the name *xs*, to denote a list of values.). Moreover, by using this (implicit!) convention, it is easier to deal with cases such as that of the Hartley transform (a close relative of the Fourier transform), which does not change the type of the input function, but rather the *interpretation* of that type. We prefer to always give explicit typings rather than relying on syntactical conventions, and to use type synonyms for the case in which we have different interpretations of the same type. In the example of the Laplace transformation, this leads to

```
type T = Real
type S = ℂ
ℒ : (T → ℂ) → (S → ℂ)
```

1.2 Types of data

Dividing up the world (or problem domain) into values of different types is one of the guiding principles of this course. We will see that keeping track of types can guide the development of theories, languages, programs and proofs. To start out we introduce some of the ways types are described in Haskell.

TODO: make the following few paragraphs flow better

TODO: Introduce: newtype, products, sums, recursion, parameters

1.2.1 type / newtype / data

There are three keywords in Haskell involved in naming types: **type**, **newtype**, and **data**.

type – abbreviating type expressions The **type** keyword is used to create a type synonym - just another name for a type expression.

```
type Heltal = Integer
type Foo = (Maybe [String], [[Heltal]])
type BinOp = Heltal → Heltal → Heltal
type Env v s = [(v, s)]
```

The new name for the type on the RHS does not add type safety, just readability (if used wisely). The *Env* example shows that a type synonym can have type parameters.

newtype – more protection A simple example of the use of **newtype** in Haskell is to distinguish values which should be kept apart. A simple example is

```
newtype Age = Ag Int -- Age in years
newtype Shoe = Sh Int -- Shoe size (EU)
```

Which introduces two new types, *Age* and *Shoe*, which both are internally represented by an *Int* but which are good to keep apart.

The constructor functions $Ag :: Int \rightarrow Age$ and $Sh :: Int \rightarrow Shoe$ are used to translate from plain integers to ages and shoe sizes.

In the lecture notes we used a newtype for the semantics of complex numbers as a pair of numbers in the cartesian representation but may also be useful to have another newtype for complex as a pair of numbers in the polar representation.

data – for syntax trees Some examples:

```
data N = Z | S N
```

This declaration introduces

- a new type *N* for unary natural numbers,
- a constructor $Z :: N$ to represent zero, and
- a constructor $S :: N \rightarrow N$ to represent the successor.

Examples values: $zero = Z$, $one = S\ Z$, $three = S\ (S\ one)$

1.3 A syntax for simple arithmetical expressions

```
data E = V String | P E E | T E E
```

This declaration introduces

- a new type *E* for simple arithmetic expressions,
- a constructor $V :: String \rightarrow E$ to represent variables,
- a constructor $P :: E \rightarrow E \rightarrow E$ to represent plus, and
- a constructor $T :: E \rightarrow E \rightarrow E$ to represent times.

Example values: $x = V\ "x"$, $e_1 = P\ x\ x$, $e_2 = T\ e_1\ e_1$

If you want a constructor to be used as an infix operator you need to use symbol characters and start with a colon:

```
data E' = V' String | E' :+ E' | E' :* E'
```

Example values: $y = V\ "y"$, $e_1 = y\ :+\ y$, $e_2 = x\ :* e_1$

Finally, you can add one or more type parameters to make a whole family of datatypes in one go: (TODO: perhaps say something about the constructor names)

```
data E' v = V' v | E' v :+ E' v | E' v :* E' v
```

The purpose of the parameter *v* here is to enable a free choice of type for the variables (be it *String* or *Int* or something else).

1.3.1 *Env*, *Var*, and variable lookup

The type synonym

```
type Env v s = [(v, s)]
```

is one way of expressing a partial function from v to s . As an example value of this type:

```
env1 :: Env String Int
env1 = [("hej", 17), ("du", 38)]
```

TODO: Move this paragraph later and instead just explain (or define) *lookup*. include a comment about syntax and semantics.

The *Env* type is commonly used in evaluator functions for syntax trees containing variables: TODO: explain “the evaluator” (a function from the syntax to the semantics)

TODO: handle the partial function problem using Maybe.

```
evalE' :: Eq v => Env v sem -> (E' v -> Maybe sem)
evalE' env (V' x) = case lookup x env of
  Just c -> undefined -- ...
  -- ...
```

Notice that *env* maps “syntax” (variable names) to “semantics”, just like the evaluator does.

1.4 A case study: complex numbers

We will start by an analytic reading of the introduction of complex numbers in Adams and Essex [2010]. We choose a simple domain to allow the reader to concentrate on the essential elements of our approach without the distraction of potentially unfamiliar mathematical concepts. For this section, we bracket our previous knowledge and approach the text as we would a completely new domain, even if that leads to a somewhat exaggerated attention to detail.

Adams and Essex introduce complex numbers in Appendix 1. The section *Definition of Complex Numbers* begins with:

We begin by defining the symbol i , called **the imaginary unit**, to have the property

$$i^2 = -1$$

Thus, we could also call i the square root of -1 and denote it $\sqrt{-1}$. Of course, i is not a real number; no real number has a negative square.

At this stage, it is not clear what the type of i is meant to be, we only know that i is not a real number. Moreover, we do not know what operations are possible on i , only that i^2 is another name for -1 (but it is not obvious that, say $i * i$ is related in any way with i^2 , since the operations of multiplication and squaring have only been introduced so far for numerical types such as \mathbb{N} or \mathbb{R} , and not for symbols).

For the moment, we introduce a type for the value i , and, since we know nothing about other values, we make i the only member of this type:

```
data ImagUnits = I
```

```

i :: ImagUnits
i = I

```

We use a capital *I* in the **data** declaration because a lowercase constructor name would cause a syntax error in Haskell.

Next, we have the following definition:

Definition: A **complex number** is an expression of the form

$$a + bi \quad \text{or} \quad a + ib,$$

where *a* and *b* are real numbers, and *i* is the imaginary unit.

This definition clearly points to the introduction of a syntax (notice the keyword “form”). This is underlined by the presentation of *two* forms, which can suggest that the operation of juxtaposing *i* (multiplication?) is not commutative.

A profitable way of dealing with such concrete syntax in functional programming is to introduce an abstract representation of it in the form of a datatype:

```

data ComplexA = CPlus1  $\mathbb{R}$   $\mathbb{R}$  ImagUnits
               | CPlus2  $\mathbb{R}$  ImagUnits  $\mathbb{R}$ 

```

We can give the translation from the abstract syntax to the concrete syntax as a function *showCA*:

```

showCA :: ComplexA → String
showCA (CPlus1 x y i) = show x ++ " + " ++ show y ++ "i"
showCA (CPlus2 x i y) = show x ++ " + " ++ "i" ++ show y

```

Notice that the type \mathbb{R} is not implemented yet and it is not really even exactly implementable but we want to focus on complex numbers so we will approximate \mathbb{R} by double precision floating point numbers for now.

```

type  $\mathbb{R}$  = Double

```

The text continues with examples:

For example, $3 + 2i$, $\frac{7}{2} - \frac{2}{3}i$, $i\pi = 0 + i\pi$ and $-3 = -3 + 0i$ are all complex numbers. The last of these examples shows that every real number can be regarded as a complex number.

The second example is somewhat problematic: it does not seem to be of the form $a + bi$. Given that the last two examples seem to introduce shorthand for various complex numbers, let us assume that this one does as well, and that $a - bi$ can be understood as an abbreviation of $a + (-b)i$. With this provision, in our notation the examples are written as in Table 1.

We interpret the sentence “The last of these examples ...” to mean that there is an embedding of the real numbers in *ComplexA*, which we introduce explicitly:

```

toComplex ::  $\mathbb{R}$  → ComplexA
toComplex x = CPlus1 x 0 I

```

Again, at this stage there are many open questions. For example, we can assume that *i* 1 stands for the complex number *CPlus₂* 0 *I* 1, but what about *i* by itself? If juxtaposition is meant to

Mathematics	Haskell
$3 + 2i$	<code>CPlus₁ 3 2 I</code>
$\frac{7}{2} - \frac{2}{3}i = \frac{7}{2} + \frac{-2}{3}i$	<code>CPlus₁ (7 / 2) (-2 / 3) I</code>
$i\pi = 0 + i\pi$	<code>CPlus₂ 0 I π</code>
$-3 = -3 + 0i$	<code>CPlus₁ (-3) 0 I</code>

Table 1: Examples of notation and abstract syntax for some complex numbers.

denote some sort of multiplication, then perhaps 1 can be considered as a unit, in which case we would have that i abbreviates $i \ 1$ and therefore `CPlus2 0 I 1`. But what about, say, $2 \ i$? Abbreviations with i have only been introduced for the ib form, and not for the bi one!

The text then continues with a parenthetical remark which helps us dispel these doubts:

(We will normally use $a + bi$ unless b is a complicated expression, in which case we will write $a + ib$ instead. Either form is acceptable.)

This remark suggests strongly that the two syntactic forms are meant to denote the same elements, since otherwise it would be strange to say “either form is acceptable”. After all, they are acceptable by definition.

Given that $a + ib$ is only “syntactic sugar” for $a + bi$, we can simplify our representation for the abstract syntax, eliminating one of the constructors:

data `ComplexB` = `CPlusB` $\mathbb{R} \ \mathbb{R} \ \text{ImagUnits}$

In fact, since it doesn’t look as though the type `ImagUnits` will receive more elements, we can dispense with it altogether:

data `ComplexC` = `CPlusC` $\mathbb{R} \ \mathbb{R}$

(The renaming of the constructor to `CPlusC` serves as a guard against the case we have suppressed potentially semantically relevant syntax.)

We read further:

It is often convenient to represent a complex number by a single letter; w and z are frequently used for this purpose. If a , b , x , and y are real numbers, and $w = a + bi$ and $z = x + yi$, then we can refer to the complex numbers w and z . Note that $w = z$ if and only if $a = x$ and $b = y$.

First, let us notice that we are given an important semantic information: to check equality for complex numbers, it is enough to check equality of the components (the arguments to the constructor `CPlusC`). (Another way of saying this is that `CPlusC` is injective.) The equality on complex numbers is what we would obtain in Haskell by using **deriving** `Eq`.

This shows that the set of complex numbers is, in fact, isomorphic with the set of pairs of real numbers, a point which we can make explicit by re-formulating the definition in terms of a **newtype**:

newtype `ComplexD` = `CD` (\mathbb{R}, \mathbb{R}) **deriving** `Eq`

The point of the somewhat confusing discussion of using “letters” to stand for complex numbers is to introduce a substitute for *pattern matching*, as in the following definition:

Definition: If $z = x + yi$ is a complex number (where x and y are real), we call x the **real part** of z and denote it $Re(z)$. We call y the **imaginary part** of z and denote it $Im(z)$:

$$\begin{aligned} Re(z) &= Re(x + yi) = x \\ Im(z) &= Im(x + yi) = y \end{aligned}$$

This is rather similar to Haskell’s *as-patterns*:

$$\begin{aligned} re &:: ComplexD \rightarrow \mathbb{R} \\ re \ z@(CD(x, y)) &= x \\ im &:: ComplexD \rightarrow \mathbb{R} \\ im \ z@(CD(x, y)) &= y \end{aligned}$$

a potential source of confusion being that the symbol z introduced by the as-pattern is not actually used on the right-hand side of the equations.

The use of as-patterns such as “ $z = x + yi$ ” is repeated throughout the text, for example in the definition of the algebraic operations on complex numbers:

The sum and difference of complex numbers

If $w = a + bi$ and $z = x + yi$, where a , b , x , and y are real numbers, then

$$\begin{aligned} w + z &= (a + x) + (b + y) i \\ w - z &= (a - x) + (b - y) i \end{aligned}$$

With the introduction of algebraic operations, the language of complex numbers becomes much richer. We can describe these operations in a *shallow embedding* in terms of the concrete datatype *ComplexD*, for example:

$$\begin{aligned} (+.) &:: ComplexD \rightarrow ComplexD \rightarrow ComplexD \\ (CD(a, b)) +. (CD(x, y)) &= CD((a + x), (b + y)) \end{aligned}$$

or we can build a datatype of “syntactic” complex numbers from the algebraic operations to arrive at a *deep embedding* as seen in the next section.

Exercises:

- implement $(*)$ for *ComplexD*

At this point we can sum up the “evolution” of the datatypes introduced so far. Starting from *ComplexA*, the type has evolved by successive refinements through *ComplexB*, *ComplexC*, ending up in *ComplexD* (see Fig. 1).

1.5 A syntax for (complex) arithmetical expressions

So far we have tried to find a datatype to represent the intended *semantics* of complex numbers. That approach is called “shallow embedding”. Now we turn to the *syntax* instead (“deep embedding”).

We want a datatype *ComplexE* for the abstract syntax tree of expressions. The syntactic expressions can later be evaluated to semantic values:

```

data    ImagUnits = I
data    ComplexA = CPlus1  $\mathbb{R}$   $\mathbb{R}$  ImagUnits
              | CPlus2  $\mathbb{R}$  ImagUnits  $\mathbb{R}$ 
data    ComplexB = CPlusB  $\mathbb{R}$   $\mathbb{R}$  ImagUnits
data    ComplexC = CPlusC  $\mathbb{R}$   $\mathbb{R}$ 
newtype ComplexD = CD ( $\mathbb{R}$ ,  $\mathbb{R}$ ) deriving Eq

```

Figure 1: Complex number datatype refinement.

evalE :: *ComplexE* → *ComplexD*

The datatype *ComplexE* should collect ways of building syntactic expression representing complex numbers and we have so far seen the symbol *i*, an embedding from \mathbb{R} , plus and times. We make these four *constructors* in one recursive datatype as follows:

```

data ComplexE = ImagUnit
                | ToComplex  $\mathbb{R}$ 
                | Plus ComplexE ComplexE
                | Times ComplexE ComplexE
deriving (Eq, Show)

```

Note that, in *ComplexA* above, we also had a constructor for “plus”, but it was another “plus”. They are distinguished by type: *CPlus*₁ took (basically) two real numbers, while *Plus* here takes two complex numbers as arguments.

We can implement the evaluator *evalE* by pattern matching on the syntax tree and recursion.

TODO: mention “wishful thinking”

```

evalE ImagUnit      = CD (0, 1)
evalE (ToComplex r) = CD (r, 0)
evalE (Plus c1 c2) = evalE c1 +. evalE c2
evalE (Times c1 c2) = evalE c1 *. evalE c2


```

We also define a function to embed a semantic complex number in the syntax:

```

fromCD :: ComplexD → ComplexE
fromCD (CD (x, y)) = Plus (ToComplex x) (Times (ToComplex y) ImagUnit)
testE1 = Plus (ToComplex 3) (Times (ToComplex 2) ImagUnit)
testE2 = Times ImagUnit ImagUnit

```



This is *testE1* as an abstract syntax tree:

There are certain laws we would like to hold for operations on complex numbers. The simplest is perhaps $i^2 = -1$ from the start of the lecture,

TODO: Explain the use of functions to `Bool` as specifications where the reading should be “forall inputs, this should return `True`”

```
propImagUnit :: Bool
propImagUnit = Times ImagUnit ImagUnit === ToComplex (-1)
(===) :: ComplexE → ComplexE → Bool
z === w = evalE z == evalE w
```

TODO: format `===` in a similar way as `==`

and that `fromCD` is an embedding:

```
propFromCD :: ComplexD → Bool
propFromCD c = evalE (fromCD c) == c
```

but we also have that *Plus* and *Times* should be associative and commutative and *Times* should distribute over *Plus*:

```
propAssocPlus x y z = Plus (Plus x y) z === Plus x (Plus y z)
propAssocTimes x y z = Times (Times x y) z === Times x (Times y z)
propDistTimesPlus x y z = Times x (Plus y z) === Plus (Times x y) (Times x z)
```

These three laws actually fail, but not because of the implementation of `evalE`. We will get back to that later but let us first generalise the properties a bit by making the operator a parameter:

```
propAssocA :: Eq a => (a → a → a) → a → a → a → Bool
propAssocA (+?) x y z = (x +? y) +? z == x +? (y +? z)
```

Note that `propAssocA` is a higher order function: it takes a function (a binary operator) as its first parameter. It is also polymorphic: it works for many different types *a* (all types which have an `==` operator).

Thus we can specialise it to *Plus*, *Times* and other binary operators. In Haskell there is a type class *Num* for different types of “numbers” (with operations `(+)`, `(*)`, etc.). We can try out `propAssocA` for a few of them.

```
propAssocAInt = propAssocA (+) :: Int → Int → Int → Bool
propAssocADouble = propAssocA (+) :: Double → Double → Double → Bool
```

The first is fine, but the second fails due to rounding errors. QuickCheck can be used to find small examples - I like this one best:

```
notAssocEvidence :: (Double, Double, Double, Bool)
notAssocEvidence = (lhs, rhs, lhs - rhs, lhs == rhs)
  where lhs = (1 + 1) + 1 / 3
        rhs = 1 + (1 + 1 / 3)
```

For completeness: this is the answer:

```
(2.3333333333333335      -- Notice the five at the end
,2.3333333333333333,      -- which is not present here.
,4.440892098500626e-16    -- The difference
,False)
```


This is actually the underlying reason why some of the laws failed for complex numbers: the approximative nature of *Double*. But to be sure there is no other bug hiding we need to make one more version of the complex number type: parameterise on the underlying type for \mathbb{R} . At the same time we generalise *ToComplex* to *FromCartesian*:

TODO: mention the parameterised **newtype** *ComplexSem* $r = CS\ (r, r)$ **deriving** *Eq* as well. Currently imported. Perhaps in the figure summarizing the evolution.

```

data ComplexSyn r = FromCartesian r r
    | ComplexSyn r :+: ComplexSyn r
    | ComplexSyn r :*: ComplexSyn r

toComplexSyn :: Num a => a -> ComplexSyn a
toComplexSyn x = FromCartesian x (fromInteger 0)

-- From CSem: newtype ComplexSem r = CS (r , r) deriving Eq
evalCSyn :: Num r => ComplexSyn r -> CSem.ComplexSem r
evalCSyn (FromCartesian x y) = CS (x, y)
evalCSyn (l :+: r) = evalCSyn l CSem.+ . evalCSyn r
evalCSyn (l :*: r) = evalCSyn l CSem.* . evalCSyn r

instance Num a => Num (ComplexSyn a) where
    (+) = (:+ :)
    (*) = (:* :)
    fromInteger = fromIntegerCS
    -- TODO: add a few more operations (hint: extend ComplexSyn as well)
    -- TODO: also extend eval

fromIntegerCS :: Num r => Integer -> ComplexSyn r
fromIntegerCS = toComplexSyn o fromInteger

```

TODO: place this paragraph properly.

From syntax to semantics and back We have seen evaluation functions from abstract syntax to semantics (*eval* :: *Syn* -> *Sem*). Often a partial inverse is also available: *embed* :: *Sem* -> *Syn*. For our complex numbers we have

TODO: fill in a function from *ComplexSem* $r \rightarrow ComplexSyn\ r$. (Roughly *embed* (*CS* (*x*, *y*)) = *Plus* (*ToC* *x*) (*Times* *I* (*ToC* *y*)).)

TODO: draw diagram of the types and the functions *eval* and *embed* to give an intuition for the “round-trip” property

The embedding should satisfy a round-trip property: *eval* (*embed* *s*) == *s* for all *s*.

TODO: Add typed quantification

Exercise: What about the opposite direction? When is *embed* (*eval* *e*) == *e*? (Step 0: type the quantification. Step 1: what equality is suitable here? Step 2: if you use “equality up to eval” - how is the resulting property related to the first round-trip property?)

We can also state and check properties relating the semantic and the syntactic operations:

$a + b = eval\ (Plus\ (embed\ a)\ (embed\ b))$ for all *a* and *b*.

1.5.1 Other

TODO: find a good place for this part

Lifting operations to a parameterised type When we define addition on complex numbers (represented as pairs of real and imaginary components) we can do that for any underlying type r which supports addition.

```

type CS = CSem.ComplexSem  -- for shorter type expressions below
liftCS :: (  r  →    r  →    r ) →
          (CS r → CS r → CS r)
liftCS (+) (CS (x, y)) (CS (x', y')) = CS (x + x', y + y')

```

Note that *liftCS* takes $(+)$ as its first parameter and uses it twice on the RHS.

Laws TODO: Associative, Commutative, Distributive, ...


Associative $(+) = \forall a, b, c. (a + b) + c = a + (b + c)$

Commutative $(+) = \forall a, b. a + b = b + a$

Non-examples: division is not commutative, average is commutative but not associative.

TODO: Talk more about the properties tested above in *propDistTimesPlus*, ... (The underlying set matters: $(+)$ for \mathbb{R} has some properties, $(+)$ for *Double* has other, ... Approximation is often necessary, but makes many laws false. Thus, attempt to do it late.)

Distributive $(*) (+) = \forall a, b, c. (a + b) * c = (a * c) + (b * c)$



(In the language of section 4.2.1, distributivity means that $(*c)$ is a $(+)$ -homomorphism.)

Exercise: Find some operator $(\#)$ which satisfies *Distributive* $(+)$ $(\#)$

Exercise: Find other pairs of operators satisfying a distributive law.

1.6 More Haskell

1.6.1 Function composition

The infix operator `.` in Haskell is an implementation of the mathematical operation of function composition.

$$f \circ g = \lambda x \rightarrow f (g x)$$

The period is an ASCII approximation of the composition symbol \circ typically used in mathematics. (The symbol \circ is encoded as U+2218 and called RING OPERATOR in Unicode, `∘` in HTML, `\circ` in \TeX , etc.)

The type is perhaps best illustrated by a diagram with types as nodes and functions (arrows) as directed edges:

In Haskell we get the following type:

$$(\circ) :: (b \rightarrow c) \rightarrow (a \rightarrow b) \rightarrow (a \rightarrow c)$$

which may take a while to get used to.




Figure 2: Function composition diagram

1.6.2 Notation and abstract syntax for (infinite) sequences

As a bit of preparation for the language of sequences and limits in later lectures we here spend a few lines on the notation and abstract syntax of sequences.

Common math book notation: $\{a_i\}_{i=0}^{\infty}$ or just $\{a_i\}$ and (not always) an indication of the type X of the a_i . Note that the a at the center of this notation actually carries all of the information: an infinite family of values $a_i : X$. If we interpret “subscript” as function application we can see that $a : \mathbb{N} \rightarrow X$ is a useful typing of a sequence. Some examples:

```

type  $\mathbb{N}$       = Natural    -- imported from Numeric.Natural
type  $\mathbb{Q}$       = Ratio  $\mathbb{N}$   -- imported from Data.Ratio
type Seq  $a$  =  $\mathbb{N} \rightarrow a$ 

idSeq :: Seq  $\mathbb{N}$ 
idSeq  $i$  =  $i$                 -- 0, 1, 2, 3, ...

invSeq :: Seq  $\mathbb{Q}$ 
invSeq  $i$  =  $1 \div (1 + i)$     -- 1/1, 1/2, 1/3, 1/4, ...

pow2 :: Num  $r \Rightarrow$  Seq  $r$ 
pow2 =  $(2^{\wedge})$               -- 1, 2, 4, 8, ...

conSeq ::  $a \rightarrow$  Seq  $a$ 
conSeq  $c$   $i$  =  $c$             --  $c, c, c, c, \dots$ 

```

What operations can be perform on sequences? We have seen the first one: given a value c we can generate a constant sequence with *conSeq* c . We can also add sequences componentwise (also called “pointwise”):

```

addSeq :: Num  $a \Rightarrow$  Seq  $a \rightarrow$  Seq  $a \rightarrow$  Seq  $a$ 
addSeq  $f$   $g$   $i$  =  $f\ i + g\ i$ 

```

and in general lift any binary operation $op :: a \rightarrow b \rightarrow c$ to the corresponding, pointwise, operation of sequences:

```

liftSeq2 ::  $(a \rightarrow b \rightarrow c) \rightarrow$  Seq  $a \rightarrow$  Seq  $b \rightarrow$  Seq  $c$ 
liftSeq2  $op$   $f$   $g$   $i$  =  $op\ (f\ i)\ (g\ i)$  --  $op\ (f\ 0)\ (g\ 0), op\ (f\ 1)\ (g\ 1), \dots$ 

```

Similarly we can lift unary operations, and “nullary” operations:

```

liftSeq1 :: (a → b) → Seq a → Seq b
liftSeq1 h f i = h (f i)  -- h (f 0), h (f 1), h (f 2), ...
liftSeq0 :: a → Seq a
liftSeq0 c i = c

```

TODO: convert each "embedded exercise" to a forward pointer (some like "Related exercise: ??")

Exercise: what does function composition do to a sequence? (composition on the left?, on the right?)

(TODO: perhaps mention limits, sums, just a teasers for later chapters)

Another common mathematical operator on sequences is the limit. We will get back to limits in a later chapter, but here we just analyse the notation and typing. This definition is slightly adapted from Wikipedia (2017-11-08):

We call L the limit of the sequence $\{x_n\}$ if the following condition holds: For each real number $\epsilon > 0$, there exists a natural number N such that, for every natural number $n \geq N$, we have $|x_n - L| < \epsilon$.

If so, we say that the sequence converges to L and write

$$L = \lim_{i \rightarrow \infty} x_i$$

There are (at least) two things to note here. First, with this syntax, the $\lim_{i \rightarrow \infty} x_i$ expression form binds i in the expression x_i . We could just as well say that \lim takes a function $x :: \mathbb{N} \rightarrow X$ as its only argument. Second, an arbitrary x , may or may not have a limit. Thus the customary use of $L =$ is a bit of abuse of notation, because the right hand side may not be well defined. One way to capture that is to give \lim the type $(\mathbb{N} \rightarrow X) \rightarrow \text{Maybe } X$. Then $L = \lim_{i \rightarrow \infty} x_i$ would mean *Just* $L = \lim x$. We will return to limits and their proofs in Sec. 2.11 after we have reviewed some logic. Here we just define one more common operation: the sum of a sequence (like $\sigma = \sum_{i=0}^{\infty} 1/i!$).

Just as not all sequences have a limit, not all have a sum either. But for every sequence we can define a new sequence of partial sums:

```

sums :: Num a ⇒ Seq a → Seq a
sums = scan 0 (+)
scan :: a → (a → a → a) → Seq a → Seq a
scan z (+) f = s
  where s 0 = z
        s i = s (i - 1) + f i

```

And by combining this with limits we can state formally that the sum of a sequence a exists and is S iff the limit of $\text{sums } a$ exists and is S . As a formula we get *Just* $S = \lim (\text{sums } a)$, and for our example it turns out that it converges and that $\sigma = \sum_{i=0}^{\infty} 1/i! = e$ but we will not get to that until Sec. 8.1.

We will also return to limits in Sec. 3.3 about derivatives where we explore variants of the classical definition

$$f'(x) = \lim_{h \rightarrow 0} \frac{f(x+h) - f(x)}{h}$$

TODO: sum up the results in terms of a DSL: a type $(\text{Seq } a)$, some operations (conSeq , addSeq , fmap , sums , ...) and some "run functions" or predicates (like \lim)

1.7 Some helper functions (can be skipped)

```
propAssocAdd :: (Eq a, Num a) => a -> a -> a -> Bool
propAssocAdd = propAssocA (+)

(*) :: ComplexD -> ComplexD -> ComplexD
CD (ar, ai) *. CD (br, bi) = CD (ar * br - ai * bi, ar * bi + ai * br)

instance Show ComplexD where
    show = showCD

showCD :: ComplexD -> String
showCD (CD (x, y)) = show x ++ " + " ++ show y ++ "i"
```

1.8 Exercises: complex numbers and DSLs

Exercise 1.1. Consider the following data type for simple arithmetic expressions:

```
data Exp = Con Integer
        | Exp 'Plus' Exp
        | Exp 'Minus' Exp
        | Exp 'Times' Exp
    deriving (Eq, Show)
```

(Note the use of “backticks” around *Plus* etc. which makes it possible to use a name as an infix operator.)

1. Write the following expressions in Haskell, using the *Exp* data type:
 - (a) $a_1 = 2 + 2$
 - (b) $a_2 = a_1 + 7 * 9$
 - (c) $a_3 = 8 * (2 + 11) - (3 + 7) * (a_1 + a_2)$
2. Create a function $eval :: Exp \rightarrow Integer$ that takes a value of the *Exp* data type and returns the corresponding number (for instance, $eval ((Con\ 3)\ 'Plus'\ (Con\ 3)) = 6$). Try it on the expressions from the first part, and verify that it works as expected.
3. Consider the following expression:

$$c_1 = (x - 15) * (y + 12) * z$$

where $x = 5$
 $y = 8$
 $z = 13$

In order to represent this with our *Exp* data type, we are going to have to make some modifications:

- (a) Update the *Exp* data type with a new constructor *Var String* that allows variables with strings as names to be represented. Use the updated *Exp* to write an expression for c_1 in Haskell.
- (b) Create a function $varVal :: String \rightarrow Integer$ that takes a variable name, and returns the value of that variable. For now, the function just needs to be defined for the variables in the expression above, i.e. $varVal\ "x"$ should return 5, $varVal\ "y"$ should return 8, and $varVal\ "z"$ should return 13.

- (c) Update the *eval* function so that it supports the new *Var* constructor, and use it get a numeric value of the expression *c1*.

Exercise 1.2. We will now look at a slightly more generalized version of the *Exp* type from the previous exercise:

```
data E2 a = Con a
          | Var String
          | E2 a 'Plus' E2 a
          | E2 a 'Minus' E2 a
          | E2 a 'Times' E2 a
          deriving (Eq, Show)
```

The type has now been parametrized, so that it is no longer limited to representing expressions with integers, but can instead represent expressions with any type. For instance, we could have an *E2 Double* to represent expressions with doubles, or an *E2 ComplexD* to represent expressions with complex numbers.

- Write the following expressions in Haskell, using the new *E2* data type.
 - $a_1 = 2.0 + a$
 - $a_2 = 5.3 + b * c$
 - $a_3 = a * (b + c) - (d + e) * (f + a)$
- In order to evaluate these expressions, we will need a way of translating a variable name into the value. The following table shows the value of each variable in the expressions above:

Name	Value
a	1.5
b	4.8
c	2.4
d	7.4
e	5.8
f	1.7

In Haskell, we can represent this table using a value of type *Env String a = [(String, a)]*, which is a list of pairs of variable names and values, where each entry in the list corresponds to a row in the table.

- Express the table above in Haskell by creating *vars :: Env Double*.
- Create a function *varVal :: Env a → String → a* that returns the value of a variable, given an *Env* and a variable name. For instance, *varVal vars "d"* should return 7.4
- Create a function *eval :: Num a ⇒ Env a → E2 → a* that takes a value of the new *E2* data type and returns the corresponding number. For instance, *eval vars ((Con 2) 'Plus' (Var "a"))* == 3.5. Try it on the expressions from the first part, and verify that it works as expected.

TODO: Add an exercise to implement some property as a boolean functions (for example some from the semiring properties).

Exercise 1.3. *From exam 2017-08-22*

A semiring is a set R equipped with two binary operations $+$ and \cdot , called addition and multiplication, such that:

- $(R, +, 0)$ is a commutative monoid with identity element 0:

$$\begin{aligned}(a + b) + c &= a + (b + c) \\ 0 + a &= a + 0 = a \\ a + b &= b + a\end{aligned}$$

- $(R, \cdot, 1)$ is a monoid with identity element 1:

$$\begin{aligned}(a \cdot b) \cdot c &= a \cdot (b \cdot c) \\ 1 \cdot a &= a \cdot 1 = a\end{aligned}$$

- Multiplication left and right distributes over $(R, +, 0)$:

$$\begin{aligned}a \cdot (b + c) &= (a \cdot b) + (a \cdot c) \\ (a + b) \cdot c &= (a \cdot c) + (b \cdot c) \\ a \cdot 0 &= 0 \cdot a = 0\end{aligned}$$

1. Define a datatype $SR\ v$ for the language of semiring expressions (with variables of type v). (These are expressions formed from applying the semiring operations to the appropriate number of arguments, e.g., all the left hand sides and right hand sides of the above equations.)
2. Give a type signature for, and define, a general evaluator for $SR\ v$ expressions on the basis of an assignment function.

Exercise 1.4. *From exam 2016-03-15*

A *lattice* is a set L together with two operations \vee and \wedge (usually pronounced “sup” and “inf”) such that

- \vee and \wedge are associative:

$$\begin{aligned}\forall x, y, z \in L. \quad (x \vee y) \vee z &= x \vee (y \vee z) \\ \forall x, y, z \in L. \quad (x \wedge y) \wedge z &= x \wedge (y \wedge z)\end{aligned}$$

- \vee and \wedge are commutative:

$$\begin{aligned}\forall x, y \in L. \quad x \vee y &= y \vee x \\ \forall x, y \in L. \quad x \wedge y &= y \wedge x\end{aligned}$$

- \vee and \wedge satisfy the *absorption laws*:

$$\begin{aligned}\forall x, y \in L. \quad x \vee (x \wedge y) &= x \\ \forall x, y \in L. \quad x \wedge (x \vee y) &= x\end{aligned}$$

1. Define a datatype for the language of lattice expressions.
2. Define a general evaluator for *Lattice* expressions on the basis of an assignment function.

Exercise 1.5. *From exam 2016-08-23*

An *abelian monoid* is a set M together with a constant (nullary operation) $0 \in M$ and a binary operation $\oplus : M \rightarrow M \rightarrow M$ such that:

- 0 is a unit of \oplus

$$\forall x \in M. \ 0 \oplus x = x \oplus 0 = x$$

- \oplus is associative

$$\forall x, y, z \in M. \ x \oplus (y \oplus z) = (x \oplus y) \oplus z$$

- \oplus is commutative

$$\forall x, y \in M. \ x \oplus y = y \oplus x$$

1. Define a datatype *AbMonoidExp* for the language of abelian monoid expressions. (These are expressions formed from applying the monoid operations to the appropriate number of arguments, e.g., all the left hand sides and right hand sides of the above equations.)
2. Define a general evaluator for *AbMonoidExp* expressions on the basis of an assignment function.

Exercise 1.6. Read the full chapter and complete the definition of the instance for *Num* for the datatype *ComplexSyn*. Also add a constructor for variables to enable writing expressions like *(Var "z") :: toComplex 1*.

Exercise 1.7. Read the next few pages of Appendix I (in [Adams and Essex, 2010]) defining the polar view of Complex Numbers and try to implement complex numbers again, this time based on magnitude and phase for the semantics.

Exercise 1.8. Implement a simplifier *simp :: ComplexSyn r → ComplexSyn r* that handles a few cases like $0 * x = 0$, $1 * x = x$, $(a + b) * c = a * c + b * c$, ... What class context do you need to add to the type of *simp*?

Exercise 1.9. Functions and pairs (the “tupling transform”). From one function $f :: a \rightarrow (b, c)$ returning a pair, you can always make a pair of two functions $pf :: (a \rightarrow b, a \rightarrow c)$. Implement this transform:

$$f2p :: (a \rightarrow (b, c)) \rightarrow (a \rightarrow b, a \rightarrow c)$$

Also implement the opposite transform:

$$p2f :: (a \rightarrow b, a \rightarrow c) \rightarrow (a \rightarrow (b, c))$$

This kind of transformation is often useful, and it works also for n -tuples.

Exercise 1.10. There is also a “dual” to the tupling transform: to show this, implement these functions:

$$\begin{aligned} s2p &:: (Either\ b\ c \rightarrow a) \rightarrow (b \rightarrow a, c \rightarrow a) \\ p2s &:: (b \rightarrow a, c \rightarrow a) \rightarrow (Either\ b\ c \rightarrow a) \end{aligned}$$

Exercise 1.11. Counting values. Now assume we have *f2p*, *s2f*, etc used with three finite types with cardinalities *A*, *B*, and *C*. (For example, the cardinality of *Bool* is 2, the cardinality of *Weekday* is 7, etc.) Then what is the cardinality of *Either a b*? *(a, b)*? *a → b*? etc. These rules for computing the cardinality suggests that *Either* is similar to sum, *(,)* is similar to product and *(→)* to (flipped) power. These rules show that we can use many intuitions from high-school algebra when working with types.

Exercise 1.12. Functions as tuples. For any type *t* the type *Bool → t* is basically “the same” as the type *(t, t)*. Implement the two functions *isoR* and *isoL* forming an isomorphism:

$$\begin{aligned} isoR &:: (Bool \rightarrow t) \rightarrow (t, t) \\ isoL &:: (t, t) \rightarrow (Bool \rightarrow t) \end{aligned}$$

and show that $isoL \circ isoR = id$ and $isoR \circ isoL = id$.

Exercise 1.13. (From section 1.6.2): How is *liftSeq₁* related to *fmap*? *liftSeq₀* to *conSeq*?

TODO: remove or use some left overs:

```
data ComplexSy v r = Var v
                  | FromCart r r
                  | ComplexSy v r :+ ComplexSy v r
                  | ComplexSy v r :** ComplexSy v r
```

and the second parameter *r* makes is possible to express “complex numbers over” different base types (like *Double*, *Float*, *Integer*, etc.).

2 Logic and calculational proofs

Course learning outcomes:

- Knowledge and understanding
 - design and implement a DSL (Domain Specific Language) for a new domain
 - organize areas of mathematics in DSL terms
 - explain main concepts of elementary real and complex analysis, algebra, and linear algebra
- Skills and abilities
 - develop adequate notation for mathematical concepts
 - perform calculational proofs
 - use power series for solving differential equations
 - use Laplace transforms for solving differential equations
- Judgement and approach
 - discuss and compare different software implementations of mathematical concepts

This week we focus on “develop adequate notation for mathematical concepts” and “perform calculational proofs” (still in the context of “organize areas of mathematics in DSL terms”).

There will be a fair bit of theory: introducing propositional and first order logic, but also “applications” to mathematics: prime numbers, (ir)rational numbers, limit points, limits, etc.

module *DSLsofMath.W02* **where**

2.1 Propositional Calculus

Now we turn to the main topic of this week: logic and proofs.

TODO: tidy up the notes below

Swedish: Satslogik

a, b, c, \dots	names of propositions	
$False, True$	Constants	
And	\wedge	$\&$
Or	\vee	$ $
$Implies$	\Rightarrow	
Not	\neg	

Example:

```
sw :: PropCalc
sw = ((a -&- b) ==> (b -&- a))
  where a = Name "a"
        b = Name "b"
```

The example is based on the following embedding of propositional calculus terms:

```

data PropCalc = Con Bool
                | Name String
                | And PropCalc PropCalc
                | Or PropCalc PropCalc
                | Implies PropCalc PropCalc
                | Not PropCalc

(-&-) = And
(==>) = Implies

```

With this datatype we can write an evaluator to *Bool* which computes the truth value of a term given an environment:

```

type Name = String
evalPC :: (Name → Bool) → PropCalc → Bool
evalPC = error "Exercise"

```

The function *evalPC* translates from the syntactic to the semantic domain. (The evaluation function for a DSL describing a logic is often called *check* instead of *eval* but here we stick to *eval*.) Here *PropCalc* is the (abstract) *syntax* of the language of propositional calculus and *Bool* is the *semantic domain*. Alternatively, we can view $(Name \rightarrow Bool) \rightarrow Bool$ as the semantic domain. A value of this type is a mapping from a truth table to *Bool*. This mapping is often also tabulated as a truth table with one more “output” column.

As a first example, consider the proposition $t = \text{Implies } (Con \text{ False}) \ a$. We will use the shorter notation with just *T* for true and *F* for false. The truth table semantics of *t* is usually drawn as follows: one column for the name *a* listing all combinations of *T* and *F*, and one column for the result of evaluating the expression.

a	t
F	T
T	T

If we continue with the example *sw* from above we have two names *a* and *b* which together can have any of four combinations of true and false. After the name-columns are filled, we fill in the rest of the table one operation (column) at a time. The *&* columns become *F F F T* and finally the \Rightarrow column becomes true everywhere.

<i>a</i>	<i>&</i>	<i>b</i>	\Rightarrow	<i>b</i>	<i>&</i>	<i>a</i>
F	F	F	T	F	F	F
F	F	T	T	T	F	F
T	F	F	T	F	F	T
T	T	T	T	T	T	T

A proposition whose truth table is always true is called a *tautology*. Truth table verification is only viable for propositions with few names because of the exponential growth in the number of cases to check: we get 2^n cases for *n* names. (There are very good heuristic algorithms to look for tautologies even for thousands of names — but that is not part of this course.)

What we call “names” are often called “(propositional) variables” but we will soon add another kind of variables (and quantification over them) to the calculus.

2.2 First Order Logic (predicate logic)

Swedish: Första ordningens logik = predikatlogik

We now add *terms* as another datatype to the calculus. A *term* is either a (term) *variable* (like *x*, *y*, *z*), or the application of a *function symbol* (like *f*, *g*) to a suitable number of terms. If we

have the function symbols f of arity 2 and g of arity 3 we can form terms like $f(x, x)$, $g(y, z, z)$, $g(x, y, f(x, y))$, etc.

TODO: Add simple datatype for terms. (Perhaps *Rat* from further down?)

The names from the propositional calculus are generalised to *predicate symbols* of different arity. The predicate symbols can only be applied to terms, not to other predicate symbols or formulas. If we have the predicate symbols N of arity 0, P of arity 1 and Q of arity 2 we can form *formulas* like N , $P(x)$, $Q(f(x, x), y)$, etc.

Note that we have two separate layers: terms cannot contain formulas, but formulas normally contain terms.

The formulas introduced so far are all *atomic formulas* but we will add two more concepts: first the logical connectives from the propositional calculus: *And*, *Or*, *Implies*, *Not*, and then two quantifiers: “forall” (\forall) and “exists” (\exists).

An example FOL formula:

$$\forall x. P(x) \Rightarrow (\exists y. Q(f(x, x), y))$$

Note that FOL can only quantify over *term* variables, not over predicates. (Second order logic and higher order logic allow quantification over predicates.)

Another example: a formula stating that $+$ is commutative:

$$\forall x. \forall y. (x + y) = (y + x)$$

Note that $=$ is a binary predicate symbol while $+$ is a binary function symbol. Here is the same formula without infix operators:

$$\forall x. \forall y. Eq(plus(x, y), plus(y, x))$$

TODO: perhaps add simple **data FOL** for the formulas.

Forall quantification can be seen as a generalisation of *And*. First we can generalise the binary operator to an n -ary version: And_n . To prove $And_n A_1 A_2 \dots A_n$ we need a proof of each A_i . Thus we could define $And_n A_1 A_2 \dots A_n = A_1 \& A_2 \& \dots \& A_n$ where $\&$ is the infix version of binary *And*. The next step is to note that the formulas A_i can be generalised to $A(i)$ where i is a term variable and A is a unary predicate symbol. We can think of i ranging over an infinite collection of constant terms i_0, i_1, \dots . Then the final step is to introduce the notation $\forall i. A(i)$ for $A(i_1) \& A(i_2) \& \dots$. Now to prove $\forall x. P(x)$ it would be difficult to provide an infinite collection of proofs of $P(x_i)$. Instead the standard procedure is to introduce a fresh constant term a and prove $P(a)$. Another way to view this is to say that a proof of $\forall x. P x$ is a function f from terms to proofs such that $f t$ is a proof of $P t$ for all terms t .

Note that the syntactic rule for $\forall x. b$ is similar to the rule for a function definition, $f x = b$, and for anonymous functions, $\lambda x \rightarrow b$. Just as in those cases we say that the variable x is *bound* in b and that the *scope* of the variable binding extends until the end of b (but not further). The scoping of x in $\exists x. b$ is the same as in $\forall x. b$.

One common source of confusion in mathematical (and other semi-formal) texts is that variable binding sometimes is implicit. A typical example is equations: $x^2 + 2 * x + 1 = 0$ usually means roughly $\exists x. x^2 + 2 * x + 1 = 0$. We write “roughly” here because the scope of x very often extends to some text after the equation where something more is said about the solution x .

TODO: Add an example showing how to prove

$$\forall a. \forall b. \forall c. (a \rightarrow (b, c)) \Leftrightarrow (a \rightarrow b, a \rightarrow c)$$

2.3 An aside: Pure set theory

One way to build mathematics from the ground up is to start from pure set theory and define all concepts by translation to sets. We will only work with this as a mathematical domain to study, not as “the right way” of doing mathematics (there are other ways). The core of the language of pure set theory has the Empty set, the one-element set constructor Singleton, set Union, and Intersection. There are no “atoms” or “elements” to start from except for the empty set but it turns out that quite a large part of mathematics can still be expressed.

Natural numbers To talk about things like natural numbers in pure set theory they need to be encoded. Here is one such encoding (which is explored further in the first hand-in assignment).

$$\begin{aligned} \text{vonNeumann } 0 &= \text{Empty} \\ \text{vonNeumann } (n + 1) &= \text{Union } (\text{vonNeumann } n) \\ &\quad (\text{Singleton } (\text{vonNeumann } n)) \end{aligned}$$

Pairs Definition: A pair (a, b) is encoded as $\{\{a\}, \{a, b\}\}$.

TODO: merge the text below and above

As an example term language we can use pure (untyped) set theory. We have a nullary function symbol $\{\}$ for the empty set (sometimes written \emptyset) and a unary function symbol S for the function that builds a singleton set from an “element”. In pure set theory we don’t actually have any “elements” to start from: every term denotes a set. All non-variable terms so far are $\{\}$, $S \{\}$, $S (S \{\})$, \dots . The first set is empty but all the others are one-element sets.

Next we add two binary function symbols for union and intersection of sets (denoted by terms). Using union we can build sets of more than one element, for example $\text{Union } (S \{\}) (S (S \{\}))$ which has two “elements”: $\{\}$ and $S \{\}$.

FOL does not have function definitions or recursion, but in a suitable meta-language (like Haskell) we can write a function that creates a set with n elements (for any natural number n) as a term in FOL:

$$\begin{aligned} vN \ 0 &= \{\} \\ vN \ (n + 1) &= \text{step } (vN \ n) \\ \text{step } x &= \text{Union } x \ (S \ x) \end{aligned}$$

If we use conventional set notation we get $vN \ 0 = \{\}$, $vN \ 1 = \{\{\}\}$, $vN \ 2 = \{\{\}, \{\{\}\}\}$, $vN \ 3 = \{\{\}, \{\{\}\}, \{\{\}, \{\{\}\}\}$, etc. If we use the shorthand \bar{n} for $vN \ n$ we see that $\bar{0} = \{\}$, $\bar{1} = \{\bar{0}\}$, $\bar{2} = \{\bar{0}, \bar{1}\}$, $\bar{3} = \{\bar{0}, \bar{1}, \bar{2}\}$ and, in general, that \bar{n} has cardinality n . The function vN is explored in more detail in the first assignment of the DSLsofMath course.

The constructions presented so far show that, even starting from no elements, we can embed all natural numbers in pure set theory. We can also embed unordered pairs: $\{a, b\} \stackrel{\text{def}}{=} \text{Union } (S \ a) (S \ b)$ and normal ordered pairs: $(a, b) \stackrel{\text{def}}{=} \{S \ a, \{a, b\}\}$. With a bit more machinery it is possible to step by step encode \mathbb{N} , \mathbb{Z} , \mathbb{Q} , \mathbb{R} , \mathbb{C} .

TODO: cite The Haskell Road to Logic, Math and Programming Kees Doets and Jan van Eijck <https://fdit-www.cs.uni-dortmund.de/~peter/PS07/HR.pdf>

2.4 Back to quantifiers

After this detour through untyped set land let us get back to the most powerful concept of FOL: the quantifiers. We have already seen how the “forall” quantifier can be seen as a generalisation of *And* and in the same way we can see the “exists” quantifier as a generalisation of *Or*.

First we generalise the binary Or to an n -ary Or_n . To prove $Or_n A_1 A_2 \dots A_n$ is enough (and necessary) to find one i for which we can prove A_i . As before we then take the step from a family of formulas A_i to one unary predicate A expressing the formulas $A(i)$ for the term variable i . Then the final step is to “or” all these formulas to obtain $\exists i. A i$.

At this point it is good to sum up and compare the two quantifiers and how to prove them:

(t, b_t) is a proof of $\exists x. P(x)$ if b_t is a proof of $P(t)$.
 f is a proof of $\forall x. P(x)$ if $f t$ is a proof of $P(t)$ for all t .

If we abbreviate “is a proof” as $:$ and use the Haskell convention for function application we get

$(t, b_t) : \exists x. P x$ **if** $b_t : P t$
 $f : \forall x. P x$ **if** $f t : P t$ for all t

This now very much looks like type rules, and that is not a coincidence. The Curry-Howard correspondence says that we can think of propositions as types and proofs as “programs”. These typing judgments are not part of FOL, but the correspondence is used quite a bit in this course to keep track of proofs.

TODO: Add more about Curry-Howard (the binary logical connectives, etc.)

TODO: find the right place for the a note that the type of tuples is isomorphic to the (dependent) function type $\{i : 1..n\} \rightarrow A_i$.

TODO: Add typed quantification for Exists.

(Roughly: $(\exists x : T. P x) = (\exists x. T x \ \& \ P x)$.)

2.5 Proof by contradiction

Let’s try to express and prove the irrationality of the square root of 2. We have two main concepts involved: the predicate “irrational” and the function “square root of”. The square root function (for positive real numbers) can be specified by $r = \sqrt{s}$ iff $r^2 = s$ and $r : \mathbb{R}$. The formula “ x is irrational” is just $\neg (R x)$ where R is the predicate “is rational”.

$$R x = \exists a : \mathbb{N}. \exists b : \mathbb{N}_{>0}. b * x = a \ \& \ GCD(a, b) = 1$$

The classical way to prove a negation $\neg P$ is to assume P and derive something absurd (some Q and $\neg Q$, for example). Lets take $P = R r$ and $Q = GCD(a, b) = 1$. Assuming P we immediately get Q so what we need is to prove $\neg Q$, that is $GCD(a, b) \neq 1$. We can use the equations $b * r = a$ and $r^2 = 2$. Squaring the first equation and using the second we get $b^2 * 2 = a^2$. Thus a^2 is even, which means that a is even, thus $a = 2 * c$ for some c . But then $b^2 * 2 = a^2 = 4 * c^2$ which means that $b^2 = 2 * c^2$. By the same reasoning again we have that also b is even. But then $GCD(a, b) \geq 2$ which implies $\neg Q$.

To sum up: by assuming P we can prove both Q and $\neg Q$. Thus, by contradiction $\neg P$ must hold.

2.6 Proof by cases

As another example, let’s prove that there are two irrational numbers p and q such that $p^2 q$ is rational.

$$S = \exists p. \exists q. \neg (R p) \ \& \ \neg (R q) \ \& \ R(p^2 q)$$

We know from above that $r = \sqrt{2}$ is irrational, so as a first attempt we could set $p = q = r$. Then we have satisfied two of the three clauses ($\neg (R\ p)$ and $\neg (R\ q)$). What about the third clause: is $x = p \wedge q = r \wedge r$ rational? We can reason about two possible cases, one of which has to hold: $R\ x$ or $\neg (R\ x)$.

Case 1: $R\ x$ holds. Then we have a proof of S with $p = q = r = \sqrt{2}$.

Case 2: $\neg (R\ x)$ holds. Then we have another irrational number x to play with. Let's try $p = x$ and $q = r$. Then $p \wedge q = x \wedge r = (r \wedge r) \wedge r = r \wedge (r * r) = r \wedge 2 = 2$ which is clearly rational. Thus, also in this case we have a proof of S , but now with $p = r \wedge r$ and $q = r$.

To sum up: yes, there are irrational numbers such that their power is rational. We can prove the existence without knowing what numbers p and q actually are!

2.7 Functions as proofs

To prove a formula $P \Rightarrow Q$ we assume a proof $p : P$ and derive a proof $q : Q$. Such a proof can be expressed as $(\lambda p \rightarrow q) : (P \Rightarrow Q)$: a proof of an implication is a function from proofs to proofs.

As we saw earlier, a similar rule holds for the “forall” quantifier: a function f from terms t to proofs of $P\ t$ is a proof of $\forall x. P\ x$.

A very common kind of formula is “typed quantification”: if a type (a set) S of terms can be described as those that satisfy the unary predicate T we can introduce the short-hand notation

$$(\forall x : T. P\ x) = (\forall x. T\ x \Rightarrow P\ x)$$

A proof of this is a two-argument function p which takes a term and a proof to a proof.

In pseudo-Haskell we can express the implication laws as follows:

$$\begin{aligned} \text{impIntro} &: (A \rightarrow B) \rightarrow (A \Rightarrow B) \\ \text{impElim} &: (A \Rightarrow B) \rightarrow (A \rightarrow B) \end{aligned}$$

It should come as no surprise that this “API” can be implemented by $(\Rightarrow) = (\rightarrow)$, which means that both *impIntro* and *impElim* can be implemented as *id*.

Similarly we can express the universal quantification laws as:

$$\begin{aligned} \forall\text{-Intro} &: ((a : \text{Term}) \rightarrow P\ a) \rightarrow (\forall x. P\ x) \\ \forall\text{-Elim} &: (\forall x. P\ x) \rightarrow ((a : \text{Term}) \rightarrow P\ a) \end{aligned}$$

To actually implement this we need a *dependent* function type, which Haskell does not provide. But we can still use it as a tool for understanding and working with logic formulas and mathematical proofs.

Haskell supports limited forms of dependent types and more is coming every year but for proper dependently typed programming I recommend the language Agda.

2.8 Proofs for *And* and *Or*

TODO: textify

$$\begin{aligned} \text{andIntro} &: P \rightarrow Q \rightarrow P \ \& \ Q \\ \text{andElimL} &: P \ \& \ Q \rightarrow P \\ \text{andElimR} &: P \ \& \ Q \rightarrow Q \end{aligned}$$

If we see these introduction and elimination rules as an API, what would be a reasonable implementation of the datatype $P \& Q$? A type of pairs! Then we see that the corresponding Haskell functions would be

```
pair :: p → q → (p, q)  -- andIntro
fst  :: (p, q) → p       -- andElimL
snd  :: (p, q) → q       -- andElimR

orIntroL : P → P | Q
orIntroR : Q → P | Q
orElim   : (P ⇒ R) → (Q ⇒ R) → ((P | Q) ⇒ R)
```

Here the implementation type can be a labelled sum type, also called disjoint union and in Haskell: *Either*.

```
data Either p q = Left p | Right q
  -- Left is orIntroL, Right is orIntroR
either :: (p → r) → (q → r) → Either p q → r
either l r (Left x)  = l x
either l r (Right y) = r y
```

TODO: include *Either* in the Haskell intro at the end of the previous chapter

2.9 Case study: there is always another prime

As an example of combining forall, exists and implication let us turn to one statement of the fact that there are infinitely many primes. If we assume we have a unary predicate expressing that a number is prime and a binary (infix) predicate ordering the natural numbers we can define a formula *IP* for “Infinite many Primes” as follows:

$$IP = \forall n. \text{ Prime } n \Rightarrow \exists m. \text{ Prime } m \ \& \ m > n$$

Combined with the fact that there is at least one prime (like 2) we can repeatedly refer to this statement to produce a never-ending stream of primes.

To prove this formula we first translate from logic to programs as described above. We can translate step by step, starting from the top level. The forall-quantifier translates to a (dependent) function type $(n : \text{Term}) \rightarrow$ and the implication to a normal function type $\text{Prime } n \rightarrow$. The exists-quantifier translates to a (dependent) pair type $((m : \text{Term}), \dots)$ and finally the $\&$ translates into a pair type. Putting all this together we get a type signature for any *proof* of the theorem:

$$\text{proof} : (n : \text{Term}) \rightarrow \text{Prime } n \rightarrow ((m : \text{Term}), (\text{Prime } m, m > n))$$

Now we can start filling in the definition of *proof* as a two-argument function returning a nested pair:

```
proof n np = (m, (pm, gt))
  where m' = 1 + factorial n
        m = {- some non-trivial prime factor of m' -}
        pm = {- a proof that m is prime -}
        gt = {- a proof that m > n -}
```

The proof *pm* is the core of the theorem. First, we note that for any $2 \leq p \leq n$ we have

$$\begin{aligned}
m' \% p &= \\
(1 + n!) \% p &= \\
1 \% p + (n!) \% p &= \\
1 + 0 &= \\
1
\end{aligned}$$

Thus m' is not divisible by any number from 2 to n . But is it a prime? If m' is prime then $m = m'$ and the proof is done (because $1 + n! \geq 1 + n > n$). Otherwise, let m be a prime factor of m' (thus $m' = m * q$, $q > 1$). Then $1 = m' \% p = (m \% p) * (q \% p)$ which means that neither m nor q are divisible by p (otherwise the product would be zero). Thus they must both be $> n$. QED.

Note that the proof can be used to define a somewhat useful function which takes any prime number to some larger prime number. We can compute a few example values:

$$\begin{aligned}
2 &\mapsto 3 & (1+2!) \\
3 &\mapsto 7 & (1+3!) \\
5 &\mapsto 11 & (1+5! = 121 = 11*11) \\
7 &\mapsto 71 & \dots
\end{aligned}$$

2.10 Existential quantification as a pair type

We mentioned before that existential quantification can be seen as as a “big *Or*” of a family of formulas $P a$ for all terms a . This means that to prove the quantification, we only need exhibit one witness and one proof for that member of the family.

$$\exists\text{-Intro} : (a : \text{Term}) \rightarrow P a \rightarrow (\exists x. P x)$$

For binary *Or* the “family” only had two members, one labelled L for *Left* and one R for *Right*, and we used one introduction rule for each. Here, for the generalisation of *Or*, we have unified the two rules into one with an added parameter a corresponding to the label which indicates the family member.

In the other direction, if we look at the binary elimination rule, we see the need for two arguments to be sure of how to prove the implication for any family member of the binary *Or*.

$$\text{orElim} : (P \Rightarrow R) \rightarrow (Q \Rightarrow R) \rightarrow ((P \mid Q) \Rightarrow R)$$

The generalisation unifies these two to one family of arguments. If we can prove R for each member of the family, we can be sure to prove R when we encounter some family member:

$$\exists\text{-Elim} : ((a : \text{Term}) \rightarrow P a \Rightarrow R) \rightarrow (\exists x. P x) \Rightarrow R$$

The datatype corresponding to $\exists x. P x$ is a pair of a witness a and a proof of $P a$. We sometimes write this type $(a : \text{Term}, P a)$.

2.11 Basic concepts of calculus

Now we have built up quite a bit of machinery to express logic formulas and proofs. It is time time to apply it to some concepts in calculus. We start we the concept of “limit point” which is used in the formulation of different properties of limits of functions.

TODO: Perhaps start with the “expression” $\lim_{x \rightarrow x_0} f(x)$ and explain that not all $x _ \theta$ make sense, etc. [For context and motivation.]

TODO: Or talk a bit about open and closed sets. (Open set = every point is internal = there is some wiggle-room around each point in the set. Closed set contains all its limit points.)

Limit point Definition (adapted from Rudin [1964], page 28): Let X be a subset of \mathbb{R} . A point $p \in \mathbb{R}$ is a limit point of X iff for every $\epsilon > 0$, there exists $q \in X$ such that $q \neq p$ and $|q - p| < \epsilon$.

To express “Let X be a subset of \mathbb{R} ” we write $X : \mathcal{P} \mathbb{R}$. In general, the operator \mathcal{P} takes a set (here \mathbb{R}) to the set of all its subsets.

$$\begin{aligned} \text{Limp} : \mathbb{R} &\rightarrow \mathcal{P} \mathbb{R} \rightarrow \text{Prop} \\ \text{Limp } p \ X &= \forall \epsilon > 0. \ \exists q \in X - \{p\}. \ |q - p| < \epsilon \end{aligned}$$

Notice that q depends on ϵ . Thus by introducing a function $\text{get}q$ we can move the \exists out.

$$\begin{aligned} \text{type } Q &= \mathbb{R}_{>0} \rightarrow (X - \{p\}) \\ \text{Limp } p \ X &= \exists \text{get}q : Q. \ \forall \epsilon > 0. \ |\text{get}q \ \epsilon - p| < \epsilon \end{aligned}$$

Next: introduce the “open ball” function B .

$$\begin{aligned} B : \mathbb{R} &\rightarrow \mathbb{R}_{>0} \rightarrow \mathcal{P} \mathbb{R} \\ B \ c \ r &= \{x \mid |x - c| < r\} \end{aligned}$$

$B \ c \ r$ is often called an “open ball” around c of radius r . On the real line this “open ball” is just an open interval, but with complex c or in more dimensions the term feels more natural. In every case $B \ c \ r$ is an open set of values (points) of distance less than r from c . The open balls around c are special cases of *neighbourhoods of c* which can have other shapes but must contain some open ball.

Using B we get

$$\text{Limp } p \ X = \exists \text{get}q : Q. \ \forall \epsilon > 0. \ \text{get}q \ \epsilon \in B \ p \ \epsilon$$

Example 1: Is $p = 1$ a limit point of $X = \{1\}$? No! $X - \{p\} = \{\}$ (there is no $q \neq p$ in X), thus there cannot exist a function $\text{get}q$ because it would have to return elements in the empty set!

Example 2: Is $p = 1$ a limit point of the open interval $X = (0, 1)$? First note that $p \notin X$, but it is “very close” to X . A proof needs a function $\text{get}q$ which from any ϵ computes a point $q = \text{get}q \ \epsilon$ which is in both X and $B \ 1 \ \epsilon$. We need a point q which is in X and *closer* than ϵ from 1. We can try with $q = 1 - \epsilon / 2$ because $|1 - (1 - \epsilon / 2)| = |\epsilon / 2| = \epsilon / 2 < \epsilon$ which means $q \in B \ 1 \ \epsilon$. We also see that $q \neq 1$ because $\epsilon > 0$. The only remaining thing to check is that $q \in X$. This is true for sufficiently small ϵ but the function $\text{get}q$ must work for all positive reals. We can use any value in X (for example $17 / 38$) for ϵ which are “too big” ($\epsilon \geq 2$). Thus our function can be

$$\begin{aligned} \text{get}q \ \epsilon \mid \epsilon < 2 &= 1 - \epsilon / 2 \\ \mid \text{otherwise} &= 17 / 38 \end{aligned}$$

A slight variation which is often useful would be to use \max to define $\text{get}q \ \epsilon = \max (17/38, 1 - \epsilon/2)$. Similarly, we can show that any internal point (like $1 / 2$) is a limit point.

Example 3: limit of an infinite discrete set X

$$X = \{1 / n \mid n \in \mathbb{N}_{>0}\}$$

Show that 0 is a limit point of X . Note (as above) that $0 \notin X$.

We want to prove $\text{Limp } 0 \ X$ which is the same as $\exists \text{get}q : Q. \ \forall \epsilon > 0. \ \text{get}q \ \epsilon \in B \ 0 \ \epsilon$. Thus, we need a function $\text{get}q$ which takes any $\epsilon > 0$ to an element of $X - \{0\} = X$ which is less than ϵ away from 0. Or, equivalently, we need a function $\text{get}n : \mathbb{R}_{>0} \rightarrow \mathbb{N}_{>0}$ such that $1 / n < \epsilon$. Thus, we need to find an n such that $1 / \epsilon < n$. If $1 / \epsilon$ would be an integer we could use the next integer $(1 + 1 / \epsilon)$, so the only step remaining is to round up:

$$\begin{aligned} \text{getq } \epsilon &= 1 / \text{getn } \epsilon \\ \text{getn } \epsilon &= 1 + \text{ceiling } (1 / \epsilon) \end{aligned}$$

Exercise: prove that 0 is the *only* limit point of X .

Proposition: If X is finite, then it has no limit points.

$$\forall p \in \mathbb{R}. \neg (\text{Limp } p \ X)$$

This is a good exercise in quantifier negation!

$$\begin{aligned} &\neg (\text{Limp } p \ X) \\ &= \{- \text{Def. of Limp} -\} \\ &\neg (\exists \text{getq} : Q. \forall \epsilon > 0. \text{getq } \epsilon \in B \ p \ \epsilon) \\ &= \{- \text{Negation of existential} -\} \\ &\forall \text{getq} : Q. \neg (\forall \epsilon > 0. \text{getq } \epsilon \in B \ p \ \epsilon) \\ &= \{- \text{Negation of universal} -\} \\ &\forall \text{getq} : Q. \exists \epsilon > 0. \neg (\text{getq } \epsilon \in B \ p \ \epsilon) \\ &= \{- \text{Simplification} -\} \\ &\forall \text{getq} : Q. \exists \epsilon > 0. |\text{getq } \epsilon - p| \geq \epsilon \end{aligned}$$

Thus, using the “functional interpretation” of this type we see that a proof needs a function *noLim*

$$\text{noLim} : (\text{getq} : Q) \rightarrow \mathbb{R}_{>0}$$

such that **let** $\epsilon = \text{noLim } \text{getq}$ **in** $|\text{getq } \epsilon - p| \geq \epsilon$.

Note that *noLim* is a *higher-order* function: it takes a function *getq* as an argument. How can we analyse this function to find a suitable ϵ ? The key here is that the range of *getq* is $X - \{p\}$ which is a finite set (not containing p). Thus we can enumerate all the possible results in a list $xs = [x_1, x_2, \dots, x_n]$, and measure their distances to p : $ds = \text{map } (\lambda x \rightarrow |x - p|) \ xs$. Now, if we let $\epsilon = \text{minimum } ds$ we can be certain that $|\text{getq } \epsilon - p| \geq \epsilon$ just as required (and $\epsilon \neq 0$ because $p \notin xs$).

Exercise: If $\text{Limp } p \ X$ we now know that X is infinite. Show how to construct an infinite sequence $a : \mathbb{N} \rightarrow \mathbb{R}$ of points in $X - \{p\}$ which gets arbitrarily close to p . Note that this construction can be seen as a proof of $\text{Limp } p \ X \Rightarrow \text{Infinite } X$.

The limit of a sequence TODO: transcribe the 2016 notes

Now we can move from limit points to the more familiar limit of a sequence. A sequence a is a function from \mathbb{N} to \mathbb{R} and we define a binary infix predicate *haslim* using a helper predicate P :

$$\begin{aligned} a \text{ haslim } L &= \forall \epsilon > 0. P \ a \ L \ \epsilon \\ P \ a \ L \ \epsilon &= \exists N : \mathbb{N}. \forall n \geq N. |a_n - L| < \epsilon \end{aligned}$$

TODO: perhaps swap the argument order in the definition of *Limp* to make it fit better with *haslim*.

Exercise: prove that $(a_1 \text{ haslim } L_1) \ \& \ (a_2 \text{ haslim } L_2)$ implies $(a_1 + a_2) \text{ haslim } (L_1 + L_2)$.

When we are not interested in the exact limit, just that it exists, we say that a sequence a is *convergent* when $\exists L. a \text{ haslim } L$.

2.12 Questions and answers from the exercise sessions week 2

Variables, *Env* and *lookup* This was frequently a source of confusion already the first week so there is already a question + answers earlier in this text. But here is an additional example to help clarify the matter.

```
data Rat v = RV v | FromI Integer | RPlus (Rat v) (Rat v) | RDiv (Rat v) (Rat v)
  deriving (Eq, Show)
newtype RatSem = RSem (Integer, Integer)
```

We have a type *Rat v* for the syntax trees of rational number expressions and a type *RatSem* for the semantics of those rational number expressions as pairs of integers. The constructor *RV* :: *v* → *Rat v* is used to embed variables with names of type *v* in *Rat v*. We could use *String* instead of *v* but with a type parameter *v* we get more flexibility at the same time as we get better feedback from the type checker. To evaluate some *e* :: *Rat v* we need to know how to evaluate the variables we encounter. What does “evaluate” mean for a variable? Well, it just means that we must be able to translate a variable name (of type *v*) to a semantic value (a rational number in this case). To “translate a name to a value” we can use a function (of type *v* → *RatSem*) so we can give the following implementation of the evaluator:

```
evalRat1 :: (v → RatSem) → (Rat v → RatSem)
evalRat1 ev (RV v)      = ev v
evalRat1 ev (FromI i)   = fromISem i
evalRat1 ev (RPlus l r) = plusSem (evalRat1 ev l) (evalRat1 ev r)
evalRat1 ev (RDiv l r)  = divSem (evalRat1 ev l) (evalRat1 ev r)
```

Notice that we simply added a parameter *ev* for “evaluate variable” to the evaluator. The rest of the definition follows a common pattern: recursively translate each subexpression and apply the corresponding semantic operation to combine the results: *RPlus* is replaced by *plusSem*, etc.

```
fromISem :: Integer → RatSem
fromISem i = RSem (i, 1)

plusSem :: RatSem → RatSem → RatSem
plusSem = undefined -- TODO: exercise

-- Division of rational numbers
divSem :: RatSem → RatSem → RatSem
divSem (RSem (a, b)) (RSem (c, d)) = RSem (a * d, b * c)
```

Often the first argument *ev* to the eval function is constructed from a list of pairs:

```
type Env v s = [(v, s)]
envToFun :: (Show v, Eq v) ⇒ Env v s → (v → s)
envToFun [] v = error ("envToFun: variable " ++ show v ++ " not found")
envToFun ((w, s) : env) v
  | w == v    = s
  | otherwise = envToFun env v
```

Thus, *Env v s* can be seen as an implementation of a “lookup table”. It could also be implemented using hash tables or binary search trees, but efficiency is not the point here. Finally, with *envToFun* in our hands we can implement a second version of the evaluator:

```
evalRat2 :: (Show v, Eq v) ⇒ (Env v RatSem) → (Rat v → RatSem)
evalRat2 env e = evalRat1 (envToFun env) e
```

The law of the excluded middle Many had problems with implementing the “law of the excluded middle” in the exercises and it is indeed a tricky property to prove. The key to implementing it lies in double negation and as that is encoded with higher order functions it gets a bit hairy.

TODO[Daniel]: more explanation

SET and PRED Several groups have had trouble grasping the difference between *SET* and *PRED*. This is understandable, because we have so far in the lectures mostly talked about term syntax + semantics, and not so much about predicate syntax and semantics. The one example of terms + predicates covered in the lectures is Predicate Logic and I never actually showed how eval (for the expressions) and check (for the predicates) is implemented.

As an example we can take our terms to be the rational number expressions defined above and define a type of predicates over those terms:

```

type Term v = Rat v
data RPred v = Equal    (Term v) (Term v)
              | LessThan (Term v) (Term v)
              | Positive  (Term v)
              | AND      (RPred v) (RPred v)
              | NOT      (RPred v)
deriving (Eq, Show)

```

Note that the first three constructors, *Eq*, *LessThan*, and *Positive*, describe predicates or relations between terms (which can contain term variables) while the two last constructors, *AND* and *NOT*, just combine such relations together. (Terminology: I often mix the words “predicate” and “relation”).

We have already defined the evaluator for the *Term v* type but we need to add a corresponding “evaluator” (called *check*) for the *RPred v* type. Given values for all term variables the predicate checker should just determine if the predicate is true or false.

```

checkRP :: (Eq v, Show v) => Env v RatSem -> RPred v -> Bool
checkRP env (Equal    t1 t2) = eqSem      (evalRat2 env t1) (evalRat2 env t2)
checkRP env (LessThan t1 t2) = lessThanSem (evalRat2 env t1) (evalRat2 env t2)
checkRP env (Positive  t1)   = positiveSem (evalRat2 env t1)
checkRP env (AND p q) = (checkRP env p) & (checkRP env q)
checkRP env (NOT p)  = not (checkRP env p)

```

Given this recursive definition of *checkRP*, the semantic functions *eqSem*, *lessThanSem*, and *positiveSem* can be defined by just working with the rational number representation:

```

eqSem      :: RatSem -> RatSem -> Bool
lessThanSem :: RatSem -> RatSem -> Bool
positiveSem :: RatSem -> Bool
eqSem      = error "TODO"
lessThanSem = error "TODO"
positiveSem = error "TODO"

```

2.13 More general code for first order languages

“överkurs”

It is possible to make one generic implementation which can be specialised to any first order language.

TODO: add explanatory text

- *Term* = Syntactic terms
- *n* = names (of atomic terms)
- *f* = function names
- *v* = variable names
- *WFF* = Well Formed Formulas
- *p* = predicate names

```

data Term n f v = N n | F f [Term n f v] | V v
deriving Show
data WFF n f v p =
  P p [Term n f v]
| Equal (Term n f v) (Term n f v)
| And (WFF n f v p) (WFF n f v p)
| Or (WFF n f v p) (WFF n f v p)
| Equiv (WFF n f v p) (WFF n f v p)
| Impl (WFF n f v p) (WFF n f v p)
| Not (WFF n f v p)
| FORALL v (WFF n f v p)
| EXISTS v (WFF n f v p)
deriving Show

```

TODO: Perhaps introduce GADT datatype notation in exercises

2.14 Exercises: abstract FOL

```

{-# LANGUAGE EmptyCase #-}
import AbstractFOL

```

2.14.1 Exercises

The propositional fragment of FOL is given by the rules for \wedge , \rightarrow , \longleftrightarrow , \neg , \vee .

We can use the Haskell type checker to check proofs in this fragment, using the functional models for introduction and elimination rules. Examine Fig. 3 (also available in the file **AbstractFOL.lhs**), which introduces an empty datatype for every connective (except \longleftrightarrow), and corresponding types for the introduction and elimination rules. The introduction and elimination rules are explicitly left “undefined”, but we can still combine them and type check the results. For example:

```

example0 :: And p q → And q p
example0 evApq = andIntro (andElimR evApq) (andElimL evApq)

```

(The variable name *evApq* is a mnemonic for “evidence of *And p q*”.)

```

module AbstractFOL (And, Not, Or, Impl,
                    andIntro, andElimL, andElimR,
                    orIntroL, orIntroR, orElim,
                    notIntro, notElim,
                    implIntro, implElim) where

data And  p q
data Not  p
data Or   p q
data Impl p q

andIntro  :: p → q → And p q
andElimL :: And p q → p
andElimR :: And p q → q

orIntroL  :: p → Or p q
orIntroR  :: q → Or p q
orElim    :: Or p q → (p → r) → (q → r) → r

notElim   :: Not (Not p) → p
notIntro  :: (p → And q (Not q)) → Not p

implIntro :: (p → q) → Impl p q
implElim  :: Impl p q → p → q

andIntro   = undefined; andElimL = undefined; andElimR = undefined
orIntroL   = undefined; orIntroR = undefined; orElim    = undefined
notElim    = undefined; notIntro  = undefined
implIntro  = undefined; implElim  = undefined

```

Figure 3: The Haskell module *AbstractFOL*.

Notice that Haskell will not accept

$$example0 \text{ evApq} = andIntro (andElimL \text{ evApq}) (andElimR \text{ evApq})$$

unless we change the type.

Another example:

$$\begin{aligned} example1 &:: And\ q\ (Not\ q) \rightarrow p \\ example1\ \text{evAqnq} &= notElim\ (notIntro\ (\lambda hyp \rightarrow \text{evAqnq})) \end{aligned}$$

Short technical note For these exercises, you might find it useful to take a look at typed holes, a feature which is enabled by default in GHC and available (the same way as the language extension `EmptyCase` above) from version 7.8.1 onwards: https://wiki.haskell.org/GHC/Typed_holes.

If you are familiar with Agda, these will be familiar to use. In summary, when trying to code up the definition of some expression (which you have already typed) you can get GHC’s type checker to help you out a little in seeing how far you might be from forming the expression you want. That is, how far you are from constructing something of the appropriate type.

Take *example0* below, and say you are writing:

$$example0\ e = andIntro\ (_ \ e)\ _$$

When loading the module, GHC will tell you which types your holes (marked by “_”) should have for the expression to be type correct.

On to the exercises.

Exercise 2.1. Prove these three theorems (for arbitrary p and q):

$$\begin{aligned} &Impl\ (And\ p\ q)\ q \\ &Or\ p\ q \rightarrow Or\ q\ p \\ &Or\ p\ (Not\ p) \end{aligned}$$

Exercise 2.2. Translate to Haskell and prove the De Morgan laws:

$$\begin{aligned} \neg (p \vee q) &\longleftrightarrow \neg p \wedge \neg q \\ \neg (p \wedge q) &\longleftrightarrow \neg p \vee \neg q \end{aligned}$$

(translate equivalence to conjunction of two implications).

Exercise 2.3. So far, the implementation of the datatypes has played no role. To make this clearer: define the types for connectives in *AbstractFol* in any way you wish, e.g.:

$$\begin{aligned} And\ p\ q &= A\ () \\ Not\ p &= B\ p \end{aligned}$$

etc. as long as you still export only the data types, and not the constructors. Convince yourself that the proofs given above still work and that the type checker can indeed be used as a poor man’s proof checker.

Exercise 2.4. The introduction and elimination rules suggest that some implementations of the datatypes for connectives might be more reasonable than others. We have seen that the type of evidence for $p \rightarrow q$ is very similar to the type of functions $p \rightarrow q$, so it would make sense to define

type *Impl* $p\ q = (p \rightarrow q)$

Similarly, \wedge -*ElimL* and \wedge -*ElimR* behave like the functions *fst* and *snd* on pairs, so we can take

type *And* $p\ q = (p, q)$

while the notion of proof by cases is very similar to that of writing functions by pattern-matching on the various clauses, making $p \vee q$ similar to *Either*:

type *Or* $p\ q = \text{Either } p\ q$

1. Define and implement the corresponding introduction and implementation rules as functions.
2. Compare proving the distributivity laws

$$\begin{aligned} (p \wedge q) \vee r &\longleftrightarrow (p \vee r) \wedge (q \vee r) \\ (p \vee q) \wedge r &\longleftrightarrow (p \wedge r) \vee (q \wedge r) \end{aligned}$$

using the “undefined” introduction and elimination rules, with writing the corresponding functions with the given implementations of the datatypes. The first law, for example, requires a pair of functions:

$$\begin{aligned} &(\text{Either } (p, q)\ r \rightarrow (\text{Either } p\ r, \text{Either } q\ r) \\ &, (\text{Either } p\ r, \text{Either } q\ r) \rightarrow \text{Either } (p, q)\ r \\ &) \end{aligned}$$

Moral: The natural question is: is it true that every time we find an implementation using the “pairs, \rightarrow , *Either*” translation of sentences, we can also find one using the “undefined” introduction and elimination rules? The answer, perhaps surprisingly, is *yes*, as long as the functions we write are total. This result is known as *the Curry-Howard isomorphism*.

Exercise 2.5. Can we extend the Curry-Howard isomorphism to formulas with \neg ? In other words, is there a type that we could use to define *Not* p , which would work together with pairs, \rightarrow , and *Either* to give a full translation of sentential logic?

Unfortunately, we cannot. The best that can be done is to define an empty type

data *Empty*

and define *Not* as

type *Not* $p = p \rightarrow \text{Empty}$

The reason for this definition is: when p is *Empty*, the type *Not* p is not empty: it contains the identity

$$\begin{aligned} \text{idEmpty} &:: \text{Empty} \rightarrow \text{Empty} \\ \text{isEmpty } \text{evE} &= \text{evE} \end{aligned}$$

When p is not *Empty* (and therefore is true), there is no (total, defined) function of type $p \rightarrow \text{Empty}$, and therefore *Not* p is false.

Moreover, mathematically, an empty set acts as a contradiction: there is exactly one function from the empty set to any other set, namely the empty function. Thus, if we had an element of the empty set, we could obtain an element of any other set.

Now to the exercise:

Implement *notIntro* using the definition of *Not* above, i.e., find a function

$$\text{notIntro} :: (p \rightarrow (q, q \rightarrow \text{Empty})) \rightarrow (p \rightarrow \text{Empty})$$

Using

$$\begin{aligned} \text{contraHey} &:: \text{Empty} \rightarrow p \\ \text{contraHey } \text{evE} &= \text{case } \text{evE} \text{ of } \{ \} \end{aligned}$$

prove

$$q \wedge \neg q \rightarrow p$$

You will, however, not be able to prove $p \vee \neg p$ (try it!).

Prove

$$\neg p \vee \neg q \rightarrow \neg(p \wedge q)$$

but you will not be able to prove the converse.

Exercise 2.6. The implementation *Not* $p = p \rightarrow \text{Empty}$ is not adequate for representing the sentential fragment of FOL, but it is adequate for *constructive logic* (also known as *intuitionistic*). In constructive logic, the $\neg p$ is *defined* as $p \rightarrow \perp$, and the following elimination rule is given for \perp

TODO: This notation has not been introduced: replace with the stype used above.

$$\begin{aligned} &\dots \\ i. &\perp \\ &\dots \\ j. &p \text{ } (\perp\text{-Elim} : i) \end{aligned}$$

corresponding to the principle that everything follows from a contradiction (“if you believe \perp , you believe everything”).

Every sentence provable in constructive logic is provable in classical logic, but the converse, as we have seen in the previous exercise, does not hold. On the other hand, there is no sentence in classical logic which would be contradicted in constructive logic. In particular, while we cannot prove $p \vee \neg p$, we *can* prove (constructively!) that there is no p for which $\neg(p \vee \neg p)$, i.e., that the sentence $\neg \neg (p \vee \neg p)$ is always true.

Show this by implementing the following function:

$$\text{noContra} :: (\text{Either } p \text{ } (p \rightarrow \text{Empty}) \rightarrow \text{Empty}) \rightarrow \text{Empty}$$

Exercise 2.7. From exam 2016-08-23

Consider the classical definition of continuity:

Definition: Let $X \subseteq \mathbb{R}$, and $c \in X$. A function $f : X \rightarrow \mathbb{R}$ is *continuous at c* if for every $\epsilon > 0$, there exists $\delta > 0$ such that, for every x in the domain of f , if $|x - c| < \delta$, then $|f x - f c| < \epsilon$.

1. Write the definition formally, using logical connectives and quantifiers.
2. Introduce functions and types to simplify the definition.
3. Prove the following proposition: If f and g are continuous at c , $f + g$ is continuous at c .

Exercise 2.8. From exam 2017-08-22

Adequate notation for mathematical concepts and proofs (or “50 shades of continuity”).

A formal definition of “ $f : X \rightarrow \mathbb{R}$ is continuous” and “ f is continuous at c ” can be written as follows (using the helper predicate Q):

$$\begin{aligned} C(f) &= \forall c : X. \text{Cat}(f, c) \\ \text{Cat}(f, c) &= \forall \epsilon > 0. \exists \delta > 0. Q(f, c, \epsilon, \delta) \\ Q(f, c, \epsilon, \delta) &= \forall x : X. |x - c| < \delta \Rightarrow |f x - f c| < \epsilon \end{aligned}$$

By moving the existential quantifier outwards we can introduce the function $\text{get}\delta$ which computes the required δ from c and ϵ :

$$C'(f) = \exists \text{get}\delta : X \rightarrow \mathbb{R}_{>0} \rightarrow \mathbb{R}_{>0}. \forall c : X. \forall \epsilon > 0. Q(f, c, \epsilon, \text{get}\delta c \epsilon)$$

Now, consider this definition of *uniform continuity*:

Definition: Let $X \subseteq \mathbb{R}$. A function $f : X \rightarrow \mathbb{R}$ is *uniformly continuous* if for every $\epsilon > 0$, there exists $\delta > 0$ such that, for every x and y in the domain of f , if $|x - y| < \delta$, then $|f x - f y| < \epsilon$.

1. Write the definition of $UC(f)$ = “ f is uniformly continuous” formally, using logical connectives and quantifiers. Try to use Q .
2. Transform $UC(f)$ into a new definition $UC'(f)$ by a transformation similar to the one from $C(f)$ to $C'(f)$. Explain the new function $\text{new}\delta$ introduced.
3. Prove that $\forall f : X \rightarrow \mathbb{R}. UC'(f) \Rightarrow C'(f)$. Explain your reasoning in terms of $\text{get}\delta$ and $\text{new}\delta$.

2.14.2 More exercises

Preliminary remarks

- when asked to “sketch an implementation” of a function, you must explain how the various results might be obtained from the arguments, in particular, why the evidence required as output may result from the evidence given as input. You may use all the facts you know (for instance, that addition is monotonic) without formalisation.

- to keep things short, let us abbreviate a significant chunk of the definition of a *haslim* L (see section 2.11) by

$$\begin{aligned} P : Seq\ X \rightarrow X \rightarrow \mathbb{R}_{>0} \rightarrow Prop \\ P\ a\ L\ \epsilon = \exists\ N : \mathbb{N}. \ \forall\ n : \mathbb{N}. \ (n \geq N) \rightarrow (|a_n - L| < \epsilon) \end{aligned}$$

Exercise 2.9. Consider the statement:

The sequence $\{a_n\} = (0, 1, 0, 1, \dots)$ does not converge.

- Define the sequence $\{a_n\}$ as a function $a : \mathbb{N} \rightarrow \mathbb{R}$.
- The statement “the sequence $\{a_n\}$ is convergent” is formalised as

$$\exists\ L : \mathbb{R}. \ \forall\ \epsilon > 0. \ P\ a\ L\ \epsilon$$

The formalisation of “the sequence $\{a_n\}$ is not convergent” is therefore

$$\neg \exists\ L : \mathbb{R}. \ \forall\ \epsilon > 0. \ P\ a\ L\ \epsilon$$

Simplify this expression using the rules

$$\begin{aligned} \neg (\exists\ x. \ P\ x) &\longleftrightarrow (\forall\ x. \ \neg (P\ x)) \\ \neg (\forall\ x. \ P\ x) &\longleftrightarrow (\exists\ x. \ \neg (P\ x)) \\ \neg (P \rightarrow Q) &\longleftrightarrow P \wedge \neg Q \end{aligned}$$

The resulting formula should have no \neg in it (that’s possible because the negation of $<$ is \geq).

- Give a functional interpretation of the resulting formula.
- Sketch an implementation of the function, considering two cases: $L \neq 0$ and $L = 0$.

Exercise 2.10. Consider the statement:

The limit of a convergent sequence is unique.

- There are many ways of formalising this in FOL. For example:

$$\begin{aligned} \text{let } Q\ a\ L = \forall\ \epsilon > 0. \ P\ a\ L\ \epsilon \\ \text{in } \forall\ L_1 : \mathbb{R}. \ \forall\ L_2 : \mathbb{R}. \ (Q\ a\ L_1 \wedge Q\ a\ L_2) \rightarrow L_1 = L_2 \end{aligned}$$

i.e., if the sequence converges to two limits, then they must be equal, or

$$\forall\ L_1 : \mathbb{R}. \ \forall\ L_2 : \mathbb{R}. \ Q\ a\ L_1 \wedge L_1 \neq L_2 \rightarrow \neg Q\ a\ L_2$$

i.e., if a sequence converges to a limit, then it doesn’t converge to anything that isn’t the limit.

Simplify the latter alternative to eliminate the negation and give functional representations of both.

- Choose one of the functions and sketch an implementation of it.

3 Types in Mathematics

```
{-# LANGUAGE FlexibleInstances #-}  
module DSLsofMath.W03 where  
import Data.Char (toUpper)
```

3.1 Types in mathematics

Types are sometimes mentioned explicitly in mathematical texts:

- $x \in \mathbb{R}$
- $\sqrt{} : \mathbb{R}_{\geq 0} \rightarrow \mathbb{R}_{\geq 0}$
- $(_)^2 : \mathbb{R} \rightarrow \mathbb{R}$ or, alternatively but *not* equivalently
- $(_)^2 : \mathbb{R} \rightarrow \mathbb{R}_{\geq 0}$

The types of “higher-order” operators are usually not given explicitly:

- $\lim : (\mathbb{N} \rightarrow \mathbb{R}) \rightarrow \mathbb{R}$ for $\lim_{n \rightarrow \infty} \{a_n\}$
- $d/dt : (\mathbb{R} \rightarrow \mathbb{R}) \rightarrow (\mathbb{R} \rightarrow \mathbb{R})$
- sometimes, instead of df/dt one sees f' or \dot{f} or $D f$
- $\partial f / \partial x_i : (\mathbb{R}^n \rightarrow \mathbb{R}) \rightarrow (\mathbb{R}^n \rightarrow \mathbb{R})$
- we mostly see $\partial f / \partial x$, $\partial f / \partial y$, $\partial f / \partial z$ etc. when, in the context, the function f has been given a definition of the form $f(x, y, z) = \dots$
- a better notation (by Landau) which doesn’t rely on the names given to the arguments was popularised in Landau [1934] (English edition Landau [2001]): D_1 for the partial derivative with respect to x_1 , etc.
- Exercise: for $f : \mathbb{R}^2 \rightarrow \mathbb{R}$ define D_1 and D_2 using only D .

3.2 Typing Mathematics: derivative of a function

Let’s start simple with the classical definition of the derivative from Adams and Essex [2010]:

The **derivative** of a function f is another function f' defined by

$$f'(x) = \lim_{h \rightarrow 0} \frac{f(x+h) - f(x)}{h}$$

at all points x for which the limit exists (i.e., is a finite real number). If $f'(x)$ exists, we say that f is **differentiable** at x .

We can start by assigning types to the expressions in the definition. Let's write X for the domain of f so that we have $f : X \rightarrow \mathbb{R}$ and $X \subseteq \mathbb{R}$ (or, equivalently, $X : \mathcal{P} \mathbb{R}$). If we denote with Y the subset of X for which f is differentiable we get $f' : Y \rightarrow \mathbb{R}$. Thus, the operation which maps f to f' has type $(X \rightarrow \mathbb{R}) \rightarrow (Y \rightarrow \mathbb{R})$. Unfortunately, the only notation for this operation given (implicitly) in the definition is a postfix prime. To make it easier to see we use a prefix D instead and we can thus write $D : (X \rightarrow \mathbb{R}) \rightarrow (Y \rightarrow \mathbb{R})$. We will often assume that $X = Y$ so that we can see D as preserving the type of its argument.

Now, with the type of D sorted out, we can turn to the actual definition of the function $D f$. The definition is given for a fixed (but arbitrary) x . Here we make small detour to define *limit*. Here is the definition from Adams and Essex [2010].

A formal definition of limit

We say that $f(x)$ **approaches the limit** L as x **approaches** a , and we write

$$\lim_{x \rightarrow a} f(x) = L,$$

if the following condition is satisfied:

for every number $\epsilon > 0$ there exists a number $\delta > 0$, possibly depending on ϵ , such that if $0 < |x - a| < \delta$, then x belongs to the domain of f and

$$|f(x) - L| < \epsilon.$$

The *lim* notation has four components: a variable name x , a point a an expression $f(x)$ and the limit L . TODO: cont.

Coming back to the definition of the derivative we see that the *lim* expression is using the (anonymous) function $g \ h = \frac{f(x+h) - f \ x}{h}$ and that the limit of g is taken at 0. Note that g is defined in the scope of x and that its definition uses x so it can be seen as having x as an implicit, first argument. To be more explicit we write $\varphi \ x \ h = \frac{f(x+h) - f \ x}{h}$ and take the limit of $\varphi \ x$ at 0. So, to sum up, $D f \ x = \lim (\varphi \ x) \ 0$.²

The key here is that we name, type, and specify the operation of computing the derivative (of a one-argument function). We will use this operation quite a bit in the rest of the book, but here are just a few examples to get used to the notation.

```
sq x = x^2
double x = 2 * x
c2 x = 2
sq' = D sq = D (\x → x^2) = D (^2) = (2*) = double
sq'' = D sq' = D double = c2 = const 2
```

What we cannot do at this stage is to actually *implement* D in Haskell. If we only have a function $f : \mathbb{R} \rightarrow \mathbb{R}$ as a “black box” we cannot really compute the actual derivative $f' : \mathbb{R} \rightarrow \mathbb{R}$, only numerical approximations. But if we also have access to the “source code” of f , then we can apply the usual rules we have learnt in calculus. We will get back to this question in section 3.7.

²We could go one step further by noting that f is in the scope of φ and used in its definition. Thus the function $\psi \ f \ x \ h = \varphi \ x \ h$, or $\psi \ f = \varphi$, is used. With this notation, and $\limAt \ x \ f = \lim \ f \ x$, we obtain a point-free definition that can come in handy: $D f = \limAt \ 0 \circ \psi \ f$.

3.3 Typing Mathematics: partial derivative

Continuing on our quest of typing the elements of mathematical textbook definitions we now turn to a functions of more than one argument. Our example here will be from page 169 of Mac Lane [1986], where we read

1 [...] a function $z = f(x, y)$ for all points (x, y) in some open set U of the cartesian
 2 (x, y) -plane. [...] If one holds y fixed, the quantity z remains just a function of x ; its
 3 derivative, when it exists, is called the *partial derivative* with respect to x . Thus at a
 4 point (x, y) in U this derivative for $h \neq 0$ is

$$5 \quad \partial z / \partial x = f'_x(x, y) = \lim_{h \rightarrow 0} (f(x + h, y) - f(x, y)) / h$$

What are the types of the elements involved? We have

$U \subseteq \mathbb{R} \times \mathbb{R}$ -- cartesian plane
 $f : U \rightarrow \mathbb{R}$
 $z : U \rightarrow \mathbb{R}$ -- but see below
 $f'_x : U \rightarrow \mathbb{R}$

The x in the subscript of f' is *not* a real number, but a symbol (a *Char*).

The expression (x, y) has six occurrences. The first two (on line 1) denote variables of type U , the third (on line 2) is just a name $((x, y)$ -plane). The fourth (at line 4) denotes a variable of type U bound by a universal quantifier: “a point (x, y) in U ” as text which would translate to $\forall(x, y) \in U$ as a formula fragment.

The variable h appears to be a non-zero real number, bound by a universal quantifier (“for $h \neq 0$ ” on line 4), but that is incorrect. In fact, h is used as a local variable introduced in the subscript of *lim*. This variable h is a parameter of an anonymous function, whose limit is then taken at 0.

That function, which we can name φ , has the type $\varphi : U \rightarrow (\mathbb{R} - \{0\}) \rightarrow \mathbb{R}$ and is defined by

$$\varphi(x, y) h = (f(x + h, y) - f(x, y)) / h$$

The limit is then $\lim(\varphi(x, y)) 0$. Note that 0 is a limit point of $\mathbb{R} - \{0\}$, so the type of *lim* is the one we have discussed:

$$\lim : (X \rightarrow \mathbb{R}) \rightarrow \{p \mid p \in \mathbb{R}, \text{Limp } p \ X\} \rightarrow \mathbb{R}$$

On line 1, $z = f(x, y)$ probably does not mean that $z \in \mathbb{R}$, although the phrase “the quantity z ” (on line 2) suggests this. A possible interpretation is that z is used to abbreviate the expression $f(x, y)$; thus, everywhere we can replace z with $f(x, y)$. In particular, $\partial z / \partial x$ becomes $\partial f(x, y) / \partial x$, which we can interpret as $\partial f / \partial x$ applied to (x, y) (remember that (x, y) is bound in the context by a universal quantifier on line 4). There is the added difficulty that, just like the subscript in f'_x , the x in ∂x is not the x bound by the universal quantifier, but just a symbol.

3.4 Type inference and understanding: Lagrangian case study

From (Sussman and Wisdom 2013):

A mechanical system is described by a Lagrangian function of the system state (time, coordinates, and velocities). A motion of the system is described by a path that gives the coordinates for each moment of time. A path is allowed if and only if it satisfies the Lagrange equations. Traditionally, the Lagrange equations are written

$$\frac{d}{dt} \frac{\partial L}{\partial \dot{q}} - \frac{\partial L}{\partial q} = 0$$

What could this expression possibly mean?

To start answering the question, we start typing the elements involved:

- a. The use of notation for “partial derivative”, $\partial L / \partial q$, suggests that L is a function of at least a pair of arguments:

$$L : \mathbb{R}^i \rightarrow \mathbb{R}, i \geq 2$$

This is consistent with the description: “Lagrangian function of the system state (time, coordinates, and velocities)”. So, if we let “coordinates” be just one coordinate, we can take $i = 3$:

$$L : \mathbb{R}^3 \rightarrow \mathbb{R}$$

The “system state” here is a triple (of type $S = (T, Q, V) = \mathbb{R}^3$) and we can call the the three components $t : T$ for time, $q : Q$ for coordinate, and $v : V$ for velocity. (We use $T = Q = V = \mathbb{R}$ in this example but it can help the reading to remember the different uses of \mathbb{R} .)

- b. Looking again at the same derivative, $\partial L / \partial q$ suggests that q is the name of a real variable, one of the three arguments to L . In the context, which we do not have, we would expect to find somewhere the definition of the Lagrangian as

$$\begin{array}{l} L : (T, Q, V) \rightarrow \mathbb{R} \\ L(t, q, v) = \dots \end{array}$$

- c. therefore, $\partial L / \partial q$ should also be a function of the same triple of arguments:

$$(\partial L / \partial q) : (T, Q, V) \rightarrow \mathbb{R}$$

It follows that the equation expresses a relation between *functions*, therefore the 0 on the right-hand side is *not* the real number 0, but rather the constant function *const* 0:

$$\begin{array}{l} \text{const } 0 : (T, Q, V) \rightarrow \mathbb{R} \\ \text{const } 0(t, q, v) = 0 \end{array}$$

- d. We now have a problem: d / dt can only be applied to functions of *one* real argument t , and the result is a function of one real argument:

$$(d / dt)(\partial L / \partial \dot{q}) : T \rightarrow \mathbb{R}$$

Since we subtract from this the function $\partial L / \partial q$, it follows that this, too, must be of type $T \rightarrow \mathbb{R}$. But we already typed it as $(T, Q, V) \rightarrow \mathbb{R}$, contradiction!

- e. The expression $\partial L / \partial \dot{q}$ appears to also be malformed. We would expect a variable name where we find \dot{q} , but \dot{q} is the same as dq / dt , a function.

- f. Looking back at the description above, we see that the only immediate candidate for an application of d/dt is “a path that gives the coordinates for each moment of time”. Thus, the path is a function of time, let us say

$$w : T \rightarrow Q \quad \text{-- with } T = \mathbb{R} \text{ for time and } Q = \mathbb{R} \text{ for coordinates } (q : Q)$$

We can now guess that the use of the plural form “equations” might have something to do with the use of “coordinates”. In an n -dimensional space, a position is given by n coordinates. A path would then be a function

$$w : T \rightarrow Q \quad \text{-- with } Q = \mathbb{R}^n$$

which is equivalent to n functions of type $T \rightarrow \mathbb{R}$, each computing one coordinate as a function of time. We would then have an equation for each of them. We will use $n = 1$ for the rest of this example.

- g. Now that we have a path, the coordinates at any time are given by the path. And as the time derivative of a coordinate is a velocity, we can actually compute the trajectory of the full system state (T, Q, V) starting from just the path.

$$\begin{aligned} q : T &\rightarrow Q \\ q \, t &= w \, t \quad \text{-- or, equivalently, } q = w \\ \dot{q} : T &\rightarrow V \\ \dot{q} \, t &= dw / dt \quad \text{-- or, equivalently, } \dot{q} = D \, w \end{aligned}$$

We combine these in the “combinator” *expand*, given by

$$\begin{aligned} \text{expand} : (T \rightarrow Q) &\rightarrow (T \rightarrow (T, Q, V)) \\ \text{expand } w \, t &= (t, w \, t, D \, w \, t) \end{aligned}$$

- h. With *expand* in our toolbox we can fix the typing problem in item 4 above. The Lagrangian is a “function of the system state (time, coordinates, and velocities)” and the “expanded path” (*expand w*) computes the state from just the time. By composing them we get a function

$$L \circ (\text{expand } w) : T \rightarrow \mathbb{R}$$

which describes how the Lagrangian would vary over time if the system would evolve according to the path w .

This particular composition is not used in the equation, but we do have

$$(\partial L / \partial q) \circ (\text{expand } w) : T \rightarrow \mathbb{R}$$

which is used inside d / dt .

- i. We now move to using D for d / dt , D_2 for $\partial / \partial q$, and D_3 for $\partial / \partial \dot{q}$. In combination with *expand w* we find these type correct combinations for the two terms in the equation:

$$\begin{aligned} D ((D_2 \, L) \circ (\text{expand } w)) &: T \rightarrow \mathbb{R} \\ (D_3 \, L) \circ (\text{expand } w) &: T \rightarrow \mathbb{R} \end{aligned}$$

The equation becomes

$$D ((D_3 \, L) \circ (\text{expand } w)) - (D_2 \, L) \circ (\text{expand } w) = \text{const } 0$$

or, after simplification:

$$D (D_3 \, L \circ \text{expand } w) = D_2 \, L \circ \text{expand } w$$

where both sides are functions of type $T \rightarrow \mathbb{R}$.

- j. “A path is allowed if and only if it satisfies the Lagrange equations” means that this equation is a predicate on paths (for a particular L):

$$\text{Lagrange } (L, w) = D (D_3 L \circ \text{expand } w) == D_2 L \circ \text{expand } w$$

where we use $(==)$ to avoid confusion with the equality sign $(=)$ used for the definition of the predicate.

So, we have figured out what the equation “means”, in terms of operators we recognise. If we zoom out slightly we see that the quoted text means something like: If we can describe the mechanical system in terms of “a Lagrangian” ($L : S \rightarrow \mathbb{R}$), then we can use the equation to check if a particular candidate path $w : T \rightarrow \mathbb{R}$ qualifies as a “motion of the system” or not. The unknown of the equation is the path w , and as the equation involves partial derivatives it is an example of a partial differential equation (a PDE). We will not dig into how to solve such PDEs, but they are widely used in physics.

3.5 Type in Mathematics (Part II)

3.5.1 Type classes

The kind of type inference we presented in the last lecture becomes automatic with experience in a domain, but is very useful in the beginning.

The “trick” of looking for an appropriate combinator with which to pre- or post-compose a function in order to make types match is often useful. It is similar to the casts one does automatically in expressions such as $4 + 2.5$.

One way to understand such casts from the point of view of functional programming is via *type classes*. As a reminder, the reason $4 + 2.5$ works is because floating point values are members of the class *Num*, which includes the member function

fromInteger :: *Integer* → *a*

which converts integers to the actual type *a*.

Type classes are related to mathematical structures which, in turn, are related to DSLs. The structuralist point of view in mathematics is that each mathematical domain has its own fundamental structures. Once these have been identified, one tries to push their study as far as possible *on their own terms*, i.e., without introducing other structures. For example, in group theory, one starts by exploring the consequences of just the group structure, before one introduces, say, an order structure and monotonicity.

The type classes of Haskell seem to have been introduced without relation to their mathematical counterparts, perhaps because of pragmatic considerations. For now, we examine the numerical type classes *Num*, *Fractional*, and *Floating*.

```
class (Eq a, Show a) => Num a where
  (+), (-), (*) :: a -> a -> a
  negate      :: a -> a
  abs, signum :: a -> a
  fromInteger :: Integer -> a
```

This is taken from the Haskell documentation³ but it appears that *Eq* and *Show* are not necessary, because there are meaningful instances of *Num* which don’t support them:

³Fig. 6.2 in section 6.4 of the Haskell 2010 report: [Marlow, ed., Sect. 6.4].

```

instance Num a => Num (x -> a) where
  f + g      = λx -> f x + g x
  f - g      = λx -> f x - g x
  f * g      = λx -> f x * g x
  negate f   = negate ∘ f
  abs f      = abs ∘ f
  signum f   = signum ∘ f
  fromInteger = const ∘ fromInteger

```

This instance for functions allows us to write expressions like $\sin + \cos :: \text{Double} \rightarrow \text{Double}$ or $\text{sq} * \text{double} :: \text{Integer} \rightarrow \text{Integer}$. As another example:

$$\sin^2 = \lambda x \rightarrow (\sin x)^\wedge (\text{const } 2\ x) = \lambda x \rightarrow (\sin x)^\wedge 2$$

thus the typical math notation \sin^2 works fine in Haskell. (Note that there is a clash with another use of superscript for functions: sometimes f^n means *composition* of f with itself n times. With that reading \sin^2 would mean $\lambda x \rightarrow \sin (\sin x)$.)

Exercise: play around with this a bit in ghci.

3.5.2 Overloaded integers literals

In Haskell, every use of an integer literal like 2, 1738, etc., is actually implicitly an application of *fromInteger* to the literal. This means that the same program text can have different meaning depending on the type of the context. The literal *two* = 2, for example, can be used as an integer, a real number, a complex number, or even as a (constant) function (by the instance *Num (x -> a)*).

TODO: Start of aside - make it fit in the text better

The instance declaration above looks recursive, but is not. The same pattern appeared already in section 1.5, which near the end included roughly the following lines:

```

instance Num r => Num (ComplexSyn r) where
  -- ... several other methods and then
  fromInteger = toComplexSyn ∘ fromInteger

```

To see why this is not a recursive definition we need to expand the type and to do this I will introduce a name for the right hand side (RHS): *fromIntC*.

```

--      ComplexSyn r <----- r <----- Integer
fromIntC =      toComplexSyn . fromInteger

```

I have placed the types in the comment, with “backwards-pointing” arrows indicating that *fromInteger* :: *Integer* → *r* and *toComplexSyn* :: *r* → *ComplexSyn r* while the resulting function is *fromIntC* :: *Integer* → *ComplexSyn r*. The use of *fromInteger* at type *r* means that the full type of *fromIntC* must refer to the *Num* class. Thus we arrive at the full type:

$$\text{fromIntC} :: \text{Num } r \Rightarrow \text{Integer} \rightarrow \text{ComplexSyn } r$$

As an example we have that

```

(fromInteger 3) :: ComplexSyn Double ==
toComplexSyn (fromInteger 3)         ==
toComplexSyn 3.0                     ==
3.0 + 0 i

```

TODO: End of aside - connect back to the $x \rightarrow a$ instance

Next we have *Fractional* for when we also have division:

```
class Num a  $\Rightarrow$  Fractional a where
  (/)      :: a  $\rightarrow$  a  $\rightarrow$  a
  recip    :: a  $\rightarrow$  a    -- (1/)
  fromRational :: Rational  $\rightarrow$  a
```

and *Floating* when we can implement the “standard” functions from calculus:

```
class Fractional a  $\Rightarrow$  Floating a where
   $\pi$           :: a
  exp, log,  $\sqrt{\cdot}$  :: a  $\rightarrow$  a
  (**), logBase :: a  $\rightarrow$  a  $\rightarrow$  a
  sin, cos, tan  :: a  $\rightarrow$  a
  asin, acos, atan :: a  $\rightarrow$  a
  sinh, cosh, tanh :: a  $\rightarrow$  a
  asinh, acosh, atanh :: a  $\rightarrow$  a
```

We can instantiate these type classes for functions in the same way we did for *Num*:

```
instance Fractional a  $\Rightarrow$  Fractional (x  $\rightarrow$  a) where
  recip f      = recip  $\circ$  f
  fromRational = const  $\circ$  fromRational
```

```
instance Floating a  $\Rightarrow$  Floating (x  $\rightarrow$  a) where
   $\pi$       = const  $\pi$ 
  exp f    = exp  $\circ$  f
  f ** g =  $\lambda x \rightarrow (f\ x) ** (g\ x)$ 
  -- and so on
```

Exercise: complete the instance declarations.

These type classes represent an abstract language of algebraic and standard operations, abstract in the sense that the exact nature of the elements involved is not important from the point of view of the type class, only from that of its implementation.

3.6 Type classes in Haskell

We now abstract from *Num* and look at what a type class is and how it is used. One view of a type class is as a set of types. For *Num* that is the set of “numeric types”, for *Eq* the set of “types with computable equality”, etc. The types in this set are called instances and are declared by **instance** declarations. When a class *C* is defined, there are no types in this set (no instances). In each Haskell module where *C* is in scope there is a certain collection of instance declarations. Here is an example of a class with just two instances:

```
class C a where
  foo :: a  $\rightarrow$  a
instance C Integer where
  foo = (1+)
instance C Char where
  foo = toUpper
```

Here we see the second view of a type class: as a collection of overloaded methods (here just *foo*). Overloaded here means that the same symbol can be used with different meaning at different types. If we use *foo* with an integer it will add one, but if we use it with a character it will convert it to upper case. The full type of *foo* is $C\ a \Rightarrow a \rightarrow a$ and this means that it can be used at any type *a* for which there is an instance of *C* in scope.

Instance declarations can also be parameterised:

```
instance C a => C [a] where
  foo xs = map foo xs
```

This means that for any type *a* which is already an instance of *C* we also make the type *[a]* an instance (recursively). Thus, we now have an infinite collection of instances of *C*: *Char*, *[Char]*, *[[Char]]*, etc. Similarly, with the function instance for *Num* above, we immediately make the types $x \rightarrow Double$, $x \rightarrow (y \rightarrow Double)$, etc. into instances (for all *x*, *y*, ...).

TODO: perhaps make the “looks recursive” *fromInteger* example talk about *foo* instead? (And then just mention *fromInteger* shortly.)

3.7 Computing derivatives

TODO: intro text

The “little language” of derivatives:

```
D (f + g)  = D f + D g
D (f * g)  = D f * g + f * D g
D (f ∘ g) x = D f (g x) * D g x  -- the chain rule
D (const a) = const 0
D id        = const 1
D (^n) x    = (n - 1) * (x^(n - 1))
D sin x     = cos x
D cos x     = - (sin x)
D exp x     = exp x
```

and so on.

We observe that we can compute derivatives for any expressions made out of arithmetical functions, standard functions, and their compositions. In other words, the computation of derivatives is based on a DSL of expressions (representing functions in one variable):

```
expression ::= const ℝ
            | id
            | expression + expression
            | expression * expression
            | exp expression
            | ...
```

etc.

We can implement this in a datatype:

```
data FunExp = Const Double
            | Id
            | FunExp :+: FunExp
```

```

| FunExp *: FunExp
| Exp FunExp
-- and so on
deriving Show

```

The intended meaning of elements of the *FunExp* type is functions:

```

eval :: FunExp → Double → Double
eval (Const α) = const α
eval Id        = id
eval (e1 :+: e2) = eval e1 + eval e2 -- note the use of “lifted +”
eval (e1 **: e2) = eval e1 * eval e2 -- “lifted *”
eval (Exp e1)    = exp (eval e1)     -- and “lifted exp”
-- and so on

```

An example:

```

f1 :: Double → Double
f1 x = exp (x^2)
e1 :: FunExp
e1 = Exp (Id **: Id)

```

We can implement the derivative of such expressions using the rules of derivatives. We want to implement a function *derive* :: *FunExp* → *FunExp* which makes the following diagram commute:

$$\begin{array}{ccc}
\text{FunExp} & \xrightarrow{\text{eval}} & \text{Func} \\
\downarrow \text{derive} & & \downarrow D \\
\text{FunExp} & \xrightarrow{\text{eval}} & \text{Func}
\end{array}$$

In other words we want

$$\text{eval} \circ \text{derive } e = D \circ \text{eval}$$

or, in other words, for any expression *e*, we want

$$\text{eval} (\text{derive } e) = D (\text{eval } e)$$

For example, let us derive the *derive* function for *Exp e*:

```

eval (derive (Exp e))           = {- specification of derive above -}
D (eval (Exp e))                = {- def. eval -}
D (exp (eval e))                = {- def. exp for functions -}
D (exp ∘ eval e)                = {- chain rule -}
(D exp ∘ eval e) * D (eval e)   = {- D rule for exp -}
(exp ∘ eval e) * D (eval e)     = {- specification of derive -}
(exp ∘ eval e) * (eval (derive e)) = {- def. of eval for Exp -}
(eval (Exp e)) * (eval (derive e)) = {- def. of eval for **: -}
eval (Exp e **: derive e)

```

Therefore, the specification is fulfilled by taking

$$\text{derive} (\text{Exp } e) = \text{Exp } e **: \text{derive } e$$

Similarly, we obtain

$$\begin{aligned}
\text{derive } (\text{Const } \alpha) &= \text{Const } 0 \\
\text{derive } \text{Id} &= \text{Const } 1 \\
\text{derive } (e_1 \text{ :+ } e_2) &= \text{derive } e_1 \text{ :+ } \text{derive } e_2 \\
\text{derive } (e_1 \text{ :* } e_2) &= (\text{derive } e_1 \text{ :* } e_2) \text{ :+ } (e_1 \text{ :* } \text{derive } e_2) \\
\text{derive } (\text{Exp } e) &= \text{Exp } e \text{ :* } \text{derive } e
\end{aligned}$$

Exercise: complete the *FunExp* type and the *eval* and *derive* functions.

3.8 Shallow embeddings

The DSL of expressions, whose syntax is given by the type *FunExp*, turns out to be almost identical to the DSL defined via type classes in the first part of this lecture. The correspondence between them is given by the *eval* function.

The difference between the two implementations is that the first one separates more cleanly from the semantical one. For example, *:+* *stands for* a function, while *+* *is* that function.

The second approach is called “shallow embedding” or “almost abstract syntax”. It can be more economical, since it needs no *eval*. The question is: can we implement *derive* in the shallow embedding?

Note that the reason the shallow embedding is possible is that the *eval* function is a *fold*: first evaluate the sub-expressions of *e*, then put the evaluations together without reference to the sub-expressions. This is sometimes referred to as “compositionality”.

We check whether the semantics of derivatives is compositional. The evaluation function for derivatives is

$$\begin{aligned}
\text{eval}' &:: \text{FunExp} \rightarrow \text{Double} \rightarrow \text{Double} \\
\text{eval}' &= \text{eval} \circ \text{derive}
\end{aligned}$$

For example:

$$\begin{aligned}
\text{eval}' (\text{Exp } e) &= \{- \text{ def. } \text{eval}', \text{ function composition } -\} \\
\text{eval } (\text{derive } (\text{Exp } e)) &= \{- \text{ def. } \text{derive} \text{ for } \text{Exp} -\} \\
\text{eval } (\text{Exp } e \text{ :* } \text{derive } e) &= \{- \text{ def. } \text{eval} \text{ for } \text{:*} -\} \\
\text{eval } (\text{Exp } e) * \text{eval } (\text{derive } e) &= \{- \text{ def. } \text{eval} \text{ for } \text{Exp} -\} \\
\text{exp } (\text{eval } e) * \text{eval } (\text{derive } e) &= \{- \text{ def. } \text{eval}' -\} \\
\text{exp } (\text{eval } e) * \text{eval}' e &= \{- \text{ let } f = \text{eval } e, f' = \text{eval}' e -\} \\
\text{exp } f * f' &
\end{aligned}$$

Thus, given only the derivative $f' = \text{eval}' e$, it is impossible to compute $\text{eval}' (\text{Exp } e)$. (There is no way to implement $\text{eval}' \text{Exp} :: (\mathbb{R} \rightarrow \mathbb{R}) \rightarrow (\mathbb{R} \rightarrow \mathbb{R})$.) Thus, it is not possible to directly implement *derive* using shallow embedding; the semantics of derivatives is not compositional. Or rather, *this* semantics is not compositional. It is quite clear that the derivatives cannot be evaluated without, at the same time, being able to evaluate the functions. So we can try to do both evaluations simultaneously:

$$\begin{aligned}
\text{type } \text{FD } a &= (a \rightarrow a, a \rightarrow a) \\
\text{evalD} &:: \text{FunExp} \rightarrow \text{FD Double} \\
\text{evalD } e &= (\text{eval } e, \text{eval}' e)
\end{aligned}$$

(Note: At this point, you are advised to look up and solve exercise 1.9 on the “tupling transform” in case you have not done so already.)

Is *evalD* compositional?

We compute, for example:

```

evalD (Exp e)                = {- specification of evalD -}
(eval (Exp e), eval' (Exp e)) = {- def. eval for Exp and reusing the computation above -}
(exp (eval e), exp (eval e) * eval' e) = {- introduce names for subexpressions -}
let f = eval e
    f' = eval' e
in (exp f, exp f * f')       = {- def. evalD -}
let (f, f') = evalD e
in (exp f, exp f * f')

```

This semantics *is* compositional and the *Exp* case is:

```

evalDExp :: FD Double → FD Double
evalDExp (f, f') = (exp f, exp f * f')

```

We can now define a shallow embedding for the computation of derivatives, using the numerical type classes.

```

instance Num a => Num (a → a, a → a) where
  (f, f') + (g, g') = (f + g, f' + g')
  (f, f') * (g, g') = (f * g, f' * g + f * g')
  fromInteger n = (fromInteger n, const 0)

```

Exercise: implement the rest

3.9 Exercises

Exercise 3.1. To get a feeling for the Lagrange equations, let $L(t, q, v) = m * v^2 / 2 + m * g * q$, compute *expand w*, perform the derivatives and check if the equation is satisfied for

- $w1 = id$ or
- $w2 = sin$ or
- $w3 = (q0 -) \circ (g*) \circ (/2) \circ (^2)$

3.10 Exercises from old exams

Exercise 3.2. From Practice Exam 2016

Consider the following text from MacLane’s *Mathematics: Form and Function* (page 168):

If $z = g(y)$ and $y = h(x)$ are two functions with continuous derivatives, then in the relevant range $z = g(h(x))$ is a function of x and has derivative

$$z'(x) = g'(y) * h'(x)$$

Give the types of the elements involved (x , y , z , g , h , z' , g' , h' , $*$ and $'$).

Exercise 3.3. *From exam 2016-03-16*

Consider the following text from Mac Lane's *Mathematics: Form and Function* (page 182): TODO proper citation

In these cases one tries to find not the values of x which make a given function $y = f(x)$ a minimum, but the values of a given function $f(x)$ which make a given quantity a minimum. Typically, that quantity is usually measured by an integral whose integrand is some expression F involving both x , values of the function $y = f(x)$ at interest and the values of its derivatives - say an integral

$$\int_a^b F(y, y', x) dx, \quad y = f(x).$$

Give the types of the variables involved (x, y, y', f, F, a, b) and the type of the four-argument integration operator:

$$\int \cdot d \cdot$$

Exercise 3.4. *From exam 2016-08-23*

In the simplest case of probability theory, we start with a *finite*, non-empty set Ω of *elementary events*. *Events* are subsets of Ω , i.e. elements of the powerset of Ω , (that is, $\mathcal{P}\Omega$). a *probability function* P associates to each event a real number between 0 and 1, such that

- $P \emptyset = 0, P \Omega = 1$
- A and B are disjoint (i.e., $A \cap B = \emptyset$), then: $P A + P B = P (A \cup B)$.

Conditional probabilities are defined as follows (*Elementary Probability 2nd Edition*, Stirzaker 2003):

Let A and B be events with $P B > 0$. given that B occurs, the *conditional probability* that A occurs is denoted by $P (A \mid B)$ and defined by

$$P (A \mid B) = P (A \cap B) / P B$$

- a. What are the types of the elements involved in the definition of conditional probability? ($P, \cap, /, \mid$)
- b. In the 1933 monograph that set the foundations of contemporary probability theory, Kolmogorov used, instead of $P (A \mid B)$, the expression $P_A B$. Type this expression. Which notation do you prefer (provide a *brief* explanation).

Exercise 3.5. *From exam 2017-03*

(Note that this exam question was later included as one of the examples in this chapter, see 3.3. It is kept here in case you want to check if you remember it!)

Consider the following text from page 169 of Mac Lane [1968]:

[...] a function $z = f(x, y)$ for all points (x, y) in some open set U of the cartesian (x, y) -plane. [...] If one holds y fixed, the quantity z remains just a function of x ; its derivative, when it exists, is called the *partial derivative* with respect to x . Thus at a point (x, y) in U this derivative for $h \neq 0$ is

$$\partial z / \partial x = f'_x(x, y) = \lim_{h \rightarrow 0} (f(x + h, y) - f(x, y)) / h$$

What are the types of the elements involved in the equation on the last line? You are welcome to introduce functions and names to explain your reasoning.

Exercise 3.6. *From exam 2017-08-22*

Multiplication for matrices (from the matrix algebra DSL).

Consider the following definition, from “Linear Algebra” by Donald H. Pelletier:

Definition: If A is an $m \times n$ matrix and B is an $n \times p$ matrix, then the *product*, AB , is an $m \times p$ matrix; the $(i, j)^{th}$ entry of AB is the sum of the products of the pairs that are obtained when the entries from the i^{th} row of the left factor, A , are paired with those from the j^{th} column of the right factor, B .

- a. Introduce precise types for the variables involved: A, m, n, B, p, i, j . You can write $Fin\ n$ for the type of the values $\{0, 1, \dots, n - 1\}$.
- b. Introduce types for the functions mul and $proj$ where $AB = mul\ A\ B$ and $proj\ i\ j\ M =$ “take the $(i, j)^{th}$ entry of M ”. What class constraints (if any) are needed on the type of the matrix entries in the two cases?
- c. Implement mul in Haskell. You may use the functions row and col specified by $row\ i\ M =$ “the i^{th} row of M ” and $col\ j\ M =$ “the j^{th} column of M ”. You don’t need to implement them and here you can assume they return plain Haskell lists.

Exercise 3.7. (Extra material outside the course.) In the same direction as the Lagrangian case study in section 3.4 there are two nice blog posts about Hamiltonian dynamics: one introductory and one more advanced. It is a good exercise to work through the examples in these posts.

4 Compositional Semantics and Algebraic Structures


```
{-# LANGUAGE FlexibleInstances, GeneralizedNewtypeDeriving #-}
module DSLsofMath.W04 where
import Prelude hiding (Monoid)
```

4.1 Compositional semantics

4.1.1 A simpler example of a non-compositional function

Consider a very simple datatype of integer expressions:

```
data E = Add E E | Mul E E | Con Integer deriving Eq
e1, e2 :: E           -- 1 + 2 * 3
e1 = Add (Con 1) (Mul (Con 2) (Con 3)) -- 1 + (2 * 3)
e2 = Mul (Add (Con 1) (Con 2)) (Con 3) -- (1 + 2) * 3
```



When working with expressions it is often useful to have a “pretty-printer” to convert the abstract syntax trees to strings like “1+2*3”.

```
pretty :: E → String
```

We can view *pretty* as an alternative *eval* function for a semantics using *String* as the semantic domain instead of the more natural *Integer*. We can implement *pretty* in the usual way as a “fold” over the syntax tree using one “semantic constructor” for each syntactic constructor:

```
pretty (Add x y) = prettyAdd (pretty x) (pretty y)
pretty (Mul x y) = prettyMul (pretty x) (pretty y)
pretty (Con c)   = prettyCon c

prettyAdd :: String → String → String
prettyMul :: String → String → String
prettyCon :: Integer → String
```

Now, if we try to implement the semantic constructors without thinking too much we would get the following:

```
prettyAdd xs ys = xs ++ "+" ++ ys
prettyMul xs ys = xs ++ "*" ++ ys
prettyCon c     = show c

p1, p2 :: String
p1 = pretty e1
p2 = pretty e2

trouble :: Bool
trouble = p1 == p2
```

Note that both e_1 and e_2 are different but they pretty-print to the same string. There are many ways to fix this, some more “pretty” than others, but the main problem is that some information is lost in the translation.

TODO(perhaps): Explain using three pretty printers for the three “contexts”: at top level, inside *Add*, inside *Mul*, ... then combine them with the tupling transform just as with *evalD*. The result is the following:

```

prTop :: E → String
prTop e = let (pTop, _, _) = prVersions e
           in pTop

prVersions = foldE prVerAdd prVerMul prVerCon

prVerAdd (xTop, xInA, xInM) (yTop, yInA, yInM) =
  let s = xInA ++ "+" ++ yInA    -- use InA because we are “in Add”
  in (s, paren s, paren s)        -- parens needed except at top level

prVerMul (xTop, xInA, xInM) (yTop, yInA, yInM) =
  let s = xInM ++ "*" ++ yInM    -- use InM because we are “in Mul”
  in (s, s, paren s)             -- parens only needed inside Mul

prVerCon i =
  let s = show i
  in (s, s, s)                   -- parens never needed


paren :: String → String
paren s = "(" ++ s ++ ")"

```

Exercise: Another way to make this example go through is to refine the semantic domain from *String* to *Precedence* → *String*. This can be seen as another variant of the result after the tupling transform: if *Precedence* is an n -element type then *Precedence* → *String* can be seen as an n -tuple. In our case a three-element *Precedence* would be enough.

4.1.2 Compositional semantics in general

In general, for a syntax *Syn*, and a possible semantics (a type *Sem* and an *eval* function of type *Syn* → *Sem*), we call the semantics *compositional* if we can implement *eval* as a fold. Informally a “fold” is a recursive function which replaces each abstract syntax constructor C_i of *Syn* with a “semantic constructor” ci . Thus, in our datatype *E*, a compositional semantics means that *Add* maps to *add*, *Mul* ↦ *mul*, and *Con* ↦ *con* for some “semantic functions” *add*, *mul*, and *con*.



As an example we can define a general *foldE* for the integer expressions:

```

foldE :: (s → s → s) → (s → s → s) → (Integer → s) → (E → s)
foldE add mul con = rec
  where rec (Add x y) = add (rec x) (rec y)
        rec (Mul x y) = mul (rec x) (rec y)
        rec (Con i)   = con i

```

Notice that *foldE* has three function arguments corresponding to the three constructors of *E*. The “natural” evaluator to integers is then easy:

```
evalE1 :: E → Integer
evalE1 = foldE (+) (*) id
```

and with a minimal modification we can also make it work for other numeric types:

```
evalE2 :: Num a ⇒ E → a
evalE2 = foldE (+) (*) fromInteger
```

Another thing worth noting is that if we replace each abstract syntax constructor with itself we get the identity function (a “deep copy”):

```
idE :: E → E
idE = foldE Add Mul Con
```

Finally, it is often useful to capture the semantic functions (the parameters to the fold) in a type class:

```
class IntExp t where
  add :: t → t → t
  mul :: t → t → t
  con :: Integer → t
```

In this way we can make “hide” the arguments to the fold:

```
foldIE :: IntExp t ⇒ E → t
foldIE = foldE add mul con

instance IntExp E where
  add = Add
  mul = Mul
  con = Con

instance IntExp Integer where
  add = (+)
  mul = (*)
  con = id

idE' :: E → E
idE' = foldIE

evalE' :: E → Integer
evalE' = foldIE
```

And we can also see *pretty* as instance:

```
instance IntExp String where
  add = prettyAdd
  mul = prettyMul
  con = prettyCon


pretty' :: E → String
pretty' = foldIE
```

4.1.3 Back to derivatives and evaluation

TODO: perhaps not include this here. The background is that this material did not quite fit in the previous lecture. Also some repetition was needed.

Review section 3.8 again with the definition of $eval'$ being non-compositional (just like *pretty*) and $evalD$ a more complex, but compositional, semantics.

We want to implement $eval' = eval \circ derive$ in the following diagram:



As we saw in section 3.8 this does not work. Instead we need to use the “tupling transform” to compute a pair of f and Df at once.

4.2 Algebraic Structures and DSLs

In this lecture, we continue exploring the relationship between type classes, mathematical structures, and DSLs.

4.2.1 Algebras, homomorphisms

TODO: rewrite to make it smoother

From Wikipedia:

In universal algebra, an algebra (or algebraic structure) is a set A together with a collection of operations on A .

Example:

```

class Monoid a where
  unit :: a
  op    :: a -> a -> a

```

After the operations have been specified, the nature of the algebra can be further limited by axioms, which in universal algebra often take the form of identities, or *equational laws*.

Example: Monoid equations

A monoid is an algebra which has an associative operation 'op' and a unit. The laws can be formulated as the following equations:

$$\begin{aligned}
 \forall x : a. (unit \text{ 'op' } x == x \wedge x \text{ 'op' } unit == x) \\
 \forall x, y, z : a. (x \text{ 'op' } (y \text{ 'op' } z)) == (x \text{ 'op' } y) \text{ 'op' } z
 \end{aligned}$$

Examples of monoids include numbers with additions, $(\mathbb{R}, 0, (+))$, numbers with multiplication $(\mathbb{R}_{>0}, 1, (*))$, and even endofunctions with composition $(a \rightarrow a, id, (\circ))$. It is a good exercise to check that the laws are satisfied.

In mathematics, as soon as there are several examples of a structure, the question of what “translation” between them comes up. An important class of such “translations” are “structure preserving maps” called *homomorphisms*. As two examples, we have the homomorphisms *log* and *exp*:

$$\begin{aligned}
\exp : \mathbb{R} &\rightarrow \mathbb{R}_{>0} \\
\exp 0 &= 1 \\
\exp (a + b) &= \exp a * \exp b \\
\log : \mathbb{R}_{>0} &\rightarrow \mathbb{R} \\
\log 1 &= 0 \\
\log (a * b) &= \log a + \log b
\end{aligned}$$

More formally, a homomorphism between two algebras A and B is a function $h : A \rightarrow B$ from the set A to the set B such that, for every operation fA of A and corresponding fB of B (of arity, say, n), $h (fA (x_1, \dots, x_n)) = fB (h (x_1), \dots, h (x_n))$.

Example: Monoid homomorphism

$$\begin{aligned}
h \text{ unit} &= \text{unit} && \text{-- } h \text{ takes units to units} \\
h (x \text{ 'op' } y) &= h x \text{ 'op' } h y && \text{-- and distributes over op}
\end{aligned}$$

```

newtype ANat    = A Int deriving (Show, Num, Eq)
instance Monoid ANat where
    unit          = A 0
    op (A m) (A n) = A (m + n)
newtype MNat    = M Int deriving (Show, Num, Eq)
instance Monoid MNat where
    unit          = M 1
    op (M m) (M n) = M (m * n)

```

Exercise: characterise the homomorphisms from $ANat$ to $MNat$.

Solution:

Let $h : ANat \rightarrow MNat$ be a homomorphism. Then

$$\begin{aligned}
h 0 &= 1 \\
h (x + y) &= h x * h y
\end{aligned}$$

For example $h (x + x) = h x * h x = (h x)^2$ which for $x = 1$ means that $h 2 = (h 1)^2$.

More generally, every n in $ANat$ is equal to the sum of n ones: $1 + 1 + \dots + 1$. Therefore

$$h n = (h 1)^n$$

Every choice of $h 1$ “induces a homomorphism”. This means that the value of the function h is fully determined by its value for 1.

Exercise: show that $const$ is a homomorphism. The distribution law can be shown as follows:

$$\begin{aligned}
h a + h b &= \{- h = const \text{ in this case -}\} \\
const a + const b &= \{- \text{By def. of } (+) \text{ on functions -}\} \\
(\lambda x \rightarrow const a x + const b x) &= \{- \text{By def. of } const, \text{ twice -}\} \\
(\lambda x \rightarrow a + b) &= \{- \text{By def. of } const -\} \\
const (a + b) &= \{- h = const -\} \\
h (a + b) &
\end{aligned}$$

We now have a homomorphism from values to functions, and you may wonder if there is a homomorphism in the other direction. The answer is “Yes, many”. Exercise: Show that $apply\ c$ is a homomorphism for all c , where $apply\ x\ f = f\ x$.

4.2.2 Homomorphism and compositional semantics

Last time, we saw that *eval* is compositional, while *eval'* is not. Another way of phrasing that is to say that *eval* is a homomorphism, while *eval'* is not. To see this, we need to make explicit the structure of *FunExp*:

```
instance Num FunExp where
  (+)      = (:+ :)
  (*)      = (:* :)
  fromInteger = Const ∘ fromInteger
instance Fractional FunExp where
instance Floating FunExp where
  exp      = Exp
```

and so on.

Exercise: complete the type instances for *FunExp*.

For instance, we have

$$\begin{aligned} \text{eval } (e_1 \text{ :* } e_2) &= \text{eval } e_1 * \text{eval } e_2 \\ \text{eval } (\text{Exp } e) &= \text{exp } (\text{eval } e) \end{aligned}$$

These properties do not hold for *eval'*, but do hold for *evalD*.

The numerical classes in Haskell do not fully do justice to the structure of expressions, for example, they do not contain an identity operation, which is needed to translate *Id*, nor an embedding of doubles, etc. If they did, then we could have evaluated expressions more abstractly:

$$\text{eval} :: \text{GoodClass } a \Rightarrow \text{FunExp} \rightarrow a$$

where *GoodClass* gives exactly the structure we need for the translation.

Exercise: define *GoodClass* and instantiate *FunExp* and *Double* \rightarrow *Double* as instances of it. Find another instance of *GoodClass*.

```
class GoodClass t where
  addF :: t → t → t
  mulF :: t → t → t
  expF :: t → t
  -- ... TODO: continue to mimic the FunExp datatype as a class
newtype FD a = FD (a → a, a → a)
instance Num a ⇒ GoodClass (FD a) where
  addF = evalDApp
  mulF = evalDMul
  expF = evalDExp
  -- ... TODO
  evalDApp = error "TODO"
  evalDMul = error "TODO"
  evalDExp = error "TODO"
```

Therefore, we can always define a homomorphism from *FunExp* to *any* instance of *GoodClass*, in an essentially unique way. In the language of category theory, *FunExp* is an initial algebra.

Let us explore this in the simpler context of *Monoid*. The language of monoids is given by


```

type Var    = String
data MExpr = Unit | Op MExpr MExpr | V Var

```

Alternatively, we could have parametrised *MExpr* over the type of variables.

Just as in the case of FOL terms, we can evaluate an *MExpr* in a monoid instance if we are given a way of interpreting variables, also called an assignment:

$$\text{evalM} :: \text{Monoid } a \Rightarrow (\text{Var} \rightarrow a) \rightarrow (\text{MExpr} \rightarrow a)$$

Once given an $f :: \text{Var} \rightarrow a$, the homomorphism condition defines *evalM*:

```

evalM f Unit      = unit
evalM f (Op e1 e2) = op (evalM f e1) (evalM f e2)
evalM f (V x)     = f x

```

(Observation: In *FunExp*, the role of variables was played by *Double*, and the role of the assignment by the identity.)

The following correspondence summarises the discussion so far:

Computer Science	Mathematics
DSL	structure (category, algebra, ...)
deep embedding, abstract syntax	initial algebra
shallow embedding	any other algebra
semantics	homomorphism from the initial algebra

The underlying theory of this table is a fascinating topic but mostly out of scope for the DSLsof-Math course. See Category Theory and Functional Programming for a whole course around this (lecture notes are available on github).

4.2.3 Other homomorphisms

Last time, we defined a *Num* instance for functions with a *Num* codomain. If we have an element of the domain of such a function, we can use it to obtain a homomorphism from functions to their codomains:

$$\text{Num } a \Rightarrow x \rightarrow (x \rightarrow a) \rightarrow a$$

As suggested by the type, the homomorphism is just function application:

```

apply :: a -> (a -> b) -> b
apply a = \f -> f a

```

Indeed, writing $h = \text{apply } c$ for some fixed c , we have

```

h (f + g) = {- def. apply -}
(f + g) c = {- def. + for functions -}
f c + g c = {- def. apply -}
h f + h g

```

etc.

Can we do something similar for FD ?

The elements of $FD\ a$ are pairs of functions, so we can take

$$\begin{aligned} applyFD &:: a \rightarrow FD\ a \rightarrow (a, a) \\ applyFD\ c\ (f, f') &= (f\ c, f'\ c) \end{aligned}$$

We now have the domain of the homomorphism ($FD\ a$) and the homomorphism itself ($applyFD\ c$), but we are missing the structure on the codomain, which now consists of pairs (a, a) . In fact, we can *compute* this structure from the homomorphism condition. For example:

$$\begin{aligned} &h\ ((f, f') * (g, g')) \\ &= \{- \text{ def. } * \text{ for } FD\ a \ -\} \\ &\quad h\ (f * g, f' * g + f * g') \\ &= \{- \text{ def. } h = applyFD\ c \ -\} \\ &\quad ((f * g)\ c, (f' * g + f * g')\ c) \\ &= \{- \text{ def. } * \text{ and } + \text{ for functions } -\} \\ &\quad (f\ c * g\ c, f'\ c * g\ c + f\ c * g'\ c) \\ &= \{- \text{ homomorphism condition from step 1 } -\} \\ &\quad h\ (f, f') \otimes h\ (g, g') \\ &= \{- \text{ def. } h = applyFD\ c \ -\} \\ &\quad (f\ c, f'\ c) \otimes (g\ c, g'\ c) \end{aligned}$$

The identity will hold if we take

$$(x, x') \otimes (y, y') = (x * y, x' * y + x * y')$$

Exercise: complete the instance declarations for $(Double, Double)$.

Note: As this computation goes through also for the other cases we can actually work with just pairs of values (at an implicit point $c :: a$) instead of pairs of functions. Thus we can redefine FD to be

$$\text{type } FD\ a = (a, a)$$

Hint: Something very similar can be used for Assignment 2.

4.3 Summing up: definitions and representation

We defined a *Num* structure on pairs $(Double, Double)$ by requiring the operations to be compatible with the interpretation $(f\ a, f'\ a)$. For example

$$(x, x') \otimes (y, y') = (x * y, x' * y + x * y')$$

There is nothing in the “nature” of pairs of *Double* that forces this definition upon us. We chose it, because of the intended interpretation.

This multiplication is obviously not the one we need for *complex numbers*:

$$(x, x') * (y, y') = (x * y - x' * y', x * y' + x' * y)$$

Again, there is nothing in the nature of pairs that foists this operation on us. In particular, it is, strictly speaking, incorrect to say that a complex number *is* a pair of real numbers. The correct

interpretation is that a complex number can be *represented* by a pair of real numbers, provided we define the operations on these pairs in a suitable way.

The distinction between definition and representation is similar to the one between specification and implementation, and, in a certain sense, to the one between syntax and semantics. All these distinctions are frequently obscured, for example, because of prototyping (working with representations / implementations / concrete objects in order to find out what definition / specification / syntax is most adequate). They can also be context-dependent (one man's specification is another man's implementation). Insisting on the difference between definition and representation can also appear quite pedantic (as in the discussion of complex numbers above). In general though, it is a good idea to be aware of these distinctions, even if they are suppressed for reasons of brevity or style. We will see this distinction again in section 5.1.

4.3.1 Some helper functions

```
instance Num E where  -- Some abuse of notation (no proper negate, etc.)
  (+) = Add
  (*) = Mul
  fromInteger = Con
  negate = negateE
  negateE (Con c) = Con (negate c)
  negateE _ = error "negate: not supported"
```

4.4 Exercises

Exercise 4.1. Complete the instance declarations for *FunExp* (for *Num*, *Fractional*, and *Floating*).

Exercise 4.2. Complete the instance declarations for (*Double*, *Double*), deriving them from the homomorphism requirement for *apply* (in section 4.2.3).

Exercise 4.3. We now have three different ways of computing the derivative of a function such as $f\ x = \sin x + \exp(\exp x)$ at a given point, say $x = \pi$.

- Find $e :: \text{FunExp}$ such that $\text{eval } e = f$ and use eval' .
- Find an expression of (the first version of) type *FD Double* and use *apply*.
- Apply f directly to the appropriate (x, x') and use *snd*.

Do you get the same result?

Exercise 4.4. From exam 2017-08-22

In exercise 1.3 we looked at the datatype *SR v* for the language of semiring expressions. We will now use some of the concepts discussed in this chapter to expand on this language.

- Define a type class *SemiRing* that corresponds to the semiring structure.
- Define a *SemiRing* instance for the datatype *SR v* that you defined in exercise 1.3.
- Find two other instances of the *SemiRing* class.
- Specialise the evaluator that you defined in exercise 1.3 to the two *SemiRing* instances defined above. Take three semiring expressions of type *SR String*, give the appropriate assignments and compute the results of evaluating, in each case, the three expressions.

Exercise 4.5. *From exam 2016-03-15*

In exercise 1.4, we looked at a datatype for the language of lattice expressions. We will now use some of the concepts discussed in this chapter to expand on this language.

- Define a type class *Lattice* that corresponds to the lattice structure.
- Define a *Lattice* instance for the datatype for lattice expressions that you defined in 1.4.1.
- Find two other instances of the *Lattice* class.
- Specialise the evaluator you defined in exercise 1.4.2 to the two *Lattice* instances defined above. Take three lattice expressions, give the appropriate assignments and compute the results of evaluating, in each case, the three expressions.

Exercise 4.6. *From exam 2016-08-23*

In exercise 1.5, we looked at a datatype for the language of abelian monoid expressions. We will now use some of the concepts discussed in this chapter to expand on this language.

- Define a type class *AbMonoid* that corresponds to the abelian monoid structure.
- Define an *AbMonoid* instance for the datatype for abelian monoid expressions that you defined in exercise 1.5.1.
- Find one other instance of the *AbMonoid* class and give an example which is *not* an instance of *AbMonoid*.
- Specialise the evaluator that you defined in exercise 1.5.2 to the *AbMonoid* instance defined above. Take three ‘AbMonoidExp’ expressions, give the appropriate assignments and compute the results of evaluating the three expressions.

Exercise 4.7. (Closely related to exam question)

A *ring* is a set A together with two constants (or nullary operations), 0 and 1, one unary operation, *negate*, and two binary operations, + and *, such that

- 0 is the neutral element of +

$$\forall x \in A. \ x + 0 = 0 + x = x$$

- + is associative

$$\forall x, y, z \in A. \ x + (y + z) = (x + y) + z$$

- negate* inverts elements with respect to addition

$$\forall x \in A. \ x + \text{negate } x = \text{negate } x + x = 0$$

- + is commutative

$$\forall x, y \in A. \ x + y = y + x$$

- 1 is the unit of *

$$\forall x \in A. \ x * 1 = 1 * x = x$$

f. $*$ is associative

$$\forall x, y, z \in A. \quad x * (y * z) = (x * y) * z$$

g. $*$ distributes over $+$

$$\begin{aligned} \forall x, y, z \in A. \quad x * (y + z) &= (x * y) + (x * z) \\ \forall x, y, z \in A. \quad (x + y) * z &= (x * z) + (y * z) \end{aligned}$$

Remarks:

- a. and b. say that $(A, 0, +)$ is a monoid
 - a–c. say that $(A, 0, +, \text{negate})$ is a group
 - a–d. say that $(A, 0, +, \text{negate})$ is a commutative group
 - e. and f. say that $(A, 1, *)$ is a monoid
- a. Define a type class ‘Ring’ that corresponds to the ring structure.
 - b. Define a datatype for the language of ring expressions (including variables) and define a ‘Ring’ instance for it.
 - c. Find two other instances of the ‘Ring’ class.
 - d. Define a general evaluator for ‘Ring’ expressions on the basis of a given assignment function.
 - e. Specialise the evaluator to the two ‘Ring’ instances defined at point 3. Take three ring expressions, give the appropriate assignments and compute the results of evaluating, in each case, the three expressions.

Exercise 4.8. *From exam 2017-03-14*

Recall the type of expressions

```
data FunExp = Const Rational
           | Id
           | FunExp :+: FunExp
           | FunExp **: FunExp
           | FunExp :/: FunExp
           | Exp FunExp
           | Sin FunExp
           | Cos FunExp
           -- and so on
deriving Show
```

and consider the function

$$f :: \text{Double} \rightarrow \text{Double} \quad f \, x = \exp (\sin x) + x$$

- a. Find an expression e such that $\text{eval } e == f$ and show this using equational reasoning.

- b. Implement a function *deriv2* such that, for any $f : \text{Fractional } a \Rightarrow a \rightarrow a$ constructed with the grammar of *FunExp* and any x in the domain of f , we have that *deriv2* f x computes the second derivative of f at x . Use the function *derive* $:: \text{FunExp} \rightarrow \text{FunExp}$ from the lectures (*eval* (*derive* e) is the derivative of *eval* e). What instance declarations do you need?

The type of *deriv2* f should be $\text{Fractional } a \Rightarrow a \rightarrow a$.

Exercise 4.9. Based on the lecture notes, complete all the instance and datatype declarations and definitions in the files

FunNumInst.lhs,
FunExp.lhs,
Derive.lhs,
EvalD.lhs, and
ShallowD.lhs.

Exercise 4.10. Write a function

simplify $:: \text{FunExp} \rightarrow \text{FunExp}$

to simplify the expression resulted from *derive*. For example

simplify (*Const* 0 $:: \text{Exp Id}$) = *Const* 0 *simplify* (*Const* 0 $:: \text{Exp Id}$) = *Exp Id* *simplify* (*Const* 2 $:: \text{Const 1}$) = *Const* 2

5 Polynomials and Power Series

```
{-# LANGUAGE TypeSynonymInstances #-}
module DSLsofMath.W05 where
```

5.1 Polynomials

From Adams and Essex [2010], page 55:

A **polynomial** is a function P whose value at x is

$$P(x) = a_n x^n + a_{n-1} x^{n-1} + \cdots + a_1 x + a_0$$

where a_n, a_{n-1}, \dots, a_1 , and a_0 , called the **coefficients** of the polynomial [original spelling], are constants and, if $n > 0$, then $a_n \neq 0$. The number n , the degree of the highest power of x in the polynomial, is called the **degree** of the polynomial. (The degree of the zero polynomial is not defined.)

This definition raises a number of questions, for example “what is the zero polynomial?”.

The types of the elements involved in the definition appear to be

$$P : \mathbb{R} \rightarrow \mathbb{R}, x \in \mathbb{R}, a_0, \dots, a_n \in \mathbb{R} \text{ with } a_n \neq 0 \text{ if } n > 0$$

The phrasing should be “whose value at *any* x is”. The remark that the a_i are constants is probably meant to indicate that they do not depend on x , otherwise every function would be a polynomial. The zero polynomial is, according to this definition, the *const* 0 function. Thus, what is meant is

A **polynomial** is a function $P : \mathbb{R} \rightarrow \mathbb{R}$ which is either constant zero, or there exist $a_0, \dots, a_n \in \mathbb{R}$ with $a_n \neq 0$ such that, for any $x \in \mathbb{R}$

$$P(x) = a_n x^n + a_{n-1} x^{n-1} + \cdots + a_1 x + a_0$$

Given the coefficients a_i we can evaluate P at any given x . Assuming the coefficients are given as

$$as = [a_0, a_1, \dots, a_n]$$

(we prefer counting up), then the evaluation function is written

$$\begin{aligned} eval &:: [\mathbb{R}] \rightarrow \mathbb{R} \rightarrow \mathbb{R} \\ eval \quad [] &\quad x = 0 \\ eval \quad (a : as) &\quad x = a + x * eval \ as \ x \end{aligned}$$

Note that we can read the type as $eval :: [\mathbb{R}] \rightarrow (\mathbb{R} \rightarrow \mathbb{R})$ and thus identify $[\mathbb{R}]$ as the type for the (abstract) syntax and $(\mathbb{R} \rightarrow \mathbb{R})$ as the type of the semantics. Exercise: Show that this evaluation function gives the same result as the formula above. Exercise: Use the *Num* instance for functions to rewrite *eval* into a one-argument function. As an example, the polynomial which is usually written just x is represented by the list $[0, 1]$ and the polynomial $\lambda x \rightarrow x^2 - 1$ is represented by the list $[-1, 0, 1]$.

It is worth noting that the definition of a what is called a “polynomial function” is semantic, not syntactic. A syntactic definition would talk about the form of the expression (a sum of coefficients

times natural powers of x). This semantic definition only requires that the function P *behaves like* such a sum. (Has the same value for all x .) This may seem pedantic, but here is an interesting example of a family of functions which syntactically looks very trigonometric: $T_n(x) = \cos(n * \arccos(x))$. It can be shown that T_n is a polynomial function of degree n . Exercise: show this by induction on n using the rule for $\cos(\alpha + \beta)$. Start by computing T_0 , T_1 , and T_2 by hand to get a feeling for how it works.

Not every list of coefficients is valid according to the definition. In particular, the empty list is not a valid list of coefficients, so we have a conceptual, if not empirical, type error in our evaluator.

The valid lists are those *finite* lists in the set

$$\{[0]\} \cup \{(a : as) \mid \text{last } (a : as) \neq 0\}$$

We cannot express the $\text{last } (a : as) \neq 0$ in Haskell, but we can express the condition that the list should not be empty:

```
data Poly a = Single a | Cons a (Poly a)
deriving (Eq, Ord)
```

(TODO: show the version and motivation for using just $[a]$ as well. Basically, one can use $[]$ as the syntax for the “zero polynomial” and $(c : cs)$ for all other.)

The relationship between $\text{Poly } a$ and $[a]$ is given by the following functions:

```
toList :: Poly a -> [a]
toList (Single a) = a : []
toList (Cons a as) = a : toList as

fromList :: [a] -> Poly a
fromList (a : []) = Single a
fromList (a0 : a1 : as) = Cons a0 (fromList (a1 : as))

instance Show a => Show (Poly a) where
    show = show o toList
```

Since we only use the arithmetical operations, we can generalise our evaluator:

```
evalPoly :: Num a => Poly a -> (a -> a)
evalPoly (Single a) x = a
evalPoly (Cons a as) x = a + x * evalPoly as x
```

Since we have $\text{Num } a$, there is a Num structure on $a \rightarrow a$, and evalPoly looks like a homomorphism. Question: is there a Num structure on $\text{Poly } a$, such that evalPoly is a homomorphism?

For example, the homomorphism condition gives for $(+)$

$$\text{evalPoly } as + \text{evalPoly } bs = \text{evalPoly } (as + bs)$$

Both sides are functions, they are equal iff they are equal for every argument. For an arbitrary x

$$\begin{aligned} & (\text{evalPoly } as + \text{evalPoly } bs) x = \text{evalPoly } (as + bs) x \\ \Leftrightarrow & \{- + \text{ on functions is defined point-wise -}\} \\ & \text{evalPoly } as x + \text{evalPoly } bs x = \text{evalPoly } (as + bs) x \end{aligned}$$

To proceed further, we need to consider the various cases in the definition of evalPoly . We give here the computation for the last case (where as has at least one Cons), using the traditional list notation $(:)$ for brevity.

$$\text{evalPoly } (a : \text{as}) \ x + \text{evalPoly } (b : \text{bs}) \ x = \text{evalPoly } ((a : \text{as}) + (b : \text{bs})) \ x$$

For the left-hand side, we have:

$$\begin{aligned} \text{evalPoly } (a : \text{as}) \ x + \text{evalPoly } (b : \text{bs}) \ x &= \{- \text{def. evalPoly} -\} \\ (a + x * \text{evalPoly } \text{as} \ x) + (b + x * \text{evalPoly } \text{bs} \ x) &= \{- \text{properties of } +, \text{ valid in any ring} -\} \\ (a + b) + x * (\text{evalPoly } \text{as} \ x + \text{evalPoly } \text{bs} \ x) &= \{- \text{homomorphism condition} -\} \\ (a + b) + x * (\text{evalPoly } (\text{as} + \text{bs}) \ x) &= \{- \text{def. evalPoly} -\} \\ \text{evalPoly } ((a + b) : (\text{as} + \text{bs})) \ x & \end{aligned}$$

The homomorphism condition will hold for every x if we define

$$(a : \text{as}) + (b : \text{bs}) = (a + b) : (\text{as} + \text{bs})$$

We leave the derivation of the other cases and operations as an exercise. Here, we just give the corresponding definitions.

```
instance Num a => Num (Poly a) where
  (+) = polyAdd
  (*) = polyMul
  negate = polyNeg
  fromInteger = Single o fromInteger

polyAdd :: Num a => Poly a -> Poly a -> Poly a
polyAdd (Single a) (Single b) = Single (a + b)
polyAdd (Single a) (Cons b bs) = Cons (a + b) bs
polyAdd (Cons a as) (Single b) = Cons (a + b) as
polyAdd (Cons a as) (Cons b bs) = Cons (a + b) (polyAdd as bs)

polyMul :: Num a => Poly a -> Poly a -> Poly a
polyMul (Single a) (Single b) = Single (a * b)
polyMul (Single a) (Cons b bs) = Cons (a * b) (polyMul (Single a) bs)
polyMul (Cons a as) (Single b) = Cons (a * b) (polyMul as (Single b))
polyMul (Cons a as) (Cons b bs) = Cons (a * b) (polyAdd (polyMul as (Cons b bs))
                                                         (polyMul (Single a) bs))

polyNeg :: Num a => Poly a -> Poly a
polyNeg = fmap negate
```

Therefore, we *can* define a ring structure (the mathematical counterpart of *Num*) on *Poly a*, and we have arrived at the canonical definition of polynomials, as found in any algebra book (see, for example, Rotman [2006] for a very readable text):

Given a commutative ring A , the commutative ring given by the set *Poly A* together with the operations defined above is the ring of **polynomials** with coefficients in A .

The functions *evalPoly as* are known as *polynomial functions*.

Caveat: The canonical representation of polynomials in algebra does not use finite lists, but the equivalent

$$\text{Poly}' A = \{ a : \mathbb{N} \rightarrow A \mid \{- a \text{ has only a finite number of non-zero values} -\} \}$$

Exercise: what are the ring operations on $Poly' A$? For example, here is addition:

$$a + b = c \Leftrightarrow a\ n + b\ n = c\ n \quad -- \quad \forall n : \mathbb{N}$$

Observations:

- a. Polynomials are not, in general, isomorphic (in one-to-one correspondence) with polynomial functions. For any finite ring A , there is a finite number of functions $A \rightarrow A$, but there is a countable number of polynomials. That means that the same polynomial function on A will be the evaluation of many different polynomials.

For example, consider the ring \mathbb{Z}_2 ($\{0, 1\}$ with addition and multiplication modulo 2). In this ring, we have

$$evalPoly\ [0, 1, 1] = const\ 0 = evalPoly\ [0]\{-\ in\ \mathbb{Z}_2 \rightarrow \mathbb{Z}_2\ -\}$$

but

$$[0, 1, 1] \neq [0]\{-\ in\ Poly\ \mathbb{Z}_2\ -\}$$

Therefore, it is not generally a good idea to confuse polynomials with polynomial functions.

- b. In keeping with the DSL terminology, we can say that the polynomial functions are the semantics of the language of polynomials. We started with polynomial functions, we wrote the evaluation function and realised that we have the makings of a homomorphism. That suggested that we could create an adequate language for polynomial functions. Indeed, this turns out to be the case; in so doing, we have recreated an important mathematical achievement: the algebraic definition of polynomials.

Let

$$\begin{aligned} x &:: Num\ a \Rightarrow Poly\ a \\ x &= Cons\ 0\ (Single\ 1) \end{aligned}$$

Then (again, using the list notation for brevity) for any polynomial $as = [a_0, a_1, \dots, a_n]$ we have

$$as = a_0 + a_1 * x + a_2 * x^2 + \dots + a_n * x^n$$

Exercise: check this.

This justifies the standard notation

$$as = \sum_{i=0}^n a_i * x^i$$

5.2 Polynomial degree as a homomorphism

TODO: textify black board notes

It is often the case that a certain function is *almost* a homomorphism and the domain or range *almost* a monoid. In the section on *eval* and *eval'* for *FunExp* we have seen “tupling” as one way to fix such a problem and here we will introduce another way.

The *degree* of a polynomial is a good candidate for being a homomorphism: if we multiply two polynomials we can normally add their degrees. If we try to check that $degree :: Poly\ a \rightarrow \mathbb{N}$ is

the function underlying a monoid morphism we need to decide on the monoid structure to use for the source and for the target, and we need to check the homomorphism laws. We can use $unit = Single\ 1$ and $op = polyMul$ for the source monoid and we can try to use $unit = 0$ and $op = (+)$ for the target monoid. Then we need to check that

$$\begin{aligned} degree\ (Single\ 1) &= 0 \\ \forall x, y. degree\ (x\ 'op'\ y) &= degree\ x + degree\ y \end{aligned}$$

The first law is no problem and for most polynomials the second law is also straightforward to prove (exercise: prove it). But we run into trouble with one special case: the zero polynomial.

Looking back at the definition from Adams and Essex [2010], page 55 it says that the degree of the zero polynomial is not defined. Let's see why that is the case and how we might "fix" it. Assume there is a z such that $degree\ 0 = z$ and that we have some polynomial p with $degree\ p = n$. Then we get

$$\begin{aligned} z &= \{-\text{assumption}-\} \\ degree\ 0 &= \{-\text{simple calculation}-\} \\ degree\ (0 * p) &= \{-\text{homomorphism condition}-\} \\ degree\ 0 + degree\ p &= \{-\text{assumption}-\} \\ z + n \end{aligned}$$

Thus we need to find a z such that $z = z + n$ for all natural numbers n ! At this stage we could either give up, or think out of the box. Intuitively we could try to use $z = -Infinity$, which would seem to satisfy the law but which is not a natural number. More formally what we need to do is to extend the monoid $(\mathbb{N}, 0, +)$ by one more element. In Haskell we can do that using the *Maybe* type constructor:

```
class Monoid a where
  unit :: a
  op    :: a -> a -> a
instance Monoid a => Monoid (Maybe a) where
  unit = Nothing
  op   = opMaybe
  opMaybe Nothing m      = m
  opMaybe m      Nothing = m
  opMaybe (Just m1) (Just m2) = Just (op m1 m2)
```

We quote the Haskell prelude implementation:

```
Lift a semigroup into Maybe forming a Monoid according to http://en.wikipedia.org/wiki/Monoid: "Any semigroup  $S$  may be turned into a monoid simply by adjoining an element  $e$  not in  $S$  and defining  $e * e = e$  and  $e * s = s = s * e$  for all  $s \in S$ ."
Since there is no Semigroup typeclass [...], we use Monoid instead.
```

Thus, to sum up, *degree* is a monoid homomorphism from $(Poly\ a, 1, *)$ to $(Maybe\ \mathbb{N}, Nothing, opMaybe)$.

TODO: check all the properties.

5.3 Power Series

Power series are obtained from polynomials by removing in *Poly'* the restriction that there should be a *finite* number of non-zero coefficients; or, in the case of *Poly*, by going from lists to streams.

$$\text{PowerSeries}' a = \{f : \mathbb{N} \rightarrow a\}$$

type *PowerSeries* *a* = *Poly a* -- finite and infinite non-empty lists

The operations are still defined as before. If we consider only infinite lists, then only the equations which do not contain the patterns for singleton lists will apply.

Power series are usually denoted

$$\sum_{n=0}^{\infty} a_n * x^n$$

the interpretation of x being the same as before. The simplest operation, addition, can be illustrated as follows:

$$\begin{aligned} \sum_{i=0}^{\infty} a_i * x^i &\cong [a_0, & a_1, & \dots] \\ \sum_{i=0}^{\infty} b_i * x^i &\cong [b_0, & b_1, & \dots] \\ \sum_{i=0}^{\infty} (a_i + b_i) * x^i &\cong [a_0 + b_0, & a_1 + b_1, & \dots] \end{aligned}$$

The evaluation of a power series represented by $a : \mathbb{N} \rightarrow A$ is defined, in case the necessary operations make sense on A , as a function

$$\begin{aligned} \text{eval } a : A &\rightarrow A \\ \text{eval } a \ x &= \lim s \text{ where } s \ n = \sum_{i=0}^n a_i * x^i \end{aligned}$$

Note that $\text{eval } a$ is, in general, a partial function (the limit might not exist).

We will consider, as is usual, only the case in which $A = \mathbb{R}$ or $A = \mathbb{C}$.

The term *formal* refers to the independence of the definition of power series from the ideas of convergence and evaluation. In particular, two power series represented by a and b , respectively, are equal only if $a = b$ (as functions). If $a \neq b$, then the power series are different, even if $\text{eval } a = \text{eval } b$.

Since we cannot in general compute limits, we can use an “approximative” *eval*, by evaluating the polynomial resulting from an initial segment of the power series.

$$\begin{aligned} \text{eval} &:: \text{Num } a \Rightarrow \text{Integer} \rightarrow \text{PowerSeries } a \rightarrow (a \rightarrow a) \\ \text{eval } n \text{ as } x &= \text{evalPoly } (\text{takePoly } n \text{ as}) \ x \\ \text{takePoly} &:: \text{Integer} \rightarrow \text{PowerSeries } a \rightarrow \text{Poly } a \\ \text{takePoly } n \ (\text{Single } a) &= \text{Single } a \\ \text{takePoly } n \ (\text{Cons } a \text{ as}) &= \text{if } n \leq 1 \\ &\quad \text{then } \text{Single } a \\ &\quad \text{else } \text{Cons } a \ (\text{takePoly } (n - 1) \text{ as}) \end{aligned}$$

Note that $\text{eval } n$ is not a homomorphism: for example:

$$\begin{aligned} \text{eval } 2 \ (x * x) \ 1 &= \\ \text{evalPoly } (\text{takePoly } 2 \ [0, 0, 1]) \ 1 &= \\ \text{evalPoly } [0, 0] \ 1 &= \\ 0 & \end{aligned}$$

but

$$\begin{array}{lcl}
(eval\ 2\ x\ 1) & = & \\
evalPoly\ (takePoly\ 2\ [0,1])\ 1 & = & \\
evalPoly\ [0,1]\ 1 & = & \\
1 & &
\end{array}$$

and thus $eval\ 2\ (x * x)\ 1 = 0 \neq 1 = 1 * 1 = (eval\ 2\ x\ 1) * (eval\ 2\ x\ 1)$.

5.4 Operations on power series

Power series have a richer structure than polynomials. For example, we also have division (this is similar to the move from \mathbb{Z} to \mathbb{Q}). We start with a special case: trying to compute $p = \frac{1}{1-x}$ as a power series. The specification of $a / b = c$ is $a = c * b$, thus in our case we need to find a p such that $1 = (1 - x) * p$. For polynomials there is no solution to this equation. One way to see that is by using the homomorphism *degree*: the degree of the left hand side is 0 and the degree of the RHS is $1 + degree\ p \neq 0$. But there is still hope if we move to formal power series.

Remember that p is then represented by a stream of coefficients $[p_0, p_1, \dots]$. We make a table of the coefficients of the RHS $= (1 - x) * p = p - x * p$ and of the LHS $= 1$ (seen as a power series).

$$\begin{array}{lcl}
p & == & [p_0, p_1, \quad p_2, \quad \dots] \\
x * p & == & [0, \quad p_0, \quad p_1, \quad \dots] \\
p - x * p & == & [p_0, p_1 - p_0, p_2 - p_1, \dots] \\
1 & == & [1, \quad 0, \quad 0, \quad \dots]
\end{array}$$

Thus, to make the last two lines equal, we are looking for coefficients satisfying $p_0 = 1, p_1 - p_0 = 0, p_2 - p_1 = 0, \dots$. The solution is unique: $1 = p_0 = p_1 = p_2 = \dots$ but only exists for streams (infinite lists) of coefficients. In the common math notation we have just computed

$$\frac{1}{1-x} = \sum_{i=0}^{\infty} x^i$$

Note that this equation holds when we interpret both sides as formal power series, but not necessarily if we try to evaluate the expressions for a particular x . That works for $|x| < 1$ but not for $x = 2$, for example.

For a more general case of power series division p / q with $p = a : as, q = b : bs$, we assume that $a * b \neq 0$. Then we want to find, for any given $(a : as)$ and $(b : bs)$, the series $(c : cs)$ satisfying

$$\begin{array}{lcl}
(a : as) / (b : bs) = (c : cs) & \Leftrightarrow & \{- \text{ def. of division } -\} \\
(a : as) = (c : cs) * (b : bs) & \Leftrightarrow & \{- \text{ def. of } * \text{ for } Cons -\} \\
(a : as) = (c * b) : (cs * (b : bs) + [c] * bs) & \Leftrightarrow & \{- \text{ equality on compnents, def. of division } -\} \\
c = a / b & \{- \text{ and } -\} & \\
as = cs * (b : bs) + [a / b] * bs & \Leftrightarrow & \{- \text{ arithmetics } -\} \\
c = a / b & \{- \text{ and } -\} & \\
cs = (as - [a / b] * bs) / (b : bs) & &
\end{array}$$

This leads to the implementation:

```

instance (Eq a, Fractional a) => Fractional (PowerSeries a) where
  (/) = divPS
  fromRational = Single o fromRational
  divPS :: (Eq a, Fractional a) => PowerSeries a -> PowerSeries a -> PowerSeries a

```

$$\begin{aligned}
\text{divPS } as \quad (\text{Single } b) &= as * \text{Single } (1 / b) \\
\text{divPS } (\text{Single } 0) \quad (\text{Cons } b \ bs) &= \text{Single } 0 \\
\text{divPS } (\text{Single } a) \quad (\text{Cons } b \ bs) &= \text{divPS } (\text{Cons } a \ (\text{Single } 0)) \ (\text{Cons } b \ bs) \\
\text{divPS } (\text{Cons } a \ as) \quad (\text{Cons } b \ bs) &= \text{Cons } c \ (\text{divPS } (as - (\text{Single } c) * bs) \ (\text{Cons } b \ bs)) \\
&\quad \text{where } c = a / b
\end{aligned}$$

The first two equations allow us to also use division on polynomials, but the result will, in general, be a power series, not a polynomial. The first one should be self-explanatory. The second one extends a constant polynomial, in a process similar to that of long division.

For example:

$$\begin{aligned}
ps0, ps1, ps2 &:: (\text{Eq } a, \text{Fractional } a) \Rightarrow \text{PowerSeries } a \\
ps0 &= 1 / (1 - x) \\
ps1 &= 1 / (1 - x)^2 \\
ps2 &= (x^2 - 2 * x + 1) / (x - 1)
\end{aligned}$$

Every *ps* is the result of a division of polynomials: the first two return power series, the third is a polynomial (almost: it has a trailing 0.0).

$$\begin{aligned}
\text{example0} &= \text{takePoly } 10 \ ps0 \\
\text{example01} &= \text{takePoly } 10 \ (ps0 * (1 - x))
\end{aligned}$$

5.5 Formal derivative

Considering the analogy between power series and polynomial functions (via polynomials), we can arrive at a formal derivative for power series through the following computation:

$$\begin{aligned}
\left(\sum_{n=0}^{\infty} a_n * x^n \right)' &= \sum_{n=0}^{\infty} (a_n * x^n)' = \sum_{n=0}^{\infty} a_n * (x^n)' = \sum_{n=0}^{\infty} a_n * (n * x^{n-1}) \\
&= \sum_{n=0}^{\infty} (n * a_n) * x^{n-1} = \sum_{n=1}^{\infty} (n * a_n) * x^{n-1} = \sum_{m=0}^{\infty} ((m+1) * a_{m+1}) * x^m
\end{aligned} \tag{1}$$

Thus the *m*th coefficient of the derivative is $(m+1) * a_{m+1}$.

TODO: redo to arrive at the recursive formulation.

We can implement this, for example, as

$$\begin{aligned}
\text{deriv } (\text{Single } a) &= \text{Single } 0 \\
\text{deriv } (\text{Cons } a \ as) &= \text{deriv}' \ as \ 1 \\
&\quad \text{where } \text{deriv}' \ (\text{Single } a) \ n = \text{Single } (n * a) \\
&\quad \text{deriv}' \ (\text{Cons } a \ as) \ n = \text{Cons } (n * a) \ (\text{deriv}' \ as \ (n+1))
\end{aligned}$$

Side note: we cannot in general implement a Boolean equality test for *PowerSeries*. For example, we know that *deriv ps0* equals *ps1* but we cannot compute *True* in finite time by comparing the coefficients of the two power series.

$$\begin{aligned}
\text{checkDeriv} &:: \text{Integer} \rightarrow \text{Bool} \\
\text{checkDeriv } n &= \text{takePoly } n \ (\text{deriv } ps0) == \text{takePoly } n \ ps1
\end{aligned}$$

Recommended reading: the Functional pearl: “Power series, power serious” McIlroy [1999].

5.6 Helpers

```
instance Functor Poly where
  fmap = fmapPoly
fmapPoly :: (a → b) → (Poly a → Poly b)
fmapPoly f (Single a)  = Single (f a)
fmapPoly f (Cons a as) = Cons (f a) (fmapPoly f as)
po1 :: Num a ⇒ Poly a
po1 = 1 + x^2 - 3 * x^4
```

5.7 Exercises

Exercise 5.1. Fill in the gaps in lectures on polynomials and power series. In particular:

- Give a derivation for the definition of $*$ for polynomials
- Prove that, with the definition of $x = [0, 1]$ we really have

$$as = a_0 + a_1 * x + a_2 * x^2 + \dots + a_n * x^n$$

Exercise 5.2. Complete the following definition

```
instance Num a ⇒ Num [a] where
  (+) = addP
  (*) = mulP
  -- ... TODO
addP :: Num a ⇒ [a] → [a] → [a]
addP = zipWith' (+)
mulP :: Num a ⇒ [a] → [a] → [a]
mulP = -- TODO
```

Note that *zipWith'* is almost, but not quite, the definition of *zipWith* from the standard Haskell prelude.

TODO: more exercises!

```

{-# LANGUAGE FlexibleInstances #-}
{-# LANGUAGE TypeSynonymInstances #-}
module DSLsofMath.W06 where
import DSLsofMath.W05

```

6 Higher-order Derivatives and their Applications

6.1 Review

- key notion *homomorphism*: $S1 \rightarrow S2$
- questions (“equations”):
 - $S1 \stackrel{?}{\leftarrow} S2$ what is the homomorphism between two given structures
 - e.g., $apply : Num\ a \rightarrow Num\ (x \rightarrow a)$
 - $S1? \rightarrow S2$ what is $S1$ compatible with a given homomorphism
 - e.g., $eval : Poly\ a \rightarrow (a \rightarrow a)$
 - $S1 \rightarrow S2?$ what is $S2$ compatible with a given homomorphism
 - e.g., $applyFD : FD\ a \rightarrow (a, a)$
 - $S1 \stackrel{?}{\leftarrow} S2?$ can we find a good structure on $S2$ so that it becomes homomorphic w. $S1?$
 - e.g., $evalD : FunExp \rightarrow FD\ a$
- The importance of the last two is that they offer “automatic differentiation”, i.e., any function constructed according to the grammar of *FunExp*, can be “lifted” to a function that computes the derivative (e.g., a function on pairs).

Example $f\ x = \sin\ x + 2 * x$

We have: $f\ 0 = 0$, $f\ 2 = 4.909297426825682$, etc.

The type of f is $f :: Floating\ a \Rightarrow a \rightarrow a$.

How do we compute, say, $f'\ 2$?

We have several choices.

a. Using *FunExp*

Recall (week 3):

```

data FunExp = Const Rational
            | Id
            | FunExp :+: FunExp
            | FunExp **: FunExp
            | FunExp :/: FunExp
            | Exp FunExp
            | Sin FunExp
            | Cos FunExp
            -- and so on
deriving Show

```

What is the expression e for which $f = eval\ e$?

We have

$$\begin{aligned}
& eval\ e\ x = f\ x \\
& \Leftrightarrow \\
& eval\ e\ x = sin\ x + 2 * x \\
& \Leftrightarrow \\
& eval\ e\ x = eval\ (Sin\ Id)\ x + eval\ (Const\ 2\ :*\ Id)\ x \\
& \Leftrightarrow \\
& eval\ e\ x = eval\ ((Sin\ Id) :+ (Const\ 2\ :*\ Id))\ x \\
& \Leftarrow \\
& e = Sin\ Id :+ (Const\ 2\ :*\ Id)
\end{aligned}$$

Finally, we can apply *derive* and obtain

$$f'\ 2 = eval\ (derive\ e)\ 2$$

This can hardly be called "automatic", look at all the work we did in deducing *e*!

However, consider this:

$$e = f\ Id$$

(Perhaps it would have been better to use, in the definition of *FunExp*, *X* instead of *Id*.)

In general, to find the value of the derivative of a function *f* at a given *x*, we can use

$$drv\ f\ x = evalFunExp\ (derive\ (f\ Id))\ x$$

b. Using *FD*

Recall

$$\begin{aligned}
\mathbf{type}\ FD\ a &= (a \rightarrow a, a \rightarrow a) \\
applyFD\ (f, g)\ x &= (f\ x, g\ x)
\end{aligned}$$

The operations on *FD a* are such that, if *eval e = f*, then

$$(eval\ e, eval'\ e) = (f, f')$$

We are looking for (g, g') such that

$$f\ (g, g') = (f, f') \quad -- (*)$$

so we can then do

$$f'\ 2 = snd\ (applyFD\ (f\ (g, g'))\ 2)$$

We can fulfill (*) if we can find a (g, g') that is a sort of "unit" for *FD a*:

$$\begin{aligned}
sin\ (g, g') &= (sin, cos) \\
exp\ (g, g') &= (exp, exp)
\end{aligned}$$

and so on.

In general, the chain rule gives us

$$f (g, g') = (f \circ g, (f' \circ g) * g')$$

Therefore, we need: $g = id$ and $g' = const\ 1$.

Finally

$$f' \ 2 = snd \ (applyFD \ (f \ (id, const\ 1)) \ 2)$$

In general

$$drvFD \ f \ x = snd \ (applyFD \ (f \ (id, const\ 1)) \ x)$$

computes the derivative of f at x .

$$\begin{aligned} f_1 &:: FD \ Double \rightarrow FD \ Double \\ f_1 &= f \end{aligned}$$

c. Using pairs

We have **instance** *Floating* $a \Rightarrow Floating \ (a, a)$, moreover, the instance declaration looks exactly the same as that for *FD a*:

$$\begin{aligned} &\mathbf{instance} \ Floating \ a \Rightarrow Floating \ (FD \ a) \ \mathbf{where} \\ &\quad exp \ (f, f') = (exp \ f, (exp \ f) * f') \\ &\quad sin \ (f, f') = (sin \ f, (cos \ f) * f') \\ &\quad cos \ (f, f') = (cos \ f, -(sin \ f) * f') \\ &\mathbf{instance} \ Floating \ a \Rightarrow Floating \ (a, a) \ \mathbf{where} \\ &\quad exp \ (f, f') = (exp \ f, (exp \ f) * f') \\ &\quad sin \ (f, f') = (sin \ f, cos \ f * f') \\ &\quad cos \ (f, f') = (cos \ f, -(sin \ f) * f') \end{aligned}$$

In fact, the latter represents a generalisation of the former. It is also the “maximally general” such generalisation (discounting the “noise” generated by the less-than-clean design of *Num*, *Fractional*, *Floating*).

Still, we need to use this machinery. We are now looking for a pair of values (g, g') such that

$$f \ (g, g') = (f \ 2, f' \ 2)$$

In general

$$f \ (g, g') = (f \ g, (f' \ g) * g')$$

Therefore

$$\begin{aligned} f \ (g, g') &= (f \ 2, f' \ 2) \\ \Leftrightarrow \\ (f \ g, (f' \ g) * g') &= (f \ 2, f' \ 2) \\ \Leftarrow \\ g &= 2, g' = 1 \end{aligned}$$

Introducing

$$var \ x = (x, 1)$$

we can, as in the case of *FD*, simplify matters a little:

$$f' \ x = \text{snd} \ (f \ (\text{var} \ x))$$

In general

$$\text{drvP} \ f \ x = \text{snd} \ (f \ (x, 1))$$

computes the derivative of f at x .

$$\begin{aligned} f_2 &:: (\text{Double}, \text{Double}) \rightarrow (\text{Double}, \text{Double}) \\ f_2 &= f \end{aligned}$$

6.2 Higher-order derivatives

Consider

$$[f, f', f'', \dots]$$

representing the evaluation of an expression and its derivatives:

$$\text{evalAll} \ e = (\text{evalFunExp} \ e) : \text{evalAll} \ (\text{derive} \ e)$$

Notice that, if

$$[f, f', f'', \dots] = \text{evalAll} \ e$$

then

$$[f', f'', \dots] = \text{evalAll} \ (\text{derive} \ e)$$

We want to define the operations on lists of functions in such a way that *evalAll* is a homomorphism. For example:

$$\text{evalAll} \ (e_1 :*: e_2) = \text{evalAll} \ e_1 * \text{evalAll} \ e_2$$

where the $(*)$ sign stands for the multiplication of infinite lists of functions, the operation we are trying to determine.

We have, writing *eval* for *evalFunExp* in order to save ink

$$\begin{aligned} & \text{evalAll} \ (e_1 :*: e_2) = \text{evalAll} \ e_1 * \text{evalAll} \ e_2 \\ \Leftrightarrow & \\ & \text{eval} \ (e_1 :*: e_2) : \text{evalAll} \ (\text{derive} \ (e_1 :*: e_2)) = \\ & \text{eval} \ e_1 : \text{evalAll} \ (\text{derive} \ e) * \text{eval} \ e_1 : \text{evalAll} \ (\text{derive} \ e_2) \\ \Leftrightarrow & \\ & (\text{eval} \ e_1 * \text{eval} \ e_2) : \text{evalAll} \ (\text{derive} \ (e_1 :*: e_2)) = \\ & \text{eval} \ e_1 : \text{evalAll} \ (\text{derive} \ e) * \text{eval} \ e_1 : \text{evalAll} \ (\text{derive} \ e_2) \\ \Leftrightarrow & \\ & (\text{eval} \ e_1 * \text{eval} \ e_2) : \text{evalAll} \ (\text{derive} \ e_1 :*: e_2 :+ e_1 * \text{derive} \ e_2) = \\ & \text{eval} \ e_1 : \text{evalAll} \ (\text{derive} \ e) * \text{eval} \ e_1 : \text{evalAll} \ (\text{derive} \ e_2) \\ \Leftarrow & \\ & (a : as) * (b : bs) = (a * b) : (as * (b : bs) + (a : as) * bs) \end{aligned}$$

The final line represents the definition of $(*)$ needed for ensuring the conditions are met.

As in the case of pairs, we find that we do not need any properties of functions, other than their *Num* structure, so the definitions apply to any infinite list of *Num* a :

```
instance Num a  $\Rightarrow$  Num [a] where
  (a : as) + (b : bs) = (a + b) : (as + bs)
  (a : as) * (b : bs) = (a * b) : (as * (b : bs) + (a : as) * bs)
```

Exercise: complete the instance declarations for *Fractional* and *Floating*. Write a general derivative computation, similar to *drv* functions above:

```
drvList k f x = undefined -- kth derivative of f at x
```

Exercise: Compare the efficiency of different ways of computing derivatives.

6.3 Polynomials

```
data Poly a = Single a | Cons a (Poly a)
deriving (Eq, Ord)

evalPoly :: Num a  $\Rightarrow$  Poly a  $\rightarrow$  a  $\rightarrow$  a
evalPoly (Single a) x = a
evalPoly (Cons a as) x = a + x * evalPoly as x
```

6.4 Power series

No need for a separate type in Haskell

```
type PowerSeries a = Poly a -- finite and infinite non-empty lists
```

Now we can divide, as well as add and multiply.

We can also derive:

```
deriv (Single a) = Single 0
deriv (Cons a as) = deriv' as 1
where deriv' (Single a) n = Single (n * a)
      deriv' (Cons a as) n = Cons (n * a) (deriv' as (n + 1))
```

and integrate:

```
integ :: Fractional a  $\Rightarrow$  PowerSeries a  $\rightarrow$  a  $\rightarrow$  PowerSeries a
integ as a0 = Cons a0 (integ' as 1)
where integ' (Single a) n = Single (a / n)
      integ' (Cons a as) n = Cons (a / n) (integ' as (n + 1))
```

Note that a_0 is the constant that we need due to indefinite integration.

All these definitions make sense, irrespective of convergence, hence “formal”.

If the power series involved do converge, then *eval* is a morphism between the formal structure and that of the functions represented:

```
eval as + eval bs = eval (as + bs)
eval as * eval bs = eval (as * bs)
eval (deriv as) = D (eval as)
eval (integ as c) x =  $\int_0^x$  (eval as t) dt + c
```

6.5 Simple differential equations

Many first-order differential equations have the structure

$$f' x = g f x, \quad f 0 = f_0$$

i.e., they are defined in terms of g .

The fundamental theorem of calculus gives us

$$f x = \int_0^x (g f t) dt + f_0$$

If $f = eval\ as$

$$eval\ as\ x = \int_0^x (g (eval\ as) t) dt + f_0$$

Assuming that g is a polymorphic function that commutes with $eval$

$$eval\ as\ x = \int_0^x (eval\ (g\ as)\ t) dt + f_0$$

$$eval\ as\ x = eval\ (integ\ (g\ as)\ f_0)\ x$$

$$as = integ\ (g\ as)\ f_0$$

Which functions g commute with $eval$? All the ones in *Num*, *Fractional*, *Floating*, by construction; additionally, as above, *deriv* and *integ*.

Therefore, we can implement a general solver for these simple equations:

$$solve :: Fractional\ a \Rightarrow (PowerSeries\ a \rightarrow PowerSeries\ a) \rightarrow a \rightarrow PowerSeries\ a$$

$$solve\ g\ f_0 = f \quad \text{-- solves } f' = g\ f, f\ 0 = f_0$$

$$\textbf{where } f = integ\ (g\ f)\ f_0$$

$$idx = solve\ (\lambda f \rightarrow 1)\ 0$$

$$idf = eval\ 100\ idx$$

$$expx = solve\ (\lambda f \rightarrow f)\ 1$$

$$expf = eval\ 100\ expx$$

$$sinx = solve\ (\lambda f \rightarrow cosx)\ 0$$

$$cosx = solve\ (\lambda f \rightarrow -sinx)\ 1$$

$$sinf = eval\ 100\ sinx$$

$$cosf = eval\ 100\ cosx$$

$$idx, expx, sinx, cosx :: Fractional\ a \Rightarrow PowerSeries\ a$$

$$idf, expf, sinf, cosf :: Fractional\ a \Rightarrow a \rightarrow a$$

6.6 The *Floating* structure of *PowerSeries*

Can we compute $exp\ as$?

Specification:

$$eval\ (exp\ as) = exp\ (eval\ as)$$

Differentiating both sides, we obtain

$$D\ (eval\ (exp\ as)) = exp\ (eval\ as) * D\ (eval\ as)$$

$$\Leftrightarrow \{-\ eval\ morphism\ -\}$$

$$\begin{aligned}
& eval (deriv (exp as)) = eval (exp as * deriv as) \\
\Leftarrow \\
& deriv (exp as) = exp as * deriv as
\end{aligned}$$

Adding the “initial condition” $eval (exp as) 0 = exp (head as)$, we obtain

$$exp as = integ (exp as * deriv as) (exp (val as))$$

Note: we cannot use *solve* here, because the *g* function uses both *exp as* and *as* (it “looks inside” its argument).

instance (*Eq a, Floating a*) \Rightarrow *Floating (PowerSeries a)* **where**

$$\begin{aligned}
\pi &= Single \pi \\
exp fs &= integ (exp fs * deriv fs) (exp (val fs)) \\
sin fs &= integ (cos fs * deriv fs) (sin (val fs)) \\
cos fs &= integ (-sin fs * deriv fs) (cos (val fs)) \\
val :: PowerSeries a &\rightarrow a \\
val (Single a) &= a \\
val (Cons a as) &= a
\end{aligned}$$

In fact, we can implement *all* the operations needed for evaluating *FunExp* functions as power series!

$$\begin{aligned}
evalP :: (Eq r, Floating r) &\Rightarrow FunExp \rightarrow PowerSeries r \\
evalP (Const x) &= Single (fromRational x) \\
evalP (e_1 :+: e_2) &= evalP e_1 + evalP e_2 \\
evalP (e_1 **: e_2) &= evalP e_1 * evalP e_2 \\
evalP (e_1 :/: e_2) &= evalP e_1 / evalP e_2 \\
evalP Id &= idx \\
evalP (Exp e) &= exp (evalP e) \\
evalP (Sin e) &= sin (evalP e) \\
evalP (Cos e) &= cos (evalP e)
\end{aligned}$$

6.7 Taylor series

If $f = eval [a_0, a_1, \dots, a_n, \dots]$, then

$$\begin{aligned}
f 0 &= a_0 \\
f' &= eval (deriv [a_0, a_1, \dots, a_n, \dots]) \\
&= eval ([a_1, 2 * a_2, 3 * a_3, \dots, n * a_n, \dots]) \\
\Rightarrow \\
f' 0 &= a_1 \\
f'' &= eval (deriv [a_1, 2 * a_2, \dots, n * a_n, \dots]) \\
&= eval ([2 * a_2, 3 * 2 * a_3, \dots, n * (n - 1) * a_n, \dots]) \\
\Rightarrow \\
f'' 0 &= 2 * a_2
\end{aligned}$$

In general:

$$f^{(k)} 0 = fact k * a_k$$

Therefore

$$f = \text{eval } [f\ 0, f'\ 0, f''\ 0 / 2, \dots, f^{(n)}0 / (\text{fact } n), \dots]$$

The series $[f\ 0, f'\ 0, f''\ 0 / 2, \dots, f^{(n)}0 / (\text{fact } n), \dots]$ is called the Taylor series centred in 0, or the Maclaurin series.

Therefore, if we can represent f as a power series, we can find the value of all derivatives of f at 0!

```

derivs :: Num a => PowerSeries a -> PowerSeries a
derivs as = derivs1 as 0 1
  where
    derivs1 (Cons a as) n factn = Cons (a * factn)
                                         (derivs1 as (n + 1) (factn * (n + 1)))
    derivs1 (Single a)   n factn = Single (a * factn)
    -- remember that x = Cons 0 (Single 1)
    ex3 = takePoly 10 (derivs (x^3 + 2 * x))
    ex4 = takePoly 10 (derivs sinx)

```

In this way, we can compute all the derivatives at 0 for all functions f constructed with the grammar of *FunExp*. That is because, as we have seen, we can represent all of them by power series!

What if we want the value of the derivatives at $a \neq 0$?

We then need the power series of the “shifted” function g :

$$g\ x = f\ (x + a) \Leftrightarrow g = f \circ (+a)$$

If we can represent g as a power series, say $[b_0, b_1, \dots]$, then we have

$$g^{(k)}0 = \text{fact } k * b_k = f^{(k)}a$$

In particular, we would have

$$f\ x = g\ (x - a) = \sum b_n * (x - a)^n$$

which is called the Taylor expansion of f at a .

Example:

We have that $\text{id}x = [0, 1]$, thus giving us indeed the values

$$[\text{id}\ 0, \text{id}'\ 0, \text{id}''\ 0, \dots]$$

In order to compute the values of

$$[\text{id}\ a, \text{id}'\ a, \text{id}''\ a, \dots]$$

for $a \neq 0$, we compute

$$\text{id}a\ a = \text{takePoly } 10\ (\text{derivs}\ (\text{evalP}\ (\text{Id} \text{ :+ } \text{Const } a)))$$

More generally, if we want to compute the derivative of a function f constructed with *FunExp* grammar, at a point a , we need the power series of $g\ x = f\ (x + a)$:

$$d\ f\ a = \text{takePoly } 10\ (\text{derivs}\ (\text{evalP}\ (f\ (\text{Id} \text{ :+ } \text{Const } a))))$$

Use, for example, our $f\ x = \sin\ x + 2 * x$ above.

As before, we can use directly power series:

$$dP\ f\ a = \text{takePoly } 10\ (\text{derivs}\ (f\ (\text{id}x + \text{Single } a)))$$

6.8 Associated code

```

instance Num a  $\Rightarrow$  Num (x  $\rightarrow$  a) where
  f + g      =  $\lambda x \rightarrow f\ x + g\ x$ 
  f - g      =  $\lambda x \rightarrow f\ x - g\ x$ 
  f * g      =  $\lambda x \rightarrow f\ x * g\ x$ 
  negate f   = negate  $\circ$  f
  abs f      = abs  $\circ$  f
  signum f   = signum  $\circ$  f
  fromInteger = const  $\circ$  fromInteger

instance Fractional a  $\Rightarrow$  Fractional (x  $\rightarrow$  a) where
  recip f     = recip  $\circ$  f
  fromRational = const  $\circ$  fromRational

instance Floating a  $\Rightarrow$  Floating (x  $\rightarrow$  a) where
   $\pi$          = const  $\pi$ 
  exp f      = exp  $\circ$  f
  sin f      = sin  $\circ$  f
  cos f      = cos  $\circ$  f
  f ** g     =  $\lambda x \rightarrow (f\ x) ** (g\ x)$ 
  -- and so on

evalFunExp :: Floating a  $\Rightarrow$  FunExp  $\rightarrow$  a  $\rightarrow$  a
evalFunExp (Const  $\alpha$ ) = const (fromRational  $\alpha$ )
evalFunExp Id          = id
evalFunExp (e1 :+: e2) = evalFunExp e1 + evalFunExp e2 -- note the use of “lifted +”
evalFunExp (e1 **: e2) = evalFunExp e1 * evalFunExp e2 -- “lifted *”
evalFunExp (Exp e1)    = exp (evalFunExp e1)           -- and “lifted exp”
evalFunExp (Sin e1)    = sin (evalFunExp e1)
evalFunExp (Cos e1)    = cos (evalFunExp e1)
  -- and so on

derive (Const  $\alpha$ ) = Const 0
derive Id          = Const 1
derive (e1 :+: e2) = derive e1 :+: derive e2
derive (e1 **: e2) = (derive e1 **: e2) :+: (e1 **: derive e2)
derive (Exp e)      = Exp e **: derive e
derive (Sin e)      = Cos e **: derive e
derive (Cos e)      = Const (-1) **: Sin e **: derive e

instance Num FunExp where
  (+) = (:+:)
  (*) = (:*)
  fromInteger n = Const (fromInteger n)

instance Fractional FunExp where
  (/) = (:/)

instance Floating FunExp where
  exp = Exp
  sin = Sin

```

6.8.1 Not included to avoid overlapping instances

```

instance Num a  $\Rightarrow$  Num (FD a) where
  (f, f') + (g, g') = (f + g, f' + g')

```



```

(f, f') * (g, g') = (f * g, f' * g + f * g')
fromInteger n = (fromInteger n, const 0)
instance Fractional a => Fractional (FD a) where
  (f, f') / (g, g') = (f / g, (f' * g - g' * f) / (g * g))
instance Floating a => Floating (FD a) where
  exp (f, f')      = (exp f, (exp f) * f')
  sin (f, f')      = (sin f, (cos f) * f')
  cos (f, f')      = (cos f, -(sin f) * f')

```

6.8.2 This is included instead

```

instance Num a => Num (a, a) where
  (f, f') + (g, g') = (f + g, f' + g')
  (f, f') * (g, g') = (f * g, f' * g + f * g')
  fromInteger n = (fromInteger n, fromInteger 0)
instance Fractional a => Fractional (a, a) where
  (f, f') / (g, g') = (f / g, (f' * g - g' * f) / (g * g))
instance Floating a => Floating (a, a) where
  exp (f, f')      = (exp f, (exp f) * f')
  sin (f, f')      = (sin f, cos f * f')
  cos (f, f')      = (cos f, -(sin f) * f')

```

6.9 Exercises

Exercise 6.1. As shown at the start of the chapter, we can find expressions $e :: FunExp$ such that $eval\ e = f$ automatically using the assignment $e = f\ Id$. This is possible thanks to the *Num*, *Fractional*, and *Floating* instances of *FunExp*. Use this method to find *FunExp* representations of the functions below, and show step by step how the application of the function to *Id* is evaluated in each case.

- $f_1\ x = x^2 + 4$
- $f_2\ x = 7 * exp\ (2 + 3 * x)$
- $f_3\ x = 1 / (sin\ x + cos\ x)$

Exercise 6.2. For each of the expressions $e :: FunExp$ you found in exercise 6.1, use *derive* to find an expression $e' :: FunExp$ representing the derivative of the expression, and verify that e' is indeed the derivative of e .

Exercise 6.3. At the start of this chapter, we saw three different ways of computing the value of the derivative of a function at a given point:

- Using *FunExp*
- Using *FD*
- Using pairs

Try using each of these methods to find the values of $f_1'\ 2$, $f_2'\ 2$, and $f_3'\ 2$, i.e. the derivatives of each of the functions in exercise 6.1, evaluated at the point 2. You can verify that the result is correct by comparing it with the expressions e_1' , e_2' and e_3' that you found in 6.2.

Exercise 6.4. The exponential function $\exp t = e^t$ has the property that $\int \exp t \, dt = \exp t + C$. Use this fact to express the functions below as *PowerSeries* using *integ*. *Hint: the definitions will be recursive.*

- a. $\lambda t \rightarrow \exp t$
- b. $\lambda t \rightarrow \exp (3 * t)$
- c. $\lambda t \rightarrow 3 * \exp (2 * t)$

Exercise 6.5. In the chapter, we saw that a representation $\exp x :: \text{PowerSeries}$ of the exponential function can be implemented using *solve* as $\exp x = \text{solve } (\lambda f \rightarrow f) \, 1$. Use the same method to implement power series representations of the following functions:

- a. $\lambda t \rightarrow \exp (3 * t)$
- b. $\lambda t \rightarrow 3 * \exp (2 * t)$

Exercise 6.6.

- a. Implement $\text{id}x'$, $\text{sin}x'$ and $\text{cos}x'$ using *solve*
- b. Complete the instance *Floating (PowerSeries a)*

Exercise 6.7. Consider the following differential equation:

$$f'' t + f' t - 2 * f t = e^{3*t}, \quad f 0 = 1, \quad f' 0 = 2$$

We will solve this equation assuming that f can be expressed by a power series fs , and finding the three first coefficients of fs .

- a. Implement $\exp x3 :: \text{PowerSeries}$, a power series representation of e^{3*t}
- b. Find an expression for fs'' , the second derivative of fs , in terms of $\exp x3$, fs' , and fs .
- c. Find an expression for fs' in terms of fs'' , using *integ*.
- d. Find an expression for fs in terms of fs' , using *integ*.
- e. Use *takePoly* to find the first three coefficients of fs . You can check that your solution is correct using a tool such as MATLAB or WolframAlpha, by first finding an expression for $f t$, and then getting the Taylor series expansion for that expression.

Exercise 6.8. *From exam 2016-03-15*

Consider the following differential equation:

$$f'' t - 2 * f' t + f t = e^{2*t}, \quad f 0 = 2, \quad f' 0 = 3$$

Solve the equation assuming that f can be expressed by a power series fs , that is, use *deriv* and *integ* to compute fs . What are the first three coefficients of fs ?

Exercise 6.9. *From exam 2016-08-23*

Consider the following differential equation:

$$f'' t - 5 * f' t + 6 * f t = e^t, \quad f 0 = 1, \quad f' 0 = 4$$

Solve the equation assuming that f can be expressed by a power series fs , that is, use *deriv* and *integ* to compute fs . What are the first three coefficients of fs ?

Exercise 6.10. *From practice exam*

Consider the following differential equation:

$$f'' t - 2 * f' t + f t - 2 = 3 * e^{2*t}, \quad f 0 = 5, \quad f' 0 = 6$$

Solve the equation assuming that f can be expressed by a power series fs , that is, use *deriv* and *integ* to compute fs . What are the first three coefficients of fs ?

Exercise 6.11. *From exam 2017-03-14*

Consider the following differential equation:

$$f'' t + 4 * f t = 6 * \cos t, \quad f 0 = 0, \quad f' 0 = 0$$

Solve the equation assuming that f can be expressed by a power series fs , that is, use *integ* and the differential equation to express the relation between fs , fs' , fs'' , and rhs where rhs is the power series representation of $(6*) \circ \cos$. What are the first four coefficients of fs ?

Exercise 6.12. *From exam 2017-08-22*

Consider the following differential equation:

$$f'' t - 3\sqrt{2} * f' t + 4 * f t = 0, \quad f 0 = 2, \quad f' 0 = 3\sqrt{2}$$

Solve the equation assuming that f can be expressed by a power series fs , that is, use *integ* and the differential equation to express the relation between fs , fs' , and fs'' . What are the first three coefficients of fs ?

```

{-# LANGUAGE FlexibleInstances #-}
{-# LANGUAGE UndecidableInstances #-}
module DSLsofMath.W07 where

```

7 Matrix algebra and linear transformations

Often, especially in engineering textbooks, one encounters the “definition”: a vector is an $n + 1$ -tuple of real or complex numbers, arranged as a column:

$$\mathbf{v} = \begin{pmatrix} v_0 \\ \vdots \\ v_n \end{pmatrix}$$

Other times, this is supplemented by the definition of a “row vector”:

$$\mathbf{v} = v_0, \dots, v_n$$

The v_i s are real or complex numbers, or, more generally, elements of a *field* (analogous to being an instance of *Fractional*). Vectors can be added “point-wise” and multiplied with “scalars”, i.e., elements of the field:

$$\begin{aligned} \mathbf{v} + \mathbf{w} &= \begin{pmatrix} v_0 \\ \vdots \\ v_n \end{pmatrix} + \begin{pmatrix} w_0 \\ \vdots \\ w_n \end{pmatrix} = \begin{pmatrix} v_0 + w_0 \\ \vdots \\ v_n + w_n \end{pmatrix} \\ \mathbf{s} * \mathbf{v} &= \begin{pmatrix} s*v_0 \\ \vdots \\ s*v_n \end{pmatrix} \end{aligned}$$

The scalar s “scales” all the components of \mathbf{v} .

In fact, the most important feature of vectors is that they can be *uniquely* expressed as a simple sort of combination of other vectors:

$$\mathbf{v} = \begin{pmatrix} v_0 \\ \vdots \\ v_n \end{pmatrix} = v_0 \begin{pmatrix} 1 \\ 0 \\ \vdots \\ 0 \end{pmatrix} + \dots + v_n \begin{pmatrix} 0 \\ \vdots \\ 1 \end{pmatrix}$$

We denote by

```

      0
      0
ek  =  .
      .
      0
      1      <-- position k
      0
      .
      .
      0

```

the vector that is everywhere 0 except at position k , where it is 1, so that

$$v = v_0 * e0 + \dots + v_n * en$$

There is a temptation to model vectors by lists or tuples, but the simplest (at least conceptually) way is to view them as *functions*:

```

type S      = ...  -- the scalars, forming a field ( $\mathbb{R}$ , or Complex, or  $Zn$ , etc.)
type Vector G = G → S

```

Usually, G is finite, i.e., *Bounded* and *Enumerable*.

We know from the previous lectures that if S is an instance of *Num*, *Fractional*, etc. then so is $G \rightarrow S$, with the pointwise definitions. In particular, the instance declarations for $+$, multiplication, and embedding of constants, give us exactly the structure needed for the vector operations. For example

```

      s * v
= {- s is promoted to a function -}
      const s * v
= {- Num instance definition -}
      λg → (const s) g * v g
= {- definition of const -}
      λg → s * v g

```

The set G is typically $\{0, 1, \dots, n\}$. The basis vectors are then

$$e\ i : G \rightarrow S, e\ i\ g = i\ 'is'\ g$$

Implementation:

```

is a b = if a == b then 1 else 0
e g = λ(G g') → g 'is' g'
toL v = [v g | g <- [minBound .. maxBound]]  -- so we can actually see them

```

and every

$$v : G \rightarrow S$$

is trivially a linear combination of $e\ is$:

$$v = v\ 0 * e\ 0 + \dots + v\ n * e\ n$$

7.1 Functions on vectors

What if we have another vector space, $Vector\ G' = G' \rightarrow S$? We are interested in functions $f : Vector\ G \rightarrow Vector\ G'$:

$$f\ v = f\ (v\ 0 * e\ 0 + \dots + v\ n * e\ n)$$

A “good” function should translate the operations in $Vector\ G$ into operations in $Vector\ G'$, i.e., should be a homomorphism:

$$f\ v = f\ (v\ 0 * e\ 0 + \dots + v\ n * e\ n) = v\ 0 * f\ (e\ 0) + \dots + v\ n * f\ (e\ n)$$

But this means that we can determine the values of $f : (G \rightarrow S) \rightarrow (G' \rightarrow S)$ from just the values of $f \circ e : G \rightarrow (G' \rightarrow S)$, a much “smaller” function. Let $m = f \circ e$. Then

$$f\ v = v\ 0 * m\ 0 + \dots + v\ n * m\ n$$

Each of $m\ k$ is a $Vector\ G'$, as is the resulting $f\ v$. We have

$$\begin{aligned} & f\ v\ g' \\ &= \{- \text{ as above } -\} \\ & \quad v\ 0 * m\ 0 + \dots + v\ n * m\ n \\ &= \{- * \text{ and } + \text{ for functions are def. pointwise } -\} \\ & \quad v\ 0 * m\ 0\ g' + \dots + v\ n * m\ n\ g' \\ &= \{- \text{ using } sum -\} \\ & \quad sum\ [v\ i * m\ i\ g' \mid i \leftarrow [minBound .. maxBound]] \end{aligned}$$

Implementation:

This is the almost the standard “vector-matrix” multiplication:

$$M = [m\ 0 \mid \dots \mid m\ n]$$

The columns of M are the images of the canonical base vectors $e\ i$ through f . Every $m\ k$ has $card\ G'$ rows, and it has become standard to use $M\ i\ j$ to mean the i th element of the j th column, i.e., $m\ j\ i$, so that

$$(M * v)\ i = sum\ [M\ i\ j * v\ j \mid j \leftarrow [0 .. n]]$$

$$mul\ m\ v\ g' = sum\ [m\ g'\ g * v\ g \mid g \leftarrow [minBound .. maxBound]]$$

Example:

$$(M * e\ k)\ i = sum\ [M\ i\ j * e\ k\ j \mid j \leftarrow [0 .. n]] = sum\ [M\ i\ k] = M\ i\ k$$

i.e., $e\ k$ extracts the k th column from M (hence the notation “e” for “extract”).

Given an arbitrary matrix M , we can define

$$f\ v = M * v$$

and obtain a linear transformation $(M*)$. Moreover $((M*) \circ e)\ g\ g' = M\ g'\ g$, i.e., the matrix constructed as above for f is precisely M .

Exercise: compute $((M*) \circ e) g g'$.

Therefore, every linear transformation is of the form $(M*)$ and every $(M*)$ is a linear transformation.

Matrix-matrix multiplication is defined in order to ensure that

$$(M' * M) * v = M' * (M * v)$$

that is

$$((M' * M)*) = (M'*) \circ (M*)$$

Exercise: work this out in detail.

Exercise: show that matrix-matrix multiplication is associative.

Perhaps the simplest vector space is obtained for $G = ()$, the singleton set. In this case, the vectors $s : () \rightarrow S$ are functions that can take exactly one argument, therefore have exactly one value: $s ()$, so they are often identified with S . But, for any $v : G \rightarrow S$, we have a function $fv : G \rightarrow (() \rightarrow S)$, namely

$$fv g () = v g$$

fv is similar to our m function above. The associated matrix is

$$M = [m\ 0 \mid \dots \mid m\ n] = [fv\ 0 \mid \dots \mid fv\ n]$$

having $n + 1$ columns (the dimension of *Vector* G) and one row (dimension of *Vector* $()$). Let $w :: \text{Vector } G$:

$$M * w = w\ 0 * fv\ 0 + \dots + w\ n * fv\ n$$

$M * v$ and each of the $fv\ k$ are “almost scalars”: functions of type $() \rightarrow S$, thus, the only component of $M * w$ is

$$(M * w) () = w\ 0 * fv\ 0 () + \dots + w\ n * fv\ n () = w\ 0 * v\ 0 + \dots + w\ n * v\ n$$

i.e., the scalar product of v and w .

Remark: I have not discussed the geometrical point of view. For the connection between matrices, linear transformations, and geometry, I warmly recommend binge-watching the “Essence of linear algebra” videos on youtube (start here: <https://www.youtube.com/watch?v=kjBOesZCoqc>).

7.2 Examples of matrix algebra

7.2.1 Derivative

We have represented polynomials of degree $n + 1$ by the list of their coefficients. This is quite similar to “standard” geometrical vectors represented by $n + 1$ coordinates. This suggests that polynomials of degree $n + 1$ form a vector space, and we could interpret that as $\{0, \dots, n\} \rightarrow \mathbb{R}$ (or, more generally, $\text{Field } a \Rightarrow \{0, \dots, n\} \rightarrow a$. The operations $+$ and $*$ are defined in the same way as they are for functions.

The *derive* function takes polynomials of degree $n + 1$ to polynomials of degree n , and since $D(f + g) = Df + Dg$ and $D(s * f) = s * Df$, we expect it to be a linear transformation. What is its associated matrix?

To answer that, we must first determine the canonical base vectors. As for geometrical vectors, they are

$$e_i : \{0, \dots, n\} \rightarrow \text{Real}, e_i(j) = \delta_{ij}$$

The evaluation of e_i returns the function $\lambda x \rightarrow x^i$, as expected.

The associated matrix will be

$$M = [D(e_0), D(e_1), \dots, D(e_n)]$$

where each $D(e_i)$ has length n . Vector e_{i+1} represents x^{i+1} , therefore

$$D(e_{i+1})(j) = (i+1) * x^i(j)$$

i.e.

$$D(e_{i+1})(j) = \text{if } i == j \text{ then } i+1 \text{ else } 0$$

and

$$D(e_0) = 0$$

Example: $n+1 = 3$:

$$M = \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

Take the polynomial

$$3 * X^2 + 2 * X + 1$$

as a vector

$$v = \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

and we have

$$M * v = \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 2 \end{bmatrix} * \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} = \begin{bmatrix} 2 \\ 6 \end{bmatrix}$$


representing the polynomial $6 * X^2 + 2 * X$.

Exercise: write the (infinite-dimensional) matrix representing D for power series.

Exercise: write the matrix associate with integration of polynomials.

7.2.2 Simple deterministic systems (transition systems)

Simple deterministic systems are given by endo-functions on a finite set $f : G \rightarrow G$. They can often be conveniently represented as a graph, for example



Here, $G = \{0, \dots, 6\}$. A node in the graph represents a state. A transition $i \rightarrow j$ means $f i = j$. Since f is an endo-function, every node must be the source of exactly one arrow.

We can take as vectors the characteristic functions of subsets of G , i.e., $G \rightarrow \{0, 1\}$. $\{0, 1\}$ is not a field w.r.t. the standard arithmetical operations (it is not even closed w.r.t. addition), and the standard trick to avoid this is to extend the type of the functions to \mathbb{R} .

The canonical basis vectors are, as usual, $e i = \lambda j \rightarrow i \text{ 'is' } j$. Each $e i$ is the characteristic function of a singleton set, $\{i\}$. Thus, the inputs to f are canonical vectors.

To write the matrix associated to f , we have to compute what vector is associated to each canonical base vector vector:

$$M = [f(e_0), f(e_1), \dots, f(e_n)]$$

Therefore:

$$M = \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 1 & 1 & 0 & 0 \end{bmatrix}$$

Starting with a canonical base vector $e i$, we obtain $M * e i = e(f i)$, as we would expect.

It is more interesting if we start with a non-base vector. For example, $e 2 + e 4$, which represents the subset $\{2, 4\}$.

The more interesting thing is if we start with something different from a basis vector, say $[0, 1, 1]$. We obtain $\{f 2, f 4\} = \{5, 6\}$, the image of $\{2, 4\}$ through f . In a sense, we can say that the two

computations were done in parallel. But that is not quite accurate: if start with $\{3, 4\}$, we no longer get the characteristic function of $\{f\ 3, f\ 4\} = \{6\}$, instead, we get a vector that does not represent a characteristic function at all: $[0, 0, 0, 0, 0, 2]$. In general, if we start with an arbitrary vector, we can interpret this as starting with various quantities of some unspecified material in each state, simultaneously. If f were injective, the respective quantities would just get shifted around, but in our case, we get a more interesting behaviour.

What if we do want to obtain the characteristic function of the image of a subset? In that case, we need to use other operations than the standard arithmetical ones, for example *min* and *max*. The problem is that $(\{0, 1\}, \max, \min)$ is not a field, and neither is (\mathbb{R}, \max, \min) . This is not a problem if all we want is to compute the evolutions of possible states, but we cannot apply most of the “deeper” results of linear algebra.

In the example above, we have:

```
newtype G = G Int deriving (Eq, Show)
instance Bounded G where
    minBound = G 0
    maxBound = G 6
instance Enum G where
    toEnum      = G
    fromEnum (G n) = n
```

The transition function:

```
f1 0 = 1
f1 1 = 3
f1 2 = 5
f1 3 = 6
f1 4 = 6
f1 5 = 4
f1 6 = 5
```

The associated matrix:

$$m_1\ (G\ g')\ (G\ g) = g' \text{ 'is' } f_1\ g$$

Test:


$$t_1 = toL\ (mul\ m_1\ (e\ 3 + e\ 4))$$

7.2.3 Non-deterministic systems

Another interpretation of the application of M to characteristic functions of a subset is the following: assuming that all I know is that the system is in one of the states of the subset, where can it end up after one step? (this assumes the *max-min* algebra as above).

In this case, the uncertainty is entirely caused by the fact that we do not know the exact initial state. However, there are cases in which the output of f is not known, even when the input is known. Such situations are modelled by endo-relations: $R: G \rightarrow G$, with $g\ R\ g'$ if g' is a potential successor of g . Endo-relations can also be pictured as graphs, but the restriction that every node should be the source of exactly one arrow is lifted. Every node can be the source of one, none, or many arrows.

For example:



Now, starting in 0 we might end up either in 1 or 2 (but not both!). Starting in 6, the system breaks down: there is no successor state.

The matrix associated to R is built in the same fashion: we need to determine what vectors the canonical base vectors are associated with:

$$\begin{array}{rcl}
 & 0 & 0 & 0 & 0 & 0 & 0 \\
 & 1 & 0 & 0 & 0 & 1 & 0 & 0 \\
 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
 M & = & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\
 & 0 & 0 & 1 & 0 & 0 & 1 & 0 \\
 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\
 & 0 & 0 & 0 & 1 & 1 & 0 & 0
 \end{array}$$

Exercise: start with $e_2 + e_3$ and iterate a number of times, to get a feeling for the possible evolutions. What do you notice? What is the largest number of steps you can make before the result is the origin vector? Now invert the arrow from 2 to 4 and repeat the exercise. What changes? Can you prove it?

Implementation:

The transition function has type $G \rightarrow (G \rightarrow Bool)$:

$$\begin{array}{l}
 f_2 \ 0 \ g = g == 1 \vee g == 2 \\
 f_2 \ 1 \ g = g == 3 \\
 f_2 \ 2 \ g = g == 4 \vee g == 5 \\
 f_2 \ 3 \ g = g == 6 \\
 f_2 \ 4 \ g = g == 1 \vee g == 6 \\
 f_2 \ 5 \ g = g == 4 \\
 f_2 \ 6 \ g = False
 \end{array}$$

The associated matrix:

$$m_2 \ (G \ g') \ (G \ g) = f_2 \ g \ g'$$

We need a *Num* instance for *Bool* (not a field!):

instance *Num Bool* **where**


```
(+) = (∨)
(*) = (∧)
fromInteger 0 = False
fromInteger 1 = True
negate        = ¬
abs           = id
signum        = id
```

Test:

```
t2 = toL (mul m2 (e 3 + e 4))
```

7.2.4 Stochastic systems

Quite often, we have more information about the transition to possible future states. In particular, we can have *probabilities* of these transitions. For example



One could say that this case is a generalisation of the previous one, in which we can take all probabilities to be equally distributed among the various possibilities. While this is plausible, it is not entirely correct. For example, we have to introduce a transition from state 6 above. The nodes must be sources of *at least* one arrow.

In the case of the non-deterministic example, the “legitimate” inputs were characteristic functions, i.e., the “vector space” was $G \rightarrow \{0, 1\}$ (the scare quotes are necessary because, as discussed, the target is not a field). In the case of stochastic systems, the inputs will be *probability distributions* over G , that is, functions $p : G \rightarrow [0, 1]$ with the property that

$$\text{sum } [p \ g \mid g \leftarrow [0..6]] = 1$$

If we know the current probability distributions over states, then we can compute the next one by using the *total probability formula*, normally expressed as

$$p \ a = \text{sum } [p \ (a \mid b) * p \ b \mid b \leftarrow [0..6]]$$

This formula in itself would be worth a lecture. For one thing, the notation is extremely suspicious. $(a \mid b)$, which is usually read “ a , given b ”, is clearly not of the same type as a or b , so cannot really be an argument to p . For another, the $p \ a$ we are computing with this formula is not the $p \ a$ which must eventually appear in the products on the right hand side. I do not know how this notation came about: it is neither in Bayes’ memoir, nor in Kolmogorov’s monograph.

The conditional probability $p \ (a \mid b)$ gives us the probability that the next state is a , given that the current state is b . But this is exactly the information summarised in the graphical representation. Moreover, it is clear that, at least formally, the total probability formula is identical to a matrix-vector multiplication.

As usual, we write the associated matrix by looking at how the canonical base vectors are transformed. In this case, the canonical base vector $e \ i = \lambda j \rightarrow i$ ‘is’ j is the probability distribution *concentrated* in i :

$$M = \begin{matrix} & \begin{matrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{matrix} \\ \begin{matrix} .4 \\ .6 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{matrix} & \begin{matrix} 0 & 0 & 0 & 0 & .5 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & .7 & 0 & 0 & 1 & 0 \\ 0 & 0 & .3 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & .5 & 0 & 1 \end{matrix} \end{matrix}$$

Exercise: starting from state 0, how many steps do you need to take before the probability is concentrated in state 6? Reverse again the arrow from 2 to 4. What can you say about the long-term behaviour of the system now?

Exercise: Implement the example. You will need to define:

The transition function

$$\begin{aligned} f_3 &:: G \rightarrow (G \rightarrow Double) && \text{-- but we want only } G \rightarrow (G \rightarrow [0, 1]), \text{ the unit interval} \\ f_3 \ g \ g' &= undefined && \text{-- the probability of getting to } g' \text{ from } g \end{aligned}$$

The associated matrix

$$\begin{aligned} m_3 &:: G \rightarrow (G \rightarrow Double) \\ m_3 \ g' \ g &= undefined \end{aligned}$$

Test

$$t_3 = toL \ (mul \ m_3 \ (e \ 2 + e \ 4))$$

7.3 Monadic dynamical systems

All the examples of dynamical systems we have seen in the previous section have a similar structure. They work by taking a state (which is one of the generators) and return a structure of possible future states of type G :

- deterministic: there is exactly one possible future states: we take an element of G and return an element of G . The transition function has the type $f : G \rightarrow G$, the structure of the target is just G itself.

- non-deterministic: there is a set of possible future states, which we have implemented as a characteristic function $G \rightarrow \{0, 1\}$. The transition function has the type $f : G \rightarrow (G \rightarrow \{0, 1\})$. The structure of the target is the *powerset* of G .
- stochastic: given a state, we compute a probability distribution over possible future states. The transition function has the type $f : G \rightarrow (G \rightarrow [0, 1])$, the structure of the target is the probability distributions over G .

Therefore:

- deterministic: $f : G \rightarrow Id\ G$
- non-deterministic: $f : G \rightarrow Powerset\ G$, where $Powerset\ G = G \rightarrow \{0, 1\}$
- stochastic: $f : G \rightarrow Prob\ G$, where $Prob\ G = G \rightarrow [0, 1]$

We have represented the elements of the various structures as vectors. We also had a way of representing, as structures of possible states, those states that were known precisely: these were the canonical base vectors $e\ i$. Due to the nature of matrix-vector multiplication, what we have done was in effect:

$$\begin{aligned}
 & M * v \quad \text{-- } v \text{ represents the current possible states} \\
 & = \{-\ v \text{ is a linear combination of the base vectors } -\} \\
 & \quad M * (v\ 0 * e\ 0 + \dots + v\ n * e\ n) \\
 & = \{-\ \text{homomorphism } -\} \\
 & \quad v\ 0 * (M * e\ 0) + \dots + v\ n * (M * e\ n) \\
 & = \{-\ e\ i \text{ represents the perfectly known current state } i, \text{ therefore } M * e\ i = f\ i\ -\} \\
 & \quad v\ 0 * f\ 0 + \dots + v\ n * f\ n
 \end{aligned}$$

So, we apply f to every state, as if we were starting from precisely that state, obtaining the possible future states starting from that state, and then collect all these “hypothetical” possible future states in some way that takes into account the initial uncertainty (represented by $v\ 0, \dots, v\ n$) and the nature of the uncertainty (the specific $+$ and $*$).

If you examine the types of the operations involved

$$e : G \rightarrow Possible\ G$$

and

$$Possible\ G \rightarrow (G \rightarrow Possible\ G) \rightarrow Possible\ G$$

you see that they are very similar to the monadic operations

$$\begin{aligned}
 & return : g \rightarrow m\ g \\
 & (>>=) : m\ g \rightarrow (g \rightarrow m\ g') \rightarrow m\ g'
 \end{aligned}$$

which suggests that the representation of possible future states might be monadic. Indeed, that is the case.

Since we implemented all these as matrix-vector multiplications, this raises the question: is there a monad underlying matrix-vector multiplication, such that the above are instances of it (obtained by specialising the scalar type S)?

Exercise: write *Monad* instances for *Id*, *Powerset*, *Prob*.

7.4 The monad of linear algebra

The answer is yes, up to a point. Haskell *Monads*, just like *Functors*, require *return* and \gg to be defined for every type. This will not work, in general. Our definition will work for *finite types* only.

```

type S      = Double
data Vector g = V (g → S)
toF (V v)    = v

class      (Bounded a, Enum a, Eq a) ⇒ Finite a where
instance (Bounded a, Enum a, Eq a) ⇒ Finite a where

class FinFunc f where
  func :: (Finite a, Finite b) ⇒ (a → b) → f a → f b

instance FinFunc Vector where
  func = funcV

funcV :: (Finite g, Eq g') ⇒ (g → g') → Vector g → Vector g'
funcV f (V v) = V (λg' → sum [v g | g ← [minBound..maxBound], g' == f g])

class FinMon f where
  embed :: Finite a ⇒ a → f a
  bind  :: (Finite a, Finite b) ⇒ f a → (a → f b) → f b

instance FinMon Vector where
  embed a      = V (λa' → if a == a' then 1 else 0)
  bind (V v) f = V (λg' → sum [toF (f g) g' * v g | g ← [minBound..maxBound]])

```

A better implementation, using associated types, is in file *Vector.lhs* in the repository.

Exercises:

- Prove that the functor laws hold, i.e.

$$\begin{aligned} \text{func id} &= \text{id} \\ \text{func } (g \circ f) &= \text{func } g \circ \text{func } f \end{aligned}$$

- Prove that the monad laws hold, i.e.

$$\begin{aligned} \text{bind } v \text{ return} &= v \\ \text{bind } (\text{return } g) f &= f g \\ \text{bind } (\text{bind } v f) h &= \text{bind } v (\lambda g' \rightarrow \text{bind } (f g') h) \end{aligned}$$

- What properties of *S* have you used to prove these properties? Define a new type class *GoodClass* that accounts for these (and only these) properties.

7.5 Associated code

TODO: import from suitable earlier lecture.

```

instance Num a ⇒ Num (x → a) where
  f + g      = λx → f x + g x
  f - g      = λx → f x - g x
  f * g      = λx → f x * g x
  negate f   = negate ∘ f

```

```

    abs f      = abs ∘ f
    signum f    = signum ∘ f
    fromInteger = const ∘ fromInteger
instance Fractional a ⇒ Fractional (x → a) where
    recip f      = recip ∘ f
    fromRational = const ∘ fromRational
instance Floating a ⇒ Floating (x → a) where
    π           = const π
    exp f       = exp ∘ f
    sin f       = sin ∘ f
    cos f       = cos ∘ f
    f ** g = λx → (f x) ** (g x)
    -- and so on

```


8 Exponentials and Laplace

8.1 The Exponential Function

```
{-# LANGUAGE FlexibleInstances #-}  
{-# LANGUAGE TypeSynonymInstances #-}  
module DSLsofMath.W08 where  
import DSLsofMath.W05  
import DSLsofMath.W06
```

One of the classical analysis textbooks, Rudin's Rudin [1987] starts with a prologue on the exponential function. The first sentence is

This is undoubtedly the most important function in mathematics.

Rudin goes on

It is defined, for every complex number z , by the formula

$$\exp z = \sum (z^n / n!)$$

We have defined the exponential function as the function represented by the power series

```
expx :: Fractional a  $\Rightarrow$  PowerSeries a  
expx = integ expx 1
```

and approximated by

```
expf :: Fractional a  $\Rightarrow$  a  $\rightarrow$  a  
expf = eval 100 expx
```

It is easy to see, using the definition of *integ* that the power series *expx* is, indeed

$$\exp x = [1, 1/2, 1/(2*3), \dots, 1/(2*3*\dots*n), \dots]$$

We can compute the exponential for complex values if we can give an instance of *Fractional* for complex numbers. We could use the datatype *Data.Complex* from the Haskell standard library, but we prefer to roll our own in order to remind the basic operations on complex numbers.

As we saw in week 1, complex values can be represented as pairs of real values.

```
newtype Complex r = C (r, r) deriving (Eq, Show)  
i :: Num a  $\Rightarrow$  Complex a  
i = C (0, 1)
```

Now, we have, for example

```
ex1 :: Fractional a  $\Rightarrow$  Complex a  
ex1 = expf i
```

We have $\text{ex1} = C (0.5403023058681398, 0.8414709848078965)$. Note that

$$\begin{aligned}\cos f\ 1 &= 0.5403023058681398 \\ \sin f\ 1 &= 0.8414709848078965\end{aligned}$$

and therefore $\exp f\ i = C(\cos f\ 1, \sin f\ 1)$. Coincidence?

Instead of evaluating the sum of the terms $a_n * z^n$, let us instead collect the terms in a series:

$$\begin{aligned}\text{terms as } z &= \text{terms1 as } z\ 0 \textbf{ where} \\ \text{terms1 } (Cons\ a\ as)\ z\ n &= Cons\ (a * z^n)\ (\text{terms1 as } z\ (n + 1))\end{aligned}$$

We obtain

$$\begin{aligned}\text{ex2} &:: \text{Fractional } a \Rightarrow \text{PowerSeries } (\text{Complex } a) \\ \text{ex2} &= \text{takePoly } 10\ (\text{terms expx } i)\end{aligned}$$

$$\begin{aligned}\text{ex2} &= [C(1.0, 0.0), C(0.0, 1.0) \\ &\quad , C(-0.5, 0.0), C(0.0, -0.16666666666666666) \\ &\quad , C(4.1666666666666664e-2, 0.0), C(0.0, 8.333333333333333e-3) \\ &\quad , C(-1.388888888888887e-3, 0.0), C(0.0, -1.9841269841269839e-4) \\ &\quad , C(2.4801587301587298e-5, 0.0), C(0.0, 2.7557319223985884e-6) \\ &\quad]\end{aligned}$$

We can see that the real part of this series is the same as

$$\text{ex2R} = \text{takePoly } 10\ (\text{terms cosx } 1)$$

and the imaginary part is the same as

$$\text{ex2I} = \text{takePoly } 10\ (\text{terms sinx } 1)$$

(within approx 20 decimals). But the terms of a series evaluated at 1 are the coefficients of the series. Therefore, the coefficients of $\cos x$ are

$$[1, 0, -1 / 2!, 0, 1 / 4!, 0, -1 / 6!, \dots]$$

i.e.

$$\begin{aligned}a(2 * n) &= (-1)^n / (2 * n)! \\ a(2 * n + 1) &= 0\end{aligned}$$

and the terms of $\sin x$ are

$$[0, 1, 0, -1 / 3!, 0, 1 / 5!, 0, -1 / 7!, \dots]$$

i.e.

$$\begin{aligned}a(2 * n) &= 0 \\ a(2 * n + 1) &= (-1)^n / (2 * n + 1)!\end{aligned}$$

This can be proven from the definitions of $\cos x$ and $\sin x$. From this we obtain *Euler's formula*:

$$\exp(i * x) = \cos x + i * \sin x$$

One thing which comes out of Euler's formula is the fact that the exponential is a *periodic function*. A function $f : A \rightarrow B$ is said to be periodic if there exists $T \in A$ such that

$$f\ a = f\ (a + T) \quad -- \forall a \in A$$

(therefore, for this definition to make sense, we need addition on A ; in fact we normally assume at least group structure, i.e., addition and subtraction).

Since \sin and \cos are periodic, with period $2 * \pi$, we have, using the standard notation $a + i * b$ for $C\ (a, b)$

$$\begin{aligned} & e^{\wedge}(a + i * b + i * 2 * \pi) \\ = & e^{\wedge}(a + i * (b + 2 * \pi)) \\ = & e^{\wedge}a * e^{\wedge}(i * (b + 2 * \pi)) \\ = & e^{\wedge}a * (\cos(b + 2 * \pi) + i * \sin(b + 2 * \pi)) \\ = & e^{\wedge}a * (\cos b + i * \sin b) \\ = & e^{\wedge}a * e^{\wedge}(i * b) \\ = & e^{\wedge}(a + i * b) \end{aligned}$$

8.1.1 Exponential function: Associated code

TODO: Perhaps import from W01

```
instance Num r => Num (Complex r) where
  (+) = addC
  (*) = mulC
  fromInteger n = C (fromInteger n, 0)
  -- abs = absC -- requires Floating r as context
addC :: Num r => Complex r -> Complex r -> Complex r
addC (C (a, b)) (C (x, y)) = C ((a + x), (b + y))
mulC :: Num r => Complex r -> Complex r -> Complex r
mulC (C (ar, ai)) (C (br, bi)) = C (ar * br - ai * bi, ar * bi + ai * br)
modulusSquaredC :: Num r => Complex r -> r
modulusSquaredC (C (x, y)) = x^2 + y^2
absC :: Floating r => Complex r -> Complex r
absC c = C (sqrt(modulusSquaredC c), 0)
scale :: Num r => r -> Complex r -> Complex r
scale a (C (x, y)) = C (a * x, a * y)
conj :: Num r => Complex r -> Complex r
conj (C (x, x')) = C (x, -x')
instance Fractional r => Fractional (Complex r) where
  (/) = divC
  fromRational r = C (fromRational r, 0)
divC :: Fractional a => Complex a -> Complex a -> Complex a
divC x y = scale (1 / modSq) (x * conj y)
where modSq = modulusSquaredC y
```

8.2 The Laplace transform

This material was inspired by Quinn and Rai [2008], which is highly recommended reading.

Consider the differential equation

$$f'' x - 3 * f' x + 2 * f x = \exp(3 * x), f 0 = 1, f' 0 = 0$$

We can solve such equations with the machinery of power series:

$$\begin{aligned} fs &= \text{integ } fs' \ 1 \\ \textbf{where } fs' &= \text{integ } (\exp(3 * x) + 3 * fs' - 2 * fs) \ 0 \end{aligned}$$

We have done this by “zooming in” on the function f and representing it by a power series, $f x = \sum a_n * x^n$. This allows us to reduce the problem of finding a function $f : \mathbb{R} \rightarrow \mathbb{R}$ to that of finding a function $a : \mathbb{N} \rightarrow \mathbb{R}$ (or finding a list of sufficiently many a -values for a good approximation).

Still, recursive equations are not always easy to solve (especially without a computer), so it’s worth looking for alternatives.

We have gone from

$$a : \mathbb{N} \rightarrow \mathbb{R} \quad \text{to} \quad f : \mathbb{R} \rightarrow \mathbb{R}$$

via

$$\sum a_n * x^n$$

In “zooming out”, we would like to go one step further

$$a : \mathbb{N} \rightarrow \mathbb{R} \quad \text{to} \quad f : \mathbb{R} \rightarrow \mathbb{R} \quad \text{to} \quad ?$$

via

$$\sum a_n * x^n \quad ?$$

At least the second part is “easy”. The analogue of series for a continuous function is integral

$$\sum a_n * x^n \quad \int_0^\infty (f t) * x^t dt$$

We note that, for the integral $\int_0^\infty (f t) * x^t dt$ to converge for a larger class of functions (say, bounded functions), we have to limit ourselves to $|x| < 1$. Both this condition and the integral make sense for $x \in \mathbb{C}$, so we could take

$$a : \mathbb{N} \rightarrow \mathbb{R} \quad \text{to} \quad f : \mathbb{R} \rightarrow \mathbb{R} \quad \text{to} \quad F : \{z \mid |z| < 1\} \rightarrow \mathbb{C}$$

but let us stick to \mathbb{R} for now.

Writing, somewhat optimistically

$$\mathcal{L} f x = \int_0^\infty (f t) * x^t dt$$

we can ask ourselves what $\mathcal{L} f'$ looks like. After all, we want to solve *differential* equations by “zooming out”. We have

$$\mathcal{L} f' x = \int_0^\infty (f' t) * x^\wedge t \, dt$$

Remember that $D(f * g) = Df * g + f * Dg$, therefore

$$\begin{aligned} & \mathcal{L} f' x \\ &= \{g t = x^\wedge t; g' t = \log x * x^\wedge t\} \\ & \int_0^\infty (D(f t * x^\wedge t)) - f t * \log x * x^\wedge t \, dt \\ &= \\ & \int_0^\infty (D(f t * x^\wedge t)) \, dt - \int_0^\infty f t * \log x * x^\wedge t \, dt \\ &= \\ & \lim_{t \rightarrow \infty} (f t * x^\wedge t) - (f 0 * x^\wedge 0) - \log x * \int_0^\infty f t * x^\wedge t \, dt \\ &= \\ & -f 0 - \log x * \int_0^\infty f t * x^\wedge t \, dt \\ &= \\ & -f 0 - \log x * \mathcal{L} f x \end{aligned}$$

The factor $\log x$ is somewhat awkward. Let us therefore return to the definition of \mathcal{L} and operate a change of variables:

$$\begin{aligned} & \mathcal{L} f x = \int_0^\infty (f t) * x^\wedge t \, dt \\ & \Leftrightarrow \{x = \exp(\log x)\} \\ & \mathcal{L} f x = \int_0^\infty (f t) * (\exp(\log x))^\wedge t \, dt \\ & \Leftrightarrow \{(a^\wedge b)^\wedge c = a^\wedge(b * c)\} \\ & \mathcal{L} f x = \int_0^\infty (f t) * \exp(\log x * t) \, dt \end{aligned}$$

Since $\log x < 0$ for $|x| < 1$, we make the substitution $-s = \log x$. The condition $|x| < 1$ becomes $s > 0$ (or, in \mathbb{C} , $\text{real } s > 0$), and we have

$$\mathcal{L} f s = \int_0^\infty (f t) * \exp(-s * t) \, dt$$

This is the definition of the Laplace transform of the function f . Going back to the problem of computing $\mathcal{L} f'$, we now have

$$\begin{aligned} & \mathcal{L} f' s \\ &= \{- \text{The computation above with } s = -\log x. -\} \\ & -f 0 + s * \mathcal{L} f s \end{aligned}$$

We have obtained

$$\mathcal{L} f' s = s * \mathcal{L} f s - f 0$$

From this, we can deduce

$$\begin{aligned} & \mathcal{L} f'' s \\ &= \\ & s * \mathcal{L} f' s - f' 0 \\ &= \\ & s * (s * \mathcal{L} f s - f 0) - f' 0 \end{aligned}$$

$$= s^2 * \mathcal{L} f s - s * f 0 - f' 0$$

Exercise: what is the general formula for $\mathcal{L} f^{(k)} s$?

Returning to our differential equation, we have

$$\begin{aligned} f'' x - 3 * f' x + 2 * f x &= \exp(3 * x), f 0 = 1, f' 0 = 0 \\ \Leftrightarrow \{- \text{ point-free form } -\} \\ f'' - 3 * f' + 2 * f &= \exp \circ (3*), f 0 = 1, f' 0 = 0 \\ \Rightarrow \{- \text{ applying } \mathcal{L} \text{ to both sides } -\} \\ \mathcal{L} (f'' - 3 * f' + 2 * f) &= \mathcal{L} (\exp \circ (3*)), f 0 = 1, f' 0 = 0 \quad \text{-- Eq. (1)} \end{aligned}$$

Remark: Note that this is a necessary condition, but not a sufficient one. The Laplace transform is not injective. For one thing, it does not take into account the behaviour of f for negative arguments. Because of this, we often assume that the domain of definition for functions to which we apply the Laplace transform is $\mathbb{R}_{\geq 0}$. For another, it is known that changing the values of f for a countable number of its arguments does not change the value of the integral.

For the definition of \mathcal{L} and the linearity of the integral, we have that, for any f and g for which the transformation is defined, and for any constants α and β

$$\mathcal{L} (\alpha * f + \beta * g) = \alpha * \mathcal{L} f + \beta * \mathcal{L} g$$

Note that this is an equality between functions. (Comparing to last week we can also see f and g as vectors and \mathcal{L} as a linear transformation.)

Applying this to the left-hand side of (1), we have for any s

$$\begin{aligned} &\mathcal{L} (f'' - 3 * f' + 2 * f) s \\ = \{- \mathcal{L} \text{ is linear } -\} \\ &\mathcal{L} f'' s - 3 * \mathcal{L} f' s + 2 * \mathcal{L} f s \\ = \{- \text{ re-writing } \mathcal{L} f'' \text{ and } \mathcal{L} f' \text{ in terms of } \mathcal{L} f -\} \\ &s^2 * \mathcal{L} f s - s * f 0 - f' 0 - 3 * (s * \mathcal{L} f s - f 0) + 2 * \mathcal{L} f s \\ = \{- f 0 = 1, f' 0 = 0 -\} \\ &(s^2 - 3 * s + 2) * \mathcal{L} f s - s + 3 \end{aligned}$$

For the right-hand side, we apply the definition:

$$\begin{aligned} &\mathcal{L} (\exp \circ (3*)) s \\ = \{- \text{ Def. of } \mathcal{L} -\} \\ &\int_0^\infty \exp(3 * t) * \exp(-s * t) dt \\ = \\ &\int_0^\infty \exp((3 - s) * t) dt \\ = \\ &\lim_{t \rightarrow \infty} \frac{\exp((3-s)*t)}{3-s} - \frac{\exp((3-s)*0)}{3-s} \\ = \{- \text{ for } s > 3 -\} \\ &\frac{1}{s-3} \end{aligned}$$

Therefore, we have, writing F for $\mathcal{L} f$

$$(s^2 - 3s + 2) * F s - s + 3 = \frac{1}{s-3}$$

and therefore

$$\begin{aligned} & F s \\ &= \{- \text{Solve for } F s -\} \\ & \quad \frac{\frac{1}{s-3} + s - 3}{s^2 - 3s + 2} \\ &= \{- s^2 - 3s + 2 = (s - 1) * (s - 2) -\} \\ & \quad \frac{10 - 6s + s^2}{(s-1)*(s-2)*(s-3)} \end{aligned}$$

We now have the problem of “recovering” the function f from its Laplace transform. The standard approach is to use the linearity of \mathcal{L} to write F as a sum of functions with known inverse transforms. We know one such function:

$$\exp(\alpha * t) \{- \text{is the inverse Laplace transform of } -\} 1 / (s - \alpha)$$

In fact, in our case, this is all we need.

The idea is to write $F s$ as a sum of three fractions with denominators $s - 1$, $s - 2$, and $s - 3$ respectively, i.e., to find A , B , and C such that

$$\begin{aligned} & A / (s - 1) + B / (s - 2) + C / (s - 3) = (10 - 6s + s^2) / ((s - 1) * (s - 2) * (s - 3)) \\ & \Rightarrow \\ & A * (s - 2) * (s - 3) + B * (s - 1) * (s - 3) + C * (s - 1) * (s - 2) = 10 - 6s + s^2 \quad -- (2) \end{aligned}$$

We need this equality (2) to hold for values $s > 3$. A *sufficient* condition for this is for (2) to hold for *all* s . A *necessary* condition for this is for (2) to hold for the specific values 1, 2, and 3.

$$\begin{aligned} \text{For } s = 1 : A * (-1) * (-2) &= 10 - 6 + 1 \Rightarrow A = 2.5 \\ \text{For } s = 2 : B * 1 * (-1) &= 10 - 12 + 4 \Rightarrow B = -2 \\ \text{For } s = 3 : C * 2 * 1 &= 10 - 18 + 9 \Rightarrow C = 0.5 \end{aligned}$$

It is now easy to check that, with these values, (2) does indeed hold, and therefore that we have

$$F s = 2.5 * (1 / (s - 1)) - 2 * (1 / (s - 2)) + 0.5 * (1 / (s - 3))$$

The inverse transform is now easy:

$$f t = 2.5 * \exp t - 2 * \exp(2 * t) + 0.5 * \exp(3 * t)$$

Our mix of necessary and sufficient conditions makes it necessary to check that we have, indeed, a solution for the differential equation. The verification is in this case trivial.

9 End

TODO: sum up and close

Acknowledgements

This work was partially supported by the projects GRACeFUL (grant agreement No 640954) and CoeGSS (grant agreement No 676547), which have received funding from the European Union’s Horizon 2020 research and innovation programme.

References

- R. A. Adams and C. Essex. *Calculus: a complete course*. Pearson Canada, 7th edition, 2010.
- N. Botta, P. Jansson, and C. Ionescu. Contributions to a computational theory of policy advice and avoidability. *Journal of Functional Programming*, 27:1–52, 2017a. ISSN 0956-7968. doi: 10.1017/S0956796817000156.
- N. Botta, P. Jansson, C. Ionescu, D. R. Christiansen, and E. Brady. Sequential decision problems, dependent types and generic solutions. *Logical Methods in Computer Science*, 13(1), 2017b. doi: 10.23638/LMCS-13(1:7)2017. URL [https://doi.org/10.23638/LMCS-13\(1:7\)2017](https://doi.org/10.23638/LMCS-13(1:7)2017).
- R. Boute. The decibel done right: a matter of engineering the math. *Antennas and Propagation Magazine, IEEE*, 51(6):177–184, 2009. doi: 10.1109/MAP.2009.5433137.
- C. H. Edwards, D. E. Penney, and D. Calvis. *Elementary Differential Equations*. Pearson Prentice Hall Upper Saddle River, NJ, 6h edition, 2008.
- D. Gries and F. B. Schneider. *A logical approach to discrete math*. Springer, 1993. doi: 10.1007/978-1-4757-3837-7.
- D. Gries and F. B. Schneider. Teaching math more effectively, through calculational proofs. *American Mathematical Monthly*, pages 691–697, 1995. doi: 10.2307/2974638.
- C. Ionescu and P. Jansson. Dependently-typed programming in scientific computing: Examples from economic modelling. In R. Hinze, editor, *24th Symposium on Implementation and Application of Functional Languages (IFL 2012)*, volume 8241 of *LNCS*, pages 140–156. Springer-Verlag, 2013a. doi: 10.1007/978-3-642-41582-1_9.
- C. Ionescu and P. Jansson. Dependently-typed programming in scientific computing. In *Implementation and Application of Functional Languages*, pages 140–156. Springer Berlin Heidelberg, 2013b. doi: 10.1007/978-3-642-41582-1_9.
- C. Ionescu and P. Jansson. Domain-specific languages of mathematics: Presenting mathematical analysis using functional programming. In J. Jeuring and J. McCarthy, editors, *Proceedings of the 4th and 5th International Workshop on Trends in Functional Programming in Education, Sophia-Antipolis, France and University of Maryland College Park, USA, 2nd June 2015 and 7th June 2016*, volume 230 of *Electronic Proceedings in Theoretical Computer Science*, pages 1–15. Open Publishing Association, 2016. doi: 10.4204/EPTCS.230.1.
- C. Jaeger, P. Jansson, S. van der Leeuw, M. Resch, and J. D. Tabara. GSS: Towards a research program for Global Systems Science. <http://blog.global-systems-science.eu/?p=1512>, 2013. ISBN 978.3.94.1663-12-1. Conference Version, prepared for the Second Open Global Systems Science Conference June 10-12, 2013, Brussels.

- R. Kraft. Functions and parameterizations as objects to think with. In *Maple Summer Workshop, July 2004, Wilfrid Laurier University, Waterloo, Ontario, Canada*, 2004.
- E. Landau. *Einführung in die Differentialrechnung und Integralrechnung*. Noordhoff, 1934.
- E. Landau. *Differential and Integral Calculus*. AMS/Chelsea Publication Series. AMS Chelsea Pub., 2001.
- D. Lincke, P. Jansson, M. Zalewski, and C. Ionescu. Generic libraries in C++ with concepts from high-level domain descriptions in Haskell: A DSL for computational vulnerability assessment. In *IFIP Working Conf. on Domain Specific Languages*, volume 5658/2009 of *LNCS*, pages 236–261, 2009. doi: 10.1007/978-3-642-03034-5_12.
- S. Mac Lane. *Mathematics: Form and function*. Springer New York, 1986.
- S. Marlow (ed.). The Haskell 2010 report, 2010. <http://www.haskell.org/onlinereport/haskell2010/>.
- M. D. McIlroy. Functional pearl: Power series, power serious. *J. of Functional Programming*, 9: 323–335, 1999. doi: 10.1017/S0956796899003299.
- T. J. Quinn and S. Rai. Discovering the laplace transform in undergraduate differential equations. *PRIMUS*, 18(4):309–324, 2008.
- J. J. Rotman. *A first course in abstract algebra*. Pearson Prentice Hall, 2006.
- W. Rudin. *Principles of mathematical analysis*, volume 3. McGraw-Hill New York, 1964.
- W. Rudin. *Real and complex analysis*. Tata McGraw-Hill Education, 1987.
- J. Tolvanen. Industrial experiences on using DSLs in embedded software development. In *Proceedings of Embedded Software Engineering Kongress (Tagungsband), December 2011*, 2011. doi: 10.1.1.700.1924.
- C. Wells. Communicating mathematics: Useful ideas from computer science. *American Mathematical Monthly*, pages 397–408, 1995. doi: 10.2307/2975030.