

Domain Specific Languages of Mathematics: Lecture Notes

Patrik Jansson

Cezar Ionescu

March 5, 2017

Abstract

These notes aim to cover the lectures and exercises of the recently introduced course “Domain-Specific Languages of Mathematics” (at Chalmers and University of Gothenburg). The course was developed in response to difficulties faced by third-year computer science students in learning and applying classical mathematics (mainly real and complex analysis). The main idea is to encourage the students to approach mathematical domains from a functional programming perspective: to identify the main functions and types involved and, when necessary, to introduce new abstractions; to give calculational proofs; to pay attention to the syntax of the mathematical expressions; and, finally, to organize the resulting functions and types in domain-specific languages.

1 Week 1

This lecture is partly based on the paper [Ionescu and Jansson, 2016] from the International Workshop on Trends in Functional Programming in Education 2015. We will implement certain concepts in the functional programming language Haskell and the code for this lecture is placed in a module called *DSLsofMath.W01* that starts here:

module *DSLsofMath.W01* **where**

1.1 A case study: complex numbers

We will start by an analytic reading of the introduction of complex numbers in Adams and Essex [2010]. We choose a simple domain to allow the reader to concentrate on the essential elements of our approach without the distraction of potentially unfamiliar mathematical concepts. For this section, we bracket our previous knowledge and approach the text as we would a completely new domain, even if that leads to a somewhat exaggerated attention to detail.

Adams and Essex introduce complex numbers in Appendix 1. The section *Definition of Complex Numbers* begins with:

We begin by defining the symbol i , called **the imaginary unit**, to have the property

$$i^2 = -1$$

Thus, we could also call i the square root of -1 and denote it $\sqrt{-1}$. Of course, i is not a real number; no real number has a negative square.

At this stage, it is not clear what the type of i is meant to be, we only know that i is not a real number. Moreover, we do not know what operations are possible on i , only that i^2 is another name for -1 (but it is not obvious that, say $i * i$ is related in any way with i^2 , since the operations of multiplication and squaring have only been introduced so far for numerical types such as \mathbb{N} or \mathbb{R} , and not for symbols).

For the moment, we introduce a type for the value i , and, since we know nothing about other values, we make i the only member of this type:

```
data ImagUnits = I
i :: ImagUnits
i = I
```

We use a capital I in the **data** declaration because a lowercase constructor name would cause a syntax error in Haskell.

Next, we have the following definition:

Definition: A **complex number** is an expression of the form

$$a + bi \quad \text{or} \quad a + ib,$$

where a and b are real numbers, and i is the imaginary unit.

This definition clearly points to the introduction of a syntax (notice the keyword “form”). This is underlined by the presentation of *two* forms, which can suggest that the operation of juxtaposing i (multiplication?) is not commutative.

A profitable way of dealing with such concrete syntax in functional programming is to introduce an abstract representation of it in the form of a datatype:

```
data ComplexA = CPlus1  $\mathbb{R}$   $\mathbb{R}$  ImagUnits
              | CPlus2  $\mathbb{R}$  ImagUnits  $\mathbb{R}$ 
```

We can give the translation from the abstract syntax to the concrete syntax as a function *showCA*:

```
showCA :: ComplexA → String
showCA (CPlus1 x y i) = show x ++ " + " ++ show y ++ "i"
showCA (CPlus2 x i y) = show x ++ " + " ++ "i" ++ show y
```

Notice that the type \mathbb{R} is not implemented yet and it is not really even exactly implementable but we want to focus on complex numbers so we will approximate \mathbb{R} by double precision floating point numbers for now.

```
type  $\mathbb{R}$  = Double
```

The text continues with examples:

For example, $3 + 2i$, $\frac{7}{2} - \frac{2}{3}i$, $i\pi = 0 + i\pi$, and $-3 = -3 + 0i$ are all complex numbers. The last of these examples shows that every real number can be regarded as a complex number.

The second example is somewhat problematic: it does not seem to be of the form $a + bi$. Given that the last two examples seem to introduce shorthand for various complex numbers, let us assume that this one does as well, and that $a - bi$ can be understood as an abbreviation of $a + (-b) i$.

With this provision, in our notation the examples are written as:

```
testC1 :: [ComplexA]
testC1 = [ CPlus1 3 2 I, CPlus1 (7 / 2) (-2 / 3) I
          , CPlus2 0 I π, CPlus1 (-3) 0 I
          ]
testS1 = map showCA testC1
```

We interpret the sentence “The last of these examples ...” to mean that there is an embedding of the real numbers in *ComplexA*, which we introduce explicitly:

```
toComplex :: ℝ → ComplexA
toComplex x = CPlus1 x 0 i
```

Again, at this stage there are many open questions. For example, we can assume that i 1 stands for the complex number $CPlus_2$ 0 i 1, but what about i by itself? If juxtaposition is meant to denote some sort of multiplication, then perhaps 1 can be considered as a unit, in which case we would have that i abbreviates i 1 and therefore $CPlus_2$ 0 i 1. But what about, say, 2 i ? Abbreviations with i have only been introduced for the ib form, and not for the bi one!

The text then continues with a parenthetical remark which helps us dispel these doubts:

(We will normally use $a + bi$ unless b is a complicated expression, in which case we will write $a + ib$ instead. Either form is acceptable.)

This remark suggests strongly that the two syntactic forms are meant to denote the same elements, since otherwise it would be strange to say “either form is acceptable”. After all, they are acceptable by definition.

Given that $a + ib$ is only “syntactic sugar” for $a + bi$, we can simplify our representation for the abstract syntax, eliminating one of the constructors:

```
data ComplexB = CPlusB ℝ ℝ ImagUnits
```

In fact, since it doesn’t look as though the type *ImagUnits* will receive more elements, we can dispense with it altogether:

```
data ComplexC = CPlusC ℝ ℝ
```

(The renaming of the constructor to *CPlusC* serves as a guard against the case we have suppressed potentially semantically relevant syntax.)

We read further:

It is often convenient to represent a complex number by a single letter; w and z are frequently used for this purpose. If a , b , x , and y are real numbers, and $w = a + bi$ and $z = x + yi$, then we can refer to the complex numbers w and z . Note that $w = z$ if and only if $a = x$ and $b = y$.

First, let us notice that we are given an important semantic information: *CPlusC* is not just syntactically injective (as all constructors are), but also semantically. The equality on complex numbers is what we would obtain in Haskell by using **deriving Eq**.

This shows that complex numbers are, in fact, isomorphic with pairs of real numbers, a point which we can make explicit by re-formulating the definition in terms of a **newtype**:

```
type ComplexD = ComplexSem ℝ
newtype ComplexSem r = CS (r, r) deriving Eq
```

The point of the somewhat confusing discussion of using “letters” to stand for complex numbers is to introduce a substitute for *pattern matching*, as in the following definition:

Definition: If $z = x + yi$ is a complex number (where x and y are real), we call x the **real part** of z and denote it $Re\ (z)$. We call y the **imaginary part** of z and denote it $Im\ (z)$:

$$\begin{aligned} Re\ (z) &= Re\ (x + yi) = x \\ Im\ (z) &= Im\ (x + yi) = y \end{aligned}$$

This is rather similar to Haskell’s *as-patterns*:

```
re :: ComplexSem r → r
re z@(CS (x, y)) = x
im :: ComplexSem r → r
im z@(CS (x, y)) = y
```

a potential source of confusion being that the symbol z introduced by the as-pattern is not actually used on the right-hand side of the equations.

The use of as-patterns such as “ $z = x + yi$ ” is repeated throughout the text, for example in the definition of the algebraic operations on complex numbers:

The sum and difference of complex numbers

If $w = a + bi$ and $z = x + yi$, where a , b , x , and y are real numbers, then

$$\begin{aligned} w + z &= (a + x) + (b + y) i \\ w - z &= (a - x) + (b - y) i \end{aligned}$$

With the introduction of algebraic operations, the language of complex numbers becomes much richer. We can describe these operations in a *shallow embedding* in terms of the concrete datatype *ComplexSem*, for example:

```
(+.) :: Num r ⇒ ComplexSem r → ComplexSem r → ComplexSem r
(CS (a, b)) +. (CS (x, y)) = CS ((a + x), (b + y))
```

or we can build a datatype of “syntactic” complex numbers from the algebraic operations to arrive at a *deep embedding* as seen in the next section.

Exercises:

- implement $(*.)$ for *ComplexSem*

1.2 A syntax for arithmetical expressions

So far we have tried to find a datatype to represent the intended *semantics* of complex numbers. That approach is called “shallow embedding”. Now we turn to the *syntax* instead (“deep embedding”).

We want a datatype *ComplexE* for the abstract syntax tree of expressions. The syntactic expressions can later be evaluated to semantic values:

$$evalE :: ComplexE \rightarrow ComplexD$$

The datatype *ComplexE* should collect ways of building syntactic expression representing complex numbers and we have so far seen the symbol *i*, an embedding from \mathbb{R} , plus and times. We make these four *constructors* in one recursive datatype as follows:

```
data ComplexE = ImagUnit
               | ToComplex  $\mathbb{R}$ 
               | Plus   ComplexE ComplexE
               | Times  ComplexE ComplexE
deriving (Eq, Show)
```

And we can write the evaluator by induction over the syntax tree:

$$\begin{aligned} evalE \text{ ImagUnit} &= CS \ (0, 1) \\ evalE \ (ToComplex \ r) &= CS \ (r, 0) \\ evalE \ (Plus \ c_1 \ c_2) &= evalE \ c_1 +. evalE \ c_2 \\ evalE \ (Times \ c_1 \ c_2) &= evalE \ c_1 *. evalE \ c_2 \end{aligned}$$

We also define a function to embed a semantic complex number in the syntax:

```
fromCS :: ComplexD → ComplexE
fromCS (CS (x, y)) = Plus (ToComplex x) (Times (ToComplex y) ImagUnit)
testE1 = Plus (ToComplex 3) (Times (ToComplex 2) ImagUnit)
testE2 = Times ImagUnit ImagUnit
```

There are certain laws we would like to hold for operations on complex numbers. The simplest is perhaps $i^2 = -1$ from the start of the lecture,

$$\begin{aligned} propImagUnit &:: Bool \\ propImagUnit &= Times \ ImagUnit \ ImagUnit == ToComplex \ (-1) \\ (===) &:: ComplexE \rightarrow ComplexE \rightarrow Bool \\ z === w &= evalE \ z == evalE \ w \end{aligned}$$

and that *fromCS* is an embedding:

$$\begin{aligned} propFromCS &:: ComplexD \rightarrow Bool \\ propFromCS \ c &= evalE \ (fromCS \ c) == c \end{aligned}$$

but we also have that *Plus* and *Times* should be associative and commutative and *Times* should distribute over *Plus*:

$$\begin{aligned} propAssocPlus \ x \ y \ z &= Plus \ (Plus \ x \ y) \ z == Plus \ x \ (Plus \ y \ z) \\ propAssocTimes \ x \ y \ z &= Times \ (Times \ x \ y) \ z == Times \ x \ (Times \ y \ z) \\ propDistTimesPlus \ x \ y \ z &= Times \ x \ (Plus \ y \ z) == Plus \ (Times \ x \ y) \ (Times \ x \ z) \end{aligned}$$

These three laws actually fail, but not because of the implementation of *evalE*. We will get back to that later but let us first generalise the properties a bit by making the operator a parameter:

```
propAssocA :: Eq a => (a -> a -> a) -> a -> a -> a -> Bool
propAssocA (+?) x y z = (x +? y) +? z == x +? (y +? z)
```

Note that *propAssocA* is a higher order function: it takes a function (a binary operator) as its first parameter. It is also polymorphic: it works for many different types *a* (all types which have an *==* operator).

Thus we can specialise it to *Plus*, *Times* and other binary operators. In Haskell there is a type class *Num* for different types of “numbers” (with operations (+), (*), etc.). We can try out *propAssocA* for a few of them.

```
propAssocAInt = propAssocA (+) :: Int -> Int -> Int -> Bool
propAssocADouble = propAssocA (+) :: Double -> Double -> Double -> Bool
```

The first is fine, but the second fails due to rounding errors. QuickCheck can be used to find small examples - I like this one best:

```
notAssocEvidence :: (Double, Double, Double, Bool)
notAssocEvidence = (lhs, rhs, lhs - rhs, lhs == rhs)
  where lhs = (1 + 1) + 1 / 3
        rhs = 1 + (1 + 1 / 3)
```

For completeness: this is the answer:

```
(2.3333333333333335      -- Notice the five at the end
, 2.3333333333333333,    -- which is not present here.
, 4.440892098500626e-16  -- The difference
, False)
```

This is actually the underlying reason why some of the laws failed for complex numbers: the approximative nature of *Double*. But to be sure there is no other bug hiding we need to make one more version of the complex number type: parameterise on the underlying type for \mathbb{R} . At the same time we generalise *ToComplex* to *FromCartesian*:

```
data ComplexSyn r = FromCartesian r r
                  | ComplexSyn r :+: ComplexSyn r
                  | ComplexSyn r *: ComplexSyn r

toComplexSyn :: Num a => a -> ComplexSyn a
toComplexSyn x = FromCartesian x (fromInteger 0)

evalCSyn :: Num r => ComplexSyn r -> ComplexSem r
evalCSyn (FromCartesian x y) = CS (x, y)
evalCSyn (l :+: r) = evalCSyn l +. evalCSyn r
evalCSyn (l *: r) = evalCSyn l *. evalCSyn r

instance Num a => Num (ComplexSyn a) where
  (+) = (:+:)
  (*) = (:*)
  fromInteger = fromIntegerCS
  -- TODO: add a few more operations (hint: extend ComplexSyn as well)
  -- TODO: also extend eval

fromIntegerCS :: Num r => Integer -> ComplexSyn r
fromIntegerCS = toComplexSyn o fromInteger
```

1.3 TODO[PaJa]: Textify

Here are some notes about things scribbled on the blackboard during the first two lectures. At some point this should be made into text for the lecture notes.

1.3.1 Pitfalls with traditional mathematical notation

A function or the value at a point? Mathematical texts often talk about “the function $f(x)$ ” when “the function f ” would be more clear. Otherwise there is a clear risk of confusion between $f(x)$ as a function and $f(x)$ as the value you get from applying the function f to the value bound to the name x .

Scoping Scoping rules for the integral sign:

$$\begin{aligned} f(x) &= x^2 \\ g(x) &= \int_x^{2x} f(x)dx &= \int_x^{2x} f(y)dy \end{aligned}$$

The variable x bound on the left is independent of the variable x “bound under the integral sign”.

From syntax to semantics and back We have seen evaluation functions from abstract syntax to semantics ($eval :: Syn \rightarrow Sem$). Often a partial inverse is also available: $embed :: Sem \rightarrow Syn$. For our complex numbers we have TODO: fill in a function from $ComplexSem\ r \rightarrow ComplexSyn\ r$.

The embedding should satisfy a round-trip property: $eval\ (embed\ s) == s$ for all s . Exercise: What about the opposite direction? When is $embed\ (eval\ e) == e$?

We can also state and check properties relating the semantic and the syntactic operations:

$a + b = eval\ (Plus\ (embed\ a)\ (embed\ b))$ for all a and b .

Variable names as type hints In mathematical texts there are often conventions about the names used for variables of certain types. Typical examples include i, j, k for natural numbers or integers, x, y for real numbers and z, w for complex numbers.

The absence of explicit types in mathematical texts can sometimes lead to confusing formulations. For example, a standard text on differential equations by Edwards, Penney and Calvis Edwards et al. [2008] contains at page 266 the following remark:

The differentiation operator D can be viewed as a transformation which, when applied to the function $f(t)$, yields the new function $D\{f(t)\} = f'(t)$. The Laplace transformation \mathcal{L} involves the operation of integration and yields the new function $\mathcal{L}\{f(t)\} = F(s)$ of a new independent variable s .

This is meant to introduce a distinction between “operators”, such as differentiation, which take functions to functions of the same type, and “transforms”, such as the Laplace transform, which take functions to functions of a new type. To the logician or the computer scientist, the way of phrasing this difference in the quoted text sounds strange: surely the *name* of the independent variable does not matter: the Laplace transformation could very well return a function of the “old” variable t . We can understand that the name of the variable is used to carry semantic meaning about its type (this is also common in functional programming, for example with the conventional use of *as* to denote a list of *as*). Moreover, by using this (implicit!) convention, it is easier to deal with cases such as that of the Hartley transform (a close relative of the Fourier transform), which

does not change the type of the input function, but rather the *interpretation* of that type. We prefer to always give explicit typings rather than relying on syntactical conventions, and to use type synonyms for the case in which we have different interpretations of the same type. In the example of the Laplace transformation, this leads to

```

type T = Real
type S = ℂ
ℒ : (T → ℂ) → (S → ℂ)

```

1.3.2 Other

Lifting operations to a parameterised type When we define addition on complex numbers (represented as pairs of real and imaginary components) we can do that for any underlying type r which supports addition.

```

type CS = ComplexSem -- for shorter type expressions below
liftPlus :: (r → r → r) →
            (CS r → CS r → CS r)
liftPlus (+) (CS (x, y)) (CS (x', y')) = CS (x + x', y + y')

```

Note that *liftPlus* takes (+) as its first parameter and uses it twice on the RHS.

Laws TODO: Associative, Commutative, Distributive, ...

TODO[PaJa]: move earlier Table of examples of notation and abstract syntax for some complex numbers:

Mathematics	Haskell
$3 + 2i$	<i>CPlus</i> ₁ 3 2 <i>i</i>
$\frac{7}{2} - \frac{2}{3}i = \frac{7}{2} + \frac{-2}{3}i$	<i>CPlus</i> ₁ (7 / 2) (-2 / 3) <i>i</i>
$i\pi = 0 + i\pi$	<i>CPlus</i> ₂ 0 <i>i</i> π
$-3 = -3 + 0i$	<i>CPlus</i> ₁ (-3) 0 <i>i</i>

1.4 Questions and answers from the exercise sessions week 1

1.4.1 Function composition

The infix operator `.` in Haskell is an implementation of the mathematical operation of function composition.

$$f \circ g = \lambda x \rightarrow f (g x)$$

The period is an ASCII approximation of the composition symbol \circ typically used in mathematics. (The symbol \circ is encoded as U+2218 and called RING OPERATOR in Unicode, ∘ in HTML, `\circ` in `TEX`, etc.)

The type is perhaps best illustrated by a diagram with types as nodes and functions (arrows) as directed edges:

In Haskell we get the following type:

$$(\circ) :: (b \rightarrow c) \rightarrow (a \rightarrow b) \rightarrow (a \rightarrow c)$$

which may take a while to get used to.




Figure 1: Function composition diagram

1.4.2 fromInteger (looks recursive)

Near the end of the lecture notes there was an instance declaration including the following lines:

```
instance Num r  $\Rightarrow$  Num (ComplexSyn r) where
  -- ... several other methods and then
  fromInteger = toComplexSyn  $\circ$  fromInteger
```

This definition looks recursive, but it is not. To see why we need to expand the type and to do this I will introduce a name for the right hand side (RHS): *fromIntC*.

```
--          ComplexSyn r <----- r <----- Integer
fromIntC =          toComplexSyn . fromInteger
```

I have placed the types in the comment, with “backwards-pointing” arrows indicating that *fromInteger* :: *Integer* \rightarrow *r* and *toComplexSyn* :: *r* \rightarrow *ComplexSyn r* while the resulting function is *fromIntC* :: *Integer* \rightarrow *ComplexSyn r*. The use of *fromInteger* at type *r* means that the full type of *fromIntC* must refer to the *Num* class. Thus we arrive at the full type:

```
fromIntC :: Num r  $\Rightarrow$  Integer  $\rightarrow$  ComplexSyn r
```

1.4.3 type / newtype / data

There are three keywords in Haskell involved in naming types: **type**, **newtype**, and **data**.

type – abbreviating type expressions The **type** keyword is used to create a type synonym - just another name for a type expression.

```
type Heltal = Integer
type Foo = (Maybe [String], [[Heltal]])
type BinOp = Heltal  $\rightarrow$  Heltal  $\rightarrow$  Heltal
type Env v s = [(v, s)]
```

The new name for the type on the RHS does not add type safety, just readability (if used wisely). The *Env* example shows that a type synonym can have type parameters.

newtype – more protection A simple example of the use of **newtype** in Haskell is to distinguish values which should be kept apart. A simple example is

```
newtype Age = Ag Int -- Age in years
newtype Shoe = Sh Int -- Shoe size (EU)
```

Which introduces two new types, *Age* and *Shoe*, which both are internally represented by an *Int* but which are good to keep apart.

The constructor functions $Ag :: Int \rightarrow Age$ and $Sh :: Int \rightarrow Shoe$ are used to translate from plain integers to ages and shoe sizes.

In the lecture notes we used a newtype for the semantics of complex numbers as a pair of numbers in the cartesian representation but may also be useful to have another newtype for complex as a pair of numbers in the polar representation.

data – for syntax trees Some examples:

```
data N = Z | S N
```

This declaration introduces

- a new type *N* for unary natural numbers,
- a constructor $Z :: N$ to represent zero, and
- a constructor $S :: N \rightarrow N$ to represent the successor.

Examples values: $zero = Z$, $one = S\ Z$, $three = S\ (S\ one)$

```
data E = V String | P E E | T E E
```

This declaration introduces

- a new type *E* for simple arithmetic expressions,
- a constructor $V :: String \rightarrow E$ to represent variables,
- a constructor $P :: E \rightarrow E \rightarrow E$ to represent plus, and
- a constructor $T :: E \rightarrow E \rightarrow E$ to represent times.

Example values: $x = V\ "x"$, $e_1 = P\ x\ x$, $e_2 = T\ e_1\ e_1$

If you want a constructor to be used as an infix operator you need to use symbol characters and start with a colon:

```
data E' = V' String | E' :+: E' | E' :* E'
```

Example values: $y = V\ "y"$, $e_1 = y\ :+:\ y$, $e_2 = x\ :* e_1$

Finally, you can add one or more type parameters to make a whole family of datatypes in one go:

```
data ComplexSy v r = Var v
                    | FromCart r r
                    | ComplexSy v r :+: ComplexSy v r
                    | ComplexSy v r :** ComplexSy v r
```

The purpose of the first parameter *v* here is to enable a free choice of type for the variables (be it *String* or *Int* or something else) and the second parameter *r* makes it possible to express “complex numbers over” different base types (like *Double*, *Float*, *Integer*, etc.).

1.4.4 *Env*, *Var*, and variable lookup

The type synonym

```
type Env v s = [(v, s)]
```

is one way of expressing a partial function from v to s .

Example value:

```
env1 :: Env String Int
env1 = [("hej", 17), ("du", 38)]
```

The *Env* type is commonly used in evaluator functions for syntax trees containing variables:

```
evalCP :: Eq v => Env v (ComplexSem r) -> (ComplexSy v r -> ComplexSem r)
evalCP env (Var x) = case lookup x env of
  Just c -> undefined  -- ...
  -- ...
```

Notice that *env* maps “syntax” (variable names) to “semantics”, just like the evaluator does.

1.5 Some helper functions

```
propAssocAdd :: (Eq a, Num a) => a -> a -> a -> Bool
propAssocAdd = propAssocA (+)

(*.) :: Num r => ComplexSem r -> ComplexSem r -> ComplexSem r
CS (ar, ai) *. CS (br, bi) = CS (ar * br - ai * bi, ar * bi + ai * br)

instance Show r => Show (ComplexSem r) where
  show = showCS

showCS :: Show r => ComplexSem r -> String
showCS (CS (x, y)) = show x ++ " + " ++ show y ++ "i"
```

2 Week 2

Course learning outcomes:

- Knowledge and understanding
 - design and implement a DSL (Domain Specific Language) for a new domain
 - organize areas of mathematics in DSL terms
 - explain main concepts of elementary real and complex analysis, algebra, and linear algebra
- Skills and abilities
 - develop adequate notation for mathematical concepts
 - perform calculational proofs
 - use power series for solving differential equations

- use Laplace transforms for solving differential equations
- Judgement and approach
 - discuss and compare different software implementations of mathematical concepts

This week we focus on “develop adequate notation for mathematical concepts” and “perform calculational proofs” (still in the context of “organize areas of mathematics in DSL terms”).

module *DSLsofMath.W02* **where**

2.1 A few words about pure set theory

One way to build mathematics from the ground up is to start from pure set theory and define all concepts by translation to sets. We will only work with this as a mathematical domain to study, not as “the right way” of doing mathematics. The core of the language of pure set theory has the Empty set, the one-element set constructor Singleton, set Union, and Intersection. There are no “atoms” or “elements” to start from except for the empty set but it turns out that quite a large part of mathematics can still be expressed.

Natural numbers To talk about things like natural numbers in pure set theory they need to be encoded. Here is one such encoding (which is explored further in the first hand-in assignment).

$$\begin{aligned} \text{vonNeumann } 0 &= \text{Empty} \\ \text{vonNeumann } (n + 1) &= \text{Union } (\text{vonNeumann } n) \\ &\quad (\text{Singleton } (\text{vonNeumann } n)) \end{aligned}$$

Pairs Definition: A pair (a, b) is encoded as $\{\{a\}, \{a, b\}\}$.

2.2 Propositional Calculus

Now we turn to the main topic of this week: logic and proofs.

TODO: type up the notes + whiteboard photos

Swedish: Satslogik

False, True, And, Or, Implies

2.3 First Order Logic (predicate logic)

TODO: type up the notes + whiteboard photos

Swedish: Första ordningens logik = predikatlogik

Adds term variables and functions, predicate symbols and quantifiers (sv: kvantorer).

2.4 Basic concepts of calculus

Limit point TODO: transcribe the 2016 notes + 2017 black board pictures into notes.

Definition (adapted from Rudin [1964], page 28): Let X be a subset of \mathbb{R} . A point $p \in \mathbb{R}$ is a limit point of X if for every $\epsilon > 0$, there exists $q \in X$ such that $q \neq p$ and $|q - p| < \epsilon$.

$$\begin{aligned} \text{Limp} : \mathbb{R} &\rightarrow \mathcal{P} \mathbb{R} \rightarrow \text{Prop} \\ \text{Limp } p \ X &= \forall \epsilon > 0. \ \exists q \in X - \{p\}. \ |q - p| < \epsilon \end{aligned}$$

Notice that q depends on ϵ . Thus by introducing a function we can move the \exists out.

$$\begin{aligned} \text{type } Q &= \mathbb{R}_- \{>0\} \rightarrow (X - \{p\}) \\ \text{Limp } p \ X &= \exists q : Q. \ \forall \epsilon > 0. \ |q \ \epsilon - p| < \epsilon \end{aligned}$$

Next: introduce the “disk function” Di .

$$\begin{aligned} Di : \mathbb{R} &\rightarrow \mathbb{R}_- \{>0\} \rightarrow \mathcal{P} \mathbb{R} \\ Di \ c \ r &= \{x \mid |x - c| < r\} \end{aligned}$$

Then we get

$$\text{Limp } p \ X = \exists q : Q. \ \forall \epsilon > 0. \ q \ \epsilon \in Di \ p \ \epsilon$$

Example: limit outside the set X

$$X = \{1 / n \mid n \in \mathbb{N}_{>0}\}$$

Show that 0 is a limit point. Note that $0 \notin X$.

We want to prove $\text{Limp } 0 \ X$

$$q \ \epsilon = 1 / n \ \textbf{where } n = \textit{ceiling} (1 / \epsilon)$$

(where the definition of n comes from a calculation showing the property involving Di is satisfied.)

Exercise: prove that 0 is the *only* limit point of X .

Proposition: If X is finite, then it has no limit points.

$$\forall p \in \mathbb{R}. \ \neg (\text{Limp } p \ X)$$

Good exercise in quantifier negation!

$$f : (q : Q) \rightarrow \mathbb{R}_{>0} \{- \text{ such that } \textbf{let } \epsilon = f \ q \ \textbf{in } q \ \epsilon \notin Di \ p \ \epsilon - \}$$

Note that $q \ \epsilon$ is in (TODO: To be cont.)

The limit of a sequence TODO: transcribe the 2016 notes + 2017 black board pictures into notes.

$$P \ a \ \epsilon \ L = (\epsilon > 0) \rightarrow \exists N : \mathbb{Z}. (\forall n : \mathbb{N}. (n \geq N) \rightarrow (|a_n - L| < \epsilon))$$

2.5 Questions and answers from the exercise sessions week 2

Variables, *Env* and *lookup* This was a frequently source of confusion already the first week so there is already a question + answers earlier in this text. But here is an additional example to help clarify the matter.

```
data Rat v = RV v | FromI Integer | RPlus (Rat v) (Rat v) | RDiv (Rat v) (Rat v)
deriving (Eq, Show)
newtype RatSem = RSem (Integer, Integer)
```

We have a type *Rat v* for the syntax trees of rational number expressions and a type *RatSem* for the semantics of those rational number expressions as pairs of integers. The constructor *RV* :: *v* → *Rat v* is used to embed variables with names of type *v* in *Rat v*. We could use *String* instead of *v* but with a type parameter *v* we get more flexibility at the same time as we get better feedback from the type checker. To evaluate some *e* :: *Rat v* we need to know how to evaluate the variables we encounter. What does “evaluate” mean for a variable? Well, it just means that we must be able to translate a variable name (of type *v*) to a semantic value (a rational number in this case). To “translate a name to a value” we can use a function (of type *v* → *RatSem*) so we can give the following implementation of the evaluator:

```
evalRat1 :: (v → RatSem) → (Rat v → RatSem)
evalRat1 ev (RV v)      = ev v
evalRat1 ev (FromI i)   = fromISem i
evalRat1 ev (RPlus l r) = plusSem (evalRat1 ev l) (evalRat1 ev r)
evalRat1 ev (RDiv l r)  = divSem (evalRat1 ev l) (evalRat1 ev r)
```

Notice that we simply added a parameter *ev* for “evaluate variable” to the evaluator. The rest of the definition follows a common pattern: recursively translate each subexpression and apply the corresponding semantic operation to combine the results: *RPlus* is replaced by *plusSem*, etc.

```
fromISem :: Integer → RatSem
fromISem i = RSem (i, 1)

plusSem :: RatSem → RatSem → RatSem
plusSem = undefined -- TODO: exercise

-- Division of rational numbers
divSem :: RatSem → RatSem → RatSem
divSem (RSem (a, b)) (RSem (c, d)) = RSem (a * d, b * c)
```

Often the first argument *ev* to the eval function is constructed from a list of pairs:

```
type Env v s = [(v, s)]
envToFun :: (Show v, Eq v) ⇒ Env v s → (v → s)
envToFun [] v = error ("envToFun: variable " ++ show v ++ " not found")
envToFun ((w, s) : env) v
  | w == v      = s
  | otherwise   = envToFun env v
```

Thus, *Env v s* can be seen as an implementation of a “lookup table”. It could also be implemented using hash tables or binary search trees, but efficiency is not the point here. Finally, with *envToFun* in our hands we can implement a second version of the evaluator:

```
evalRat2 :: (Show v, Eq v) ⇒ (Env v RatSem) → (Rat v → RatSem)
evalRat2 env e = evalRat1 (envToFun env) e
```

The law of the excluded middle Many had problems with implementing the “law of the excluded middle” in the exercises and it is indeed a tricky property to prove. The key to implementing it lies in double negation and as that is encoded with higher order functions it gets a bit hairy.

TODO[Daniel]: more explanation

SET and PRED Several groups have had trouble grasping the difference between *SET* and *PRED*. This is understandable, because we have so far in the lectures mostly talked about term syntax + semantics, and not so much about predicate syntax and semantics. The one example of terms + predicates covered in the lectures is Predicate Logic and I never actually showed how `eval` (for the expressions) and `check` (for the predicates) is implemented.

As an example we can take our terms to be the rational number expressions defined above and define a type of predicates over those terms:

```

type Term v = Rat v
data RPred v = Equal    (Term v) (Term v)
              | LessThan (Term v) (Term v)
              | Positive  (Term v)
              | And       (RPred v) (RPred v)
              | Not       (RPred v)
deriving (Eq, Show)

```

Note that the first three constructors, *Eq*, *LessThan*, and *Positive*, describe predicates or relations between terms (which can contain term variables) while the two last constructors, *And* and *Not*, just combine such relations together. (Terminology: I often mix the words “predicate” and “relation”).

We have already defined the evaluator for the *Term v* type but we need to add a corresponding “evaluator” (called *check*) for the *RPred v* type. Given values for all term variables the predicate checker should just determine if the predicate is true or false.

```

checkRP :: (Eq v, Show v) => Env v RatSem -> RPred v -> Bool
checkRP env (Equal    t1 t2) = eqSem      (evalRat2 env t1) (evalRat2 env t2)
checkRP env (LessThan t1 t2) = lessThanSem (evalRat2 env t1) (evalRat2 env t2)
checkRP env (Positive  t1)   = positiveSem (evalRat2 env t1)
checkRP env (And p q)  = (checkRP env p) & (checkRP env q)
checkRP env (Not p)    = not (checkRP env p)

```

Given this recursive definition of *checkRP*, the semantic functions *eqSem*, *lessThanSem*, and *positiveSem* can be defined by just working with the rational number representation:

```

eqSem      :: RatSem -> RatSem -> Bool
lessThanSem :: RatSem -> RatSem -> Bool
positiveSem :: RatSem -> Bool
eqSem      = error "TODO"
lessThanSem = error "TODO"
positiveSem = error "TODO"

```

2.6 More general code for first order languages

“överkurs”

It is possible to make one generic implementation which can be specialised to any first order language.

TODO: add explanatory text

- $Term$ = Syntactic terms
- n = names (of atomic terms)
- f = function names
- v = variable names
- WFF = Well Formed Formulas
- p = predicate names

```

data  $Term\ n\ f\ v = N\ n \mid F\ f\ [Term\ n\ f\ v] \mid V\ v$ 
deriving Show
data  $WFF\ n\ f\ v\ p =$ 
   $P\ p\ [Term\ n\ f\ v]$ 
 $\mid Equal\ (Term\ n\ f\ v)\ (Term\ n\ f\ v)$ 
 $\mid And\ (WFF\ n\ f\ v\ p)\ (WFF\ n\ f\ v\ p)$ 
 $\mid Or\ (WFF\ n\ f\ v\ p)\ (WFF\ n\ f\ v\ p)$ 
 $\mid Equiv\ (WFF\ n\ f\ v\ p)\ (WFF\ n\ f\ v\ p)$ 
 $\mid Impl\ (WFF\ n\ f\ v\ p)\ (WFF\ n\ f\ v\ p)$ 
 $\mid Not\ (WFF\ n\ f\ v\ p)$ 
 $\mid FORALL\ v\ (WFF\ n\ f\ v\ p)$ 
 $\mid EXISTS\ v\ (WFF\ n\ f\ v\ p)$ 
deriving Show

```

3 Week 3

```

{-# LANGUAGE FlexibleInstances #-}
module DSLsofMath.W03 where

```

3.1 Types in mathematics

Types are sometimes mentioned explicitly in mathematical texts:

- $x \in \mathbb{R}$
- $\sqrt{\cdot} : \mathbb{R}_{\geq 0} \rightarrow \mathbb{R}_{\geq 0}$
- $(_)^2 : \mathbb{R} \rightarrow \mathbb{R}$ or, alternatively but *not* equivalently
- $(_)^2 : \mathbb{R} \rightarrow \mathbb{R}_{\geq 0}$

The types of “higher-order” operators are usually not given explicitly:

- $lim : (\mathbb{N} \rightarrow \mathbb{R}) \rightarrow \mathbb{R}$ for $lim_{n \rightarrow \infty} \{a_n\}$

- $d/dt : (\mathbb{R} \rightarrow \mathbb{R}) \rightarrow \mathbb{R} \rightarrow \mathbb{R}$
- sometimes, instead of df/dt one sees f' or \dot{f} or $D f$
- $\partial f / \partial x_i : (\mathbb{R}^n \rightarrow \mathbb{R}) \rightarrow \mathbb{R}^n \rightarrow \mathbb{R}$
- we mostly see $\partial f / \partial x$, $\partial f / \partial y$, $\partial f / \partial z$ etc. when, in the context, the function f has been given a definition of the form $f(x, y, z) = \dots$
- a better notation (by Landau) which doesn't rely on the names given to the arguments was popularised in Landau [1934] (English edition Landau [2001]): D_1 for the partial derivative with respect to x_1 , etc.
- Exercise: for $f : \mathbb{R}^2 \rightarrow \mathbb{R}$ define D_1 and D_2 using only D .

3.2 Typing Mathematics: partial derivative

As an example we will try to type the elements of a mathematical definition.

For example, on page 169 of Mac Lane [1986], we read

[...] a function $z = f(x, y)$ for all points (x, y) in some open set U of the cartesian (x, y) -plane. [...] If one holds y fixed, the quantity z remains just a function of x ; its derivative, when it exists, is called the *partial derivative* with respect to x . Thus at a point (x, y) in U this derivative for $h \neq 0$ is

$$\partial z / \partial x = f'_x(x, y) = \lim_{h \rightarrow 0} (f(x + h, y) - f(x, y)) / h$$

What are the types of the elements involved? We have

$U \subseteq \mathbb{R} \times \mathbb{R}$ -- cartesian plane
 $f : U \rightarrow \mathbb{R}$
 $z : U \rightarrow \mathbb{R}$ -- but see below
 $f_x : U \rightarrow \mathbb{R}$

The x in the subscript of f' is *not* a real number, but a symbol (a *Char*).

The expression (x, y) has several occurrences. The first two denote variables of type U , the third is just a name ((x, y) -plane). The third denotes a variable of type U , it is bound by a universal quantifier

$$\forall (x, y) \in U$$

The variable h appears to be a non-zero real number, bound by a universal quantifier, but that is incorrect. In fact, h is used as a variable to construct the arguments of a function, whose limit is then taken at 0.

That function, which we can denote by φ has the type $\varphi : U \rightarrow (\mathbb{R} - \{0\}) \rightarrow \mathbb{R}$ and is defined by

$$\varphi(x, y) h = (f(x + h, y) - f(x, y)) / h$$

The limit is then $\lim (\varphi(x, y)) 0$. Note that 0 is a limit point of $\mathbb{R} - \{0\}$, so the type of \lim is the one we have discussed:

$$\lim : (X \rightarrow \mathbb{R}) \rightarrow \{p \mid p \in \mathbb{R}, \text{Limp } p \ X\} \rightarrow \mathbb{R}$$

$z = f(x, y)$ probably does not mean that $z \in \mathbb{R}$, although the phrase “the quantity z ” suggests this. A possible interpretation is that z is used to abbreviate the expression $f(x, y)$; thus, everywhere we can replace z with $f(x, y)$. In particular, $\partial z / \partial x$ becomes $\partial f(x, y) / \partial x$, which we can interpret as $\partial f / \partial x$ applied to (x, y) (remember that (x, y) is bound in the context by a universal quantifier). There is the added difficulty that, just like x , the x in ∂x is not the x bound by the universal quantifier, but just a symbol.

3.3 Type inference and understanding: Lagrangian case study

From (Sussman and Wisdom 2013):

A mechanical system is described by a Lagrangian function of the system state (time, coordinates, and velocities). A motion of the system is described by a path that gives the coordinates for each moment of time. A path is allowed if and only if it satisfies the Lagrange equations. Traditionally, the Lagrange equations are written

$$\frac{d}{dt} \frac{\partial L}{\partial \dot{q}} - \frac{\partial L}{\partial q} = 0$$

What could this expression possibly mean?

To start answering the question, we start typing the elements involved:

1. $\partial L / \partial q$ suggests that L is a function of at least a pair of arguments:

$$L : \mathbb{R}^n \rightarrow \mathbb{R}, n \geq 2$$

This is consistent with the description: “Lagrangian function of the system state (time, coordinates, and velocities)”. So we can take $n = 3$:

$$L : \mathbb{R}^3 \rightarrow \mathbb{R}$$

2. $\partial L / \partial q$ suggests that q is the name of a real variable, one of the three arguments to L . In the context, which we do not have, we would expect to find somewhere the definition of the Lagrangian as

$$L(t, q, v) = \dots$$

3. therefore, $\partial L / \partial q$ should also be a function of a triple of arguments:

$$\partial L / \partial q : \mathbb{R}^3 \rightarrow \mathbb{R}$$

It follows that the equation expresses a relation between *functions*, therefore the 0 on the right-hand side is *not* the real number 0, but rather the constant function 0:

$$\begin{aligned} \text{const } 0 : \mathbb{R}^3 &\rightarrow \mathbb{R} \\ \text{const } 0(t, q, v) &= 0 \end{aligned}$$

4. We now have a problem: d/dt can only be applied to functions of *one* real argument t , and the result is a function of one real argument:

$$\frac{d}{dt} \frac{\partial L}{\partial \dot{q}} : \mathbb{R} \rightarrow \mathbb{R}$$

Since we subtract from this the function $\partial L / \partial q$, it follows that this, too, must be of type $\mathbb{R} \rightarrow \mathbb{R}$, contradiction.

5. The expression $\partial L / \partial \dot{q}$ appears to also be malformed. We would expect a variable name where we find \dot{q} , but \dot{q} is the same as dq/dt , a function.
6. Looking back at the description above, we see that the only candidate for an application of d/dt is “a path that gives the coordinates for each moment of time”. Thus, the path is a function of time, let us say

$$w : \mathbb{R} \rightarrow \mathbb{R}, \text{ where } w(t) \text{ is a coordinate at time } t$$

We can now guess that the use of the plural form “equations” might have something to do with the use of “coordinates”. In an n -dimensional space, a position is given by n coordinates. A path would be a function

$$w : \mathbb{R} \rightarrow \mathbb{R}^n$$

which is equivalent to n functions of type $\mathbb{R} \rightarrow \mathbb{R}$. We would then have an equation for each of them. We will use $n = 1$ for the rest of this example.

7. The Lagrangian is a “function of the system state (time, coordinates, and velocities)”. If we have a path, then the coordinates at any time are given by the path. The velocity is the derivative of the path, also fixed by the path:

$$\begin{aligned} q &: \mathbb{R} \rightarrow \mathbb{R} \\ q \ t &= w \ t \\ \dot{q} &: \mathbb{R} \rightarrow \mathbb{R} \\ \dot{q} \ t &= dw / dt \end{aligned}$$

The equations do not use a function $L : \mathbb{R}^3 \rightarrow \mathbb{R}$, but rather

$$L \circ \text{expand } w : \mathbb{R} \rightarrow \mathbb{R}$$

where the “combinator” *expand* is given by

$$\begin{aligned} \text{expand} &: (\mathbb{R} \rightarrow \mathbb{R}) \rightarrow \mathbb{R} \rightarrow \mathbb{R}^3 \\ \text{expand } w \ t &= (t, w \ t, D \ w \ t) \end{aligned}$$

8. Similarly, using D_1 , D_2 , D_3 instead of $\partial L / \partial t$ etc., we have that, instead of $\partial L / \partial q$ what is meant is

$$D_2 \ L \circ \text{expand } w : \mathbb{R} \rightarrow \mathbb{R}$$

and instead of $\partial L / \partial \dot{q}$

$$D_3 \ L \circ \text{expand } w : \mathbb{R} \rightarrow \mathbb{R}$$

The equation becomes

$$D \ (D_3 \ L \circ \text{expand } w) - D_2 \ L \circ \text{expand } w = 0$$

a relation between functions of type $\mathbb{R} \rightarrow \mathbb{R}$. In particular, the right-hand 0 is the constant function

$$\text{const } 0 : \mathbb{R} \rightarrow \mathbb{R}$$

3.4 Types in Mathematics (Part II)

3.4.1 Type classes

The kind of type inference we presented in the last lecture becomes automatic with experience in a domain, but is very useful in the beginning.

The “trick” of looking for an appropriate combinator with which to pre- or post-compose a function in order to makes types match is often useful. It is similar to the casts one does automatically in expressions such as $4 + 2.5$.

One way to understand such casts from the point of view of functional programming is via *type classes*. As a reminder, the reason $4 + 2.5$ works is because floating point values are members of the class *Num*, which includes the member function

fromInteger :: *Integer* → *a*

which converts integers to the actual type *a*.

Type classes are related to mathematical structures which, in turn, are related to DSLs. The structuralist point of view in mathematics is that each mathematical domain has its own fundamental structures. Once these have been identified, one tries to push their study as far as possible *on their own terms*, i.e., without introducing other structures. For example, in group theory, one starts by exploring the consequences of just the group structure, before one introduces, say, an order structure and monotonicity.

The type classes of Haskell seem to have been introduced without relation to their mathematical counterparts, perhaps because of pragmatic considerations. For now, we examine the numerical type classes *Num*, *Fractional*, and *Floating*.

```
class (Eq a, Show a) => Num a where
  (+), (-), (*) :: a -> a -> a
  negate      :: a -> a
  |.|, signum :: a -> a
  fromInteger :: Integer -> a
```

TODO: insert proper citation ?

This is taken from the Haskell documentation¹ but it appears that *Eq* and *Show* are not necessary, because there are meaningful instances of *Num* which don't support them:

```
instance Num a => Num (x -> a) where
  f + g      = \x -> f x + g x
  f - g      = \x -> f x - g x
  f * g      = \x -> f x * g x
  negate f   = negate o f
  |f|        = |.| o f
  signum f   = signum o f
  fromInteger = const o fromInteger
```

Next we have *Fractional* for when we also have division:

```
class Num a => Fractional a where
  (/) :: a -> a -> a
```

¹Fig. 6.2 in section 6.4 of the Haskell 2010 report: <https://www.haskell.org/onlinereport/haskell2010/haskellch6.html>.

```

recip :: a → a
fromRational :: Rational → a

```

and *Floating* when we can implement the “standard” functions from calculus:

```

class Fractional a ⇒ Floating a where
  π :: a
  exp, log, √ · :: a → a
  (**), logBase :: a → a → a
  sin, cos, tan :: a → a
  asin, acos, atan :: a → a
  sinh, cosh, tanh :: a → a
  asinh, acosh, atanh :: a → a

```

We can instantiate these type classes for functions in the same way we did for *Num*:

```

instance Fractional a ⇒ Fractional (x → a) where
  recip f = recip ∘ f
  fromRational = const ∘ fromRational

```

```

instance Floating a ⇒ Floating (x → a) where
  π = const π
  exp f = exp ∘ f
  f ** g = λx → (f x) ** (g x)
  -- and so on

```

Exercise: complete the instance declarations.

These type classes represent an abstract language of algebraic and standard operations, abstract in the sense that the exact nature of the elements involved is not important from the point of view of the type class, only from that of its implementation.

3.5 Computing derivatives

The “little language” of derivatives:

```

D (f + g) = D f + D g
D (f * g) = D f * g + f * D g
D (f ∘ g) x = D f (g x) * D g x -- the chain rule
D (const a) = const 0
D id = const 1
D (n) x = (n − 1) * (x^(n − 1))
D sin x = cos x
D cos x = − (sin x)
D exp x = exp x

```

and so on.

We observe that we can compute derivatives for any expressions made out of arithmetical functions, standard functions, and their compositions. In other words, the computation of derivatives is based on a DSL of expressions (representing functions in one variable):

```

expression ::= const ℝ
              | id

```

```

      | expression + expression
      | expression * expression
      | exp expression
      | ...

```

etc.

We can implement this in a datatype:

```

data FunExp = Const Double
           | Id
           | FunExp :+: FunExp
           | FunExp **: FunExp
           | Exp FunExp
           -- and so on
deriving Show

```

The intended meaning of elements of the *FunExp* type is functions:

```

eval :: FunExp → Double → Double
eval (Const α) = const α
eval Id = id
eval (e1 :+: e2) = eval e1 + eval e2 -- note the use of “lifted +”
eval (e1 **: e2) = eval e1 * eval e2 -- “lifted *”
eval (Exp e1) = exp (eval e1) -- and “lifted exp”
-- and so on

```

We can implement the derivative of such expressions using the rules of derivatives. We want to implement a function *derive* :: *FunExp* → *FunExp* which makes the following diagram commute:

$$\begin{array}{ccc}
 \text{FunExp} & \xrightarrow{\text{eval}} & \text{Func} \\
 \downarrow \text{derive} & & \downarrow D \\
 \text{FunExp} & \xrightarrow{\text{eval}} & \text{Func}
 \end{array}$$

In other words, for any expression *e*, we want

$$\text{eval} (\text{derive } e) = D (\text{eval } e)$$

For example, let us derive the *derive* function for *Exp e*:

$$\begin{aligned}
 & \text{eval} (\text{derive} (\text{Exp } e)) \\
 = & \{- \text{specification of } \text{derive} \text{ above} -\} \\
 & D (\text{eval} (\text{Exp } e)) \\
 = & \{- \text{def. } \text{eval} -\} \\
 & D (\text{exp} (\text{eval } e)) \\
 = & \{- \text{def. } \text{exp} \text{ for functions} -\} \\
 & D (\text{exp} \circ \text{eval } e) \\
 = & \{- \text{chain rule} -\} \\
 & (D \text{ exp} \circ \text{eval } e) * D (\text{eval } e) \\
 = & \{- D \text{ rule for } \text{exp} -\} \\
 & (\text{exp} \circ \text{eval } e) * D (\text{eval } e)
 \end{aligned}$$

$$\begin{aligned}
&= \{- \text{specification of } \textit{derive} -\} \\
&\quad (\textit{exp} \circ \textit{eval} \ e) * (\textit{eval} \ (\textit{derive} \ e)) \\
&= \{- \text{def. of } \textit{eval} \text{ for } \textit{Exp} -\} \\
&\quad (\textit{eval} \ (\textit{Exp} \ e)) * (\textit{eval} \ (\textit{derive} \ e)) \\
&= \{- \text{def. of } \textit{eval} \text{ for } \textit{:}* -\} \\
&\quad \textit{eval} \ (\textit{Exp} \ e \textit{:}*} \ \textit{derive} \ e)
\end{aligned}$$

Therefore, the specification is fulfilled by taking

$$\textit{derive} \ (\textit{Exp} \ e) = \textit{Exp} \ e \textit{:}*} \ \textit{derive} \ e$$

Similarly, we obtain

$$\begin{aligned}
\textit{derive} \ (\textit{Const} \ \alpha) &= \textit{Const} \ 0 \\
\textit{derive} \ \textit{Id} &= \textit{Const} \ 1 \\
\textit{derive} \ (e_1 \textit{:}+ \ e_2) &= \textit{derive} \ e_1 \textit{:}+ \ \textit{derive} \ e_2 \\
\textit{derive} \ (e_1 \textit{:}*} \ e_2) &= (\textit{derive} \ e_1 \textit{:}*} \ e_2) \textit{:}+ \ (e_1 \textit{:}*} \ \textit{derive} \ e_2) \\
\textit{derive} \ (\textit{Exp} \ e) &= \textit{Exp} \ e \textit{:}*} \ \textit{derive} \ e
\end{aligned}$$

Exercise: complete the *FunExp* type and the *eval* and *derive* functions.

3.6 Shallow embeddings

The DSL of expressions, whose syntax is given by the type *FunExp*, turns out to be almost identical to the DSL defined via type classes in the first part of this lecture. The correspondence between them is given by the *eval* function.

The difference between the two implementations is that the first one separates more cleanly from the semantical one. For example, *:+* *stands for* a function, while *+* *is* that function.

The second approach is called “shallow embedding” or “almost abstract syntax”. It can be more economical, since it needs no *eval*. The question is: can we implement *derive* in the shallow embedding?

Note that the reason the shallow embedding is possible is that the *eval* function is a *fold*: first evaluate the sub-expressions of *e*, then put the evaluations together without reference to the sub-expressions. This is sometimes referred to as “compositionality”.

We check whether the semantics of derivatives is compositional. The evaluation function for derivatives is

$$\begin{aligned}
\textit{eval}' &:: \textit{FunExp} \rightarrow \textit{Double} \rightarrow \textit{Double} \\
\textit{eval}' &= \textit{eval} \circ \textit{derive}
\end{aligned}$$

For example:

$$\begin{aligned}
&\textit{eval}' \ (\textit{Exp} \ e) \\
&= \{- \text{def. } \textit{eval}', \text{ function composition} -\} \\
&\quad \textit{eval} \ (\textit{derive} \ (\textit{Exp} \ e)) \\
&= \{- \text{def. } \textit{derive} \text{ for } \textit{Exp} -\} \\
&\quad \textit{eval} \ (\textit{Exp} \ e \textit{:}*} \ \textit{derive} \ e) \\
&= \{- \text{def. } \textit{eval} \text{ for } \textit{:}*} -\}
\end{aligned}$$

```

    eval (Exp e) :: eval (derive e)
= {- def. eval for Exp -}
    exp (eval e) * eval (derive e)
= {- def. eval' -}
    exp (eval e) * eval' e

```

and the first e doesn't go away. The semantics of derivatives is not compositional.

Or rather, *this* semantics is not compositional. It is quite clear that the derivatives cannot be evaluated without, at the same time, being able to evaluate the functions. So we can try to do both evaluations simultaneously:

```

type FD a = (a → a, a → a)
evalD :: FunExp → FD Double
evalD e      = (eval e, eval' e)

```

Is *evalD* compositional?

We compute, for example:

```

    evalD (Exp e)
= {- specification of evalD -}
    (eval (Exp e), eval' (Exp e))
= {- def. eval for Exp and reusing the computation above -}
    (exp (eval e), exp (eval e) * eval' e)
= {- introduce names for subexpressions -}
    let f = eval e
        f' = eval' e
    in (exp f, exp f * f')
= {- def. evalD -}
    let (f, f') = evalD e
    in (exp f, exp f * f')

```

This semantics *is* compositional. We can now define a shallow embedding for the computation of derivatives, using the numerical type classes.

```

instance Num a ⇒ Num (a → a, a → a) where
    (f, f') + (g, g') = (f + g, f' + g')
    (f, f') * (g, g') = (f * g, f' * g + f * g')
    fromInteger n = (fromInteger n, const 0)

```

Exercise: implement the rest

4 Week 4

```

{-# LANGUAGE FlexibleInstances, GeneralizedNewtypeDeriving #-}
module DSLsofMath.W04 where
import Prelude hiding (Monoid)

```


4.1 A simpler example of a non-compositional function

Consider a very simple datatype of integer expressions:

```
data E = Add E E | Mul E E | Con Integer deriving Eq
e1, e2 :: E
e1 = Add (Con 1) (Mul (Con 2) (Con 3))
e2 = Mul (Add (Con 1) (Con 2)) (Con 3)
```

When working with expressions it is often useful to have a “pretty-printer” to convert the abstract syntax trees to strings like “1+2*3”.

```
pretty :: E → String
```

We can view *pretty* as an alternative *eval* function for a semantics using *String* as the semantic domain instead of the more natural *Integer*. We can implement *pretty* in the usual way as a “fold” over the syntax tree using one “semantic constructor” for each syntact constructor:

```
pretty (Add x y) = prettyAdd (pretty x) (pretty y)
pretty (Mul x y) = prettyMul (pretty x) (pretty y)
pretty (Con c)   = prettyCon c
prettyAdd :: String → String → String
prettyMul :: String → String → String
prettyCon :: Integer → String
```

Now, if we try to implement the semantic constructors without thinking too much we would get the following:

```
prettyAdd sx sy = sx ++ "+" ++ sy
prettyMul sx sy = sx ++ "*" ++ sy
prettyCon i     = show i
p1, p2 :: String
p1 = pretty e1
p2 = pretty e2
trouble :: Bool
trouble = p1 == p2
```

Note that both *e₁* and *e₂* are different but they pretty-print to the same string. There are many ways to fix this, some more “pretty” than others, but the main problem is that some information is lost in the translation.

TODO(perhaps): Explain using three pretty printers for the three “contexts”: at top level, inside *Add*, inside *Mul*, ... then combine them with the tupling transform just as with *evalD*. The result is the following:

```
prTop :: E → String
prTop e = let (pTop, -, -) = prVersions e
         in pTop
prVersions = foldE prVerAdd prVerMul prVerCon
prVerAdd (xTop, xInA, xInM) (yTop, yInA, yInM) =
  let s = xInA ++ "+" ++ yInA    -- use InA because we are “in Add”
  in (s, paren s, paren s)       -- parens needed except at top level
prVerMul (xTop, xInA, xInM) (yTop, yInA, yInM) =
  let s = xInM ++ "*" ++ yInM    -- use InM because we are “in Mul”
  in (s, s, paren s)            -- parens only needed inside Mul
```

```

prVerCon i =
  let s = show i
  in (s, s, s)           -- parens never needed
paren :: String → String
paren s = "(" ++ s ++ ")"


```

Exercise: Another way to make this example go through is to refine the semantic domain from *String* to *Precedence* → *String*. This can be seen as another variant of the result after the tupling transform: if *Precedence* is an *n*-element type then *Precedence* → *String* can be seen as an *n*-tuple. In our case a three-element *Precedence* would be enough.

4.2 Compositional semantics in general

In general, for a syntax *Syn*, and a possible semantics (a type *Sem* and an *eval* function of type *Syn* → *Sem*), we call the semantics *compositional* if we can implement *eval* as a fold. Informally a “fold” is a recursive function which replaces each abstract syntax constructor *Ci* of *Syn* with a “semantic constructor” *ci*.

TODO: Picture to illustrate



As an example we can define a general *foldE* for the integer expressions:

```

foldE :: (t → t → t) → (t → t → t) → (Integer → t) →
  E → t
foldE add mul con = rec
  where rec (Add x y) = add (rec x) (rec y)
        rec (Mul x y) = mul (rec x) (rec y)
        rec (Con i)   = con i

```

Notice that *foldE* has three function arguments corresponding to the three constructors of *E*. The “natural” evaluator to integers is then easy:

```

evalE1 :: E → Integer
evalE1 = foldE (+) (*) id

```

and with a minimal modification we can also make it work for other numeric types:

```

evalE2 :: Num a ⇒ E → a
evalE2 = foldE (+) (*) fromInteger

```

Another thing worth noting is that if we replace each abstract syntax constructor with itself we get the identity function (a “deep copy”):

```

idE :: E → E
idE = foldE Add Mul Con

```

Finally, it is often useful to capture the semantic functions (the parameters to the fold) in a type class:

```
class IntExp t where
```

```
  add :: t → t → t
```

```
  mul :: t → t → t
```

```
  con :: Integer → t
```

In this way we can make “hide” the arguments to the fold:

```
foldIE :: IntExp t ⇒ E → t
```

```
foldIE = foldE add mul con
```

```
instance IntExp E where
```

```
  add = Add
```

```
  mul = Mul
```

```
  con = Con
```

```
instance IntExp Integer where
```

```
  add = (+)
```

```
  mul = (*)
```

```
  con = id
```

```
idE' :: E → E
```

```
idE' = foldIE
```

```
evalE' :: E → Integer
```

```
evalE' = foldIE
```

4.3 Back to derivatives and evaluation

Review section 3.6 again with the definition of *eval'* being non-compositional (just like *pretty*) and *evalD* a more complex, but compositional, semantics.

5 Algebraic Structures and DSLs

In this lecture, we continue exploring the relationship between type classes, mathematical structures, and DSLs.

5.1 Algebras, homomorphisms

From Wikipedia:

In universal algebra, an algebra (or algebraic structure) is a set A together with a collection of operations on A .

Example:

```
class Monoid a where
```

```
  unit :: a
```

```
  op   :: a → a → a
```

After the operations have been specified, the nature of the algebra can be further limited by axioms, which in universal algebra often take the form of identities, or *equational laws*.

Example: Monoid equations

$$\begin{aligned} \forall x : a. & \text{ (unit 'op' } x == x \wedge x \text{ 'op' unit} == x) \\ \forall x, y, z : a. & \text{ (} x \text{ 'op' (} y \text{ 'op' } z) == (x \text{ 'op' } y) \text{ 'op' } z) \end{aligned}$$

A homomorphism between two algebras A and B is a function $h : A \rightarrow B$ from the set A to the set B such that, for every operation fA of A and corresponding fB of B (of arity, say, n), $h (fA (x_1, \dots, x_n)) = fB (h (x_1), \dots, h (x_n))$.

Example: Monoid homomorphism

$$\begin{aligned} h \text{ unit} &= \text{unit} \\ h (x \text{ 'op' } y) &= h x \text{ 'op' } h y \end{aligned}$$

```
newtype ANat    = A Int deriving (Show, Num, Eq)
instance Monoid ANat where
    unit          = A 0
    op (A m) (A n) = A (m + n)
newtype MNat    = M Int deriving (Show, Num, Eq)
instance Monoid MNat where
    unit          = M 1
    op (M m) (M n) = M (m * n)
```

Exercise: characterise the homomorphisms from $ANat$ to $MNat$.

Solution:

Let $h : ANat \rightarrow MNat$ be a homomorphism. Then

$$\begin{aligned} h 0 &= 1 \\ h (x + y) &= h x * h y \end{aligned}$$

For example $h (x + x) = h x * h x = (h x)^2$ which for $x = 1$ means that $h 2 = (h 1)^2$.

More generally, every n in $ANat$ is equal to the sum of n ones: $1 + 1 + \dots + 1$. Therefore

$$h n = (h 1)^n$$

Every choice of $h 1$ “induces a homomorphism”. This means that the value of the function h is fully determined by its value for 1.

5.2 Homomorphism and compositional semantics

Last time, we saw that $eval$ is compositional, while $eval'$ is not. Another way of phrasing that is to say that $eval$ is a homomorphism, while $eval'$ is not. To see this, we need to make explicit the structure of $FunExp$:

```
instance Num FunExp where
    (+)      = (:+ :)
    (*)      = (:* :)
    fromInteger = Const o fromInteger
instance Fractional FunExp where
instance Floating FunExp where
    exp      = Exp
```

and so on.

Exercise: complete the type instances for *FunExp*.

For instance, we have

$$\begin{aligned} eval (e_1 :*: e_2) &= eval\ e_1 * eval\ e_2 \\ eval (Exp\ e) &= exp (eval\ e) \end{aligned}$$

These properties do not hold for *eval'*, but do hold for *evalD*.

The numerical classes in Haskell do not fully do justice to the structure of expressions, for example, they do not contain an identity operation, which is needed to translate *Id*, nor an embedding of doubles, etc. If they did, then we could have evaluated expressions more abstractly:

$$eval :: GoodClass\ a \Rightarrow FunExp \rightarrow a$$

where *GoodClass* gives exactly the structure we need for the translation.

Exercise: define *GoodClass* and instantiate *FunExp* and *Double* \rightarrow *Double* as instances of it. Find another instance of *GoodClass*.

Therefore, we can always define a homomorphism from *FunExp* to *any* instance of *GoodClass*, in an essentially unique way. In the language of category theory, *FunExp* is an initial algebra.

Let us explore this in the simpler context of *Monoid*. The language of monoids is given by

```
type Var    = String
data MExpr = Unit | Op MExpr MExpr | V Var
```

Alternatively, we could have parametrised *MExpr* over the type of variables.

Just as in the case of FOL terms, we can evaluate an *MExpr* in a monoid instance if we are given a way of interpreting variables, also called an assignment:

$$evalM :: Monoid\ a \Rightarrow (Var \rightarrow a) \rightarrow (MExpr \rightarrow a)$$

Once given an $f :: Var \rightarrow a$, the homomorphism condition defines *evalM*:

$$\begin{aligned} evalM\ f\ Unit &= unit \\ evalM\ f\ (Op\ e_1\ e_2) &= op\ (evalM\ f\ e_1)\ (evalM\ f\ e_2) \\ evalM\ f\ (V\ x) &= f\ x \end{aligned}$$

(Observation: In *FunExp*, the role of variables was played by *Double*, and the role of the assignment by the identity.)

The following correspondence summarises the discussion so far:

Computer Science	Mathematics
DSL	structure (category, algebra, ...)
deep embedding, abstract syntax	initial algebra
shallow embedding	any other algebra
semantics	homomorphism from the initial algebra

The underlying theory of this table is a fascinating topic but mostly out of scope for the DSLsof-Math course. See Category Theory and Functional Programming for a whole course around this (lecture notes are available on github).

5.3 Other homomorphisms

Last time, we defined a *Num* instance for functions with a *Num* codomain. If we have an element of the domain of such a function, we can use it to obtain a homomorphism from functions to their codomains:

$$\text{Num } a \Rightarrow x \rightarrow (x \rightarrow a) \rightarrow a$$

As suggested by the type, the homomorphism is just function application:

$$\begin{aligned} \text{apply} &:: a \rightarrow (a \rightarrow b) \rightarrow b \\ \text{apply } a &= \lambda f \rightarrow f \ a \end{aligned}$$

Indeed, writing $h = \text{apply } c$ for some fixed c , we have

$$\begin{aligned} &h (f + g) \\ &= \{- \text{ def. } \text{apply } - \} \\ &\quad (f + g) \ c \\ &= \{- \text{ def. } + \text{ for functions } - \} \\ &\quad f \ c + g \ c \\ &= \{- \text{ def. } \text{apply } - \} \\ &\quad h \ f + h \ g \end{aligned}$$

etc.

Can we do something similar for *FD*?

The elements of *FD* are pairs of functions, so we can take

$$\begin{aligned} \text{apply} &:: a \rightarrow \text{FD } a \rightarrow (a, a) \\ \text{apply } c \quad (f, f') &= (f \ c, f' \ c) \end{aligned}$$

We now have the domain of the homomorphism ($\text{FD } a$) and the homomorphism itself ($\text{apply } c$), but we are missing the structure on the codomain, which now consists of pairs (a, a) . In fact, we can *compute* this structure from the homomorphism condition. For example:

$$\begin{aligned} &h ((f, f') * (g, g')) \\ &= \{- \text{ def. } * \text{ for } \text{FD } a - \} \\ &\quad h (f * g, f' * g + f * g') \\ &= \{- \text{ def. } h = \text{apply } c - \} \\ &\quad ((f * g) \ c, (f' * g + f * g') \ c) \\ &= \{- \text{ def. } * \text{ and } + \text{ for functions } - \} \\ &\quad (f \ c * g \ c, f' \ c * g \ c + f \ c * g' \ c) \\ &= \{- \text{ homomorphism condition from step 1 } - \} \\ &\quad h (f, f') \otimes h (g, g') \\ &= \{- \text{ def. } h = \text{apply } c - \} \\ &\quad (f \ c, f' \ c) \otimes (g \ c, g' \ c) \end{aligned}$$

The identity will hold if we take

$$(x, x') \otimes (y, y') = (x * y, x' * y + x * y')$$

Exercise: complete the instance declarations for $(Double, Double)$.

Note: As this computation goes through also for the other cases we can actually work with just pairs of values (at an implicit point $c :: a$) instead of pairs of functions. Thus we can redefine FD to be

```
type FD a = (a, a)
```

Hint: Something very similar can be used for Assignment 2.

6 Some helper functions

```
instance Num E where -- Some abuse of notation (no proper negate, etc.)
  (+) = Add
  (*) = Mul
  fromInteger = Con
  negate = negateE
  negateE (Con c) = Con (negate c)
  negateE _ = error "negate: not supported"
```

7 Week 5: Polynomials and Power Series

```
{-# LANGUAGE TypeSynonymInstances #-}
module DSLsofMath.W05 where
```

7.1 Preliminaries

Last time, we defined a *Num* structure on pairs $(Double, Double)$ by requiring the operations to be compatible with the interpretation $(f\ a, f'\ a)$. For example

$$(x, x') \otimes (y, y') = (x * y, x' * y + x * y')$$

There is nothing in the “nature” of pairs of *Double* that forces this definition upon us. We chose it, because of the intended interpretation.

This multiplication is obviously not the one we need for *complex numbers*:

$$(x, x') * (y, y') = (x * y - x' * y', x * y' + x' * y)$$

Again, there is nothing in the nature of pairs that foists this operation on us. In particular, it is, strictly speaking, incorrect to say that a complex number *is* a pair of real numbers. The correct interpretation is that a complex number can be *represented* by a pair of real numbers, provided we define the operations on these pairs in a suitable way.

The distinction between definition and representation is similar to the one between specification and implementation, and, in a certain sense, to the one between syntax and semantics. All these distinctions are frequently obscured, for example, because of prototyping (working with representations / implementations / concrete objects in order to find out what definition / specification / syntax is most adequate). They can also be context-dependent (one man’s specification is another man’s implementation). Insisting on the difference between definition and representation can also appear quite pedantic (as in the discussion of complex numbers above). In general though, it is a good idea to be aware of these distinctions, even if they are suppressed for reasons of brevity or style.

7.2 Polynomials

From Adams and Essex [2010], page 55:

A **polynomial** is a function P whose value at x is

$$Px = a_n x^n + a_{n-1} x^{n-1} + \dots a_1 x + a_0$$

where a_n, a_{n-1}, \dots, a_1 , and a_0 , called the **coefficients** of the polynomial [original spelling], are constants and, if $n > 0$, then $a_n \neq 0$. The number n , the degree of the highest power of x in the polynomial, is called the **degree** of the polynomial. (The degree of the zero polynomial is not defined.)

This definition raises a number of questions, for example “what is the zero polynomial?”.

The types of the elements involved in the definition appear to be

$$P : \mathbb{R} \rightarrow \mathbb{R}, x \in \mathbb{R}, a_0, \dots, a_n \in \mathbb{R} \text{ with } a_n \neq 0 \text{ if } n > 0$$

The phrasing should be “whose value at *any* x is”. The remark that the a_i are constants is probably meant to indicate that they do not depend on x , otherwise every function would be a polynomial. The zero polynomial is, according to this definition, the ‘const 0’ function. Thus, what is meant is

A **polynomial** is a function $P : \mathbb{R} \rightarrow \mathbb{R}$ which is either constant zero, or there exist $a_0, \dots, a_n \in \mathbb{R}$ with $a_n \neq 0$ such that, for any $x \in \mathbb{R}$

$$Px = a_n x^n + a_{n-1} x^{n-1} + \dots a_1 x + a_0$$

Obviously, given the coefficients a_i we can evaluate P at any given x . Assuming the coefficients are given as

$$as = [a_0, a_1, \dots, a_n]$$

(we prefer counting up), then the evaluation function is written

$$\begin{aligned} eval &:: [Real] \rightarrow Real \rightarrow Real \\ eval \quad [] \quad x &= 0 \\ eval \quad (a : as) \quad x &= a + x * eval \, as \, x \end{aligned}$$

Not every list is valid according to the definition. In particular, the empty list is not a valid list of coefficients, so we have a conceptual, if not empirical, type error in our evaluator.

The valid lists are those **finite** lists in the set

$$\{[0]\} \cup \{(a : as) \mid last \, (a : as) \neq 0\}$$

We cannot express the $last \, (a : as) \neq 0$ in Haskell, but we can express the condition that the list should not be empty:

$$\begin{aligned} \text{data } Poly \, a &= Single \, a \mid Cons \, a \, (Poly \, a) \\ &\text{deriving } (Eq, Ord) \end{aligned}$$

(TODO: show the version and motivation for using just $[a]$ as well. Basically, one can use $[]$ as the syntax for the “zero polynomial” and $(c : cs)$ for all other.)

The relationship between $Poly\ a$ and $[a]$ is given by the following functions:

```

toList :: Poly a → [a]
toList (Single a) = a : []
toList (Cons a as) = a : toList as

fromList :: [a] → Poly a
fromList (a : []) = Single a
fromList (a₀ : a₁ : as) = Cons a₀ (fromList (a₁ : as))

instance Show a ⇒ Show (Poly a) where
  show = show ∘ toList

```

Since we only use the arithmetical operations, we can generalise our evaluator:

```

evalPoly :: Num a ⇒ Poly a → a → a
evalPoly (Single a) x = a
evalPoly (Cons a as) x = a + x * evalPoly as x

```

Since we have $Num\ a$, there is a Num structure on $a \rightarrow a$, and $evalPoly$ looks like a homomorphism. Question: is there a Num structure on $Poly\ a$, such that $evalPoly$ is a homomorphism?

For example, the homomorphism condition gives for $(+)$

$$evalPoly\ as + evalPoly\ bs = evalPoly\ (as + bs)$$

Both sides are functions, they are equal iff they are equal for every argument. For an arbitrary x

$$\begin{aligned}
& (evalPoly\ as + evalPoly\ bs)\ x = evalPoly\ (as + bs)\ x \\
& \Leftrightarrow \{-\text{ } + \text{ on functions is defined point-wise -}\} \\
& evalPoly\ as\ x + evalPoly\ bs\ x = evalPoly\ (as + bs)\ x
\end{aligned}$$

To proceed further, we need to consider the various cases in the definition of $evalPoly$. We give here the computation for the last case (where as has at least one $Cons$), using the traditional list notation $(:)$ for brevity.

$$evalPoly\ (a : as)\ x + evalPoly\ (b : bs)\ x = evalPoly\ ((a : as) + (b : bs))\ x$$

For the left-hand side, we have:

$$\begin{aligned}
& evalPoly\ (a : as)\ x + evalPoly\ (b : bs)\ x \\
& = \{-\text{ def. } evalPoly \text{ -}\} \\
& (a + x * evalPoly\ as\ x) + (b + x * evalPoly\ bs\ x) \\
& = \{-\text{ properties of } +, \text{ valid in any ring -}\} \\
& (a + b) + x * (evalPoly\ as\ x + evalPoly\ bs\ x) \\
& = \{-\text{ homomorphism condition -}\} \\
& (a + b) + x * (evalPoly\ (as + bs)\ x) \\
& = \{-\text{ def. } evalPoly \text{ -}\} \\
& evalPoly\ ((a + b) : (as + bs))\ x
\end{aligned}$$

The homomorphism condition will hold for every x if we define

$$(a : as) + (b : bs) = (a + b) : (as + bs)$$

We leave the derivation of the other cases and operations as an exercise. Here, we just give the corresponding definitions.

```

instance Num a  $\Rightarrow$  Num (Poly a) where
  (+) = polyAdd
  (*) = polyMul
  negate = polyNeg
  fromInteger = Single  $\circ$  fromInteger

polyAdd :: Num a  $\Rightarrow$  Poly a  $\rightarrow$  Poly a  $\rightarrow$  Poly a
polyAdd (Single a) (Single b) = Single (a + b)
polyAdd (Single a) (Cons b bs) = Cons (a + b) bs
polyAdd (Cons a as) (Single b) = Cons (a + b) as
polyAdd (Cons a as) (Cons b bs) = Cons (a + b) (polyAdd as bs)

polyMul :: Num a  $\Rightarrow$  Poly a  $\rightarrow$  Poly a  $\rightarrow$  Poly a
polyMul (Single a) (Single b) = Single (a * b)
polyMul (Single a) (Cons b bs) = Cons (a * b) (polyMul (Single a) bs)
polyMul (Cons a as) (Single b) = Cons (a * b) (polyMul as (Single b))
polyMul (Cons a as) (Cons b bs) = Cons (a * b) (polyAdd (polyMul as (Cons b bs))
                                                         (polyMul (Single a) bs))

polyNeg :: Num a  $\Rightarrow$  Poly a  $\rightarrow$  Poly a
polyNeg = fmap negate

```

Therefore, we *can* define a ring structure (the mathematical counterpart of *Num*) on *Poly a*, and we have arrived at the canonical definition of polynomials, as found in any algebra book (see, for example, Rotman [2006] for a very readable text):

Given a commutative ring *A*, the commutative ring given by the set *Poly A* together with the operations defined above is the ring of **polynomials** with coefficients in *A*.

The functions *evalPoly as* are known as *polynomial functions*.

Caveat: The canonical representation of polynomials in algebra does not use finite lists, but the equivalent

$$\text{Poly}' A = \{ a : \mathbb{N} \rightarrow A \mid \{- a \text{ has only a finite number of non-zero values -} \} \}$$

Exercise: what are the ring operations on *Poly' A*? For example, here is addition:

$$a + b = c \Leftrightarrow a\ n + b\ n = c\ n \quad -- \ \forall n : \mathbb{N}$$

Observations:

1. Polynomials are not, in general, isomorphic (in one-to-one correspondence) with polynomial functions. For any finite ring *A*, there is a finite number of functions $A \rightarrow A$, but there is a countable number of polynomials. That means that the same polynomial function on *A* will be the evaluation of many different polynomials.

For example, consider the ring \mathbb{Z}_2 ($\{0, 1\}$ with addition and multiplication modulo 2). In this ring, we have

$$evalPoly [0, 1, 1] = const\ 0 = evalPoly [0] \{- \text{ in } \mathbb{Z}_2 \rightarrow \mathbb{Z}_2 \ -\}$$

but

$$[0, 1, 1] \neq [0] \{- \text{ in } Poly\ \mathbb{Z}_2 \ -\}$$

Therefore, it is not generally a good idea to confuse polynomials with polynomial functions.

2. In keeping with the DSL terminology, we can say that the polynomial functions are the semantics of the language of polynomials. We started with polynomial functions, we wrote the evaluation function and realised that we have the makings of a homomorphism. That suggested that we could create an adequate language for polynomial functions. Indeed, this turns out to be the case; in so doing, we have recreated an important mathematical achievement: the algebraic definition of polynomials.

Let

$$\begin{aligned} x &:: Num\ a \Rightarrow Poly\ a \\ x &= Cons\ 0\ (Single\ 1) \end{aligned}$$

Then (again, using the list notation for brevity) for any polynomial $as = [a_0, a_1, \dots, a_n]$ we have

$$as = a_0 + a_1 * x + a_2 * x^2 + \dots + a_n * x^n$$

Exercise: check this.

This justifies the standard notation

$$as = \sum_{i=0}^n a_i * x^i$$

7.3 Polynomial degree as a homomorphism

TODO: textify black board notes

It is often the case that a certain function is *almost* a homomorphism and the domain or range *almost* a monoid. In the section on *eval* and *eval'* for *FunExp* we have seen “tupling” as one way to fix such a problem and here we will introduce another way.

The *degree* of a polynomial is a good candidate for being a homomorphism: if we multiply two polynomials we can normally add their degrees. If we try to check that $degree :: Poly\ a \rightarrow \mathbb{N}$ is the function underlying a monoid morphism we need to decide on the monoid structure to use for the source and for the target, and we need to check the homomorphism laws. We can use $unit = Single\ 1$ and $op = polyMul$ for the source monoid and we can try to use $unit = 0$ and $op = (+)$ for the target monoid. Then we need to check that

$$\begin{aligned} degree\ (Single\ 1) &= 0 \\ \forall x, y. degree\ (x\ 'op'\ y) &= degree\ x + degree\ y \end{aligned}$$

The first law is no problem and for most polynomials the second law is also straightforward to prove (exercise: prove it) except for one special case: the zero polynomial.

Looking back at the definition from Adams and Essex [2010], page 55 it says that the degree of the zero polynomial is not defined. Let’s see why that is the case and how we might “fix” it. Assume there is a z such that $degree\ 0 = z$ and that we have some polynomial p with $degree\ p = n$. Then we get

$$\begin{aligned}
& z \\
&= \{- \text{assumption } -\} \\
&\quad \text{degree } 0 \\
&= \{- \text{simple calculation } -\} \\
&\quad \text{degree } (0 * p) \\
&= \{- \text{homomorphism condition } -\} \\
&\quad \text{degree } 0 + \text{degree } p \\
&= \{- \text{assumption } -\} \\
&\quad z + n
\end{aligned}$$

Thus we need to find a z such that $z = z + n$ for all natural numbers n ! At this stage we could either give up, or think out of the box. Intuitively we could try to use $z = -\text{Infinity}$, which would seem to satisfy the law but which is not a natural number. More formally what we need to do is to extend the monoid $(\mathbb{N}, 0, +)$ by one more element. In Haskell we can do that using the *Maybe* type constructor:

```

class Monoid a where
  unit :: a
  op    :: a -> a -> a
instance Monoid a => Monoid (Maybe a) where
  unit = Nothing
  op    = opMaybe
opMaybe Nothing m      = m
opMaybe m      Nothing = m
opMaybe (Just m1) (Just m2) = Just (op m1 m2)

```

We quote the Haskell prelude implementation:

Lift a semigroup into *Maybe* forming a *Monoid* according to <http://en.wikipedia.org/wiki/Monoid>: "Any semigroup S may be turned into a monoid simply by adjoining an element e not in S and defining $e * e = e$ and $e * s = s = s * e$ for all $s \in S$." Since there is no *Semigroup* typeclass [...], we use *Monoid* instead.

Thus, to sum up, *degree* is a monoid homomorphism from $(\text{Poly } a, 1, *)$ to $(\text{Maybe } \mathbb{N}, \text{Nothing}, \text{opMaybe})$.

TODO: check all the properties.

8 Power Series

Power series are obtained from polynomials by removing in *Poly'* the restriction that there should be a *finite* number of non-zero coefficients; or, in, the case of *Poly*, by going from lists to streams.

$$\text{PowerSeries}' a = \{f : \mathbb{N} \rightarrow a\}$$

```

type PowerSeries a = Poly a  -- finite and infinite non-empty lists

```

The operations are still defined as before. If we consider only infinite lists, then only the equations which do not contain the patterns for singleton lists will apply.

Power series are usually denoted

$$\sum_{n=0}^{\infty} a_n * x^n$$

the interpretation of x being the same as before.

The evaluation of a power series represented by $a : \mathbb{N} \rightarrow A$ is defined, in case the necessary operations make sense on A , as a function

$$\begin{aligned} eval\ a &: A \rightarrow A \\ eval\ a\ x &= \lim s \text{ where } s\ n = \sum_{i=0}^n a_i * x^i \end{aligned}$$

Note that $eval\ a$ is, in general, a partial function (the limit might not exist).

We will consider, as is usual, only the case in which $A = \mathbb{R}$ or $A = \mathbb{C}$.

The term *formal* refers to the independence of the definition of power series from the ideas of convergence and evaluation. In particular, two power series represented by a and b , respectively, are equal only if $a = b$ (as functions). If $a \neq b$, then the power series are different, even if $eval\ a = eval\ b$.

Since we cannot in general compute limits, we can use an “approximative” *eval*, by evaluating the polynomial resulting from an initial segment of the power series.

$$\begin{aligned} eval &:: Num\ a \Rightarrow Integer \rightarrow PowerSeries\ a \rightarrow (a \rightarrow a) \\ eval\ n\ as\ x &= evalPoly\ (takePoly\ n\ as)\ x \\ takePoly &:: Integer \rightarrow PowerSeries\ a \rightarrow Poly\ a \\ takePoly\ n\ (Single\ a) &= Single\ a \\ takePoly\ n\ (Cons\ a\ as) &= \text{if } n \leq 1 \\ &\quad \text{then } Single\ a \\ &\quad \text{else } Cons\ a\ (takePoly\ (n - 1)\ as) \end{aligned}$$

Note that $eval\ n$ is not a homomorphism: for example $eval\ 2\ (x * x)\ 1 = 0 \neq 1 = 1 * 1 = (eval\ 2\ x\ 1) * (eval\ 2\ x\ 1)$.

9 Operations on power series

Power series have a richer structure than polynomials. For example, we also have division (this is similar to the move from \mathbb{Z} to \mathbb{Q}). Assume that $a * b \neq 0$. Then (again, using list notation for brevity), we want to find, for any given $(a : as)$ and $(b : bs)$, the series $(c : cs)$ satisfying

$$\begin{aligned} (a : as) / (b : bs) &= (c : cs) \\ \Leftrightarrow \{- \text{def. of division} -\} \\ (a : as) &= (c : cs) * (b : bs) \\ \Leftrightarrow \{- \text{def. of } * \text{ for } Cons -\} \\ (a : as) &= (c * b) : (cs * (b : bs) + [c] * bs) \\ \Leftrightarrow \{- \text{equality on compnents, def. of division} -\} \\ c &= a / b \{- \text{and} -\} \\ as &= cs * (b : bs) + [a / b] * bs \\ \Leftrightarrow \{- \text{arithmetics} -\} \\ c &= a / b \{- \text{and} -\} \\ cs &= (as - [a / b] * bs) / (b : bs) \end{aligned}$$

This leads to the implementation:

$$\begin{aligned} \text{instance } (Eq\ a, Fractional\ a) &\Rightarrow Fractional\ (PowerSeries\ a) \text{ where} \\ (/) &= divPS \end{aligned}$$

```

fromRational = Single ∘ fromRational
divPS :: (Eq a, Fractional a) ⇒ PowerSeries a → PowerSeries a → PowerSeries a
divPS as (Single b) = as * Single (1 / b)
divPS (Single 0) (Cons b bs) = Single 0
divPS (Single a) (Cons b bs) = divPS (Cons a (Single 0)) (Cons b bs)
divPS (Cons a as) (Cons b bs) = Cons c (divPS (as - (Single c) * bs) (Cons b bs))
                                where c = a / b

```

The first two equations allow us to also use division on polynomials, but the result will, in general, be a power series, not a polynomial. The first one should be self-explanatory. The second one extends a constant polynomial, in a process similar to that of long division.

For example:

```

ps0, ps1, ps2 :: (Eq a, Fractional a) ⇒ PowerSeries a
ps0 = 1 / (1 - x)
ps1 = 1 / (1 - x)^2
ps2 = (x^2 - 2 * x + 1) / (x - 1)

```

Every *ps* is the result of a division of polynomials: the first two return power series, the third is a polynomial (almost: it has a trailing 0.0).

```

example0 = takePoly 10 ps0
example01 = takePoly 10 (ps0 * (1 - x))

```

10 Formal derivative

Considering the analogy between power series and polynomial functions (via polynomials), we can define a formal derivative for power series according to the formula

$$\left(\sum_{n=0}^{\infty} a_n * x^n\right)' = \sum_{n=0}^{\infty} (a_n * x^n)' = \sum_{n=0}^{\infty} (a_n * (n * x^{n-1})) = \sum_{n=0}^{\infty} ((n * a_n) * x^{n-1})$$

We can implement this, for example, as

```

deriv (Single a) = Single 0
deriv (Cons a as) = deriv' as 1
  where deriv' (Single a) n = Single (n * a)
        deriv' (Cons a as) n = Cons (n * a) (deriv' as (n + 1))

```

Side note: we cannot in general implement a Boolean equality test for *PowerSeries*. For example, we know that *deriv ps0* equals *ps1* but we cannot compute *True* in finite time by comparing the coefficients of the two power series.

```

checkDeriv :: Integer → Bool
checkDeriv n = takePoly n (deriv ps0) == takePoly n ps1

```

Recommended reading: the Functional pearl: “Power series, power serious” McIlroy [1999].

11 Signals and Shapes

Shallow and deep embeddings of a DSL

TODO: perhaps textify DSL/

12 Helpers

```
instance Functor Poly where
  fmap = fmapPoly
fmapPoly :: (a → b) → (Poly a → Poly b)
fmapPoly f (Single a)  = Single (f a)
fmapPoly f (Cons a as) = Cons (f a) (fmapPoly f as)
po1 :: Num a ⇒ Poly a
po1 = 1 + x^2 - 3 * x^4

{-# LANGUAGE FlexibleInstances #-}
{-# LANGUAGE TypeSynonymInstances #-}
module DSLsofMath.W06 where
import DSLsofMath.W05
```

13 Week 6

13.1 Review

- key notion *homomorphism*: $S1 \rightarrow S2$
- questions (“equations”):
 - $S1 \stackrel{?}{\leftarrow} S2$ what is the homomorphism between two given structures
- e.g., $apply : Num\ a \rightarrow Num\ (x \rightarrow a)$
 - $S1? \rightarrow S2$ what is $S1$ compatible with a given homomorphism
- e.g., $eval : Poly\ a \rightarrow (a \rightarrow a)$
 - $S1 \rightarrow S2?$ what is $S2$ compatible with a given homomorphism
- e.g., $applyFD : FD\ a \rightarrow (a, a)$
 - $S1 \stackrel{?}{\leftarrow} S2?$ can we find a good structure on $S2$ so that it becomes homomorphic w. $S1?$
- e.g., $evalD : FunExp \rightarrow FD\ a$
- The importance of the last two is that they offer “automatic differentiation”, i.e., any function constructed according to the grammar of *FunExp*, can be “lifted” to a function that computes the derivative (e.g., a function on pairs).

Example $f\ x = \sin\ x + 2 * x$

We have: $f\ 0 = 1$, $f\ 2 = 4.909297426825682$ s, etc.

The type of f is $f :: Floating\ a \Rightarrow a \rightarrow a$.

How do we compute, say, $f'\ 2$?

We have several choices.

1. Using *FunExp*

Recall (week 3):

```
data FunExp = Const Double
           | Id
           | FunExp :+: FunExp
           | FunExp *: FunExp
           | FunExp :/: FunExp
           | Exp FunExp
           | Sin FunExp
           | Cos FunExp
           -- and so on
deriving Show
```

What is the expression e for which $f = \text{eval } e$?

We have

$$\begin{aligned} \text{eval } e \ x &= f \ x \\ \Leftrightarrow \\ \text{eval } e \ x &= \sin x + 2 * x \\ \Leftrightarrow \\ \text{eval } e \ x &= \text{eval } (\text{Sin } \text{Id}) \ x + \text{eval } (\text{Const } 2 \text{ } *: \text{Id}) \ x \\ \Leftrightarrow \\ \text{eval } e \ x &= \text{eval } ((\text{Sin } \text{Id}) \text{ } :+: (\text{Const } 2 \text{ } *: \text{Id})) \ x \\ \Leftarrow \\ e &= \text{Sin } \text{Id} \text{ } :+: (\text{Const } 2 \text{ } *: \text{Id}) \end{aligned}$$

Finally, we can apply *derive* and obtain

$$f' \ 2 = \text{eval } (\text{derive } e) \ 2$$

This can hardly be called "automatic", look at all the work we did in deducing e !

However, consider this:

$$e = f \ \text{Id}$$

(Perhaps it would have been better to use, in the definition of *FunExp*, X instead of Id .)

In general, to find the value of the derivative of a function f at a given x , we can use

$$\text{drv } f \ x = \text{evalFunExp } (\text{derive } (f \ \text{Id})) \ x$$

2. Using *FD*

Recall

```
type FD a = (a → a, a → a)
applyFD (f, g) x = (f x, g x)
```

The operations on $\text{FD } a$ are such that, if $\text{eval } e = f$, then

$$(\text{eval } e, \text{eval}' e) = (f, f')$$

We are looking for (g, g') such that

$$f (g, g') = (f, f') \quad -- (*)$$

so we can then do

$$f' \ 2 = snd \ (applyFD \ (f \ (g, g')) \ 2)$$

We can fulfill $(*)$ if we can find a (g, g') that is a sort of “unit” for $FD \ a$:

$$\begin{aligned} sin \ (g, g') &= (sin, cos) \\ exp \ (g, g') &= (exp, exp) \end{aligned}$$

and so on.

In general, the chain rule gives us

$$f \ (g, g') = (f \circ g, (f' \circ g) * g')$$

Therefore, we need: $g = id$ and $g' = const \ 1$.

Finally

$$f' \ 2 = snd \ (applyFD \ (f \ (id, const \ 1)) \ 2)$$

In general

$$drvFD \ f \ x = snd \ (applyFD \ (f \ (id, const \ 1)) \ x)$$

computes the derivative of f at x .

$$\begin{aligned} f_1 &:: FD \ Double \rightarrow FD \ Double \\ f_1 &= f \end{aligned}$$

3. Using pairs

We have **instance** $Floating \ a \Rightarrow Floating \ (a, a)$, moreover, the instance declaration looks exactly the same as that for $FD \ a$:

$$\begin{aligned} &\mathbf{instance} \ Floating \ a \Rightarrow Floating \ (FD \ a) \ \mathbf{where} \\ &\quad exp \ (f, f') = (exp \ f, (exp \ f) * f') \\ &\quad sin \ (f, f') = (sin \ f, (cos \ f) * f') \\ &\quad cos \ (f, f') = (cos \ f, -(sin \ f) * f') \\ &\mathbf{instance} \ Floating \ a \Rightarrow Floating \ (a, a) \ \mathbf{where} \\ &\quad exp \ (f, f') = (exp \ f, (exp \ f) * f') \\ &\quad sin \ (f, f') = (sin \ f, cos \ f * f') \\ &\quad cos \ (f, f') = (cos \ f, -(sin \ f) * f') \end{aligned}$$

In fact, the latter represents a generalisation of the former. It is also the “maximally general” such generalisation (discounting the “noise” generated by the less-than-clean design of *Num*, *Fractional*, *Floating*).

Still, we need to use this machinery. We are now looking for a pair of values (g, g') such that

$$f \ (g, g') = (f \ 2, f' \ 2)$$

In general

$$f (g, g') = (f \ g, (f' \ g) * g')$$

Therefore

$$\begin{aligned} f (g, g') &= (f \ 2, f' \ 2) \\ \Leftrightarrow \\ (f \ g, (f' \ g) * g') &= (f \ 2, f' \ 2) \\ \Leftarrow \\ g = 2, g' &= 1 \end{aligned}$$

Introducing

$$var \ x = (x, 1)$$

we can, as in the case of *FD*, simplify matters a little:

$$f' \ x = snd \ (f \ (var \ x))$$

In general

$$drvP \ f \ x = snd \ (f \ (x, 1))$$

computes the derivative of f at x .

$$\begin{aligned} f_2 &:: (Double, Double) \rightarrow (Double, Double) \\ f_2 &= f \end{aligned}$$

13.2 Higher-order derivatives

Consider

$$[f, f', f'', \dots]$$

representing the evaluation of an expression and its derivatives:

$$evalAll \ e = (evalFunExp \ e) : evalAll \ (derive \ e)$$

Notice that, if

$$[f, f', f'', \dots] = evalAll \ e$$

then

$$[f', f'', \dots] = evalAll \ (derive \ e)$$

We want to define the operations on lists of functions in such a way that *evalAll* is a homomorphism. For example:

$$evalAll \ (e_1 :*: e_2) = evalAll \ e_1 * evalAll \ e_2$$

where the $(*)$ sign stands for the multiplication of infinite lists of functions, the operation we are trying to determine.

We have, writing *eval* for *evalFunExp* in order to save ink

$$\begin{aligned}
& evalAll (e_1 :*: e_2) = evalAll e_1 * evalAll e_2 \\
\Leftrightarrow & \\
& eval (e_1 :*: e_2) : evalAll (derive (e_1 :*: e_2)) = \\
& eval e_1 : evalAll (derive e) * eval e_1 : evalAll (derive e_2) \\
\Leftrightarrow & \\
& (eval e_1 * eval e_2) : evalAll (derive (e_1 :*: e_2)) = \\
& eval e_1 : evalAll (derive e) * eval e_1 : evalAll (derive e_2) \\
\Leftrightarrow & \\
& (eval e_1 * eval e_2) : evalAll (derive e_1 :*: e_2 :+: e_1 * derive e_2) = \\
& eval e_1 : evalAll (derive e) * eval e_1 : evalAll (derive e_2) \\
\Leftarrow & \\
& (a : as) * (b : bs) = (a * b) : (as * (b : bs) + (a : as) * bs)
\end{aligned}$$

The final line represents the definition of $(*)$ needed for ensuring the conditions are met.

As in the case of pairs, we find that we do not need any properties of functions, other than their *Num* structure, so the definitions apply to any infinite list of *Num a*:

```

instance Num a => Num [a] where
  (a : as) + (b : bs) = (a + b) : (as + bs)
  (a : as) * (b : bs) = (a * b) : (as * (b : bs) + (a : as) * bs)

```

Exercise: complete the instance declarations for *Fractional* and *Floating*. Write a general derivative computation, similar to *dru* functions above:

```

druList k f x = undefined -- kth derivative of f at x

```

This is a very inefficient way of computing derivatives!

13.3 Polynomials

```

data Poly a = Single a | Cons a (Poly a)
           deriving (Eq, Ord)
evalPoly :: Num a => Poly a -> a -> a
evalPoly (Single a) x = a
evalPoly (Cons a as) x = a + x * evalPoly as x

```

13.4 Power series

No need for a separate type in Haskell

```

type PowerSeries a = Poly a -- finite and infinite non-empty lists

```

Now we can divide, as well as add and multiply.

We can also derive:

```

deriv (Single a) = Single 0
deriv (Cons a as) = deriv' as 1

```

where $\text{deriv}' (\text{Single } a) \quad n = \text{Single } (n * a)$
 $\text{deriv}' (\text{Cons } a \text{ as}) \quad n = \text{Cons } (n * a) (\text{deriv}' \text{ as } (n + 1))$

and integrate:

$\text{integ} :: \text{Fractional } a \Rightarrow \text{PowerSeries } a \rightarrow a \rightarrow \text{PowerSeries } a$
 $\text{integ } \text{as } a_0 = \text{Cons } a_0 (\text{integ}' \text{ as } 1)$
where $\text{integ}' (\text{Single } a) \quad n = \text{Single } (a / n)$
 $\text{integ}' (\text{Cons } a \text{ as}) \quad n = \text{Cons } (a / n) (\text{integ}' \text{ as } (n + 1))$

Everything here makes sense, irrespective of convergence, hence “formal”.

If the power series involved do converge, then *eval* is a morphism between the formal structure and that of the functions represented:

$\text{eval } \text{as} + \text{eval } \text{bs} = \text{eval } (\text{as} + \text{bs})$
 $\text{eval } \text{as} * \text{eval } \text{bs} = \text{eval } (\text{as} * \text{bs})$
 $\text{eval } (\text{derive } \text{as}) = D (\text{eval } \text{as})$
 $\text{eval } (\text{integ } \text{as } c) \, x = \int_0^x (\text{eval } \text{as } t) \, dt + c$

13.5 Simple differential equations

Many first-order differential equations have the structure

$$f' \, x = g \, f \, x, \quad f \, 0 = f_0$$

i.e., they are defined in terms of *g*.

The fundamental theorem of calculus gives us

$$f \, x = \int_0^x (g \, f \, t) \, dt + f_0$$

If $f = \text{eval } \text{as}$

$$\text{eval } \text{as } x = \int_0^x (g (\text{eval } \text{as}) \, t) \, dt + f_0$$

Assuming that *g* is a polymorphic function that commutes with *eval*

$$\begin{aligned} \text{eval } \text{as } x &= \int_0^x (\text{eval } (g \, \text{as}) \, t) \, dt + f_0 \\ \text{eval } \text{as } x &= \text{eval } (\text{integ } (g \, \text{as}) \, f_0) \, x \\ \text{as} &= \text{integ } (g \, \text{as}) \, f_0 \end{aligned}$$

Which functions *g* commute with *eval*? All the ones in *Num*, *Fractional*, *Floating*, by construction; additionally, as above, *deriv* and *integ*.

Therefore, we can implement a general solver for these simple equations:

$\text{solve} :: \text{Fractional } a \Rightarrow (\text{PowerSeries } a \rightarrow \text{PowerSeries } a) \rightarrow a \rightarrow \text{PowerSeries } a$
 $\text{solve } g \, f_0 = f \quad \text{-- solves } f' = g \, f, f \, 0 = f_0$
where $f = \text{integ } (g \, f) \, f_0$
 $\text{id}x = \text{solve } (\lambda f \rightarrow 1) \, 0$
 $\text{id}f = \text{eval } 100 \, \text{id}x$

```

expx = solve (λf → f) 1
expf = eval 100 expx
sinx = solve (λf → cosx) 0
cosx = solve (λf → -sinx) 1
sinf = eval 100 sinx
cosf = eval 100 cosx
idx, expx, sinx, cosx :: Fractional a ⇒ PowerSeries a
idf, expf, sinf, cosf :: Fractional a ⇒ a → a

```

13.6 The Floating structure of PowerSeries

Can we compute *exp as*?

Specification:

$$\text{eval } (\text{exp } as) = \text{exp } (\text{eval } as)$$

Differentiating both sides, we obtain

$$\begin{aligned}
D (\text{eval } (\text{exp } as)) &= \text{exp } (\text{eval } as) * D (\text{eval } as) \\
&\Leftrightarrow \{- \text{eval morphism } -\} \\
&\text{eval } (\text{deriv } (\text{exp } as)) = \text{eval } (\text{exp } as * \text{deriv } as) \\
&\Leftarrow \\
&\text{deriv } (\text{exp } as) = \text{exp } as * \text{deriv } as
\end{aligned}$$

Adding the “initial condition” $\text{eval } (\text{exp } as) 0 = \text{exp } (\text{head } as)$, we obtain

$$\text{exp } as = \text{integ } (\text{exp } as * \text{deriv } as) (\text{exp } (\text{val } as))$$

Note: we cannot use *solve* here, because the *g* function uses both *exp as* and *as* (it “looks inside” its argument).

instance (*Eq a, Floating a*) ⇒ *Floating (PowerSeries a)* **where**

```

π      = Single π
exp fs = integ (exp fs * deriv fs) (exp (val fs))
sin fs = integ (cos fs * deriv fs) (sin (val fs))
cos fs = integ (-sin fs * deriv fs) (cos (val fs))
val :: PowerSeries a → a
val (Single a)      = a
val (Cons a as)     = a

```

In fact, we can implement *all* the operations needed for evaluating *FunExp* functions as power series!

```

evalP :: FunExp → PowerSeries Double
evalP (Const x) = Single x
evalP (e1 :+: e2) = evalP e1 + evalP e2
evalP (e1 **: e2) = evalP e1 * evalP e2
evalP (e1 :/: e2) = evalP e1 / evalP e2
evalP Id         = idx
evalP (Exp e)    = exp (evalP e)
evalP (Sin e)    = sin (evalP e)
evalP (Cos e)    = cos (evalP e)

```

13.7 Taylor series

If $f = eval [a_0, a_1, \dots, a_n, \dots]$, then

$$\begin{aligned}
 f \ 0 &= a_0 \\
 f' &= eval (deriv [a_0, a_1, \dots, a_n, \dots]) \\
 &= eval ([a_1, 2 * a_2, 3 * a_3, \dots, n * a_n, \dots]) \\
 \Rightarrow \\
 f' \ 0 &= a_1 \\
 f'' &= eval (deriv [a_1, 2 * a_2, \dots, n * a_n, \dots]) \\
 &= eval ([2 * a_2, 3 * 2 * a_3, \dots, n * (n - 1) * a_n, \dots]) \\
 \Rightarrow \\
 f'' \ 0 &= 2 * a_2
 \end{aligned}$$

In general:

$$f^{(k)} 0 = fact \ k * a_k$$

Therefore

$$f = eval [f \ 0, f' \ 0, f'' \ 0 / 2, \dots, f^{(n)} 0 / (fact \ n), \dots]$$

The series $[f \ 0, f' \ 0, f'' \ 0 / 2, \dots, f^{(n)} 0 / (fact \ n), \dots]$ is called the Taylor series centred in 0, or the Maclaurin series.

Therefore, if we can represent f as a power series, we can find the value of all derivatives of f at 0!

```

derivs :: Num a => PowerSeries a -> PowerSeries a
derivs as = derivs1 as 0 1
  where
    derivs1 (Cons a as) n factn = Cons (a * factn)
                                         (derivs1 as (n + 1) (factn * (n + 1)))
    derivs1 (Single a)   n factn = Single (a * factn)
    -- remember that x = Cons 0 (Single 1)
    ex3 = takePoly 10 (derivs (x^3 + 2 * x))
    ex4 = takePoly 10 (derivs sinx)

```

In this way, we can compute all the derivatives at 0 for all functions f constructed with the grammar of *FunExp*. That is because, as we have seen, we can represent all of them by power series!

What if we want the value of the derivatives at $a \neq 0$?

We then need the power series of the “shifted” function g :

$$g \ x = f \ (x + a) \Leftrightarrow g = f \circ (+a)$$

If we can represent g as a power series, say $[b_0, b_1, \dots]$, then we have

$$g^{(k)} 0 = fact \ k * b_k = f^{(k)} a$$

In particular, we would have

$$f \ x = g \ (x - a) = \sum b_n * (x - a)^n$$

which is called the Taylor expansion of f at a .

Example:

We have that $idx = [0, 1]$, thus giving us indeed the values

$$[id\ 0, id'\ 0, id''\ 0, \dots]$$

In order to compute the values of

$$[id\ a, id'\ a, id''\ a, \dots]$$

for $a \neq 0$, we compute

$$ida\ a = takePoly\ 10\ (derivs\ (evalP\ (Id\ :+:\ Const\ a)))$$

More generally, if we want to compute the derivative of a function f constructed with *FunExp* grammar, at a point a , we need the power series of $g\ x = f\ (x + a)$:

$$d\ f\ a = takePoly\ 10\ (derivs\ (evalP\ (f\ (Id\ :+:\ Const\ a))))$$

Use, for example, our $f\ x = \sin\ x + 2 * x$ above.

As before, we can use directly power series:

$$dP\ f\ a = takePoly\ 10\ (derivs\ (f\ (idx + Single\ a)))$$

13.8 Associated code

instance *Num* $a \Rightarrow Num\ (x \rightarrow a)$ **where**

$$f + g = \lambda x \rightarrow f\ x + g\ x$$

$$f - g = \lambda x \rightarrow f\ x - g\ x$$

$$f * g = \lambda x \rightarrow f\ x * g\ x$$

$$negate\ f = negate \circ f$$

$$|f| = |\cdot| \circ f$$

$$signum\ f = signum \circ f$$

$$fromInteger = const \circ fromInteger$$

instance *Fractional* $a \Rightarrow Fractional\ (x \rightarrow a)$ **where**

$$recip\ f = recip \circ f$$

$$fromRational = const \circ fromRational$$

instance *Floating* $a \Rightarrow Floating\ (x \rightarrow a)$ **where**

$$\pi = const\ \pi$$

$$exp\ f = exp \circ f$$

$$sin\ f = sin \circ f$$

$$cos\ f = cos \circ f$$

$$f ** g = \lambda x \rightarrow (f\ x) ** (g\ x)$$

-- and so on

$$evalFunExp :: FunExp \rightarrow Double \rightarrow Double$$

$$evalFunExp\ (Const\ \alpha) = const\ \alpha$$

$$evalFunExp\ Id = id$$

$$evalFunExp\ (e_1\ :+:\ e_2) = evalFunExp\ e_1 + evalFunExp\ e_2 \quad \text{-- note the use of "lifted +"}$$

$$evalFunExp\ (e_1\ :*\ e_2) = evalFunExp\ e_1 * evalFunExp\ e_2 \quad \text{-- "lifted *"}$$

$$evalFunExp\ (Exp\ e_1) = exp\ (evalFunExp\ e_1) \quad \text{-- and "lifted exp"}$$

```

evalFunExp (Sin e1)  = sin (evalFunExp e1)
evalFunExp (Cos e1)  = cos (evalFunExp e1)
-- and so on

derive (Const α) = Const 0
derive Id         = Const 1
derive (e1 :+ e2) = derive e1 :+ derive e2
derive (e1 :* e2) = (derive e1 :* e2) :+ (e1 :* derive e2)
derive (Exp e)    = Exp e :* derive e
derive (Sin e)    = Cos e :* derive e
derive (Cos e)    = Const (-1) :* Sin e :* derive e

instance Num FunExp where
  (+) = (:+)
  (*) = (:*)
  fromInteger n = Const (fromInteger n)

instance Fractional FunExp where
  (/) = (:/)

instance Floating FunExp where
  exp = Exp
  sin = Sin

```

13.8.1 Not included to avoid overlapping instances

```

instance Num a ⇒ Num (FD a) where
  (f, f') + (g, g') = (f + g, f' + g')
  (f, f') * (g, g') = (f * g, f' * g + f * g')
  fromInteger n = (fromInteger n, const 0)

instance Fractional a ⇒ Fractional (FD a) where
  (f, f') / (g, g') = (f / g, (f' * g - g' * f) / (g * g))

instance Floating a ⇒ Floating (FD a) where
  exp (f, f') = (exp f, (exp f) * f')
  sin (f, f') = (sin f, (cos f) * f')
  cos (f, f') = (cos f, -(sin f) * f')

```

13.8.2 This is included instead

```

instance Num a ⇒ Num (a, a) where
  (f, f') + (g, g') = (f + g, f' + g')
  (f, f') * (g, g') = (f * g, f' * g + f * g')
  fromInteger n = (fromInteger n, fromInteger 0)

instance Fractional a ⇒ Fractional (a, a) where
  (f, f') / (g, g') = (f / g, (f' * g - g' * f) / (g * g))

instance Floating a ⇒ Floating (a, a) where
  exp (f, f') = (exp f, (exp f) * f')
  sin (f, f') = (sin f, cos f * f')
  cos (f, f') = (cos f, -(sin f) * f')

```

```

{-# LANGUAGE FlexibleInstances #-}
{-# LANGUAGE UndecidableInstances #-}
module DSLsofMath.W07 where

```


14 Matrix algebra and linear transformations

Often, especially in engineering textbooks, one encounters the “definition”: a vector is an $n + 1$ -tuple of real or complex numbers, arranged as a column:

$$\mathbf{v} = \begin{pmatrix} v_0 \\ \vdots \\ v_n \end{pmatrix}$$

Other times, this is supplemented by the definition of a “row vector”:

$$\mathbf{v} = v_0, \dots, v_n$$

The v_i s are real or complex numbers, or, more generally, elements of a *field* (analogous to being an instance of *Fractional*). Vectors can be added “point-wise” and multiplied with “scalars”, i.e., elements of the field:

$$\begin{aligned} \mathbf{v} + \mathbf{w} &= \begin{pmatrix} v_0 \\ \vdots \\ v_n \end{pmatrix} + \begin{pmatrix} w_0 \\ \vdots \\ w_n \end{pmatrix} = \begin{pmatrix} v_0 + w_0 \\ \vdots \\ v_n + w_n \end{pmatrix} \\ \mathbf{s} * \mathbf{v} &= \begin{pmatrix} s*v_0 \\ \vdots \\ s*v_n \end{pmatrix} \end{aligned}$$

The scalar s “scales” all the components of \mathbf{v} .

In fact, the most important feature of vectors is that they can be *uniquely* expressed as a simple sort of combination of other vectors:

$$\mathbf{v} = \begin{pmatrix} v_0 \\ \vdots \\ v_n \end{pmatrix} = v_0 \begin{pmatrix} 1 \\ 0 \\ \vdots \\ 0 \end{pmatrix} + \dots + v_n \begin{pmatrix} 0 \\ \vdots \\ 1 \end{pmatrix}$$

We denote by

$$\mathbf{e}_k = \begin{pmatrix} 0 \\ \vdots \\ 1 \text{ <-- position k} \\ \vdots \\ 0 \end{pmatrix}$$

the vector that is everywhere 0 except at position k , where it is 1, so that

$$v = v_0 * e0 + \dots + v_n * en$$

There is a temptation to model vectors by lists or tuples, but the simplest (at least conceptually) way is to view them as *functions*:

```
type S          = ...  -- the scalars, forming a field ( $\mathbb{R}$ , or Complex, or Zn, etc.)
type Vector G = G → S
```

Usually, G is finite, i.e., *Bounded* and *Enumerable*.

We know from the previous lectures that if S is an instance of *Num*, *Fractional*, etc. then so is $G \rightarrow S$, with the pointwise definitions. In particular, the instance declarations for $+$, multiplication, and embedding of constants, give us exactly the structure needed for the vector operations. For example

```
      s * v
= {- s is promoted to a function -}
      const s * v
= {- Num instance definition -}
      λg → (const s) g * v g
= {- definition of const -}
      λg → s * v g
```

The set G is typically $\{0, 1, \dots, n\}$. The basis vectors are then

$$e\ i : G \rightarrow S, e\ i\ g = i\ 'is'\ g$$

Implementation:

```
is a b = if a == b then 1 else 0
e g = λ(G g') → g 'is' g'
toL v = [v g | g ← [minBound..maxBound]]  -- so we can actually see them
```

and every

$$v : G \rightarrow S$$

is trivially a linear combination of $e\ is$:

$$v = v\ 0 * e\ 0 + \dots + v\ n * e\ n$$

14.1 Functions on vectors

What if we have another vector space, $Vector\ G' = G' \rightarrow S$? We are interested in functions $f : Vector\ G \rightarrow Vector\ G'$:

$$f\ v = f\ (v\ 0 * e\ 0 + \dots + v\ n * e\ n)$$

A “good” function should translate the operations in $Vector\ G$ into operations in $Vector\ G'$, i.e., should be a homomorphism:

$$f\ v = f\ (v\ 0 * e\ 0 + \dots + v\ n * e\ n) = v\ 0 * f\ (e\ 0) + \dots + v\ n * f\ (e\ n)$$

But this means that we can determine the values of $f : (G \rightarrow S) \rightarrow (G' \rightarrow S)$ from just the values of $f \circ e : G \rightarrow (G' \rightarrow S)$, a much “smaller” function. Let $m = f \circ e$. Then

$$f\ v = v\ 0 * m\ 0 + \dots + v\ n * m\ n$$

Each of $m\ k$ is a *Vector* G' , as is the resulting $f\ v$. We have

$$\begin{aligned} & f\ v\ g' \\ &= \{- \text{ as above } -\} \\ & \quad v\ 0 * m\ 0 + \dots + v\ n * m\ n \\ &= \{- * \text{ and } + \text{ for functions are def. pointwise } -\} \\ & \quad v\ 0 * m\ 0\ g' + \dots + v\ n * m\ n\ g' \\ &= \{- \text{ using } \textit{sum} -\} \\ & \quad \textit{sum}\ [v\ i * m\ i\ g' \mid i \leftarrow [\textit{minBound} .. \textit{maxBound}]] \end{aligned}$$

Implementation:

This is the almost the standard “vector-matrix” multiplication:

$$M = [m\ 0 \mid \dots \mid m\ n]$$

The columns of M are the images of the canonical base vectors $e\ i$ through f . Every $m\ k$ has *card* G' rows, and it has become standard to use $M\ i\ j$ to mean the i th element of the j th column, i.e., $m\ j\ i$, so that

$$(M * v)\ i = \textit{sum}\ [M\ i\ j * v\ j \mid j \leftarrow [0 .. n]]$$

$$\textit{mul}\ m\ v\ g' = \textit{sum}\ [m\ g'\ g * v\ g \mid g \leftarrow [\textit{minBound} .. \textit{maxBound}]]$$

Example:

$$(M * e\ k)\ i = \textit{sum}\ [M\ i\ j * e\ k\ j \mid j \leftarrow [0 .. n]] = \textit{sum}\ [M\ i\ k] = M\ i\ k$$

i.e., $e\ k$ extracts the k th column from M (hence the notation “e” for “extract”).

Given an arbitrary matrix M , we can define

$$f\ v = M * v$$

and obtain a linear transformation $(M*)$. Moreover $((M*) \circ e)\ g\ g' = M\ g'\ g$, i.e., the matrix constructed as above for f is precisely M .

Exercise: compute $((M*) \circ e)\ g\ g'$.

Therefore, every linear transformation is of the form $(M*)$ and every $(M*)$ is a linear transformation.

Matrix-matrix multiplication is defined in order to ensure that

$$(M' * M) * v = M' * (M * v)$$

that is

$$((M' * M)*) = (M'*) \circ (M*)$$

Exercise: work this out in detail.

Exercise: show that matrix-matrix multiplication is associative.

Perhaps the simplest vector space is obtained for $G = ()$, the singleton set. In this case, the vectors $s : () \rightarrow S$ are functions that can take exactly one argument, therefore have exactly one value: $s ()$, so they are often identified with S . But, for any $v : G \rightarrow S$, we have a function $fv : G \rightarrow (() \rightarrow S)$, namely

$$fv \ g \ () = v \ g$$

fv is similar to our m function above. The associated matrix is

$$M = [m \ 0 \mid \dots \mid m \ n] = [fv \ 0 \mid \dots \mid fv \ n]$$

having $n + 1$ columns (the dimension of *Vector* G) and one row (dimension of *Vector* $()$). Let $w :: \text{Vector } G$:

$$M * w = w \ 0 * fv \ 0 + \dots + w \ n * fv \ n$$

$M * v$ and each of the $fv \ k$ are “almost scalars”: functions of type $() \rightarrow S$, thus, the only component of $M * w$ is

$$(M * w) \ () = w \ 0 * fv \ 0 \ () + \dots + w \ n * fv \ n \ () = w \ 0 * v \ 0 + \dots + w \ n * v \ n$$

i.e., the scalar product of v and w .

Remark: I have not discussed the geometrical point of view. For the connection between matrices, linear transformations, and geometry, I warmly recommend binge-watching the “Essence of linear algebra” videos on youtube (start here: <https://www.youtube.com/watch?v=kjBOesZCoqc>).

14.2 Examples of matrix algebra

14.2.1 Derivative

We have represented polynomials of degree $n + 1$ by the list of their coefficients. This is quite similar to “standard” geometrical vectors represented by $n + 1$ coordinates. This suggests that polynomials of degree $n + 1$ form a vector space, and we could interpret that as $\{0, \dots, n\} \rightarrow \mathbb{R}$ (or, more generally, *Field* $a \Rightarrow \{0, \dots, n\} \rightarrow a$. The operations $+$ and $*$ are defined in the same way as they are for functions.

The *derive* function takes polynomials of degree $n + 1$ to polynomials of degree n , and since $D(f + g) = Df + Dg$ and $D(s * f) = s * Df$, we expect it to be a linear transformation. What is its associated matrix?

To answer that, we must first determine the canonical base vectors. As for geometrical vectors, they are

$$e \ i : \{0, \dots, n\} \rightarrow \text{Real}, e \ i \ j = i \text{ 'is' } j$$

The evaluation of $e \ i$ returns the function $\lambda x \rightarrow x^i$, as expected.

The associated matrix will be

$$M = [D(e_0), D(e_1), \dots, D(e_n)]$$

where each $D(e_i)$ has length n . Vector e_{i+1} represents x^{i+1} , therefore

$$D(e_{i+1}) = (i+1) * x^i$$

i.e.

$$D(e_{i+1})[j] = \text{if } i == j \text{ then } i+1 \text{ else } 0$$

and

$$D(e_0) = 0$$

Example: $n+1 = 3$:

$$M = \begin{pmatrix} 0 & 1 & 0 \\ 0 & 0 & 2 \end{pmatrix}$$

Take the polynomial

$$3 * X^2 + 2 * X + 1$$

as a vector

$$v = \begin{pmatrix} 1 \\ 2 \\ 3 \end{pmatrix}$$

and we have

$$M * v = \begin{pmatrix} 0 & 1 & 0 \\ 0 & 0 & 2 \end{pmatrix} * \begin{pmatrix} 1 \\ 2 \\ 3 \end{pmatrix} = \begin{pmatrix} 2 \\ 6 \end{pmatrix}$$

representing the polynomial $6 * X^2 + 2 * X$.


Exercise: write the (infinite-dimensional) matrix representing D for power series.

Exercise: write the matrix associate with integration of polynomials.

14.2.2 Simple deterministic systems (transition systems)

Simple deterministic systems are given by endo-functions on a finite set $f : G \rightarrow G$. They can often be conveniently represented as a graph, for example

$$\begin{array}{c} +---+ \\ | \ 3 \ | \\ +---+ \\ \wedge \qquad \backslash \\ / \qquad \qquad v \end{array}$$



Here, $G = \{0, \dots, 6\}$. A node in the graph represents a state. A transition $i \rightarrow j$ means $f(i) = j$. Since f is an endo-function, every node must be the source of exactly one arrow.

We can take as vectors the characteristic functions of subsets of G , i.e., $G \rightarrow \{0, 1\}$. $\{0, 1\}$ is not a field w.r.t. the standard arithmetical operations (it is not even closed w.r.t. addition), and the standard trick to avoid this is to extend the type of the functions to \mathbb{R} .

The canonical basis vectors are, as usual, $e_i = \lambda_j \rightarrow i \text{ 'is' } j$. Each e_i is the characteristic function of a singleton set, $\{i\}$. Thus, the inputs to f are canonical vectors.

To write the matrix associated to f , we have to compute what vector is associated to each canonical base vector vector:

$$M = [f(e_0), f(e_1), \dots, f(e_n)]$$

Therefore:

$$M = \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 1 & 1 & 0 & 0 \end{bmatrix}$$

Starting with a canonical base vector e_i , we obtain $M * e_i = e(f(i))$, as we would expect.

It is more interesting if we start with a non-base vector. For example, $e_2 + e_4$, which represents the subset $\{2, 4\}$.

The more interesting thing is if we start with something different from a basis vector, say $[0, 1, 1]$. We obtain $\{f(2), f(4)\} = \{5, 6\}$, the image of $\{2, 4\}$ through f . In a sense, we can say that the two computations were done in parallel. But that is not quite accurate: if start with $\{3, 4\}$, we no longer get the characteristic function of $\{f(3), f(4)\} = \{6\}$, instead, we get a vector that does not represent a characteristic function at all: $[0, 0, 0, 0, 0, 0, 2]$. In general, if we start with an arbitrary vector, we can interpret this as starting with various quantities of some unspecified material in each state, simultaneously. If f were injective, the respective quantities would just gets shifted around, but in our case, we get a more interesting behaviour.

What if we do want to obtain the characteristic function of the image of a subset? In that case, we need to use other operations than the standard arithmetical ones, for example \min and \max . The problem is that $(\{0, 1\}, \max, \min)$ is not a field, and neither is (\mathbb{R}, \max, \min) . This is not a

problem if all we want is to compute the evolutions of possible states, but we cannot apply most of the “deper” results of linear algebra.

In the example above, we have:

```

newtype  $G = G$  Int deriving (Eq, Show)
instance Bounded  $G$  where
    minBound =  $G$  0
    maxBound =  $G$  6
instance Enum  $G$  where
    toEnum      =  $G$ 
    fromEnum ( $G$   $n$ ) =  $n$ 

```

The transition function:

```

 $f_1$  0 = 1
 $f_1$  1 = 3
 $f_1$  2 = 5
 $f_1$  3 = 6
 $f_1$  4 = 6
 $f_1$  5 = 4
 $f_1$  6 = 5

```

The associated matrix:

$$m_1 (G \ g') (G \ g) = g' \text{ 'is' } f_1 \ g$$

Test:


$$t_1 = toL (mul \ m_1 \ (e \ 3 + e \ 4))$$


14.2.3 Non-deterministic systems

Another interpretation of the application of M to characteristic functions of a subset is the following: assuming that all I know is that the system is in one of the states of the subset, where can it end up after one step? (this assumes the *max-min* algebra as above).

In this case, the uncertainty is entirely caused by the fact that we do not know the exact initial state. However, there are cases in which the output of f is not known, even when the input is known. Such situations are modelled by endo-relations: $R: G \rightarrow G$, with $g \ R \ g'$ if g' is a potential successor of g . Endo-relations can also be pictured as graphs, but the restriction that every node should be the source of exactly one arrow is lifted. Every node can be the source of one, none, or many arrows.

For example:





Now, starting in 0 we might go up either in 1 or 2 (but not both!). Starting in 6, the system breaks down: there is no successor state.

The matrix associated to R is built in the same fashion: we need to determine what vectors the canonical base vectors are associated with:

$$\begin{array}{rcl}
 & & \begin{matrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
 & & 1 & 0 & 0 & 0 & 1 & 0 & 0 \\
 & & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
 \mathbf{M} & = & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\
 & & 0 & 0 & 1 & 0 & 0 & 1 & 0 \\
 & & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\
 & & 0 & 0 & 0 & 1 & 1 & 0 & 0 \end{matrix}
 \end{array}$$

Exercise: start with $e_2 + e_3$ and iterate a number of times, to get a feeling for the possible evolutions. What do you notice? What is the largest number of steps you can make before the result is the origin vector? Now invert the arrow from 2 to 4 and repeat the exercise. What changes? Can you prove it?

Implementation:

The transition function has type $G \rightarrow (G \rightarrow Bool)$:

```

f2 0 g = g == 1 ∨ g == 2
f2 1 g = g == 3
f2 2 g = g == 4 ∨ g == 5
f2 3 g = g == 6
f2 4 g = g == 1 ∨ g == 6
f2 5 g = g == 4
f2 6 g = False

```

The associated matrix:

$$m_2 (G \ g') (G \ g) = f_2 \ g \ g'$$

We need a *Num* instance for *Bool* (not a field!):

```

instance Num Bool where
  (+) = (∨)
  (*) = (∧)
  fromInteger 0 = False
  fromInteger 1 = True
  negate      = ¬

```



$$\begin{aligned} |\cdot| &= id \\ \text{signum} &= id \end{aligned}$$

Test:

$$t_2 = toL (mul\ m_2\ (e\ 3 + e\ 4))$$

14.2.4 Stochastic systems

Quite often, we have more information about the transition to possible future states. In particular, we can have *probabilities* of these transitions. For example



One could say that this case is a generalisation of the previous one, in which we can take all probabilities to be equally distributed among the various possibilities. While this is plausible, it is not entirely correct. For example, we have to introduce a transition from state 6 above. The nodes must be sources of *at least* one arrow.

In the case of the non-deterministic example, the “legitimate” inputs were characteristic functions, i.e., the “vector space” was $G \rightarrow \{0, 1\}$ (the scare quotes are necessary because, as discussed, the target is not a field). In the case of stochastic systems, the inputs will be *probability distributions* over G , that is, functions $p : G \rightarrow [0, 1]$ with the property that

$$\text{sum } [p \ g \mid g \leftarrow [0..6]] = 1$$

If we know the current probability distributions over states, then we can compute the next one by using the *total probability formula*, normally expressed as

$$p\ a = \text{sum}\ [p\ (a\ |\ b) * p\ b\ |\ b \leftarrow [0..6]]$$

This formula in itself would be worth a lecture. For one thing, the notation is extremely suspicious. $(a \mid b)$, which is usually read “ a , given b ”, is clearly not of the same type as a or b , so cannot really be an argument to p . For another, the p we are computing with this formula is not the

$p\ a$ which must eventually appear in the products on the right hand side. I do not know how this notation came about: it is neither in Bayes' memoir, nor in Kolmogorov's monograph.

The conditional probability $p\ (a\ |\ b)$ gives us the probability that the next state is a , given that the current state is b . But this is exactly the information summarised in the graphical representation. Moreover, it is clear that, at least formally, the total probability formula is identical to a matrix-vector multiplication.

As usual, we write the associated matrix by looking at how the canonical base vectors are transformed. In this case, the canonical base vector $e\ i = \lambda j \rightarrow i$ 'is' j is the probability distribution *concentrated* in i :

$$M = \begin{matrix} & \begin{matrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{matrix} \\ \begin{matrix} .4 \\ .6 \\ 0 \\ 0 \\ 0 \\ 0 \end{matrix} & \begin{matrix} 0 & 0 & 0 & 0 & .5 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & .7 & 0 & 0 & 1 & 0 \\ 0 & 0 & .3 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & .5 & 0 & 1 \end{matrix} \end{matrix}$$

Exercise: starting from state 0, how many steps do you need to take before the probability is concentrated in state 6? Reverse again the arrow from 2 to 4. What can you say about the long-term behaviour of the system now?

Exercise: Implement the example. You will need to define:

The transition function

$$\begin{aligned} f_3 &:: G \rightarrow (G \rightarrow Double) \quad \text{-- but we want only } G \rightarrow (G \rightarrow [0,1]), \text{ the unit interval} \\ f_3\ g\ g' &= undefined \quad \text{-- the probability of getting to } g' \text{ from } g \end{aligned}$$

The associated matrix

$$\begin{aligned} m_3 &:: G \rightarrow (G \rightarrow Double) \\ m_3\ g'\ g &= undefined \end{aligned}$$

Test

$$t_3 = toL\ (mul\ m_3\ (e\ 2 + e\ 4))$$

14.3 Monadic dynamical systems

All the examples of dynamical systems we have seen in the previous section have a similar structure. They work by taking a state (which is one of the generators) and return a structure of possible future states of type G :

- deterministic: there is exactly one possible future states: we take an element of G and return an element of G . The transition function has the type $f : G \rightarrow G$, the structure of the target is just G itself.
- non-deterministic: there is a set of possible future states, which we have implemented as a characteristic function $G \rightarrow \{0,1\}$. The transition function has the type $f : G \rightarrow (G \rightarrow \{0,1\})$. The structure of the target is the *powerset* of G .

- stochastic: given a state, we compute a probability distribution over possible future states. The transition function has the type $f : G \rightarrow (G \rightarrow [0,1])$, the structure of the target is the probability distributions over G .

Therefore:

- deterministic: $f : G \rightarrow Id\ G$
- non-deterministic: $f : G \rightarrow Powerset\ G$, where $Powerset\ G = G \rightarrow \{0,1\}$
- stochastic: $f : G \rightarrow Prob\ G$, where $Prob\ G = G \rightarrow [0,1]$

We have represented the elements of the various structures as vectors. We also had a way of representing, as structures of possible states, those states that were known precisely: these were the canonical base vectors $e\ i$. Due to the nature of matrix-vector multiplication, what we have done was in effect:

$$\begin{aligned}
& M * v \quad \text{-- } v \text{ represents the current possible states} \\
& = \{-\ v \text{ is a linear combination of the base vectors } -\} \\
& \quad M * (v\ 0 * e\ 0 + \dots + v\ n * e\ n) \\
& = \{-\ \text{homomorphism } -\} \\
& \quad v\ 0 * (M * e\ 0) + \dots + v\ n * (M * e\ n) \\
& = \{-\ e\ i \text{ represents the perfectly known current state } i, \text{ therefore } M * e\ i = f\ i\ -\} \\
& \quad v\ 0 * f\ 0 + \dots + v\ n * f\ n
\end{aligned}$$

So, we apply f to every state, as if we were starting from precisely that state, obtaining the possible future states starting from that state, and then collect all these “hypothetical” possible future states in some way that takes into account the initial uncertainty (represented by $v\ 0, \dots, v\ n$) and the nature of the uncertainty (the specific $+$ and $*$).

If you examine the types of the operations involved

$$e : G \rightarrow Possible\ G$$

and

$$Possible\ G \rightarrow (G \rightarrow Possible\ G) \rightarrow Possible\ G$$

you see that they are very similar to the monadic operations

$$\begin{aligned}
& return : g \rightarrow m\ g \\
& (\gg) : m\ g \rightarrow (g \rightarrow m\ g') \rightarrow m\ g'
\end{aligned}$$

which suggests that the representation of possible future states might be monadic. Indeed, that is the case.

Since we implemented all these as matrix-vector multiplications, this raises the question: is there a monad underlying matrix-vector multiplication, such that the above are instances of it (obtained by specialising the scalar type S)?

Exercise: write *Monad* instances for *Id*, *Powerset*, *Prob*.

14.4 The monad of linear algebra

The answer is yes, up to a point. Haskell *Monads*, just like *Functors*, require *return* and \gg to be defined for every type. This will not work, in general. Our definition will work for *finite types* only.

```

type S      = Double
data Vector g = V (g → S)
toF (V v)    = v

class      (Bounded a, Enum a, Eq a) ⇒ Finite a where
instance (Bounded a, Enum a, Eq a) ⇒ Finite a where

class FinFunc f where
  func :: (Finite a, Finite b) ⇒ (a → b) → f a → f b

instance FinFunc Vector where
  func = funcV

funcV :: (Finite g, Eq g') ⇒ (g → g') → Vector g → Vector g'
funcV f (V v) = V (λg' → sum [v g | g ← [minBound..maxBound], g' == f g])

class FinMon f where
  embed :: Finite a ⇒ a → f a
  bind  :: (Finite a, Finite b) ⇒ f a → (a → f b) → f b

instance FinMon Vector where
  embed a      = V (λa' → if a == a' then 1 else 0)
  bind (V v) f = V (λg' → sum [toF (f g) g' * v g | g ← [minBound..maxBound]])

```

A better implementation, using associated types, is in file *Vector.lhs* in the repository.

Exercises:

1. Prove that the functor laws hold, i.e.

$$\begin{aligned} \text{func id} &= \text{id} \\ \text{func } (g \circ f) &= \text{func } g \circ \text{func } f \end{aligned}$$

2. Prove that the monad laws hold, i.e.

$$\begin{aligned} \text{bind } v \text{ return} &= v \\ \text{bind } (\text{return } g) f &= f g \\ \text{bind } (\text{bind } v f) h &= \text{bind } v (\lambda g' \rightarrow \text{bind } (f g') h) \end{aligned}$$

3. What properties of *S* have you used to prove these properties? Define a new type class *GoodClass* that accounts for these (and only these) properties.

14.5 Associated code

TODO: import from suitable earlier lecture.

```

instance Num a ⇒ Num (x → a) where
  f + g      = λx → f x + g x
  f - g      = λx → f x - g x
  f * g      = λx → f x * g x
  negate f   = negate ∘ f

```

```

|f|      = |·| ∘ f
signum f  = signum ∘ f
fromInteger = const ∘ fromInteger

instance Fractional a ⇒ Fractional (x → a) where
  recip f      = recip ∘ f
  fromRational = const ∘ fromRational

instance Floating a ⇒ Floating (x → a) where
  π      = const π
  exp f   = exp ∘ f
  sin f   = sin ∘ f
  cos f   = cos ∘ f
  f ** g = λx → (f x) ** (g x)
  -- and so on

```

15 Week 8: Laplace

15.1 The Exponential Function

```

{-# LANGUAGE FlexibleInstances #-}
{-# LANGUAGE TypeSynonymInstances #-}
module DSLsofMath.W08 where
import DSLsofMath.W05
import DSLsofMath.W06

```

One of the classical analysis textbooks, Rudin's Rudin [1987] starts with a prologue on the exponential function. The first sentence is

This is undoubtedly the most important function in mathematics.

Rudin goes on

It is defined, for every complex number z , by the formula

$$\exp z = \sum (z^n / n!)$$

We have defined the exponential function as the function represented by the power series

```

expx :: Fractional a ⇒ PowerSeries a
expx = integ expx 1

```

and approximated by

```

expf :: Fractional a ⇒ a → a
expf = eval 100 expx

```

It is easy to see, using the definition of *integ* that the power series *expx* is, indeed

$$\text{expx} = [1, 1/2, 1/(2*3), \dots, 1/(2*3*\dots*n), \dots]$$

We can compute the exponential for complex values if we can give an instance of *Fractional* for complex numbers. We could use the datatype *Data.Complex* from the Haskell standard library, but we prefer to roll our own in order to remind the basic operations on complex numbers.

As we saw in week 1, complex values can be represented as pairs of real values.

```
newtype Complex r = C (r,r) deriving (Eq, Show)
i :: Num a => Complex a
i = C (0,1)
```

Now, we have, for example

```
ex1 :: Fractional a => Complex a
ex1 = expf i
```

We have $ex1 = C (0.5403023058681398, 0.8414709848078965)$. Note that

```
cosf 1 = 0.5403023058681398
sinf 1 = 0.8414709848078965
```

and therefore $expf\ i = C (\cosf\ 1, \sinf\ 1)$. Coincidence?

Instead of evaluating the sum of the terms $a_n * z^n$, let us instead collect the terms in a series:

```
terms as z = terms1 as z 0 where
  terms1 (Cons a as) z n = Cons (a * z^n) (terms1 as z (n + 1))
```

We obtain

```
ex2 :: Fractional a => PowerSeries (Complex a)
ex2 = takePoly 10 (terms expf i)

ex2 = [ C (1.0, 0.0), C (0.0, 1.0
      , C (-0.5, 0.0), C (0.0, -0.16666666666666666
      , C (4.1666666666666664e-2, 0.0), C (0.0, 8.333333333333333e-3
      , C (-1.388888888888887e-3, 0.0), C (0.0, -1.9841269841269839e-4)
      , C (2.4801587301587298e-5, 0.0), C (0.0, 2.7557319223985884e-6
      ]
```

We can see that the real part of this series is the same as

```
ex2R = takePoly 10 (terms cosx 1)
```

and the imaginary part is the same as

```
ex2I = takePoly 10 (terms sinx 1)
```

(within approx 20 decimals). But the terms of a series evaluated at 1 are the coefficients of the series. Therefore, the coefficients of *cosx* are

```
[1, 0, -1 / 2!, 0, 1 / 4!, 0, -1 / 6!, ...]
```

i.e.

$$\begin{aligned} a(2 * n) &= (-1)^n / (2 * n)! \\ a(2 * n + 1) &= 0 \end{aligned}$$

and the terms of $\sin x$ are

$$[0, 1, 0, -1 / 3!, 0, 1 / 5!, 0, -1 / 7!, \dots]$$

i.e.

$$\begin{aligned} a(2 * n) &= 0 \\ a(2 * n + 1) &= (-1)^n / (2 * n + 1)! \end{aligned}$$

This can be proven from the definitions of $\cos x$ and $\sin x$. From this we obtain *Euler's formula*:

$$\exp(i * x) = \cos x + i * \sin x$$

One thing which comes out of Euler's formula is the fact that the exponential is a *periodic function*. A function $f : A \rightarrow B$ is said to be periodic if there exists $T \in A$ such that

$$f a = f(a + T) \quad -- \forall a \in A$$

(therefore, for this definition to make sense, we need addition on A ; in fact we normally assume at least group structure, i.e., addition and subtraction).

Since \sin and \cos are periodic, with period $2 * \pi$, we have, using the standard notation $a + i * b$ for $C(a, b)$

$$\begin{aligned} &e^{a + i * b + i * 2 * \pi} \\ &= e^{a + i * (b + 2 * \pi)} \\ &= e^a * e^{i * (b + 2 * \pi)} \\ &= e^a * (\cos(b + 2 * \pi) + i * \sin(b + 2 * \pi)) \\ &= e^a * (\cos b + i * \sin b) \\ &= e^a * e^{i * b} \\ &= e^{a + i * b} \end{aligned}$$

15.1.1 Exponential function: Associated code

TODO: Perhaps import from W01

```
instance Num r => Num (Complex r) where
  (+) = addC
  (*) = mulC
  fromInteger n = C (fromInteger n, 0)
  -- abs = absC -- requires Floating r as context
addC :: Num r => Complex r -> Complex r -> Complex r
addC (C (a, b)) (C (x, y)) = C ((a + x), (b + y))
```

```

mulC :: Num r => Complex r -> Complex r -> Complex r
mulC (C (ar, ai)) (C (br, bi)) = C (ar * br - ai * bi, ar * bi + ai * br)

modulusSquaredC :: Num r => Complex r -> r
modulusSquaredC (C (x, y)) = x^2 + y^2

absC :: Floating r => Complex r -> Complex r
absC c = C (sqrt(modulusSquaredC c), 0)

scale :: Num r => r -> Complex r -> Complex r
scale a (C (x, y)) = C (a * x, a * y)

conj :: Num r => Complex r -> Complex r
conj (C (x, x')) = C (x, -x')

instance Fractional r => Fractional (Complex r) where
  (/) = divC
  fromRational r = C (fromRational r, 0)
divC :: Fractional a => Complex a -> Complex a -> Complex a
divC x y = scale (1 / modSq) (x * conj y)
where modSq = modulusSquaredC y

```

15.2 The Laplace transform

This material was inspired by Quinn and Rai [2008], which is highly recommended reading.

Consider the differential equation

$$f'' x - 3 * f' x + 2 * f x = \exp (3 * x), f 0 = 1, f' 0 = 0$$

We can solve such equations with the machinery of power series:

$$fs = \text{integ } fs' \ 1$$

$$\textbf{where } fs' = \text{integ } (\exp (3 * x) + 3 * fs' - 2 * fs) \ 0$$

We have done this by “zooming in” on the function f and representing it by a power series, $f x = \sum a_n * x^n$. This allows us to reduce the problem of finding a function $f : \mathbb{R} \rightarrow \mathbb{R}$ to that of finding a function $a : \mathbb{N} \rightarrow \mathbb{R}$ (or finding a list of sufficiently many a -values for a good approximation).

Still, recursive equations are not always easy to solve (especially without a computer), so it’s worth looking for alternatives.

We have gone from

$$a : \mathbb{N} \rightarrow \mathbb{R} \quad \text{to} \quad f : \mathbb{R} \rightarrow \mathbb{R}$$

via

$$\sum a_n * x^n$$

In “zooming out”, we would like to go one step further

$$a : \mathbb{N} \rightarrow \mathbb{R} \quad \text{to} \quad f : \mathbb{R} \rightarrow \mathbb{R} \quad \text{to} \quad ?$$

via

$$\sum a_n * x^n \quad ?$$

At least the second part is “easy”. The analogue of series for a continuous function is integral

$$\sum a_n * x^n \quad \int_0^\infty (f(t) * x^t) dt$$

We note that, for the integral $\int_0^\infty (f(t) * x^t) dt$ to converge for a larger class of functions (say, bounded functions), we have to limit ourselves to $|x| < 1$. Both this condition and the integral make sense for $x \in \mathbb{C}$, so we could take

$$a : \mathbb{N} \rightarrow \mathbb{R} \quad \text{to} \quad f : \mathbb{R} \rightarrow \mathbb{R} \quad \text{to} \quad F : \{z \mid |z| < 1\} \rightarrow \mathbb{C}$$

but let us stick to \mathbb{R} for now.

Writing, somewhat optimistically

$$\mathcal{L} f(x) = \int_0^\infty (f(t) * x^t) dt$$

we can ask ourselves what $\mathcal{L} f'$ looks like. After all, we want to solve *differential* equations by “zooming out”. We have

$$\mathcal{L} f'(x) = \int_0^\infty (f'(t) * x^t) dt$$

Remember that $D(f * g) = Df * g + f * Dg$, therefore

$$\begin{aligned} & \mathcal{L} f'(x) \\ &= \{g(t) = x^t; g'(t) = \log x * x^t\} \\ & \int_0^\infty (D(f(t) * x^t)) - f(t) * \log x * x^t dt \\ &= \\ & \int_0^\infty (D(f(t) * x^t)) dt - \int_0^\infty f(t) * \log x * x^t dt \\ &= \\ & \lim_{t \rightarrow \infty} (f(t) * x^t) - (f(0) * x^0) - \log x * \int_0^\infty f(t) * x^t dt \\ &= \\ & -f(0) - \log x * \int_0^\infty f(t) * x^t dt \\ &= \\ & -f(0) - \log x * \mathcal{L} f(x) \end{aligned}$$

The factor $\log x$ is somewhat awkward. Let us therefore return to the definition of \mathcal{L} and operate a change of variables:

$$\begin{aligned} & \mathcal{L} f(x) = \int_0^\infty (f(t) * x^t) dt \\ & \Leftrightarrow \{x = \exp(\log x)\} \\ & \mathcal{L} f(x) = \int_0^\infty (f(t) * (\exp(\log x))^t) dt \\ & \Leftrightarrow \{(a^b)^c = a^{(b * c)}\} \\ & \mathcal{L} f(x) = \int_0^\infty (f(t) * \exp(\log x * t)) dt \end{aligned}$$

Since $\log x < 0$ for $|x| < 1$, we make the substitution $-s = \log x$. The condition $|x| < 1$ becomes $s > 0$ (or, in \mathbb{C} , *real* $s > 0$), and we have

$$\mathcal{L} f(s) = \int_0^\infty (f(t) * \exp(-s * t)) dt$$

This is the definition of the Laplace transform of the function f . Going back to the problem of computing $\mathcal{L} f'$, we now have

$$\begin{aligned} & \mathcal{L} f' s \\ = & \{- \text{The computation above with } s = -\log x. -\} \\ & -f 0 + s * \mathcal{L} f s \end{aligned}$$

We have obtained

$$\mathcal{L} f' s = s * \mathcal{L} f s - f 0$$

From this, we can deduce

$$\begin{aligned} & \mathcal{L} f'' s \\ = & \\ & s * \mathcal{L} f' s - f' 0 \\ = & \\ & s * (s * \mathcal{L} f s - f 0) - f' 0 \\ = & \\ & s^2 * \mathcal{L} f s - s * f 0 - f' 0 \end{aligned}$$

Exercise: what is the general formula for $\mathcal{L} f^{(k)} s$?

Returning to our differential equation, we have

$$\begin{aligned} & f'' x - 3 * f' x + 2 * f x = \exp(3 * x), f 0 = 1, f' 0 = 0 \\ \Leftrightarrow & \{- \text{point-free form} -\} \\ & f'' - 3 * f' + 2 * f = \exp \circ (3*), f 0 = 1, f' 0 = 0 \\ \Rightarrow & \{- \text{applying } \mathcal{L} \text{ to both sides} -\} \\ & \mathcal{L} (f'' - 3 * f' + 2 * f) = \mathcal{L} (\exp \circ (3*)), f 0 = 1, f' 0 = 0 \quad \text{-- Eq. (1)} \end{aligned}$$

Remark: Note that this is a necessary condition, but not a sufficient one. The Laplace transform is not injective. For one thing, it does not take into account the behaviour of f for negative arguments. Because of this, we often assume that the domain of definition for functions to which we apply the Laplace transform is $\mathbb{R}_{\geq 0}$. For another, it is known that changing the values of f for a countable number of its arguments does not change the value of the integral.

For the definition of \mathcal{L} and the linearity of the integral, we have that, for any f and g for which the transformation is defined, and for any constants α and β

$$\mathcal{L} (\alpha * f + \beta * g) = \alpha * \mathcal{L} f + \beta * \mathcal{L} g$$

Note that this is an equality between functions. (Comparing to last week we can also see f and g as vectors and \mathcal{L} as a linear transformation.)

Applying this to the left-hand side of (1), we have for any s

$$\begin{aligned} & \mathcal{L} (f'' - 3 * f' + 2 * f) s \\ = & \{- \mathcal{L} \text{ is linear} -\} \\ & \mathcal{L} f'' s - 3 * \mathcal{L} f' s + 2 * \mathcal{L} f s \\ = & \{- \text{re-writing } \mathcal{L} f'' \text{ and } \mathcal{L} f' \text{ in terms of } \mathcal{L} f -\} \\ & s^2 * \mathcal{L} f s - s * f 0 - f' 0 - 3 * (s * \mathcal{L} f s - f 0) + 2 * \mathcal{L} f s \\ = & \{- f 0 = 1, f' 0 = 0 -\} \end{aligned}$$

$$(s^2 - 3s + 2) * \mathcal{L} f s - s + 3$$

For the right-hand side, we apply the definition:

$$\begin{aligned} & \mathcal{L} (exp \circ (3*)) s \\ = & \{- \text{Def. of } \mathcal{L} -\} \\ & \int_0^\infty exp(3 * t) * exp(-s * t) dt \\ = & \\ & \int_0^\infty exp((3 - s) * t) dt \\ = & \\ & \lim_{t \rightarrow \infty} 1 / (3 - s) * exp((3 - s) * t) - \\ & 1 / (3 - s) * exp((3 - s) * 0) \\ = & \{- \text{for } s > 3 -\} \\ & 1 / (s - 3) \end{aligned}$$

Therefore, we have, writing F for $\mathcal{L} f$

$$(s^2 - 3s + 2) * F s - s + 3 = 1 / (s - 3)$$

and therefore

$$\begin{aligned} & F s \\ = & \{- \text{Solve for } F s -\} \\ & (1 / (s - 3) + s - 3) / (s^2 - 3s + 2) \\ = & \{- s^2 - 3s + 2 = (s - 1) * (s - 2) -\} \\ & (10 - 6s + s^2) / ((s - 1) * (s - 2) * (s - 3)) \end{aligned}$$

We now have the problem of “recovering” the function f from its Laplace transform. The standard approach is to use the linearity of \mathcal{L} to write F as a sum of functions with known inverse transforms. We know one such function:

$$exp(\alpha * t) \{- \text{is the inverse Laplace transform of } -\} 1 / (s - \alpha)$$

In fact, in our case, this is all we need.

The idea is to write $F s$ as a sum of three fractions with denominators $s - 1$, $s - 2$, and $s - 3$ respectively, i.e., to find A , B , and C such that

$$\begin{aligned} & A / (s - 1) + B / (s - 2) + C / (s - 3) = (10 - 6s + s^2) / ((s - 1) * (s - 2) * (s - 3)) \\ \Rightarrow & \\ & A * (s - 2) * (s - 3) + B * (s - 1) * (s - 3) + C * (s - 1) * (s - 2) = 10 - 6s + s^2 \quad -- (2) \end{aligned}$$

We need this equality (2) to hold for values $s > 3$. A *sufficient* condition for this is for (2) to hold for *all* s . A *necessary* condition for this is for (2) to hold for the specific values 1, 2, and 3.

$$\begin{aligned} \text{For } s = 1 : A * (-1) * (-2) &= 10 - 6 + 1 \Rightarrow A = 2.5 \\ \text{For } s = 2 : B * 1 * (-1) &= 10 - 12 + 4 \Rightarrow B = -2 \\ \text{For } s = 3 : C * 2 * 1 &= 10 - 18 + 9 \Rightarrow C = 0.5 \end{aligned}$$

It is now easy to check that, with these values, (2) does indeed hold, and therefore that we have

$$F s = 2.5 * (1 / (s - 1)) - 2 * (1 / (s - 2)) + 0.5 * (1 / (s - 3))$$

The inverse transform is now easy:

$$f t = 2.5 * \exp t - 2 * \exp (2 * t) + 0.5 * \exp (3 * t)$$

Our mix of necessary and sufficient conditions makes it necessary to check that we have, indeed, a solution for the differential equation. The verification is in this case trivial.

References

- R. A. Adams and C. Essex. *Calculus: a complete course*. Pearson Canada, 7th edition, 2010.
- C. H. Edwards, D. E. Penney, and D. Calvis. *Elementary Differential Equations*. Pearson Prentice Hall Upper Saddle River, NJ, 6h edition, 2008.
- C. Ionescu and P. Jansson. Domain-specific languages of mathematics: Presenting mathematical analysis using functional programming. In J. Jeuring and J. McCarthy, editors, *Proceedings of the 4th and 5th International Workshop on Trends in Functional Programming in Education, Sophia-Antipolis, France and University of Maryland College Park, USA, 2nd June 2015 and 7th June 2016*, volume 230 of *Electronic Proceedings in Theoretical Computer Science*, pages 1–15. Open Publishing Association, 2016. doi: 10.4204/EPTCS.230.1.
- E. Landau. *Einführung in die Differentialrechnung und Integralrechnung*. Noordhoff, 1934.
- E. Landau. *Differential and Integral Calculus*. AMS/Chelsea Publication Series. AMS Chelsea Pub., 2001.
- S. Mac Lane. *Mathematics: Form and function*. Springer New York, 1986.
- M. D. McIlroy. Functional pearl: Power series, power serious. *J. of Functional Programming*, 9: 323–335, 1999. doi: 10.1017/S0956796899003299.
- T. J. Quinn and S. Rai. Discovering the laplace transform in undergraduate differential equations. *PRIMUS*, 18(4):309–324, 2008.
- J. J. Rotman. *A first course in abstract algebra*. Pearson Prentice Hall, 2006.
- W. Rudin. *Principles of mathematical analysis*, volume 3. McGraw-Hill New York, 1964.
- W. Rudin. *Real and complex analysis*. Tata McGraw-Hill Education, 1987.